

# SICP Chapter 1 Answers

ProducerMatt

September 4, 2022

# Contents

<b>1</b>	<b>HOW THIS DOCUMENT IS MADE</b>	<b>4</b>
1.0.1	Also consider: . . . . .	4
1.0.2	Current command for conversion . . . . .	4
1.1	Helpers for org-mode tables . . . . .	4
1.1.1	try-these . . . . .	4
1.1.2	transpose-list . . . . .	5
1.1.3	print-as-rows . . . . .	5
1.1.4	print-table . . . . .	6
1.1.5	print-table (spaces only) . . . . .	6
<b>2</b>	<b>Exercise 1.1</b>	<b>7</b>
2.1	Question . . . . .	7
2.2	Answer . . . . .	8
<b>3</b>	<b>Exercise 1.2</b>	<b>8</b>
3.1	Question . . . . .	8
3.2	Answer . . . . .	8
<b>4</b>	<b>Exercise 1.3</b>	<b>8</b>
4.1	Text . . . . .	8
4.2	Question . . . . .	9
4.3	Answer . . . . .	9
<b>5</b>	<b>Exercise 1.4</b>	<b>9</b>
5.1	Question . . . . .	9
5.2	Answer . . . . .	9
<b>6</b>	<b>Exercise 1.5</b>	<b>10</b>
6.1	Question . . . . .	10
6.2	Answer . . . . .	10
<b>7</b>	<b>Exercise 1.6</b>	<b>10</b>
7.1	Text code . . . . .	10
7.2	Question . . . . .	11
7.3	Answer . . . . .	12
<b>8</b>	<b>Exercise 1.7</b>	<b>12</b>
8.1	Text . . . . .	12
8.2	Question . . . . .	12
8.3	Diary . . . . .	12
8.3.1	Solving . . . . .	12
8.3.2	Imroving (sqrt) by avoiding extra (improve) call . . . . .	13
8.4	Answer . . . . .	14

<b>9</b>	<b>Exercise 1.8</b>	<b>15</b>
9.1	Question . . . . .	15
9.2	Diary . . . . .	15
9.3	Answer . . . . .	16
<b>10</b>	<b>Exercise 1.9</b>	<b>17</b>
10.1	Question . . . . .	17
10.2	Answer . . . . .	17
10.2.1	recursive procedure . . . . .	17
10.2.2	iterative procedure . . . . .	18
<b>11</b>	<b>Exercise 1.10</b>	<b>18</b>
11.1	Question . . . . .	18
11.2	Answer . . . . .	18
11.2.1	f . . . . .	18
11.2.2	g . . . . .	19
11.2.3	h . . . . .	19
<b>12</b>	<b>Exercise 1.11</b>	<b>20</b>
12.1	Question . . . . .	20
12.2	Answer . . . . .	20
12.2.1	Recursive . . . . .	20
12.2.2	Iterative . . . . .	20
<b>13</b>	<b>Exercise 1.12</b>	<b>22</b>
13.1	Question . . . . .	22
13.2	Answer . . . . .	22
<b>14</b>	<b>Exercise 1.13</b>	<b>23</b>
14.1	Question . . . . .	23
14.2	Answer . . . . .	23
14.2.1	Fibonacci number generator . . . . .	23
14.2.2	Various algorithms relating to the question . . . . .	23
<b>15</b>	<b>Exercise 1.14</b>	<b>25</b>
15.1	Question . . . . .	25
15.2	Answer . . . . .	25
15.3	Question 2 . . . . .	28
15.4	Answer 2 . . . . .	28
<b>16</b>	<b>Exercise 1.15</b>	<b>31</b>
16.1	Question 1 . . . . .	31
16.2	Answer 1 . . . . .	31
16.3	Question 2 . . . . .	32
16.4	Answer 2 . . . . .	32

<b>17 Exercise 1.16</b>	<b>35</b>
17.1 Text . . . . .	35
17.2 Question . . . . .	36
17.3 Diary . . . . .	36
17.4 Answer . . . . .	37
<b>18 Exercise 1.17</b>	<b>38</b>
18.1 Question . . . . .	38
18.2 Answer . . . . .	38
<b>19 Exercise 1.18</b>	<b>39</b>
19.1 Question . . . . .	39
19.2 Diary . . . . .	39
19.2.1 Comparison benchmarks: . . . . .	39
19.2.2 Hall of shame . . . . .	40
19.3 Answer . . . . .	40
<b>20 Exercise 1.19</b>	<b>40</b>
20.1 Question . . . . .	40
20.2 Diary . . . . .	41
20.3 Answer . . . . .	42
<b>21 Exercise 1.20</b>	<b>43</b>
21.1 Text . . . . .	43
21.2 Question . . . . .	43
21.3 Answer . . . . .	44
21.3.1 Applicative order . . . . .	44
<b>22 Exercise 1.21</b>	<b>46</b>
22.1 Text . . . . .	46
22.2 Question . . . . .	47
<b>23 Exercise 1.22</b>	<b>47</b>
23.1 Question . . . . .	47
23.2 Answer . . . . .	48
23.2.1 Part 1 . . . . .	48
23.2.2 Part 2 . . . . .	49
<b>24 Exercise 1.23</b>	<b>51</b>
24.1 Question . . . . .	51
24.2 A Comedy of Error (just the one) . . . . .	52
24.3 Answer . . . . .	53
<b>25 Exercise 1.24</b>	<b>57</b>
25.1 Text . . . . .	57
25.2 Question . . . . .	58
25.3 Answer . . . . .	58

<b>26 Exercise 1.25</b>	<b>61</b>
26.1 Question . . . . .	61
26.2 Answer . . . . .	62
<b>27 Exercise 1.26</b>	<b>63</b>
27.1 Question . . . . .	63
27.2 Answer . . . . .	63
<b>28 Exercise 1.27</b>	<b>64</b>
28.1 Question . . . . .	64
28.2 Answer . . . . .	64
<b>29 Exercise 1.28</b>	<b>64</b>
29.1 Question . . . . .	64
29.2 Analysis . . . . .	65
29.3 Answer . . . . .	66

# 1 HOW THIS DOCUMENT IS MADE

## TODO

```

1 (define (foo a b)
2   (+ a (* 2 b)))
3
4 (foo 5 3)

```

11

^ Dynamically evaluated when you press “enter” on the BEGIN\_SRC block!

### 1.0.1 Also consider:

- :results output for what the code prints
- :exports code or :exports results to just get one or the other

$a + (\pi \times b)$  <~ inline Latex btw :)

### 1.0.2 Current command for conversion

```

1 pandoc --from org --to latex 1.org -o 1.tex -s; xelatex 1.tex

```

## 1.1 Helpers for org-mode tables

### 1.1.1 try-these

Takes function `f` and list `testvals` and applies `f` to each item `i`. For each `i` returns a list with `i` and the result. Useful for making tables with a column for input and a column for output.

```

1 ;; Surely this could be less nightmarish
2 (define (try-these f . testvals)
3   (let ((l (if (and (= 1 (length testvals))
4                     (list? (car testvals)))
5                     (car testvals)
6                     testvals)))
7     (map (λ (i) (cons i
8                      (cons (if (list? i)
9                              (apply f i)
10                              (f i))
11                            #nil))))
12         l)))

```

### 1.1.2 transpose-list

“Rotate” a list, for example from '(1 2 3) to '('(1) '(2) '(3))

```

1 (define (transpose-list l)
2   (map list l))

```

### 1.1.3 print-as-rows

For manually printing items in rows to stdout. Can be helpful for gnuplot.

```

1 (define (p-nl a)
2   (display a)
3   (newline))
4 (define (print-spaced args)
5   (let ((a (car args))
6         (d (cdr args)))
7     (if (null? d)
8         (p-nl a)
9         (begin (display a)
10                (display " ")
11                (print-spaced d))))))
12 (define (print-as-rows . args)
13   (let ((a (car args))
14         (d (cdr args)))
15     (if (list? a)
16         (if (= 1 (length args))
17             (apply print-as-rows a)
18             (print-spaced a))
19         (p-nl a))
20     (if (null? d)
21         '()
22         (apply print-as-rows d))))

```

#### 1.1.4 print-table

Print args as a table separated by pipes. Optionally print spacer for colnames.

```

1 (use-modules (ice-9 format))
2 (define* (print-row ll #:key (mode #f))
3   (let ((fmtstr
4         (cond ((or (eq? mode #f)
5                     (equal? mode "display")
6                     (equal? mode "~a"))
7               "~a |") ;; print objects for human viewing
8         ((or (eq? mode #t)
9               (equal? mode "write")
10              (equal? mode "~s"))
11              "~s |") ;; print objects for correctly (read)ing back
12         ((string? mode)
13          mode)))) ;; pass custom format string
14   (format #t "~&|")
15   (map (λ(x) (format #t fmtstr x)) ll)
16   (format #t "~%"))
17 (define* (print-table table #:key (colnames #f) (mode #f))
18   (define (iter t)
19     (print-row (car t) #:mode mode)
20     (if colnames
21         (print-row (car t) #:mode "---|")
22         (map (λ(x) (print-row x #:mode mode)) (cdr t)))
23     (cond ((and (= 1 (length table))
24                 (list? (car table))) (iter (car table)))
25           ((<= 1 (length table)) (iter table))
26           (else error "Invalid Input??"))))
27
28 <<print-table>>
29 (let* ((l (iota 3))
30        (table (list
31                 (list 'column-1 'column-2 'column-3 'column-4)
32                 (cons 'row-a l)
33                 (cons 'row-b l)
34                 (cons 'row-c l)))))
35   (print-table table #:colnames #t ))

```

column-1	column-2	column-3	column-4
row-a	0	1	2
row-b	0	1	2
row-c	0	1	2

#### 1.1.5 print-table (spaces only)

TODO: Merge these together.

```

1 (use-modules (ice-9 format))
2 (define* (print-row ll #:key (mode #f))
3   (let ((fmtstr
4         (cond ((or (eq? mode #f)
5                     (equal? mode "display")
6                     (equal? mode "~a"))
7               "~a") ;; print objects for human viewing
8         ((or (eq? mode #t)
9               (equal? mode "write")
10              (equal? mode "~s"))
11              "~s") ;; print objects for correctly (read)ing back
12         ((string? mode)
13          mode)))) ;; pass custom format string
14
15   (format #t "~<math>\$</math>" ) ;; ensure start of new line
16   (map (lambda (x) (format #t fmtstr x)) ll)
17   (format #t "~%" ))))
18
19 (define* (print-table table #:key (colnames #f) (mode #f))
20   (define (iter t)
21     (print-row (car t) #:mode mode)
22     (map (lambda (x) (print-row x #:mode mode)) (cdr t)))
23   (cond ((and (= 1 (length table))
24               (list? (car table))) (iter (car table)))
25         ((<= 1 (length table)) (iter table))
26         (else error "Invalid Input??"))))
27
28 <<print-table-spaced>>
29 (let* ((l (iota 3))
30        (table (list
31                  (list 'column-1 'column-2 'column-3 'column-4)
32                  (cons 'row-a l)
33                  (cons 'row-b l)
34                  (cons 'row-c l))))))
35 (print-table table))

```

column-1 column-2 column-3 column-4 row-a 0 1 2 row-b 0 1 2 row-c 0 1 2

## 2 Exercise 1.1

### 2.1 Question

Below is a sequence of expressions. What is the result printed by the interpreter in response to each expression? Assume that the sequence is to be evaluated in the order in which it is presented.



## 2.2 Answer

```
1 10 ;; 10
2 (+ 5 3 4) ;; 12
3 (- 9 1) ;; 8
4 (/ 6 2) ;; 3
5 (+ (* 2 4) (- 4 6)) ;; 6
6 (define a 3) ;; a=3
7 (define b (+ a 1)) ;; b=4
8 (+ a b (* a b)) ;; 19
9 (= a b) ;; false
10 (if (and (> b a) (< b (* a b)))
11     b
12     a) ;; 4
13 (cond ((= a 4) 6)
14       ((= b 4) (+ 6 7 a))
15       (else 25)) ;; 16
16 (+ 2 (if (> b a) b a)) ;; 6
17 (* (cond ((> a b) a)
18      ((< a b) b)
19      (else -1))
20    (+ a 1)) ;; 16
```

## 3 Exercise 1.2

### 3.1 Question

Translate the following expression into prefix form:

$$\frac{5 + 2 + (2 - 3 - (6 + \frac{4}{5}))}{3(6 - 2)(2 - 7)}$$

### 3.2 Answer

```
1 (/ (+ 5 2 (- 2 3 (+ 6 (/ 4 5))))
2    (* 3 (- 6 2) (- 2 7)))
```

$$1/75$$

## 4 Exercise 1.3

### 4.1 Text

```
1 (define (square x)
2   (* x x))
```

## 4.2 Question

Define a procedure that takes three numbers as arguments and returns the sum of the squares of the two larger numbers.

## 4.3 Answer

```
1 <<square>>
2 (define (sum-square x y)
3   (+ (square x) (square y)))
4 (define (square-2of3 a b c)
5   (cond ((and (>= a b) (>= b c)) (sum-square a b))
6         ((and (>= a b) (> c b)) (sum-square a c))
7         (else (sum-square b c))))

1 <<EX1-3>>
2 <<try-these>>
3 (try-these square-2of3 '(7 5 3)
4                          '(7 3 5)
5                          '(3 5 7))
```

(7 5 3)	74
(7 3 5)	74
(3 5 7)	74

## 5 Exercise 1.4

### 5.1 Question

Observe that our model of evaluation allows for combinations whose operators are compound expressions. Use this observation to describe the behavior of the following procedure:

```
1 (define (a-plus-abs-b a b)
2   ((if (> b 0) + -) a b))
```

### 5.2 Answer

This code accepts the variables *a* and *b*, and if *b* is positive, it adds *a* and *b*. However, if *b* is zero or negative, it subtracts them. This decision is made by using the *+* and *-* procedures as the results of an *if* expression, and then evaluating according to the results of that expression. This is in contrast to a language like Python, which would do something like this:

```
1 if b > 0: a + b
2 else: a - b
```

## 6 Exercise 1.5

### 6.1 Question

Ben Bitdiddle has invented a test to determine whether the interpreter he is faced with is using applicative-order evaluation or normal-order evaluation. He defines the following two procedures:

```
1 (define (p) (p))
2
3 (define (test x y)
4   (if (= x 0)
5       0
6       y))
```

Then he evaluates the expression

```
1 (test 0 (p))
```

What behavior will Ben observe with an interpreter that uses applicative-order evaluation? What behavior will he observe with an interpreter that uses normal-order evaluation? Explain your answer. (Assume that the evaluation rule for the special form `if` is the same whether the interpreter is using normal or applicative order: The predicate expression is evaluated first, and the result determines whether to evaluate the consequent or the alternative expression.)

### 6.2 Answer

In either type of language, `(define (p) (p))` is an infinite loop. However, a normal-order language will encounter the special form, return `0`, and never evaluate `(p)`. An applicative-order language evaluates the arguments to `(test 0 (p))`, thus triggering the infinite loop.

## 7 Exercise 1.6

### 7.1 Text code

```
1 (define (abs x)
2   (if (< x 0)
3       ^^I (- x)
4       ^^I x))

1 (define (average x y)
2   (/ (+ x y) 2))

1 <<average>>
2 (define (improve guess x)
```

```

3   (average guess (/ x guess)))
4
5   <<square>>
6   <<abs>>
7   (define (good-enough? guess x)
8     (< (abs (- (square guess) x)) 0.001))
9
10  (define (sqrt-iter guess x)
11    (if (good-enough? guess x)
12        guess
13        (sqrt-iter (improve guess x) x)))
14
15  (define (sqrt x)
16    (sqrt-iter 1.0 x))

```

## 7.2 Question

Exercise 1.6: Alyssa P. Hacker doesn't see why `if` needs to be provided as a special form. "Why can't I just define it as an ordinary procedure in terms of `cond`?" she asks. Alyssa's friend Eva Lu Ator claims this can indeed be done, and she defines a new version of `if`:

```

1  (define (new-if predicate
2                then-clause
3                else-clause)
4    (cond (predicate then-clause)
5          (else else-clause)))

```

Eva demonstrates the program for Alyssa:

```

1  (new-if (= 2 3) 0 5)
2  ;; => 5
3
4  (new-if (= 1 1) 0 5)
5  ;; => 0

```

Delighted, Alyssa uses `new-if` to rewrite the square-root program:

```

1  (define (sqrt-iter guess x)
2    (new-if (good-enough? guess x)
3            guess
4            (sqrt-iter (improve guess x) x)))

```

What happens when Alyssa attempts to use this to compute square roots? Explain.

## 7.3 Answer

Using Alyssa's `new-if` leads to an infinite loop because the recursive call to `sqrt-iter` is evaluated before the actual call to `new-if`. This is because `if` and `cond` are special forms that change the way evaluation is handled; whichever branch is chosen leaves the other branches unevaluated.

## 8 Exercise 1.7

### 8.1 Text

```
1 (define (mean-square x y)
2   (average (square x) (square y)))
```

### 8.2 Question

The `good-enough?` test used in computing square roots will not be very effective for finding the square roots of very small numbers. Also, in real computers, arithmetic operations are almost always performed with limited precision. This makes our test inadequate for very large numbers. Explain these statements, with examples showing how the test fails for small and large numbers. An alternative strategy for implementing `good-enough?` is to watch how `guess` changes from one iteration to the next and to stop when the change is a very small fraction of the `guess`. Design a square-root procedure that uses this kind of end test. Does this work better for small and large numbers?

### 8.3 Diary

#### 8.3.1 Solving

My original answer was this, which compares the previous iteration until the new and old are within an arbitrary  $dx$ .

```
1 <<txt-sqrt>>
2 (define (inferior-good-enough? guess lastguess)
3   (<=
4     (abs (-
5           (/ lastguess guess)
6             1))
7     0.00000000000001)) ; dx
8 (define (new-sqrt-iter guess x lastguess) ;; Memory of previous value
9   (if (inferior-good-enough? guess lastguess)
10       guess
11       (new-sqrt-iter (improve guess x) x guess)))
12 (define (new-sqrt x)
13   (new-sqrt-iter 1.0 x 0))
```

This solution can correctly find small and large numbers:

```

1 <<inferior-good-enough>>
2 (new-sqrt 1000000000000)

3162277.6601683795

1 <<try-these>>
2 <<inferior-good-enough>>
3 (try-these new-sqrt '(0.01 0.0001 0.000001 0.00000001 0.0000000001))

0.01 0.1
0.0001 0.01
1e-06 0.001
1e-08 9.999999999999999e-05
1e-10 9.999999999999999e-06

```

However, I found this solution online that isn't just simpler but automatically reaches the precision limit of the system:

```

1 <<txt-sqrt>>
2 (define (best-good-enough? guess x)
3   (= (improve guess x) guess))

```

### 8.3.2 Improving (sqrt) by avoiding extra (improve) call

#### 1. Non-optimized

```

1 (use-modules (ice-9 format))
2 (load "../mattbench.scm")
3 (define (average x y)
4   (/ (+ x y) 2))
5 (define (improve guess x)
6   (average guess (/ x guess)))
7 (define (good-enough? guess x)
8   (= (improve guess x) guess)) ;; improve call 1
9 (define (sqrt-iter guess x)
10   (if (good-enough? guess x)
11       guess
12       (sqrt-iter (improve guess x) x))) ;; call 2
13 (define (sqrt x)
14   (sqrt-iter 1.0 x))
15 (newline)
16 (display (mattbench (lambda () (sqrt 69420)) 400000000))
17 (newline)
18 ;; 4731.30 <- Benchmark results

```

#### 2. Optimized

```

1 (use-modules (ice-9 format))
2 (load "../mattbench.scm")
3 (define (average x y)
4   (/ (+ x y) 2))
5 (define (improve guess x)
6   (average guess (/ x guess)))
7 (define (good-enough? guess nextguess x)
8   (= nextguess guess))
9 (define (sqrt-iter guess x)
10  (let ((nextguess (improve guess x)))
11    (if (good-enough? guess nextguess x)
12        guess
13        (sqrt-iter nextguess x))))
14 (define (sqrt x)
15  (sqrt-iter 1.0 x))
16 (newline)
17 (display (mattbench (lambda () (sqrt 69420)) 400000000))
18 (newline)

```

### 3. Benchmark results

Unoptimized	4731.30
Optimized	2518.44

## 8.4 Answer

The current method has decreasing accuracy with smaller numbers. Notice the steady divergence from correct answers here (should be decreasing powers of 0.1):

```

1 <<txt-sqrt>>
2 <<try-these>>
3 (try-these sqrt 0.01 0.0001 0.000001 0.00000001 0.0000000001)

```

0.01	0.10032578510960605
0.0001	0.03230844833048122
1e-06	0.031260655525445276
1e-08	0.03125010656242753
1e-10	0.03125000106562499

And for larger numbers, an infinite loop will eventually be reached.  $10^{12}$  can resolve, but  $10^{13}$  cannot.

```

1 <<txt-sqrt>>
2 (sqrt 1000000000000)

```

1000000.0

So, my definition of sqrt:

```

1 <<average>>
2 (define (improve guess x)
3   (average guess (/ x guess)))
4 (define (good-enough? guess x)
5   (= (improve guess x) guess))
6 (define (sqrt-iter guess x)
7   (if (good-enough? guess x)
8       guess
9       (sqrt-iter (improve guess x) x)))
10 (define (sqrt x)
11   (sqrt-iter 1.0 x))

1 <<try-these>>
2 <<sqrt>>
3 (try-these sqrt '(0.01 0.0001 0.000001 0.00000001 0.0000000001))

```

	0.01	0.1
	0.0001	0.01
	1e-06	0.001
	1e-08	9.999999999999999e-05
	1e-10	9.999999999999999e-06

## 9 Exercise 1.8

### 9.1 Question

Newton's method for cube roots is based on the fact that if  $y$  is an approximation to the cube root of  $x$ , then a better approximation is given by the value:

$$\frac{\frac{x}{y^2} + 2y}{3} \quad (1)$$

Use this formula to implement a cube-root procedure analogous to the square-root procedure. (In 1.3.4 we will see how to implement Newton's method in general as an abstraction of these square-root and cube-root procedures.)

### 9.2 Diary

My first attempt works, but needs an arbitrary limit to stop infinite loops:

```

1 <<square>>
2 <<try-these>>
3 (define (cb-good-enough? guess x)
4   (= (cb-improve guess x) guess))
5 (define (cb-improve guess x)
6   (/
7     (+

```



```

8      (/ x (square guess))
9      (* guess 2))
10     3))
11 (define (cbrt-iter guess x counter)
12   (if (or (cb-good-enough? guess x) (> counter 100))
13       guess
14       (begin
15         (cbrt-iter (cb-improve guess x) x (+ 1 counter))))))
16 (define (cbrt x)
17   (cbrt-iter 1.0 x 0))
18
19 (try-these cbrt 7 32 56 100)

```

7	1.912931182772389
32	3.174802103936399
56	3.825862365544778
100	4.641588833612779

However, this will hang on an infinite loop when trying to run (`cbrt 100`). I speculate it's a floating point precision issue with the “improve” algorithm. So to avoid it I'll just keep track of the last guess and stop improving when there's no more change occurring. Also while researching I discovered that (again due to floating point) (`cbrt -2`) loops forever unless you initialize your guess with a slightly different value, so let's do 1.1 instead.

### 9.3 Answer

```

1 <<square>>
2 (define (cb-good-enough? nextguess guess lastguess x)
3   (or (= nextguess guess)
4       (= nextguess lastguess)))
5 (define (cb-improve guess x)
6   (/
7     (+
8       (/ x (square guess))
9       (* guess 2))
10    3))
11 (define (cbrt-iter guess lastguess x)
12   (define nextguess (cb-improve guess x))
13   (if (cb-good-enough? nextguess guess lastguess x)
14       nextguess
15       (cbrt-iter nextguess guess x)))
16 (define (cbrt x)
17   (cbrt-iter 1.1 9999 x))

```

```

1 <<cbrt>>
2 <<try-these>>

```

```
3 (try-these cbrt 7 32 56 100 -2)
```

7	1.912931182772389
32	3.174802103936399
56	3.825862365544778
100	4.641588833612779
-2	-1.2599210498948732

## 10 Exercise 1.9

### 10.1 Question

Each of the following two procedures defines a method for adding two positive integers in terms of the procedures `inc`, which increments its argument by 1, and `dec`, which decrements its argument by 1.

```
1 (define (+ a b)
2   (if (= a 0)
3       b
4       (inc (+ (dec a) b))))
5
6 (define (+ a b)
7   (if (= a 0)
8       b
9       (+ (dec a) (inc b))))
```

Using the substitution model, illustrate the process generated by each procedure in evaluating `(+ 4 5)`. Are these processes iterative or recursive?

### 10.2 Answer

The first procedure is recursive, while the second is iterative though tail-recursion.

#### 10.2.1 recursive procedure

```
1 (+ 4 5)
2 (inc (+ 3 5))
3 (inc (inc (+ 2 5)))
4 (inc (inc (inc (+ 1 5))))
5 (inc (inc (inc (inc (+ 0 5)))))
6 (inc (inc (inc (inc 5))))
7 (inc (inc (inc 6)))
8 (inc (inc 7))
9 (inc 8)
10 9
```

### 10.2.2 iterative procedure

```
1 (+ 4 5)
2 (+ 3 6)
3 (+ 2 7)
4 (+ 1 8)
5 (+ 0 9)
6 9
```

## 11 Exercise 1.10

### 11.1 Question

The following procedure computes a mathematical function called Ackermann's function.

```
1 (define (A x y)
2   (cond ((= y 0) 0)
3         ((= x 0) (* 2 y))
4         ((= y 1) 2)
5         (else (A (- x 1)
6                   (A x (- y 1))))))
```

What are the values of the following expressions?

```
1 (A 1 10)
2 (A 2 4)
3 (A 3 3)
```

(1 10)	1024
(2 4)	65536
(3 3)	65536

```
1 <<ackermann>>
2 (define (f n) (A 0 n))
3 (define (g n) (A 1 n))
4 (define (h n) (A 2 n))
5 (define (k n) (* 5 n n))
```

Give concise mathematical definitions for the functions computed by the procedures f, g, and h for positive integer values of  $n$ . For example,  $(k\ n)$  computes  $5n^2$ .

### 11.2 Answer

#### 11.2.1 f

```
1 <<try-these>>
2 <<EX1-10-defs>>
3 (try-these f 1 2 3 10 15 20)
```

1	2
2	4
3	6
10	20
15	30
20	40

$$f(n) = 2n$$

### 11.2.2 g

```

1 <<try-these>>
2 <<EX1-10-defs>>
3 (try-these g 1 2 3 4 5 6 7 8)

```

1	2
2	4
3	8
4	16
5	32
6	64
7	128
8	256

$$g(n) = 2^n$$

### 11.2.3 h

```

1 <<try-these>>
2 <<EX1-10-defs>>
3 (try-these h 1 2 3 4)

```

1	2
2	4
3	16
4	65536

It took a while to figure this one out, just because I didn't know the term. This is repeated exponentiation. This operation is to exponentiation, what exponentiation is to multiplication. It's called either *tetration* or *hyper-4* and has no formal notation, but two common ways would be these:

$$h(n) = 2 \uparrow\uparrow n$$

$$h(n) = {}^n 2$$

## 12 Exercise 1.11

### 12.1 Question

A function  $f$  is defined by the rule that:

$$f(n) = n \text{ if } n < 3$$

and

$$f(n) = f(n-1) + 2f(n-2) + 3f(n-3) \text{ if } n \geq 3$$

Write a procedure that computes  $f$  by means of a recursive process. Write a procedure that computes  $f$  by means of an iterative process.

### 12.2 Answer

#### 12.2.1 Recursive

```
1 (define (fr n)
2   (if (< n 3)
3       n
4       (+ (fr (- n 1))
5          (* 2 (fr (- n 2)))
6          (* 3 (fr (- n 3))))))

1 <<try-these>>
2 <<EX1-11-fr>>
3 (try-these fr 1 3 5 10)
```

1	1
3	4
5	25
10	1892

#### 12.2.2 Iterative

##### 1. Attempt 1

```
1 ;; This seems like it could be better
2 (define (fi n)
3   (define (formula l)
4     (let ((a (car l))
5           (b (cadr l))
6           (c (caddr l)))
7       (+ a
8          (* 2 b)
9          (* 3 c))))
10  (define (iter l i)
```

```

11      (if (= i n)
12          (car l)
13          (iter (cons (formula l) l)
14                  (+ 1 i))))
15      (if (< n 3)
16          n
17          (iter '(2 1 0) 2)))

1  <<try-these>>
2  <<EX1-11-fi>>
3  (try-these fi 1 3 5 10)

```

1	1
3	4
5	25
10	1892

It works but it seems wasteful.

## 2. Attempt 2

```

1  (define (fi2 n)
2    (define (formula a b c)
3      (+ a
4          (* 2 b)
5          (* 3 c)))
6    (define (iter a b c i)
7      (if (= i n)
8          a
9          (iter (formula a b c)
10                 a
11                 b
12                 (+ 1 i))))
13    (if (< n 3)
14        n
15        (iter 2 1 0 2)))

1  <<try-these>>
2  <<EX1-11-fi2>>
3  (try-these fi2 1 3 5 10)

```

1	1
3	4
5	25
10	1892

I like that better.

## 13 Exercise 1.12

### 13.1 Question

The following pattern of numbers is called Pascal's triangle.

*Pretend there's a Pascal's triangle here.*

The numbers at the edge of the triangle are all 1, and each number inside the triangle is the sum of the two numbers above it. Write a procedure that computes elements of Pascal's triangle by means of a recursive process.

### 13.2 Answer

I guess I'll rotate the triangle 45 degrees to make it the top-left corner of an infinite spreadsheet.

```
1 (define (pascal x y)
2   (if (or (= x 0)
3         (= y 0))
4       1
5       (+ (pascal (- x 1) y)
6          (pascal x (- y 1)))))

1 <<try-these>>
2 <<pascal-rec>>
3 (let ((l (iota 8)))
4   (map (lambda (row)
5         (map (lambda (xy)
6               (apply pascal xy))
7               row))
8         (map (lambda (x)
9               (map (lambda (y)
10                     (list x y))
11                     l))
12               l)))
```

1	1	1	1	1	1	1	1
1	2	3	4	5	6	7	8
1	3	6	10	15	21	28	36
1	4	10	20	35	56	84	120
1	5	15	35	70	126	210	330
1	6	21	56	126	252	462	792
1	7	28	84	210	462	924	1716
1	8	36	120	330	792	1716	3432

The test code was much harder to write than the actual solution.

## 14 Exercise 1.13

### 14.1 Question

Prove that  $\text{Fib}(n)$  is the closest integer to  $\frac{n}{\sqrt{5}}$  where  $\text{Phi}$  is  $\frac{1+\sqrt{5}}{2}$ . Hint: let  $= \frac{1-\sqrt{5}}{2}$ . Use induction and the definition of the Fibonacci numbers to prove that

$$\text{Fib}(n) = \frac{\frac{n}{\sqrt{5}} - \frac{n}{\sqrt{5}}}{\sqrt{5}}$$

### 14.2 Answer

I don't know how to write a proof yet, but I can make functions to demonstrate it.

#### 14.2.1 Fibonacci number generator

```
1 (define (fib-iter n)
2   (define (iter i a b)
3     (if (= i n)
4         b
5         (iter (+ i 1)
6               b
7               (+ a b))))
8   (if (<= n 2)
9       1
10      (iter 2 1 1)))
```

#### 14.2.2 Various algorithms relating to the question

```
1 <<sqrt>>
2 (define sqrt5
3   (sqrt 5))
4 (define phi
5   (/ (+ 1 sqrt5) 2))
6 (define epsilon
7   (/ (- 1 sqrt5) 2))
8 (define (fib-phi n)
9   (/ (- (expt phi n)
10         (expt epsilon n))
11      sqrt5))

1 (use-srfis '(1))
2 <<fib-iter>>
3 <<fib-phi>>
```




```

4  <<try-these>>
5
6  (let* ((vals (drop (iota 21) 10))
7         (fibs (map fib-iter vals))
8         (approx (map fib-phi vals)))
9        (zip vals fibs approx))

```

10	55	54.99999999999999
11	89	89.0
12	144	143.99999999999997
13	233	232.99999999999994
14	377	377.00000000000006
15	610	610.0
16	987	986.9999999999998
17	1597	1596.9999999999998
18	2584	2584.0
19	4181	4181.0
20	6765	6764.999999999999

You can see they follow closely. Graphing the differences, it's just an exponential curve at very low values, presumably following the exponential increase of the Fibonacci sequence itself.



## 15 Exercise 1.14

Below is the default version of the count-change function. I'll be aggressively modifying it in order to get a graph out of it.

```
1 (define (count-change amount)
2   (cc amount 5))
3
4 (define (cc amount kinds-of-coins)
5   (cond ((= amount 0) 1)
6         ((or (< amount 0)
7              (= kinds-of-coins 0))
8          0)
9         (else
10          (+ (cc amount (- kinds-of-coins 1))
11              (cc (- amount (first-denomination
12                           kinds-of-coins))
13                  kinds-of-coins))))))
14
15 (define (first-denomination kinds-of-coins)
16   (cond ((= kinds-of-coins 1) 1)
17         ((= kinds-of-coins 2) 5)
18         ((= kinds-of-coins 3) 10)
19         ((= kinds-of-coins 4) 25)
20         ((= kinds-of-coins 5) 50)))
```

### 15.1 Question

Draw the tree illustrating the process generated by the count-change procedure of 1.2.2 in making change for 11 cents.

### 15.2 Answer

I want to generate this graph algorithmically.

```
1 ;; cursed global
2 (define bubblecounter 0)
3 ;; Returns # of ways change can be made
4 ;; "Helper" for (cc)
5 (define (count-change amount)
6   (display "digraph {\n") ;; start graph
7   (cc amount 5 0)
8   (display "}\n") ;; end graph
9   (set! bubblecounter 0))
10
11 ;; GraphViz output
12 ;; Derivative: https://stackoverflow.com/a/14806144
```

```

13 (define (cc amount kinds-of-coins oldbubble)
14   (let ((recur (lambda (new-amount new-kinds)
15     (begin
16       (display "\n") ;; Source bubble
17       (display `(\,oldbubble ,amount ,kinds-of-coins))
18       (display "\n")
19       (display " -> ") ;; arrow pointing from parent to
↪ child
20       (display "\n") ;; child bubble
21       (display `(\,bubblecounter ,new-amount ,new-kinds))
22       (display "\n")
23       (display "\n")
24       (cc new-amount new-kinds bubblecounter))))))
25   (set! bubblecounter (+ bubblecounter 1))
26   (cond ((= amount 0) 1)
27         ((or (< amount 0) (= kinds-of-coins 0)) 0)
28         (else (+
29               (recur amount (- kinds-of-coins 1))
30               (recur (- amount
31                     (first-denomination kinds-of-coins))
32                     kinds-of-coins))))))
33
34 (define (first-denomination kinds-of-coins)
35   (cond ((= kinds-of-coins 1) 1)
36         ((= kinds-of-coins 2) 5)
37         ((= kinds-of-coins 3) 10)
38         ((= kinds-of-coins 4) 25)
39         ((= kinds-of-coins 5) 50)))

```


I'm not going to include the full printout of the (count-change 11), here's an example of what this looks like via 1.

```


1 <<count-change-graphviz>>
2 (count-change 1)

1 digraph {
2   "(0 1 5)" -> "(1 1 4)"
3   "(1 1 4)" -> "(2 1 3)"
4   "(2 1 3)" -> "(3 1 2)"
5   "(3 1 2)" -> "(4 1 1)"
6   "(4 1 1)" -> "(5 1 0)"
7   "(4 1 1)" -> "(6 0 1)"
8   "(3 1 2)" -> "(7 -4 2)"
9   "(2 1 3)" -> "(8 -9 3)"
10  "(1 1 4)" -> "(9 -24 4)"
11  "(0 1 5)" -> "(10 -49 5)"
12 }

```



So, the graph of (count-change 11) is:



### 15.3 Question 2

What are the orders of growth of the space and number of steps used by this process as the amount to be changed increases?

### 15.4 Answer 2

Let's look at this via the number of function calls needed for value  $n$ . Instead of returning an integer, I'll return a pair where `car` is the number of ways to count change, and `cdr` is the number of function calls that have occurred down that branch of the tree.

```

1 (define (count-calls amount)
2   (cc-calls amount 5))
3
4 (define (cc-calls amount kinds-of-coins)
5   (cond ((= amount 0) '(1 . 1))
6         ((or (< amount 0)
7              (= kinds-of-coins 0))
8          '(0 . 1))
9         (else
10          (let ((a (cc-calls amount (- kinds-of-coins 1)))
11                (b (cc-calls (- amount (first-denomination
12                               kinds-of-coins))
13                          kinds-of-coins)))
14            (cons (+ (car a)
15                     (car b))
16                  (+ 1
17                     (cdr a)
18                     (cdr b)))))))
19
20 (define (first-denomination kinds-of-coins)
21   (cond ((= kinds-of-coins 1) 1)
22         ((= kinds-of-coins 2) 5)
23         ((= kinds-of-coins 3) 10)
24         ((= kinds-of-coins 4) 25)
25         ((= kinds-of-coins 5) 50)))
26
27 (use-srfis '(1))
28 <<cc-calls>>
29 (let* ((vals (drop (iota 101) 1))
30        (mine (map count-calls vals)))
31   (zip vals (map car mine) (map cdr mine)))

```



I believe the space to be  $\Theta(n + d)$  as the function calls count down the denominations before counting down the change. However I notice most answers describe  $\Theta(n)$  instead, maybe I'm being overly pedantic and getting the wrong answer.

My issues came finding the time. The book describes the meaning and properties of  $\Theta$  notation in Section 1.2.3. However, my lack of formal math education made realizing the significance of this passage difficult. For one, I didn't understand that  $k_1f(n) \leq R(n) \leq k_2f(n)$  means "you can find the  $\Theta$  by proving that a graph of the algorithm's resource usage is bounded by two identical functions multiplied by constants." So, the graph of resource usage for an algorithm with  $\Theta(n^2)$  will be bounded by lines of  $n^2 \times \text{someconstant}$ , the top boundary's constant being larger than the small boundary. These are arbitrarily chosen constants, you're just proving that the function behaves the way you think it does.

Overall, finding the  $\Theta$  and  $\Omega$  and  $O$  notations (they are all different btw!) is about aggressively simplifying to make a very general statement about the behavior of the algorithm.

I could tell that a "correct" way to find the  $\Theta$  would be to make a formula which describes the algorithm's function calls for given input and denominations. This is one of the biggest time sinks, although I had a lot of fun and learned a lot. In the end, with some help from Jach in a Lisp Discord, I had the following formula:

$$\sum_{i=1}^{\text{ceil}(n/\text{val}(d))} T(n - \text{val}(d) * i, d)$$

But I wasn't sure where to go from here. The graphs let me see some interesting trends, though I didn't get any closer to an answer in the process.

By reading on other websites, I knew that you could find  $\Theta$  by obtaining a formula for  $R(n)$  and removing constants to end up with a term of interest. For example, if your algorithm's resource usage is  $\frac{n^2+7n}{5}$ , this demonstrates  $\Theta(n^2)$ . So I know a formula **without** a  $\sum$  would give me the answer I wanted. It didn't occur to me that it might be possible to use calculus to remove the  $\sum$  from the equation. At this point I knew I was stuck and decided to look up a guide.

After seeing a few solutions that I found somewhat confusing, I landed on this awesome article from Codology.net. They show how you can remove the summation, and proposed this equation for count-change with 5 denominations:

$$T(n, 5) = \frac{n}{50} + 1 + \sum_{i=0}^{n/50} T(n - 50i, 1)$$

Which, when expanded and simplified, demonstrates  $\Theta(n^5)$  for 5 denominations.

Overall I'm relieved that I wasn't entirely off, given I haven't done math work like this since college. It's inspired me to restart my remedial math courses, I don't think I really grasped the nature of math as a tool of empowerment until now.

## 16 Exercise 1.15

### 16.1 Question 1

The sine of an angle (specified in radians) can be computed by making use of the approximation  $\sin x \approx x$  if  $x$  is sufficiently small, and the trigonometric identity  $\sin x = 3 \sin \frac{x}{3} - 4 \sin^3 \frac{x}{3}$  to reduce the size of the argument of  $\sin$ . (For purposes of this exercise an angle is considered "sufficiently small" if its magnitude is not greater than 0.1 radians.) These ideas are incorporated in the following procedures:

```

1 (define (cube x) (* x x x))
2 (define (p x) (- (* 3 x) (* 4 (cube x))))
3 (define (sine angle)
4   (if (not (> (abs angle) 0.1))
5       angle
6       (p (sine (/ angle 3.0)))))

```

How many times is the procedure `p` applied when `(sine 12.15)` is evaluated?

### 16.2 Answer 1

Let's find out!



```

1 (define (cube x) (* x x x))
2 (define (p x) (- (* 3 x) (* 4 (cube x))))
3 (define (sine angle)
4   (if (not (> (abs angle) 0.1))
5       (cons angle 0)
6       (let ((x (sine (/ angle 3.0))))
7         (cons (p (car x)) (+ 1 (cdr x))))))

1 <<1-15-p-measure>>
2 (let ((xy (sine 12.15)))
3   (list (car xy) (cdr xy)))

```

-0.39980345741334    5

p is evaluated 5 times.

### 16.3 Question 2

What is the order of growth in space and number of steps (as a function of a) used by the process generated by the sine procedure when (sine a) is evaluated?

### 16.4 Answer 2

```

1 (use-srfis '(1))
2 <<1-15-p-measure>>
3 (let* ((vals (iota 300 0.1 0.1))
4        (sines (map (λ (i)
5                      (cdr (sine i)))
6                      vals)))
7   (zip vals sines))

#+end_src

1 (use-srfis '(1))
2 <<1-15-p-measure>>
3 (let* ((vals (iota 10 0.1 0.1))
4        (sines (map (λ (i)
5                      (cdr (sine i)))
6                      vals)))
7   (zip vals sines))

```


Example output:

	0.1	0
	0.2	1
0.30000000000000004	2	
	0.4	2
	0.5	2
	0.6	2
0.7000000000000001	2	
	0.8	2
	0.9	2
	1.0	3

```

1  reset # helps with various issues in execution
2  set xlabel 'values of x'
3  set logscale x
4  set key top left
5  set style fill solid 1.00 border
6  set style function fillsteps below
7
8  f(x) = log(x) + 2.3
9
10 plot data using 1:2 with fillsteps title 'function calls', \
11      data using 1:(f($1)) with lines title 'log(x) + 2.3'

```



This graph shows that the number of times sine will be called is logarithmic.

- 0.1 to 0.2 are divided once

- 0.3 to 0.8 are divided twice
- 0.9 to 2.6 are divided three times
- 2.7 to 8 are divided four times
- 8.5 to 23.8 are divided five times

Given that the calls to `p` get stacked recursively, like this:

```

1 (sine 12.15)
2 (p (sine 4.05))
3 (p (p (sine 1.35)))
4 (p (p (p (sine 0.45))))
5 (p (p (p (p (sine 0.15)))))
6 (p (p (p (p (p (sine 0.05))))))
7 (p (p (p (p (p 0.05)))))
8 (p (p (p (p 0.14950000000000002))))
9 (p (p (p 0.43513455050000005)))
10 (p (p 0.9758465331678772))
11 (p -0.7895631144708228)
12 -0.39980345741334

```

So I argue the space and time is  $\Theta(\log(n))$

We can also prove this for the time by benchmarking the function:

```

1 ;; This execution takes too long for org-mode, so I'm doing it
2 ;; externally and importing the results
3 (use-srfis '(1))
4 (use-modules (ice-9 format))
5 (load "../mattbench.scm")
6 <<1-15-deps>>
7 (let* ((vals (iota 300 0.1 0.1))
8         (times (map (lambda (i)
9                       (mattbench (lambda () (sine i)) 1000000))
10                          vals)))
11       (with-output-to-file "sine-bench.dat" (lambda ()
12         (map (lambda (x y)
13               (format #t "~s~/~s~%" x y))
14               vals times))))


1 reset # helps with various issues in execution
2 set xtics 0.5
3 set xlabel 'values of x'
4 set logscale x
5 set key top left
6 set style fill solid 1.00 border
7 #set style function fillsteps below

```

```

8
9  f(x) = (log(x) * a) + b
10 fit f(x) 'Ex15/sine-bench.dat' using 1:2 via a,b
11
12 plot 'Ex15/sine-bench.dat' using 1:2 with fillsteps title
   ↳ 'time to execute', \
13   'Ex15/sine-bench.dat' using 1:(f($1)) with lines title
   ↳ sprintf('(log(x) * %.2f) + %.2f', a, b)

```



## 17 Exercise 1.16

### 17.1 Text

```

1 (define (expt-rec b n)
2   (if (= n 0)
3       1
4       (* b (expt-rec b (- n 1)))))
5
6 (define (expt-iter b n)
7   (define (iter counter product)
8     (if (= counter 0)
9         product
10        (iter (- counter 1)
11              (* b product))))
12   (iter n 1))

```

```

13
14 (define (fast-expt b n)
15   (cond ((= n 0)
16         1)
17         ((even? n)
18          (square (fast-expt b (/ n 2))))
19         (else
20          (* b (fast-expt b (- n 1))))))

```

## 17.2 Question

Design a procedure that evolves an iterative exponentiation process that uses successive squaring and uses a logarithmic number of steps, as does fast-expt. (Hint: Using the observation that  $(b^{n/2})^2 = (b^2)^{n/2}$ , keep, along with the exponent  $n$  and the base  $b$ , an additional state variable  $a$ , and define the state transformation in such a way that the product  $ab^n$  is unchanged from state to state. At the beginning of the process  $a$  is taken to be 1, and the answer is given by the value of  $a$  at the end of the process. In general, the technique of defining an *invariant quantity* that remains unchanged from state to state is a powerful way to think about the design of iterative algorithms.)

## 17.3 Diary

First I made this program which tries to use a false equivalence:

$$ab^2 = (a + 1)b^{n-1}$$

```

1 <<square>>
2 (define (fast-expt-iter b n)
3   (define (iter b n a)
4     (format #t "~&|~s~/~/|~s~/~/|~s|~%" b n a)
5     (cond ((= n 1) (begin (format #t "~&|~s~/~/|~s~/~/|~s|~%" (* b a)
6                               ↪ 1 1)
7                             (* b a))))
8     ((even? n) (iter (square b)
9                       (/ n 2)
10                      a))
11     (else (iter b (- n 1) (+ a 1))))
12   (format #t "~|~a~/|~a~/|~a|~%" "base" "power" "variable")
13   (format #t "~&|--|--|--|~%" )
14   (iter b n 1))

1 <<fast-expt-iter-fail1>>
2 <<try-these>>
3 (fast-expt-iter 2 6)

```

Here's what the internal state looks like during  $2^6$  (correct answer is 64):

base	power	variable
2	6	1
4	3	1
4	2	2
16	1	2
32	1	1

## 17.4 Answer

There are two key transforms to a faster algorithm. The first was already shown in the text:

$$ab^n \rightarrow a(b^2)^{n/2}$$

The second which I needed to deduce was this:

$$ab^n \rightarrow ((a \times b) \times b)^{n-1}$$

The solution essentially follows this logic:

- initialize  $a$  to 1
- If  $n$  is 1, return  $b * a$
- else if  $n$  is even, halve  $n$ , square  $b$ , and iterate
- else  $n$  is odd, so subtract 1 from  $n$  and  $a \rightarrow a \times b$

```

1 <<square>>
2 (define (fast-expt-iter b n)
3   (define (iter b n a)
4     (cond ((= n 1) (* b a))
5           ((even? n) (iter (square b)
6                             (/ n 2)
7                             a))
8           (else (iter b (- n 1) (* b a)))))
9   (iter b n 1))

1 <<fast-expt-iter>>
2 <<try-these>>
3 (try-these (λ(x) (fast-expt-iter 3 x)) (cdr (iota 11)))

```

1	3
2	9
3	27
4	81
5	243
6	729
7	2187
8	6561
9	19683
10	59049

## 18 Exercise 1.17

### 18.1 Question

The exponentiation algorithms in this section are based on performing exponentiation by means of repeated multiplication. In a similar way, one can perform integer multiplication by means of repeated addition. The following multiplication procedure (in which it is assumed that our language can only add, not multiply) is analogous to the `expt` procedure:

```

1 (define (* a b)
2   (if (= b 0)
3       0
4       (+ a (* a (- b 1)))))

```

This algorithm takes a number of steps that is linear in  $b$ . Now suppose we include, together with addition, operations `double`, which doubles an integer, and `halve`, which divides an (even) integer by 2. Using these, design a multiplication procedure analogous to `fast-expt` that uses a logarithmic number of steps.

### 18.2 Answer

```

1 (define (double x)
2   (+ x x))
3 (define (halve x)
4   (/ x 2))
5 (define (fast-mult-rec a b)
6   (cond ((= b 0) 0)
7         ((even? b)
8          (double (fast-mult-rec a (halve b)))) ; This was kind of a
↪ stretch to think of.G
9         ;(fast-mult (double a) (halve b))) <== My first instinct is
↪ iterative
10        (else (+ a (fast-mult-rec a (- b 1)))))

```

Proof it works:

```

1 <<fast-mult-rec>>
2 <<try-these>>
3 (try-these (λ(x) (fast-mult-rec 3 x)) (cdr (iota 11)))

```

```

1 3
2 6
3 9
4 12
5 15
6 18
7 21
8 24
9 27
10 30

```

## 19 Exercise 1.18

### 19.1 Question

Using the results of Exercise 1.16 and Exercise 1.17, devise a procedure that generates an iterative process for multiplying two integers in terms of adding, doubling, and halving and uses a logarithmic number of steps.

### 19.2 Diary

#### 19.2.1 Comparison benchmarks:

```

1 (load "../mattbench.scm")
2 <<fast-mult-iter>>
3 <<fast-mult-rec>>
4 <<print-table>>
5 (print-table (list (list "fast-mult-rec" "fast-mult-iter")
6                    (list (mattbench (λ() (fast-mult-rec 32 32)))
7                            (mattbench (λ() (fast-mult 32 32)))
8                    #:colnames #t)

```

"fast-mult-rec"	"fast-mult-iter"
196.89	166.35

So the iterative version takes 0.84 times less to do  $32 \times 32$ .



### 19.2.2 Hall of shame

Some of my *very* incorrect ideas:

$$ab = (a + 1)(b - 1)$$

$$ab = \left(a + \left(\frac{a}{2}\right)\right)(b - 1)$$

$$ab + c = (a(b - 1) + (b + c))$$

### 19.3 Answer

```
1 (define (double x)
2   (+ x x))
3 (define (halve x)
4   (/ x 2))
5 (define (fast-mult a b)
6   (define (iter a b c)
7     (cond ((= b 0) 0)
8           ((= b 1) (+ a c))
9           ((even? b)
10            (iter (double a) (halve b) c))
11            (else (iter a (- b 1) (+ a c)))))
12   (iter a b 0))

1 <<fast-mult-iter>>
2 <<try-these>>
3 (try-these (λ(x) (fast-mult 3 x)) (cdr (iota 11)))
```

1	3
2	6
3	9
4	12
5	15
6	18
7	21
8	24
9	27
10	30

## 20 Exercise 1.19

### 20.1 Question

There is a clever algorithm for computing the Fibonacci numbers in a logarithmic number of steps. Recall the transformation of the state variables *a* and *b* in the *fib-iter* process of section 1-2-2:

$$a < -a + b \text{ and } b < -a$$

Call this transformation T, and observe that applying T over and over again n times, starting with 1 and 0, produces the pair  $\text{Fib}_{(n+1)}$  and  $\text{Fib}_{(n)}$ . In other words, the Fibonacci numbers are produced by applying  $T^n$ , the nth power of the transformation T, starting with the pair (1,0). Now consider T to be the special case of  $p = 0$  and  $q = 1$  in a family of transformations  $T_{(pq)}$ , where  $T_{(pq)}$  transforms the pair (a,b) according to  $a < -bq + aq + ap$  and  $b < -bp + aq$ . Show that if we apply such a transformation  $T_{(pq)}$  twice, the effect is the same as using a single transformation  $T_{(p'q')}$  of the same form, and compute p' and q' in terms of p and q. This gives us an explicit way to square these transformations, and thus we can compute  $T^n$  using successive squaring, as in the 'fast-expt' procedure. Put this all together to complete the following procedure, which runs in a logarithmic number of steps:

```

1 (define (fib n)
2   (fib-iter 1 0 0 1 n))
3
4 (define (fib-iter a b p q count)
5   (cond ((= count 0) b)
6         ((even? count)
7          (fib-iter a
8                    b
9                    <??> ; compute p'
10                   <??> ; compute q'
11                   (/ count 2)))
12         (else (fib-iter (+ (* b q) (* a q) (* a p))
13                          (+ (* b p) (* a q))
14                          p
15                          q
16                          (- count 1))))))

```

## 20.2 Diary

More succinctly put:

$$\text{Fib}_n \begin{cases} a \leftarrow a + b \\ b \leftarrow a \end{cases}$$

$$\text{Fib-iter}_{abpq} \begin{cases} a \leftarrow bq + aq + ap \\ b \leftarrow bp + aq \end{cases}$$

(T) returns a transformation function based on the two numbers in the attached list. so (T 0 1) returns a fib function.

```

1 (define (T p q)
2   (λ (a b)
3     (cons (+ (* b q) (* a q) (* a p))
4           (+ (* b p) (* a q)))))
5
6 (define T-fib
7   (T 0 1))
8
9 ;; Repeatedly apply T functions:
10 (define (Tr f n)
11   (Tr-iter f n 0 1))
12 (define (Tr-iter f n a b)
13   (if (= n 0)
14       a
15       (let ((l (f a b)))
16         (Tr-iter f (- n 1) (car l) (cdr l)))))

```

$$T_{pq} : a, b \mapsto \begin{cases} a \leftarrow bq + aq + ap \\ b \leftarrow bp + aq \end{cases}$$

```

1 <<T-func>>
2 <<try-these>>
3 (try-these (λ (x) (Tr (T 0 1) x)) (cdr (iota 11)))

```

```

1 1
2 1
3 2
4 3
5 5
6 8
7 13
8 21
9 34
10 55

```

## 20.3 Answer

```

1 (define (fib-rec n)
2   (cond ((= n 0) 0)
3         ((= n 1) 1)
4         (else (+ (fib-rec (- n 1))
5                   (fib-rec (- n 2))))))
6 (define (fib n)
7   (fib-iter 1 0 0 1 n))

```

```

8
9 (define (fib-iter a b p q count)
10   (cond ((= count 0) b)
11         ((even? count)
12          (fib-iter a
13                    b
14                    (+ (* p p)
15                      (* q q))      ; compute p'
16                    (+ (* p q)
17                      (* q q))      ; compute q'
18                    (/ count 2)))
19         (else (fib-iter (+ (* b q) (* a q) (* a p))
20                          (+ (* b p) (* a q))
21                          p
22                          q
23                          (- count 1))))))
24

```

"n"	"fib-rec"	"fib-iter"
1	1	1
2	1	1
3	2	2
4	3	3
5	5	5
6	8	8
7	13	13
8	21	21
9	34	34

## 21 Exercise 1.20

### 21.1 Text

```

1 (define (gcd a b)
2   (if (= b 0)
3       a
4       (gcd b (remainder a b))))

```

### 21.2 Question

The process that a procedure generates is of course dependent on the rules used by the interpreter. As an example, consider the iterative `gcd` procedure given above. Suppose we were to interpret this procedure using normal-order evaluation, as discussed in 1.1.5. (The normal-order-evaluation rule for `if` is described in Exercise 1.5.) Using the substitution method (for normal order), illustrate the process generated in evaluating `(gcd 206 40)` and indicate the remainder

operations that are actually performed. How many remainder operations are actually performed in the normal-order evaluation of `(gcd 206 40)`? In the applicative-order evaluation?

### 21.3 Answer

I struggled to understand this, but the key here is that normal-order evaluation causes the unevaluated expressions to be duplicated, meaning they get evaluated multiple times.

#### 21.3.1 Applicative order

```
1  call (gcd 206 40)
2  (if)
3  (gcd 40 (remainder 206 40))
4  eval remainder before call
5  call (gcd 40 6)
6  (if)
7  (gcd 6 (remainder 40 6))
8  eval remainder before call
9  call (gcd 6 4)
10 (if)
11 (gcd 2 (remainder 4 2))
12 eval remainder before call
13 call (gcd 2 0)
14 (if)
15 ;; => 2

1  ;; call gcd
2  (gcd 206 40)
3
4  ;; eval conditional
5  (if (= 40 0)
6      206
7      (gcd 40 (remainder 206 40)))
8
9  ;; recurse
10 (gcd 40 (remainder 206 40))
11
12 ; encounter conditional
13 (if (= (remainder 206 40) 0)
14     40
15     (gcd (remainder 206 40)
16          (remainder 40 (remainder 206 40))))
17
18 ; evaluate 1 remainder
```

```

19 (if (= 6 0)
20     40
21     (gcd (remainder 206 40)
22           (remainder 40 (remainder 206 40))))
23
24 ; recurse
25 (gcd (remainder 206 40)
26       (remainder 40 (remainder 206 40)))
27
28 ; encounter conditional
29 (if (= (remainder 40 (remainder 206 40)) 0)
30     (remainder 206 40)
31     (gcd (remainder 40 (remainder 206 40))
32           (remainder (remainder 206 40) (remainder 40 (remainder 206
↪ 40)))))
33
34 ; eval 2 remainder
35 (if (= 4 0)
36     (remainder 206 40)
37     (gcd (remainder 40 (remainder 206 40))
38           (remainder (remainder 206 40) (remainder 40 (remainder 206
↪ 40)))))
39
40 ; recurse
41 (gcd (remainder 40 (remainder 206 40))
42       (remainder (remainder 206 40) (remainder 40 (remainder 206 40))))
43
44 ; encounter conditional
45 (if (= (remainder (remainder 206 40) (remainder 40 (remainder 206
↪ 40))) 0)
46     (remainder 40 (remainder 206 40))
47     (gcd (remainder (remainder 206 40) (remainder 40 (remainder 206
↪ 40)))
48           (remainder (remainder 40 (remainder 206 40)) (remainder
↪ (remainder 206 40) (remainder 40 (remainder 206 40)))))
49
50 ; eval 4 remainders
51 (if (= 2 0)
52     (remainder 40 (remainder 206 40))
53     (gcd (remainder (remainder 206 40) (remainder 40 (remainder 206
↪ 40)))
54           (remainder (remainder 40 (remainder 206 40)) (remainder
↪ (remainder 206 40) (remainder 40 (remainder 206 40)))))
55
56 ; recurse
57 (gcd (remainder (remainder 206 40) (remainder 40 (remainder 206 40)))

```

```

58     (remainder (remainder 40 (remainder 206 40)) (remainder
↪   (remainder 206 40) (remainder 40 (remainder 206 40))))))
59
60 ; encounter conditional
61 (if (= (remainder (remainder 40 (remainder 206 40)) (remainder
↪   (remainder 206 40) (remainder 40 (remainder 206 40)))) 0)
62     (remainder (remainder 206 40) (remainder 40 (remainder 206 40)))
63     (gcd (remainder (remainder 40 (remainder 206 40)) (remainder
↪   (remainder 206 40) (remainder 40 (remainder 206 40)))) (remainder
↪   a (remainder (remainder 40 (remainder 206 40)) (remainder
↪   (remainder 206 40) (remainder 40 (remainder 206 40))))))
64
65 ; eval 7 remainders
66 (if (= 0 0)
67     (remainder (remainder 206 40) (remainder 40 (remainder 206 40)))
68     (gcd (remainder (remainder 40 (remainder 206 40)) (remainder
↪   (remainder 206 40) (remainder 40 (remainder 206 40)))) (remainder
↪   a (remainder (remainder 40 (remainder 206 40)) (remainder
↪   (remainder 206 40) (remainder 40 (remainder 206 40))))))
69
70 ; eval 4 remainders
71 (remainder (remainder 206 40) (remainder 40 (remainder 206 40)))
72 ; => 2

```

So, in normal-order eval, remainder is called 18 times, while in applicative order it's called 5 times.

## 22 Exercise 1.21

### 22.1 Text

```

1  <<square>>
2  (define (smallest-divisor n)
3    (find-divisor n 2))
4
5  (define (find-divisor n test-divisor)
6    (cond ((> (square test-divisor) n)
7            n)
8          ((divides? test-divisor n)
9            test-divisor)
10         (else (find-divisor
11                 n
12                 (+ test-divisor 1)))))
13
14 (define (divides? a b)
15   (= (remainder b a) 0))

```

## 22.2 Question

Use the smallest-divisor procedure to find the smallest divisor of each of the following numbers: 199, 1999, 19999.

```
1 <<find-divisor-txt>>
2 (map smallest-divisor '(199 1999 19999))

199 1999 7
```

## 23 Exercise 1.22

### 23.1 Question

Most Lisp implementations include a primitive called runtime that returns an integer that specifies the amount of time the system has been running (measured, for example, in microseconds). The following timed-prime-test procedure, when called with an integer *n*, prints *n* and checks to see if *n* is prime. If *n* is prime, the procedure prints three asterisks followed by the amount of time used in performing the test.

```
1 <<find-divisor-txt>>
2 (define (prime? n)
3   (= n (smallest-divisor n)))

1 <<prime-smallest-divisor>>
2 (define (timed-prime-test n)
3   (newline)
4   (display n) ;; Guile compatible \downarrow
5   (start-prime-test n (get-internal-run-time)))
6 (define (start-prime-test n start-time)
7   (if (prime? n)
8       (begin
9         (report-prime (- (get-internal-run-time)
10                          start-time))
11         n)
12       #f))
13 (define (report-prime elapsed-time)
14   (display " *** "))
15   (display elapsed-time))
```

Using this procedure, write a procedure search-for-primes that checks the primality of consecutive odd integers in a specified range. Use your procedure to find the three smallest primes larger than 1000; larger than 10,000; larger than 100,000; larger than 1,000,000. Note the time needed to test each prime. Since the testing algorithm has order of growth of  $\Theta(\sqrt{n})$ , you should expect that testing for primes around 10,000 should take about  $\sqrt{10}$  times as long as



testing for primes around 1000. Do your timing data bear this out? How well do the data for 100,000 and 1,000,000 support the  $\Theta(\sqrt{n})$  prediction? Is your result compatible with the notion that programs on your machine run in time proportional to the number of steps required for the computation?

## 23.2 Answer

### 23.2.1 Part 1

So this question is a little funky, because modern machines are so fast that the single-run times can seriously vary.

```

1 <<timed-prime-test-txt>>
2 (define (search-for-primes minimum goal)
3   (define m (if (even? minimum)
4                 (+ minimum 1)
5                 (minimum)))
6   (search-for-primes-iter m '() goal))
7 (define (search-for-primes-iter n lst goal)
8   (if (= goal 0)
9       lst
10      (let ((x (timed-prime-test n)))
11        (if (not (equal? x #f))
12            (search-for-primes-iter (+ n 2) (cons x lst) (- goal 1))
13            (search-for-primes-iter (+ n 2) lst goal))))))

1 <<search-primes-basic>>
2 (let ((lt1000-1 (search-for-primes 1000 3)))
3   (list "Primes > 1000" lt1000-1))

1 1001
2 1003
3 1005
4 1007
5 1009 *** 1651
6 1011
7 1013 *** 1425
8 1015
9 1017
10 1019 *** 1375

```

There's proof it works. And here are the answers to the question:

```

1 <<search-primes-basic>>
2 (let ((lt1000-1 (search-for-primes 1000 3))
3       (lt10000-1 (search-for-primes 10000 3))
4       (lt100000-1 (search-for-primes 100000 3))

```

```

5      (lt100000000-1 (search-for-primes 1000000 3)))
6  (list
7    (list "Primes > 1000" (reverse lt1000-1))
8    (list "Primes > 10000" (reverse lt10000-1))
9    (list "Primes > 100000" (reverse lt100000-1))
10   (list "Primes > 100000000" (reverse lt100000000-1))
11   ))

Primes > 1000      (1009 1013 1019)
Primes > 10000     (10007 10009 10037)
Primes > 100000    (100003 100019 100043)
Primes > 100000000 (1000003 1000033 1000037)

```

### 23.2.2 Part 2

Repeatedly re-running, it I see it occasionally jump to twice the time. I'm not happy with this, so I'm going to refactor to use the `mattbench2` utility from the root of the project folder.

```

1  (define (mattbench2 f n)
2    ;; Executes "f" for n times, and returns how long it took.
3    ;; f is a lambda that takes no arguments, a.k.a. a "thunk"
4
5    ;; Returns a list with car(last execution results) and cadr(time
6    ↪ taken divided by iterations n)
7
8    (define (time-getter) (get-internal-run-time))
9    (define start-time (time-getter))
10   (define (how-long) (- (time-getter) start-time))
11
12   (define (iter i)
13     (f)
14     (if (<= i 0)
15         (f) ;; return the results of the last function call
16         (iter (- i 1))))
17
18   (list (iter n) ;; result of last call of f
19         (/ (how-long) (* n 1.0))));; Divide by iterations so changed
20   ↪ n has no effect

```

I'm going to get some more precise times. First, I need a prime searching variant that doesn't bother benchmarking. This will call `prime?`, which will be bound later since we'll be trying different methods.

```

1  (define (search-for-primes minimum goal)
2    (define m (if (even? minimum)
3                  (+ minimum 1)

```

```

4         (minimum)))
5     (search-for-primes-iter m '() goal))
6 (define (search-for-primes-iter n lst goal)
7     (if (= goal 0)
8         lst
9         (let ((x (prime? n)))
10            (if (not (equal? x #f))
11                (search-for-primes-iter (+ n 2) (cons n lst) (- goal 1))
12                (search-for-primes-iter (+ n 2) lst goal))))))

```

I can benchmark these functions like so:


```

1  <<mattbench2>>
2  <<prime-smallest-divisor>>
3  <<search-for-primes-untimed>>
4  <<print-table>>
5
6  (define benchmark-iterations 1000000)
7  (define (testit f)
8      (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
9            (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
10           (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
11           (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))
12
13  (print-row
14  (testit search-for-primes))

```

Here are the results (run externally from Org-Mode):

```
5425.223086  20772.332491  53577.240193  121986.712395
```



The plot for the square root function doesn't quite fit the real one and I'm not sure where the fault lies. I don't struggle to understand things like "this algorithm is slower than this other one," but when asked to find or prove the  $\Theta$  notation I'm pretty clueless;

## 24 Exercise 1.23

### 24.1 Question

The `smallest-divisor` procedure shown at the start of this section does lots of needless testing: After it checks to see if the number is divisible by 2 there is no point in checking to see if it is divisible by any larger even numbers. This suggests that the values used for `test-divisor` should not be 2, 3, 4, 5, 6, ..., but rather 2, 3, 5, 7, 11, .... To implement this change, define a procedure `next` that returns 3 if its input is equal to 2 and otherwise returns its input plus 2. Modify the `smallest-divisor` procedure to use `(next test-divisor)` instead of `(+ test-divisor 1)`. With `timed-prime-test` incorporating this modified version of `smallest-divisor`, run the test for each of the 12 primes found in Exercise 1.22. Since this modification halves the number of test steps, you should expect it to run about twice as fast. Is this expectation confirmed? If not, what is the observed ratio of the speeds of the two algorithms, and how do you explain the fact that it is different from 2?


## 24.2 A Comedy of Error (just the one)

```
1  <<square>>
2  (define (smallest-divisor n)
3    (find-divisor n 2))
4
5  (define (next n)
6    (if (= n 2)
7        3
8        (+ n 1)))
9
10 (define (find-divisor n test-divisor)
11   (cond ((> (square test-divisor) n)
12         n)
13         ((divides? test-divisor n)
14          test-divisor)
15         (else (find-divisor
16                n
17                (next test-divisor))))))
18
19 (define (divides? a b)
20   (= (remainder b a) 0))

1  <<mattbench2>>
2  <<find-divisor-faster>>
3  (define (prime? n)
4    (= n (smallest-divisor n)))
5  <<search-for-primes-untimed>>
6  <<print-table>>
7
8  (define benchmark-iterations 1000000)
9  (define (testit f)
10   (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
11         (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
12         (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
13         (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))
14
15 (print-row
16 (testit search-for-primes))
```

6456.538118 25550.757304 66746.041644 148505.580638

min	(+1)	(next)
1000	5507.42497	6366.99462
10000	20913.71497	24845.9193
100000	53778.74737	64756.73693
1000000	122135.60511	143869.63561



So it's *slower* than before. Why?  
Oh, that's why.

```

1 (define (next n)
2   (if (= n 2)
3       3
4       (+ n 1))) ;; <-- D'oh.

```

### 24.3 Answer

Ok, let's try that again.

```

1 <<square>>
2 (define (smallest-divisor n)
3   (find-divisor n 2))
4
5 (define (next n)
6   (if (= n 2)
7       3
8       (+ n 2)))

```

```

9
10 (define (find-divisor n test-divisor)
11   (cond ((> (square test-divisor) n)
12         n)
13         ((divides? test-divisor n)
14          test-divisor)
15         (else (find-divisor
16                n
17                (next test-divisor))))))
18
19 (define (divides? a b)
20   (= (remainder b a) 0))

1 <<mattbench2>>
2 <<find-divisor-faster-real>>
3 (define (prime? n)
4   (= n (smallest-divisor n)))
5 <<search-for-primes-untimed>>
6 <<print-table>>
7
8 (define benchmark-iterations 500000)
9 (define (testit f)
10  (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
11        (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
12        (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
13        (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))
14
15 (print-row
16  (testit search-for-primes))


```

```

3863.7424  13519.209814  33520.676384  73005.539932

```

min	(+1)	(next-broken)	(next-fixed)
1000	5425.223086	6456.538118	3863.7424
10000	20772.332491	25550.757304	13519.209814
100000	53577.240193	66746.041644	33520.676384
1000000	121986.712395	148505.580638	73005.539932



I had a lot of trouble getting this one to compile, I have to restart Emacs in order to get it to render.

Anyways, there's the speedup that was expected. Let's compare the ratios. Defining a new average that takes arbitrary numbers of arguments:

```
1 (define (average . args)
2   (let ((len (length args)))
3     (/ (apply + args) len)))
```

Using it for percentage comparisons:

```
1 <<average-varargs>>
2 (list (cons "% speedup for broken (next)"
3   (cons (format #f "~2$%"
4     (apply average
5       (map (lambda (x y) (* 100 (/ x y)))
6         (car smd) (car smdff))))
7     #nil))
8   (cons "% speedup for real (next)"
9     (cons (format #f "~2$%"
10       (apply average
11         (map (lambda (x y) (* 100 (/ x y)))
12           (car smd) (car smdff))))
13       #nil))))
```

```
% speedup for broken (next) 81.93%
% speedup for real (next)   155.25%
```



Since this changed algorithm cuts out almost half of the steps, you might expect something more like a 200% speedup. Let's try optimizing it further. Two observations:

1. The condition (`divides? 2 n`) only needs to be run once at the start of the program.
2. Because it only needs to be run once, it doesn't need to be a separate function at all.

```


1  <<square>>
2  (define (smallest-divisor n)
3    (if (divides? 2 n)                ;; check for division by 2
4        2
5        (find-divisor n 3)))        ;; start find-divisor at 3
6
7  (define (find-divisor n test-divisor)
8    (cond ((> (square test-divisor) n)
9            n)
10          ((divides? test-divisor n)
11             test-divisor)
12          (else (find-divisor
13                  n
14                  (+ 2 test-divisor)))))) ;; just increase by 2
15
16  (define (divides? a b)
17    (= (remainder b a) 0))

1  <<mattbench2>>
2  <<find-divisor-faster-real2>>
3  (define (prime? n)
4    (= n (smallest-divisor n)))
5  <<search-for-primes-untimed>>
6  <<print-table>>
7
8  (define benchmark-iterations 500000)
9  (define (testit f)
10    (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
11          (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
12          (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
13          (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))
14
15  (print-row
16    (testit search-for-primes))

```

3151.259574 11245.20428 27803.067944 61997.275154

min	(+1)	(next-broken)	(next-fixed)	integrated
1000	5425.223086	6456.538118	3863.7424	3151.259574
10000	20772.332491	25550.757304	13519.209814	11245.20428
100000	53577.240193	66746.041644	33520.676384	27803.067944
1000000	121986.712395	148505.580638	73005.539932	61997.275154



% speedup for broken (next) 81.93%  
 % speedup for real (next) 155.25%  
 % speedup for optimized 186.59%

## 25 Exercise 1.24

### 25.1 Text

```

1 <<square>>
2 (define (expmod base exp m)
3   (cond ((= exp 0) 1)
4         ((even? exp)
5          (remainder
6            (square (expmod base (/ exp 2) m))
7            m))
8         (else

```

```

9      (remainder
10      (* base (expmod base (- exp 1) m))
11      m))))

1  (define (fermat-test n)
2    (define (try-it a)
3      (= (expmod a n n) a))
4      (try-it (+ 1 (random (- n 1)))))

1  (define (fast-prime? n times)
2    (cond ((= times 0) #t)
3          ((fermat-test n)
4            (fast-prime? n (- times 1)))
5          (else #f)))

```

## 25.2 Question

Modify the `timed-prime-test` procedure of Exercise 1.22 to use `fast-prime?` (the Fermat method), and test each of the 12 primes you found in that exercise. Since the Fermat test has  $\Theta(\log n)$  growth, how would you expect the time to test primes near 1,000,000 to compare with the time needed to test primes near 1000? Do your data bear this out? Can you explain any discrepancy you find?

## 25.3 Answer

```


1  <<mattbench2>>
2  <<expmod>>
3  <<fermat-test>>
4  <<fast-prime>>
5  (define fermat-iterations 2)
6  (define (prime? n)
7    (fast-prime? n fermat-iterations))
8  <<search-for-primes-untimed>>
9  <<print-table>>

10
11 (define benchmark-iterations 500000)
12 (define (testit f)
13   (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
14         (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
15         (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
16         (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))
17
18 (print-row
19 (testit search-for-primes))

```

11175.799722 23518.62116 32150.745642 32679.766448

min	(+1)	integrated	fermat (2 guesses)
1000	5425.223086	3151.259574	11175.799722
10000	20772.332491	11245.20428	23518.62116
100000	53577.240193	27803.067944	32150.745642
1000000	121986.712395	61997.275154	32679.766448



It definitely looks to be advancing much slower than the other methods. I'd like to see more of the function.

```

1 <<mattbench2>>
2 <<find-divisor-faster-real>>
3 (define (prime? n)
4   (= n (smallest-divisor n)))
5 <<search-for-primes-untimed>>
6 <<print-table>>
7
8 (define benchmark-iterations 100000)
9 (define (testit f)
10   (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
11         (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
12         (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
13         (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations)))

```

```

14         (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))
15         (cadr (mattbench2 (λ() (f 10000000 3)) benchmark-iterations))
16         (cadr (mattbench2 (λ() (f 100000000 3))
↪ benchmark-iterations))
17         (cadr (mattbench2 (λ() (f 1000000000 3))
↪ benchmark-iterations))
18         (cadr (mattbench2 (λ() (f 10000000000 3))
↪ benchmark-iterations))
19         (cadr (mattbench2 (λ() (f 100000000000 3))
↪ benchmark-iterations))))
20
21 (print-row
22 (testit search-for-primes))


1 <<mattbench2>>
2 <<expmod>>
3 <<fermat-test>>
4 <<fast-prime>>
5 (define feramat-iterations 100)
6 (define (prime? n)
7   (fast-prime? n feramat-iterations))
8 <<search-for-primes-untimed>>
9 <<print-table>>

10
11 (define benchmark-iterations 100000)
12 (define (testit f)
13   (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
14         (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
15         (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
16         (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))
17         (cadr (mattbench2 (λ() (f 10000000 3)) benchmark-iterations))
18         (cadr (mattbench2 (λ() (f 100000000 3)) benchmark-iterations))
19         (cadr (mattbench2 (λ() (f 1000000000 3))
↪ benchmark-iterations))
20         (cadr (mattbench2 (λ() (f 10000000000 3))
↪ benchmark-iterations))
21         (cadr (mattbench2 (λ() (f 100000000000 3))
↪ benchmark-iterations))
22         (cadr (mattbench2 (λ() (f 1000000000000 3))
↪ benchmark-iterations))))
23
24 (print-row
25 (testit search-for-primes))

```

3802.45146	13397.91871	32948.31241	73237.64777	299326.76182	678512.75719	2064911.33345
237945.8945	319761.90842	391573.47557	448501.96232	614009.08547	661205.34772	700058.30723

min	integrated	fermat (100 guesses)
1000	3802.45146	237945.8945
10000	13397.91871	319761.90842
100000	32948.31241	391573.47557
1000000	73237.64777	448501.96232
10000000	299326.76182	614009.08547
100000000	678512.75719	661205.34772
1000000000	2064911.33345	700058.30723
10000000000	7065717.58395	2852221.29076
100000000000	20198370.27007	3717690.96246
1000000000000	60956807.83034	3995948.05596



For the life of me I have no idea what that bump is. Maybe it needs more aggressive bignum processing there?

## 26 Exercise 1.25

### 26.1 Question

Alyssa P. Hacker complains that we went to a lot of extra work in writing `expmod`. After all, she says, since we already know how to compute exponentials, we could

have simply written

```
1 (define (expmod base exp m)
2   (remainder (fast-expt base exp) m))
```

Is she correct? Would this procedure serve as well for our fast prime tester? Explain.

## 26.2 Answer

In Alyssa's version of `expmod`, the result of the `fast-expt` operation is *extremely* large. For example, in the process of checking for divisors of 1,001, the number 455 will be tried. `(expt 455 1001)` produces an integer 2,661 digits long. This is just one of the thousands of exponentiations that `smallest-divisor` will perform. It's best to avoid this, so we use to our advantage the fact that we only need to know the remainder of the exponentiations. `expmod` breaks down the exponentiation into smaller steps and performs `remainder` after every step, significantly reducing the memory requirements.

As an example, let's trace (some of) the execution of `(expmod 455 1001 1001)`:

```
1 (expmod 455 1001 1001)
2 > (even? 1001)
3 > #f
4 > (expmod 455 1000 1001)
5 > > (even? 1000)
6 > > #t
7 > > (expmod 455 500 1001)
8 > > > (even? 500)
9 > > > #t
10 ; ; ...
11 > > > x11 (expmod 455 2 1001)
12 > > > x11 > (even? 2)
13 > > > x11 > #t
14 > > > x11 > (expmod 455 1 1001)
15 > > > x11 > > (even? 1)
16 > > > x11 > > #f
17 > > > x11 > > (expmod 455 0 1001)
18 > > > x11 > > 1
19 > > > x11 > 455
20 > > > x11 > (square 455)
21 > > > x11 > 207025
22 > > > x11 819
23 ; ; ...
24 > > > (square 364)
25 > > > 132496
26 > > 364
```

```

27 > > (square 364)
28 > > 132496
29 > 364
30 455

```

You can see that the numbers remain quite manageable throughout this process. So taking these extra steps actually leads to an algorithm that performs better.

## 27 Exercise 1.26

### 27.1 Question

Louis Reasoner is having great difficulty doing Exercise 1.24. His `fast-prime?` test seems to run more slowly than his `prime?` test. Louis calls his friend Eva Lu Ator over to help. When they examine Louis's code, they find that he has rewritten the `expmod` procedure to use an explicit multiplication, rather than calling `square`:

```

1 (define (expmod base exp m)
2   (cond ((= exp 0) 1)
3         ((even? exp)
4          (remainder
5           (* (expmod base (/ exp 2) m) ; ; <== hmm.
6              (expmod base (/ exp 2) m))
7           m))
8         (else
9          (remainder
10           (* base
11              (expmod base (- exp 1) m))
12           m))))

```

“I don’t see what difference that could make,” says Louis. “I do.” says Eva. “By writing the procedure like that, you have transformed the  $\Theta(\log n)$  process into a  $\Theta(n)$  process.” Explain.

### 27.2 Answer

Making the same function call twice isn’t the same as using a variable twice – Louis’ version doubles the work, having two processes solving the exact same problem. Since the number of processes used increases exponentially, this turns  $\log n$  into  $n$ .



## 28 Exercise 1.27

### 28.1 Question

Demonstrate that the Carmichael numbers listed in Footnote 1.47 really do fool the Fermat test. That is, write a procedure that takes an integer  $n$  and tests whether  $a^n$  is congruent to  $a$  modulo  $n$  for every  $a < n$ , and try your procedure on the given Carmichael numbers.

561   1105   1729   2465   2821   6601

### 28.2 Answer

```
1 <<expmod>>
2 (define (car-test n)
3   (define (check a)
4     (= (remainder (expt a n) n)
5        (remainder (modulo a n) n)))
6   (every check
7     (cddr (iota n))))

1 <<car-test>>
2 (list (car-test 12) ; <== false (not prime)
3       (car-test 1009);<== true  (real prime)
4       (car-test 561));<== true  (not prime,
5                               ; Carmichael number)
```

## 29 Exercise 1.28

### 29.1 Question

One variant of the Fermat test that cannot be fooled is called the Miller-Rabin test (Miller 1976; Rabin 1980). This starts from an alternate form of Fermat's Little Theorem, which states that if  $n$  is a prime number and  $a$  is any positive integer less than  $n$ , then  $a$  raised to the  $(n-1)$ -st power is congruent to 1 modulo  $n$ . To test the primality of a number  $n$  by the Miller-Rabin test, we pick a random number  $a < n$  and raise  $a$  to the  $(n-1)$ -st power modulo  $n$  using the `expmod` procedure. However, whenever we perform the squaring step in `expmod`, we check to see if we have discovered a "nontrivial square root of 1 modulo  $n$ ," that is, a number not equal to 1 or  $n-1$  whose square is equal to 1 modulo  $n$ . It is possible to prove that if such a nontrivial square root of 1 exists, then  $n$  is not prime. It is also possible to prove that if  $n$  is an odd number that is not prime, then, for at least half the numbers  $a < n$ , computing  $a^{n-1}$  in this way will reveal a nontrivial square root of 1 modulo  $n$ . (This is why the Miller-Rabin test cannot be fooled.) Modify the `expmod` procedure to signal if

it discovers a nontrivial square root of 1, and use this to implement the Miller-Rabin test with a procedure analogous to `fermat-test`. Check your procedure by testing various known primes and non-primes. Hint: One convenient way to make `expmod` signal is to have it return 0.

## 29.2 Analysis

For the sake of verifying this, I want to get a bigger list of primes and Carmichael numbers to verify against. I'll save them using Guile's built in read/write functions that save Lisp lists to text:

```

1 <<find-divisor-faster-real>>
2 (define (prime? n)
3   (= n (smallest-divisor n)))
4 (call-with-output-file "Data/primes-1k_to_1mil.txt" (λ(port)
5   (write (filter prime? (iota (- 1000000 1000) 1000))
6     port))))

1 ;; fermat prime test but checks *every* value from 2 to n-1
2 (define (fermat-prime? n)
3   (define (iter a)
4     (if (= a n)
5         #f
6         (if (= (expmod a n n) a)
7             #t
8             (iter (+ 1 a)))))
9   (iter 2))

1 (use-srfis '(1))
2 <<expmod>>
3 <<fermat-prime?>>
4 <<find-divisor-faster-real>>
5 (define (prime? n)
6   (= n (smallest-divisor n)))
7 (call-with-output-file "Data/carmichael-verification.txt" (λ(port)
8   (write (filter
9     (λ(x) (and (fermat-prime? x)
10                (not (prime? x))))
11     (iota (- 1000000 1000) 1000))
12     port))))

```

This will be useful in various future functions:

```

1 (define list-of-primes (call-with-input-file
2   ↪ "Data/primes-1k_to_1mil.txt" read))
3 (define list-of-carmichaels (call-with-input-file
4   ↪ "Data/carmichael.txt" read))

```

```

1 (use-srfis '(1))
2 <<expmod>>
3 <<fermat-prime?>>
4 <<find-divisor-faster-real>>
5 (define (prime? n)
6   (= n (smallest-divisor n)))
7 <<get-lists-of-primes>>
8 (define prime-is-working
9   (and (and-map prime? list-of-primes)
10        (not (and-map prime? list-of-carmichaels))))
11 (format #t "(prime?) is working: ~a~%"
12         (if prime-is-working
13             "Yes"
14             "No"))
15 (define fermt-is-vulnerable
16   (and (and-map fermt-prime? list-of-primes)
17        (and-map fermt-prime? list-of-carmichaels)))
18 (format #t "(fermt-prime?) is vulnerable: ~a~%"
19         (if fermt-is-vulnerable
20             "Yes"
21             "No"))

```

(prime?) is working: Yes (fermat-prime?) is vulnerable: Yes

### 29.3 Answer

```

1 <<square>>
2 (define (expmod-mr base exp m)
3   (cond ((= exp 0) 1)
4         ((even? exp)
5          (let ((sqr
6                (square (expmod-mr base (/ exp 2) m))))
7            (if (= 1 (modulo sqr m))
8                0
9                (remainder sqr m))))
10        (else
11         (remainder
12          (* base (expmod-mr base (- exp 1) m))
13          m))))
1
2 (define (mr-test n)
3   (define (try-it a)
4     (let ((it (expmod-mr a n n)))
5       (or (= it a)
6           (= it 0))))
7   (try-it (+ 1 (random (- n 1)))))

```

```

1 (define (mr-prime? n times)
2   (cond ((= times 0) #t)
3         ((mr-test n)
4          (mr-prime? n (- times 1)))
5         (else #f)))

1 <<expmod-mr>>
2 <<mr-test>>
3 <<mr-prime>>
4 (define mr-times 100)
5 <<get-lists-of-primes>>
6 (format #t ~a~%mr false-positives Carmichaels: ~a~%"
  ↪ "
7   (and-map (λ(x)(mr-prime? x mr-times)) list-of-primes)
8   (and-map (λ(x)(mr-prime? x mr-times)) list-of-carmichaels))

mr detects primes: #t
mr false-positives Carmichaels: #t

```

Shoot. And I thought I did a very literal interpretation of what the book asked.

Ah, I see the problem. I need to keep track of what the pre-squaring number was and use that to determine whether the square is valid or not.

```

1 <<square>>
2 (define (expmod-mr base exp m)
3   (cond ((= exp 0) 1)
4         ((even? exp)
5          ;; Keep result and remainder separate
6          (let* ((result (expmod-mr base (/ exp 2) m))
7                (rem (remainder (square result) m)))
8            (if (and (not (= result 1))
9                    (not (= result (- m 1)))
10                   (= 1 rem))
11                0 ;; non-trivial sqrt mod 1 is found
12                rem)))
13         (else
14          (remainder
15           (* base (expmod-mr base (- exp 1) m))
16           m))))

```

Unfortunately this one has the same problem. What's the issue?

Sadly, there's a massive issue in `mr-test`.

```

1 (define (mr-test n)
2   (define (try-it a)
3     (let ((it (expmod-mr a n n))) ;; Should be "a (- n 1) n"

```

```

4      (or (= it a)      ;; Should be (= it 1)
5          (= it 0))) ;; Two strikes, you're out
6      (try-it (+ 1 (random (- n 1)))))

```

One more time.

```

1  (define (mr-test n)
2    (define (try-it a)
3      (= 1 (expmod-mr a (- n 1) n)))
4      (try-it (+ 1 (random (- n 1)))))

```

```

1  <<expmod-mr2>>
2  <<mr-test2>>
3  <<mr-prime>>
4  (define mr-times 100)
5  <<get-lists-of-primes>>
6  (format #t ~a~%mr detects primes: ~a~%mr false-positives Carmichaels: ~a~%"
  ↪      (and-map (λ(x)(mr-prime? x mr-times)) list-of-primes)
7          (and-map (λ(x)(mr-prime? x mr-times)) list-of-carmichaels))
8

```

```

      mr detects primes: #t
mr false-positives Carmichaels: #f

```