A Journey Through SICP

Notes, exercises and analyses of Abelson and Sussman

ProducerMatt

January 16, 2023

Contents

1 Introduction Notes **12**
- 1.1 Text Foreword . . . 12
- 1.2 Preface, 1e . . . 12

2 Chapter 1: Building Abstractions with Procedures **13**
- 2.1 1.1: The Elements of Programming . . . 13
- 2.2 1.1.1: Expressions . . . 14
- 2.3 1.1.3: Evaluating Combinations . . . 14
- 2.4 1.1.4: Compound Procedures . . . 15
- 2.5 1.1.5: The Substitution Model for Procedure Application . . . 15
- 2.6 1.1.6: Conditional Expressions and Predicates . . . 17
- 2.7 Exercise 1.1 . . . 17
  - 2.7.1 Question . . . 17
  - 2.7.2 Answer . . . 18
- 2.8 Exercise 1.2 . . . 18
  - 2.8.1 Question . . . 18
  - 2.8.2 Answer . . . 18
- 2.9 Exercise 1.3 . . . 18
  - 2.9.1 Question . . . 18
  - 2.9.2 Answer . . . 18
- 2.10 Exercise 1.4 . . . 19
  - 2.10.1 Question . . . 19
  - 2.10.2 Answer . . . 19
- 2.11 Exercise 1.5 . . . 19
  - 2.11.1 Question . . . 19
  - 2.11.2 Answer . . . 20
- 2.12 1.1.7: Example: Square Roots by Newton's Method . . . 20
- 2.13 1.1.8: Procedures as Black-Box Abstractions . . . 20
- 2.14 Exercise 1.6 . . . 21
  - 2.14.1 Text code . . . 21
  - 2.14.2 Question . . . 21
  - 2.14.3 Answer . . . 22
- 2.15 Exercise 1.7 . . . 22
  - 2.15.1 Text . . . 22
  - 2.15.2 Question . . . 22
  - 2.15.3 Diary . . . 23
  - 2.15.4 Answer . . . 25
- 2.16 Exercise 1.8 . . . 26
  - 2.16.1 Question . . . 26
  - 2.16.2 Diary . . . 26
  - 2.16.3 Answer . . . 27
- 2.17 1.2: Procedures and the Processes They Generate . . . 28
- 2.18 1.2.1: Linear Recursion and Iteration . . . 28
- 2.19 Exercise 1.9 . . . 29

2.19.1 Question . . . 29
2.19.2 Answer . . . 29
2.20 Exercise 1.10 . . . 30
2.20.1 Question . . . 30
2.20.2 Answer . . . 30
2.21 1.2.2: Tree Recursion . . . 31
2.21.1 Example: Counting change . . . 32
2.22 Exercise 1.11 . . . 32
2.22.1 Question . . . 32
2.22.2 Answer . . . 32
2.23 Exercise 1.12 . . . 34
2.23.1 Question . . . 34
2.23.2 Answer . . . 35
2.24 Exercise 1.13 OPTIONAL . . . 35
2.24.1 Question . . . 35
2.24.2 Answer . . . 36
2.25 1.2.3: Orders of Growth . . . 37
2.26 Exercise 1.14 . . . 38
2.26.1 Text . . . 38
2.26.2 Question A . . . 38
2.26.3 Answer . . . 38
2.26.4 Question B . . . 41
2.26.5 Answer B . . . 41
2.27 Exercise 1.15 . . . 44
2.27.1 Question A . . . 44
2.27.2 Answer A . . . 45
2.27.3 Question B . . . 45
2.27.4 Answer B . . . 45
2.28 Exercise 1.16 . . . 48
2.28.1 Text . . . 48
2.28.2 Question . . . 49
2.28.3 Diary . . . 49
2.28.4 Answer . . . 50
2.29 Exercise 1.17 . . . 51
2.29.1 Question . . . 51
2.29.2 Answer . . . 52
2.30 Exercise 1.18 . . . 52
2.30.1 Question . . . 52
2.30.2 Diary . . . 52
2.30.3 Answer . . . 53
2.31 Exercise 1.19 . . . 54
2.31.1 Question . . . 54
2.31.2 Diary . . . 55
2.31.3 Answer . . . 56
2.32 1.2.5: Greatest Common Divisor . . . 57
2.33 Exercise 1.20 . . . 57

2.33.1 Text . . . 57
2.33.2 Question . . . 57
2.33.3 Answer . . . 58
2.34 1.2.6: Example: Testing for Primality . . . 60
2.35 Exercise 1.21 . . . 61
2.35.1 Text . . . 61
2.35.2 Question . . . 61
2.35.3 Answer . . . 61
2.36 Exercise 1.22 . . . 61
2.36.1 Question . . . 61
2.36.2 Answer . . . 62
2.37 Exercise 1.23 . . . 65
2.37.1 Question . . . 65
2.37.2 A Comedy of Error (just the one) . . . 66
2.37.3 Answer . . . 67
2.38 Exercise 1.24 . . . 71
2.38.1 Text . . . 71
2.38.2 Question . . . 72
2.38.3 Answer . . . 72
2.39 Exercise 1.25 . . . 75
2.39.1 Question . . . 75
2.39.2 Answer . . . 75
2.40 Exercise 1.26 . . . 76
2.40.1 Question . . . 76
2.40.2 Answer . . . 77
2.41 Exercise 1.27 . . . 77
2.41.1 Question . . . 77
2.41.2 Answer . . . 77
2.42 Exercise 1.28 . . . 78
2.42.1 Question . . . 78
2.42.2 Analysis . . . 78
2.42.3 Answer . . . 80
2.43 1.3: Formulating Abstractions with Higher-Order Procedures . . . 82
2.44 1.3.1: Procedures as Arguments . . . 82
2.45 Exercise 1.29 . . . 82
2.45.1 Text . . . 82
2.45.2 Question . . . 83
2.45.3 Answer . . . 83
2.46 Exercise 1.30 . . . 84
2.46.1 Question . . . 84
2.46.2 Answer . . . 84
2.47 Exercise 1.31 . . . 85
2.47.1 Question A.1 . . . 85
2.47.2 Answer A.1 . . . 85
2.47.3 Question A.2 . . . 85
2.47.4 Answer A.2 . . . 85

2.47.5 Question A.3 . . . 85
2.47.6 Answer A.3 . . . 86
2.47.7 Question B . . . 86
2.47.8 Answer B . . . 86
2.48 Exercise 1.32 . . . 87
2.48.1 Question A . . . 87
2.48.2 Answer A . . . 87
2.48.3 Question B . . . 88
2.48.4 Answer B . . . 88
2.49 Exercise 1.33 . . . 88
2.49.1 Question A . . . 88
2.49.2 Answer A . . . 88
2.49.3 Question B . . . 89
2.50 1.3.2: Constructing Procedures Using lambda . . . 90
2.51 Exercise 1.34 . . . 90
2.51.1 Question . . . 90
2.51.2 Answer . . . 91
2.52 1.3.3 Procedures as General Methods . . . 91
2.53 Exercise 1.35 . . . 91
2.53.1 Text . . . 91
2.53.2 Question . . . 92
2.53.3 Answer . . . 92
2.54 Exercise 1.36 . . . 92
2.54.1 Question . . . 92
2.54.2 Answer . . . 92
2.55 Exercise 1.37 . . . 93
2.55.1 Question A . . . 93
2.55.2 Answer A . . . 94
2.55.3 Question B . . . 94
2.55.4 Answer B . . . 94
2.55.5 Question C . . . 95
2.55.6 Answer C . . . 95
2.56 Exercise 1.38 . . . 96
2.56.1 Question . . . 96
2.56.2 Answer . . . 96
2.57 Exercise 1.39 . . . 96
2.57.1 Question . . . 96
2.57.2 Answer . . . 97
2.58 1.3.4 Procedures as Returned Values . . . 97
2.59 Exercise 1.40 . . . 97
2.59.1 Text . . . 97
2.59.2 Question . . . 98
2.59.3 Answer . . . 98
2.60 Exercise 1.41 . . . 99
2.60.1 Question . . . 99
2.60.2 Answer . . . 99

2.61 Exercise 1.42 . . . 99
2.61.1 Question . . . 99
2.61.2 Answer . . . 99
2.62 Exercise 1.43 . . . 100
2.62.1 Question . . . 100
2.62.2 Answer . . . 100
2.63 Exercise 1.44 . . . 100
2.63.1 Question . . . 100
2.63.2 Answer . . . 101
2.64 Exercise 1.45 . . . 101
2.64.1 Question . . . 101
2.64.2 Answer . . . 101
2.65 Exercise 1.46 . . . 104
2.65.1 Question . . . 104
2.65.2 Answer . . . 104

3 Chapter 2: Building Abstractions with Data 106
3.1 2.1.1: Example: Arithmetic Operations for Rational Numbers . . . 106
3.2 Exercise 2.1 . . . 106
3.2.1 Text . . . 106
3.2.2 Question . . . 107
3.2.3 Answer . . . 107
3.3 Exercise 2.2 . . . 108
3.3.1 Question . . . 108
3.3.2 Answer . . . 109
3.4 Exercise 2.3 . . . 110
3.4.1 Question . . . 110
3.4.2 Answer 1 . . . 111
3.4.3 Answer 2 . . . 114
3.5 2.1.3: What Is Meant by Data? . . . 117
3.6 Exercise 2.4 . . . 117
3.6.1 Question . . . 117
3.6.2 Answer . . . 117
3.7 Exercise 2.5 OPTIONAL . . . 118
3.7.1 Question . . . 118
3.7.2 Answer . . . 118
3.8 Exercise 2.6 OPTIONAL . . . 121
3.8.1 Question . . . 121
3.8.2 Answer . . . 121
3.9 Exercise 2.7 . . . 122
3.9.1 Text . . . 122
3.9.2 Question . . . 123
3.9.3 Answer . . . 123
3.10 Exercise 2.8 . . . 123
3.10.1 Question . . . 123
3.10.2 Answer . . . 124

3.11 Exercise 2.9 . . . 124
3.11.1 Question . . . 124
3.11.2 Answer . . . 124
3.12 Exercise 2.10 . . . 126
3.12.1 Question . . . 126
3.12.2 Answer . . . 127
3.13 Exercise 2.11 . . . 127
3.13.1 Question . . . 127
3.13.2 Answer . . . 127
3.14 Exercise 2.12 . . . 131
3.14.1 Question . . . 131
3.14.2 Answer . . . 131
3.15 Exercise 2.13 OPTIONAL . . . 132
3.15.1 Question . . . 132
3.15.2 Answer . . . 132
3.16 Exercise 2.14 . . . 134
3.16.1 Question . . . 134
3.16.2 Answer . . . 134
3.17 Exercise 2.15 . . . 136
3.17.1 Question . . . 136
3.17.2 Answer . . . 136
3.18 Exercise 2.16 OPTIONAL . . . 136
3.18.1 Question . . . 136
3.18.2 Answer . . . 137
3.19 2.2: Hierarchical Data and the Closure Property . . . 137
3.20 2.2.1: Representing Sequences . . . 138
3.21 Exercise 2.17 . . . 138
3.21.1 Question . . . 138
3.21.2 Answer . . . 138
3.22 Exercise 2.18 . . . 139
3.22.1 Question . . . 139
3.22.2 Answer . . . 139
3.23 Exercise 2.19 . . . 139
3.23.1 Question . . . 139
3.23.2 Answer . . . 140
3.24 Exercise 2.20 . . . 141
3.24.1 Question . . . 141
3.24.2 Answer . . . 141
3.25 Exercise 2.21 . . . 142
3.25.1 Question . . . 142
3.25.2 Answer . . . 143
3.26 Exercise 2.22 . . . 143
3.26.1 Questions . . . 143
3.26.2 Answer . . . 144
3.27 Exercise 2.23 . . . 145
3.27.1 Question . . . 145

3.27.2 Answer . . . 145
3.28 Exercise 2.24 . . . 145
3.28.1 Text Definitions . . . 145
3.28.2 Question . . . 146
3.28.3 Answer . . . 146
3.29 Exercise 2.25 . . . 148
3.29.1 Question . . . 148
3.30 Exercise 2.26 . . . 149
3.30.1 Question . . . 149
3.30.2 Answer . . . 149
3.31 Exercise 2.27 . . . 150
3.31.1 Question . . . 150
3.31.2 Answer . . . 150
3.32 Exercise 2.28 . . . 150
3.32.1 Question . . . 150
3.32.2 Answer . . . 151
3.33 Exercise 2.29: Binary Mobiles . . . 151
3.33.1 Text Definitions . . . 151
3.33.2 Question A: Selectors . . . 151
3.33.3 Answer A . . . 151
3.33.4 Question B: total-weight . . . 152
3.33.5 Answer B . . . 152
3.33.6 Question C: Balancing . . . 152
3.33.7 Answer C . . . 153
3.33.8 Question D: Implementation shakeup . . . 155
3.33.9 Answer D . . . 155
3.34 Exercise 2.30 . . . 157
3.34.1 Question . . . 157
3.34.2 Answer . . . 157
3.35 Exercise 2.31 . . . 158
3.35.1 Question . . . 158
3.35.2 Answer . . . 159
3.36 Exercise 2.32 . . . 159
3.36.1 Question . . . 159
3.36.2 Answer . . . 159
3.37 2.2.3: Sequences as Conventional Interfaces . . . 161
3.38 Exercise 2.33: The flexibility of `accumulate` . . . 162
3.38.1 Text Definitions . . . 162
3.38.2 Question . . . 162
3.38.3 Answer . . . 162
3.39 Exercise 2.34 . . . 163
3.39.1 Question . . . 163
3.39.2 Answer . . . 164
3.40 Exercise 2.35 . . . 164
3.40.1 Question . . . 164
3.40.2 Answer . . . 164

3.41 Exercise 2.36: Accumulate across multiple lists . . . 165
3.41.1 Question . . . 165
3.41.2 Answers . . . 165
3.42 Exercise 2.37: Enter the matrices . . . 166
3.42.1 Question . . . 166
3.42.2 Answer . . . 166
3.43 Exercise 2.38: fold-right . . . 168
3.43.1 Question A . . . 168
3.43.2 Answer A . . . 168
3.43.3 Question B . . . 169
3.43.4 Answer B . . . 169
3.44 Exercise 2.39: reverse via fold . . . 169
3.44.1 Question . . . 169
3.44.2 Answer . . . 170
3.45 Exercise 2.40: `unique-pairs` . . . 170
3.45.1 Text Definitions . . . 170
3.45.2 Question . . . 171
3.45.3 Answer . . . 171
3.46 Exercise 2.41: Ordered triples of positive integers . . . 172
3.46.1 Question . . . 172
3.46.2 Answer . . . 172
3.47 Exercise 2.42: Eight Queens . . . 173
3.47.1 Question . . . 173
3.47.2 Answer . . . 174
3.48 Exercise 2.43: Louis' `queens` . . . 175
3.48.1 Question . . . 175
3.48.2 Answer . . . 175
3.49 2.2.4: Example: A Picture Language . . . 177
3.50 Exercise 2.44: `up-split` . . . 180
3.50.1 Text Definitions . . . 180
3.50.2 Question . . . 180
3.50.3 Answer . . . 180
3.51 Exercise 2.45: Generalized splitting . . . 180
3.51.1 Question . . . 180
3.51.2 Answer . . . 181
3.52 Exercise 2.46: Defining vectors . . . 181
3.52.1 Question . . . 181
3.52.2 Answer . . . 181
3.53 Exercise 2.47: Defining frames . . . 182
3.53.1 Question . . . 182
3.53.2 Answer . . . 183
3.54 Exercise 2.48: Line segments . . . 183
3.54.1 Question . . . 183
3.54.2 Answer . . . 183
3.55 Exercise 2.49: Primitive painters . . . 183
3.55.1 Text Definitions . . . 183

3.55.2 Question . . . 184
3.55.3 Answer . . . 184
3.56 Exercise 2.50: Transforming painters . . . 187
3.56.1 Text Definitions . . . 187
3.56.2 Question . . . 189
3.56.3 Answer . . . 189
3.57 Exercise 2.51 . . . 191
3.57.1 Question . . . 191
3.57.2 Answer . . . 191
3.58 2.2.4 continued . . . 193
3.59 Exercise 2.52 . . . 193
3.59.1 Question A . . . 193
3.59.2 Answer A . . . 193
3.59.3 Question B . . . 199
3.59.4 Question C . . . 201
3.59.5 Textbook Definitions . . . 201
3.59.6 Answer C . . . 202
3.60 2.3.1: Quotation . . . 204
3.61 Exercise 2.53 . . . 205
3.61.1 Question . . . 205
3.61.2 Answer . . . 205
3.62 Exercise 2.54 . . . 205
3.62.1 Answer . . . 205
3.63 Exercise 2.55 . . . 206
3.63.1 Question . . . 206
3.63.2 Answer . . . 206
3.64 2.3.2: Example: Symbolic differentiator . . . 206
3.65 Exercise 2.56: Differentiating exponentiation . . . 208
3.65.1 Text definitions . . . 208
3.65.2 Question . . . 208
3.65.3 Answer . . . 209
3.66 Exercise 2.57 . . . 210
3.66.1 Question . . . 210
3.66.2 Answer . . . 211
3.67 Exercise 2.58 . . . 212
3.67.1 Question . . . 212
3.67.2 Part 1 . . . 212
3.67.3 Answer 1 . . . 212
3.67.4 Part 2 . . . 213
3.67.5 Answer 2 . . . 214
3.68 Exercise 2.59: Representing sets . . . 216
3.68.1 Text definitions . . . 216
3.68.2 Question . . . 217
3.68.3 Answer . . . 217
3.69 Exercise 2.59: Sets with duplicates . . . 218
3.69.1 Question . . . 218

3.69.2 Answer . . . 218
3.70 Exercise 2.61: Ordered sets . . . 220
3.70.1 Question . . . 220
3.70.2 Answer . . . 220
3.71 Exercise 2.62: `union-set` ordered . . . 221
3.71.1 Question . . . 221
3.71.2 Answer . . . 221
3.72 Exercise 2.63: binary trees . . . 223
3.72.1 Text definitions . . . 223
3.72.2 Question A . . . 223
3.72.3 Answer A . . . 224
3.72.4 Question B . . . 225
3.72.5 Answer B . . . 225
3.73 Exercise 2.64: Making a balanced binary tree . . . 227
3.73.1 Question A . . . 228
3.73.2 Answer A . . . 228
3.73.3 Question B . . . 229
3.73.4 Answer B . . . 229
3.74 Exercise 2.65: Sets as binary trees . . . 229
3.74.1 Textbook Definitions . . . 229
3.74.2 Question . . . 230
3.74.3 Answer . . . 230
3.75 Exercise 2.66: binary tree `lookup` . . . 237
3.75.1 Question . . . 237
3.75.2 Answer . . . 237
3.76 Exercise 2.67: decoding Huffman tree messages . . . 238
3.76.1 Text definitions . . . 238
3.76.2 Question . . . 239
3.76.3 Answer . . . 240
3.77 Exercise 2.68: encoding Huffman tree messages . . . 240
3.77.1 Question . . . 240
3.77.2 Answer . . . 241
3.78 Exercise 2.69: Generating Huffman trees . . . 242
3.78.1 Question . . . 242
3.78.2 Answer . . . 242
3.79 Exercise 2.70 . . . 245
3.79.1 Question . . . 245
3.79.2 Answer . . . 245
3.80 Exercise 2.71 . . . 248
3.80.1 Questions . . . 248
3.80.2 Answers . . . 248

1 Introduction Notes

1.1 Text Foreword

This book centers on three areas: the human mind, collections of computer programs, and the computer.

Every program is a model of a real or mental process, and these processes are at any time only partially understood. We change these programs as our understandings of these processes evolve.

Ensuring the correctness of programs becomes a Herculean task as complexity grows. Because of this, it's important to make fundamentals that can be relied upon to support larger structures.

1.2 Preface, 1e

"Computer Science" isn't really about computers or science, in the same way that geometry isn't really about measuring the earth ('geometry' translates to 'measurement of earth').

Programming is a medium for expressing ideas about methodology. For this reason, programs should be written first for people to read, and second for machines to execute.

The essential material for introductory programming is how to control complexity when building programs.

Computer Science is about imperative knowledge, as opposed to declarative. This can be called *procedural epistemology*.

Declarative knowledge *what is true.* For example: $\sqrt{x}$ is the y such that $y^2 = x$ and $y \geq 0$

Imperative knowledge *How to follow a process.* For example: to find an approximation to $\sqrt{x}$, make a guess G, improve the guess by averaging G and x/G, keep improving until the guess is good enough.

1. Techniques for controlling complexity

   Black-box abstraction Encapsulating an operation so the details of it are irrelevant.

   The fixed point of a function $f()$ is a value y such that $f(y) = y$. Method for finding a fixed point: start with a guess for y and keep applying $f(y)$ over and over until the result doesn't change very much.

   Define a box of the method for finding the fixed point of $f()$.

   One way to find $\sqrt{x}$ is to take our function for approaching a square root `(λ(guess target) (average guess (divide target guess)))`, applying that to our method for finding a fixed point, and this creates a **procedure** to find a square root.

   Black-box abstraction

(a) Start with primitive objects of procedures and data.

(b) Combination: combine procedures with *composition*, combine data with *construction* of compound data.

(c) Abstraction: defining procedures and abstracting data. Capture common patterns by making high-order procedures composed of other procedures. Use data as procedures.

Conventional interfaces Agreed-upon ways of connecting things together.

- How do you make operations generalized?
- How do you make large-scale structure and modularity?

  Object-oriented programming thinking of your structure as a society of discrete but interacting parts.

  Operations on aggregates thinking of your structure as operating on a stream, comparable to signal processing. *(Needs clarification.)*

Metalinguistic abstractions Making new languages. This changes the way you interact with the system by letting you emphasize some parts and deemphasize other parts.

2 Chapter 1: Building Abstractions with Procedures

Computational processes are abstract 'beings' that inhabit computers. Their evolution is directed by a pattern of rules called a **program**, and processes manipulate other abstract things called **data**.

Master software engineers are able to organize programs so they can be reasonably sure the resulting process performs the task intended, without catastrophic consequences, and that any problems can be debugged.

Lisp's users have traditionally resisted attempts to select an "official" version of the language, which has enabled Lisp to continually evolve.

There are powerful program-design techniques which rely on the ability to blur the distinction between data and processes. Lisp enables these techniques by allowing processes to be represented and manipulated as data.

2.1 1.1: The Elements of Programming

A programming language isn't just a way to instruct a computer – it's also a framework for the programmer to organize their ideas. Thus it's important to consider the means the language provides for combining ideas. Every powerful language has three mechanisms for this:

primitive expressions the simplest entities the language is concerned with

means of combination how compound elements can be built from simpler ones

means of abstraction how which compound elements can be named and manipulated as units

In programming, we deal with **data** which is what we want to manipulate, and **procedures** which are descriptions of the rules for manipulating the data.

A procedure has **formal parameters**. When the procedure is applied, the formal parameters are replaced by the **arguments** it is being applied to. For example, take the following code:

```
(define (square x)
  (* x x))
```

```
<<square>>
(square 5)
```

`x` is the formal parameter and `5` is the argument.

2.2 1.1.1: Expressions

The general form of Lisp is evaluating **combinations**, denoted by parenthesis, in the form `(operator operands)`, where *operator* is a procedure and *operands* are the 0 or more arguments to the operator.

Lisp uses **prefix notation**, which is not customary mathematical notation, but provides several advantages.

1. It supports procedures that take arbitrary numbers of arguments, i.e.ă`(+ 1 2 3 4 5)`.

2. It's straightforward to nest combinations in other combinations.

2.3 1.1.3: Evaluating Combinations

The evaluator can evaluate nested expressions recursively. **Tree accumulation** is the process of evaluating nested combinations, "percolating" values upward.

The recursive evaluation of `(* (+ 2 (* 4 6)) (+ 3 5 7))` breaks down into four parts:



2.4 1.1.4: Compound Procedures

We have identified the following in Lisp:

- primitive data are numbers, primitive procedures are arithmetic operations
- Operations can be combined by nesting combinations
- Data and procedures can be abstracted by variable & procedure definitions

Procedure definitions give a name to a compound procedure.

```
(define (square x) (* x x)) ; to square something, multiply it by itself
; now it can be applied or used in other definitions:
(square 4) ; => 16

(define (sum-of-squares x y)
  (+ (square x) (square y)))
(sum-of-squares 3 4) ; => 25
```

Note how these compound procedures are used in the same way as primitive procedures.

2.5 1.1.5: The Substitution Model for Procedure Application

To understand how the interpreter works, imagine it substituting the procedure calls with the bodies of the procedure and its arguments.

```
(* (square 3) (square 4))
; has the same results as
(* (* 3 3) (* 3 3))
```

This way of understanding procedure application is called the **substitution model**. This model is to help you understand procedure substitution, and is usually not how the interpreter actually works. This book will progress through more intricate models of interpreters as it goes. This is the natural progression when learning scientific phenomena, starting with a simple model, and replace it with more refined models as the phenomena is examined in more detail.

Evaluations can be done in different orders.

Applicative order evaluates the operator and operands, and then applies the resulting procedure to the resulting arguments. In other words, reducing, then expanding, then reducing.

Normal order substitutes expressions until it obtains an expression involving only primitive operators, or until it can't substitute any further, and then evaluates. This results in expanding the expression completely before doing any reduction, which results in some repeated evaluations.

For all procedure applications that can be modeled using substitution, applicative and normal order evaluation produce the same result. Normal order becomes more complicated once dealing with procedures that can't be modeled by substitution.

Lisp uses applicative order evaluation because it helps avoid repeated work and other complications. But normal has its own advantages which will be explored in Chapter 3 and 4.

```
; Applicative evaluation
(f 5)
(sum-of-squares (+ a 1) (* a 2))
(sum-of-squares (+ 5 1) (* 5 2))
(sum-of-squares 6 10)
(+ (square x)(square y))
(+ (square 6)(square 10))
(+ (* 6 6)(* 10 10))
(+ 36 100)
136
; Normal evaluation
(f 5)
(sum-of-squares (+ a 1) (* a 2))
(sum-of-squares (+ 5 1) (* 5 2))
(+ (square (+ 5 1)) (square (* 5 2)))
(+ (* (+ 5 1) (+ 5 1)) (* (* 5 2) (* 5 2)))
(+ (* 6 6) (* 10 10))
(+ 36 100)
```

```
19 136
```

(Extra-curricular clarification: Normal order delays evaluating arguments until they're needed by a procedure, which is called lazy evaluation.)

2.6 1.1.6: Conditional Expressions and Predicates

An important aspect of programming is testing and branching depending on the results of the test. `cond` tests **predicates**, and upon encountering one, returns a **consequent**.

```
(cond
     (predicate1 consequent1)
     ...
     (predicateN consequentN))
```

A shorter form of conditional:

```
(if predicate consequent alternative)
```

If `predicate` is true, `consequent` is returned. Else, `alternative` is returned.
Combining predicates:

```
(and expression1 ... expressionN)
; if encounters false, stop eval and returns false.
(or expression1 ... expressionN)
; if encounters true, stop eval and return true. Else false.
(not expression)
; true is expression is false, false if expression is true.
```

A small clarification:

```
(define A (* 5 5))
(define (D) (* 5 5))
A ; => 25
D ; => compound procedure D
(D) ; => 25 (result of executing procedure D)
```

Special forms bring more nuances into the substitution model mentioned previously. For example, when evaluating an **`if`** expression, you evaluate the predicate and, depending on the result, either evaluate the **consequent** or the **alternative**. If you were evaluating in a standard manner, the consequent and alternative would both be evaluated, rendering the **`if`** expression ineffective.

2.7 Exercise 1.1

2.7.1 Question

> Below is a sequence of expressions. What is the result printed by the interpreter in response to each expression? Assume that the sequence is to be evaluated in the order in which it is presented.

2.7.2 Answer

```
10 ;; 10
(+ 5 3 4) ;; 12
(- 9 1) ;; 8
(/ 6 2) ;; 3
(+ (* 2 4) (- 4 6)) ;; 6
(define a 3) ;; a=3
(define b (+ a 1)) ;; b=4
(+ a b (* a b)) ;; 19
(= a b) ;; false
(if (and (> b a) (< b (* a b)))
    b
    a) ;; 4
(cond ((= a 4) 6)
      ((= b 4) (+ 6 7 a))
      (else 25)) ;; 16
(+ 2 (if (> b a) b a)) ;; 6
(* (cond ((> a b) a)
         ((< a b) b)
         (else -1))
   (+ a 1)) ;; 16
```

2.8 Exercise 1.2

2.8.1 Question

Translate the following expression into prefix form:

$$\frac{5+2+(2-3-(6+\frac{4}{5}))}{3(6-2)(2-7)}$$

2.8.2 Answer

```
(/ (+ 5 2 (- 2 3 (+ 6 (/ 4 5))))
   (* 3 (- 6 2) (- 2 7)))
```

```
1/75
```

2.9 Exercise 1.3

2.9.1 Question

Define a procedure that takes three numbers as arguments and returns the sum of the squares of the two larger numbers.

2.9.2 Answer

```
<<square>>
(define (sum-square x y)
  (+ (square x) (square y)))
(define (square-2of3 a b c)
  (cond ((and (>= a b) (>= b c)) (sum-square a b))
        ((and (>= a b) (> c b)) (sum-square a c))
        (else (sum-square b c))))
```

```
<<EX1-3>>
<<try-these>>
 (try-these square-2of3 '(7 5 3)
                        '(7 3 5)
                        '(3 5 7))
```

$$\begin{array}{ll} (7\ 5\ 3) & 74 \\ (7\ 3\ 5) & 74 \\ (3\ 5\ 7) & 74 \end{array}$$

2.10 Exercise 1.4

2.10.1 Question

> Observe that our model of evaluation allows for combinations whose operators are compound expressions. Use this observation to describe the behavior of the following procedure:

```
(define (a-plus-abs-b a b)
  ((if (> b 0) + -) a b))
```

2.10.2 Answer

This code accepts the variables `a` and `b`, and if `b` is positive, it adds `a` and `b`. However, if `b` is zero or negative, it subtracts them. This decision is made by using the `+` and `-` procedures as the results of an if expression, and then evaluating according to the results of that expression. This is in contrast to a language like Python, which would do something like this:

```
if b > 0: a + b
else: a - b
```

2.11 Exercise 1.5

2.11.1 Question

> Ben Bitdiddle has invented a test to determine whether the interpreter he is faced with is using applicative-order evaluation or normal-order evaluation. He defines the following two procedures:

```
(define (p) (p))

(define (test x y)
  (if (= x 0)
      0
      y))
```

Then he evaluates the expression:

```
(test 0 (p))
```

What behavior will Ben observe with an interpreter that uses applicative-order evaluation? What behavior will he observe with an interpreter that uses normal-order evaluation? Explain your answer. (Assume that the evaluation rule for the special form **if** is the same whether the interpreter is using normal or applicative order: The predicate expression is evaluated first, and the result determines whether to evaluate the consequent or the alternative expression.)

2.11.2 Answer

In either type of language, `(define (p) (p))` is an infinite loop. However, a normal-order language will encounter the special form, return `0`, and never evaluate `(p)`. An applicative-order language evaluates the arguments to `(test 0 (p))`, thus triggering the infinite loop.

2.12 1.1.7: Example: Square Roots by Newton's Method

Functions in the formal mathematical sense are **declarative knowledge**, while procedures like in computer science are **imperative knowledge**.

Notice that the elements of the language that have been introduced so far are sufficient for writing any purely numerical program, despite not having introduced any looping constructs like `FOR` loops.

2.13 1.1.8: Procedures as Black-Box Abstractions

Notice how the `sqrt` procedure is divided into other procedures, which mirror the division of the square root problem into sub problems.

A procedure should accomplish an identifiable task, and be ready to be used as a module in defining other procedures. This lets the programmer know how to use the procedure while not needing to know the details of how it works.

Suppressing these details are particularly helpful:

Local names. A procedure user shouldn't need to know a procedure's choices of variable names. A formal parameter of a procedure whose name is irrelevant is called a **bound variable**. A procedure definition **binds** its

parameters. A **free variable** isn't bound. The set of expressions in which a binding defines a name is the **scope** of that name.

Internal definitions and block structure. By nesting relevant definitions inside other procedures, you hide them from the global namespace. This nesting is called **block structure**. Nesting these definitions also allows relevant variables to be shared across procedures, which is called **lexical scoping**.

2.14 Exercise 1.6

2.14.1 Text code

```
(define (abs x)
  (if (< x 0)
      (- x)
      x))
```

```
(define (average x y)
  (/ (+ x y) 2))
```

```
<<average>>
(define (improve guess x)
  (average guess (/ x guess)))

<<square>>
<<abs>>
(define (good-enough? guess x)
  (< (abs (- (square guess) x)) 0.001))

(define (sqrt-iter guess x)
  (if (good-enough? guess x)
      guess
      (sqrt-iter (improve guess x) x)))

(define (sqrt x)
  (sqrt-iter 1.0 x))
```

2.14.2 Question

Alyssa P. Hacker doesn't see why **if** needs to be provided as a special form. "Why can't I just define it as an ordinary procedure in terms of cond?" she asks. Alyssa's friend Eva Lu Ator claims this can indeed be done, and she defines a new version of **if**:

```
(define (new-if predicate
                then-clause
                else-clause)
  (cond (predicate then-clause)
        (else else-clause)))
```

Eva demonstrates the program for Alyssa:

```
(new-if (= 2 3) 0 5)
;; => 5

(new-if (= 1 1) 0 5)
;; => 0
```

Delighted, Alyssa uses `new-if` to rewrite the square-root program:

```
(define (sqrt-iter guess x)
  (new-if (good-enough? guess x)
          guess
          (sqrt-iter (improve guess x) x)))
```

What happens when Alyssa attempts to use this to compute square roots? Explain.

2.14.3 Answer

Using Alyssa's `new-if` leads to an infinite loop because the recursive call to `sqrt-iter` is evaluated before the actual call to `new-if`. This is because **`if`** and **`cond`** are special forms that change the way evaluation is handled; whichever branch is chosen leaves the other branches unevaluated.

2.15 Exercise 1.7

2.15.1 Text

```
(define (mean-square x y)
  (average (square x) (square y)))
```

2.15.2 Question

The `good-enough?` test used in computing square roots will not be very effective for finding the square roots of very small numbers. Also, in real computers, arithmetic operations are almost always performed with limited precision. This makes our test inadequate for very large numbers. Explain these statements, with examples

showing how the test fails for small and large numbers. An alternative strategy for implementing `good-enough?` is to watch how `guess` changes from one iteration to the next and to stop when the change is a very small fraction of the guess. Design a square-root procedure that uses this kind of end test. Does this work better for small and large numbers?

2.15.3 Diary

1. Solving

   My original answer was this, which compares the previous iteration until the new and old are within an arbitrary dx.

```
<<txt-sqrt>>
(define (inferior-good-enough? guess lastguess)
  (<=
   (abs (-
         (/ lastguess guess)
         1))
   0.0000000000001)) ; dx
(define (new-sqrt-iter guess x lastguess) ;; Memory of previous
↪  value
  (if (inferior-good-enough? guess lastguess)
      guess
      (new-sqrt-iter (improve guess x) x guess)))
(define (new-sqrt x)
  (new-sqrt-iter 1.0 x 0))
```

   This solution can correctly find small and large numbers:

```
<<inferior-good-enough>>
(new-sqrt 10000000000000)
```

```
3162277.6601683795
```

```
<<try-these>>
<<inferior-good-enough>>
(try-these new-sqrt '(0.01 0.0001 0.000001 0.00000001 0.0000000001
))
```

0.01	0.1
0.0001	0.01
1e-06	0.001
1e-08	9.999999999999999e-05
1e-10	9.999999999999999e-06

However, I found this solution online that isn't just simpler but automatically reaches the precision limit of the system:

```
<<txt-sqrt>>
(define (best-good-enough? guess x)
   (= (improve guess x) guess))
```

2. Imroving `sqrt` by avoiding extra `improve` call

   (a) Non-optimized

```
(use-modules (ice-9 format))
(load "../mattbench.scm")
(define (average x y)
  (/ (+ x y) 2))
(define (improve guess x)
  (average guess (/ x guess)))
(define (good-enough? guess x)
   (= (improve guess x) guess)) ;; improve call 1
(define (sqrt-iter guess x)
  (if (good-enough? guess x)
      guess
      (sqrt-iter (improve guess x) x))) ;; call 2
(define (sqrt x)
  (sqrt-iter 1.0 x))
(newline)
(display (mattbench (λ() (sqrt 69420)) 400000000))
(newline)
;; 4731.30 <- Benchmark results
```

   (b) Optimized

```
(use-modules (ice-9 format))
(load "../mattbench.scm")
(define (average x y)
  (/ (+ x y) 2))
(define (improve guess x)
  (average guess (/ x guess)))
(define (good-enough? guess nextguess x)
  (= nextguess guess))
(define (sqrt-iter guess x)
  (let ((nextguess (improve guess x)))
    (if (good-enough? guess nextguess x)
        guess
        (sqrt-iter nextguess x))))
(define (sqrt x)
  (sqrt-iter 1.0 x))
(newline)
(display (mattbench (λ() (sqrt 69420)) 400000000))
```

```
18  (newline)
```

(c) Benchmark results

Unoptimized	4731.30
Optimized	2518.44

2.15.4 Answer

The current method has decreasing accuracy with smaller numbers. Notice the steady divergence from correct answers here (should be decreasing powers of 0.1):

```
1  <<txt-sqrt>>
2  <<try-these>>
3  (try-these sqrt 0.01 0.0001 0.000001 0.00000001 0.0000000001)
```

0.01	0.10032578510960605
0.0001	0.03230844833048122
1e-06	0.031260655525445276
1e-08	0.03125010656242753
1e-10	0.03125000106562499

And for larger numbers, an infinite loop will eventually be reached. 10^{12} can resolve, but 10^{13} cannot.

```
1  <<txt-sqrt>>
2  (sqrt 1000000000000)
```

```
1000000.0
```

So, my definition of `sqrt`:

```
1  <<average>>
2  (define (improve guess x)
3    (average guess (/ x guess)))
4  (define (good-enough? guess x)
5     (= (improve guess x) guess))
6  (define (sqrt-iter guess x)
7    (if (good-enough? guess x)
8        guess
9        (sqrt-iter (improve guess x) x)))
10 (define (sqrt x)
11   (sqrt-iter 1.0 x))
```

```
<<try-these>>
<<sqrt>>
(try-these sqrt '(0.01 0.0001 0.000001 0.00000001 0.0000000001))
```

0.01	0.1
0.0001	0.01
1e-06	0.001
1e-08	9.999999999999999e-05
1e-10	9.999999999999999e-06

2.16 Exercise 1.8

2.16.1 Question

Newton's method for cube roots is based on the fact that if y is an approximation to the cube root of x, then a better approximation is given by the value

$$\frac{\frac{x}{y^2} + 2y}{3}$$

Use this formula to implement a cube-root procedure analogous to the `square-root` procedure. (In 1.3.4 we will see how to implement Newton's method in general as an abstraction of these square-root and cube-root procedures.)

2.16.2 Diary

My first attempt works, but needs an arbitrary limit to stop infinite loops:

```
<<square>>
<<try-these>>
(define (cb-good-enough? guess x)
  (= (cb-improve guess x) guess))
(define (cb-improve guess x)
  (/
   (+
    (/ x (square guess))
    (* guess 2))
   3))
(define (cbrt-iter guess x counter)
  (if (or (cb-good-enough? guess x) (> counter 100))
      guess
      (begin
        (cbrt-iter (cb-improve guess x) x (+ 1 counter)))))
(define (cbrt x)
  (cbrt-iter 1.0 x 0))
```

```
18
19  (try-these cbrt 7 32 56 100)
```

7	1.912931182772389
32	3.174802103936399
56	3.825862365544778
100	4.641588833612779

However, this will hang on an infinite loop when trying to run `(cbrt 100)`. I speculate it's a floating point precision issue with the "improve" algorithm. So to avoid it I'll just keep track of the last guess and stop improving when there's no more change occurring. Also while researching I discovered that (again due to floating point) `(cbrt -2)` loops forever unless you initialize your guess with a slightly different value, so let's do 1.1 instead.

2.16.3 Answer

```
1   <<square>>
2   (define (cb-good-enough? nextguess guess lastguess x)
3     (or (= nextguess guess)
4         (= nextguess lastguess)))
5   (define (cb-improve guess x)
6     (/
7      (+
8       (/ x (square guess))
9       (* guess 2))
10     3))
11  (define (cbrt-iter guess lastguess x)
12    (define nextguess (cb-improve guess x))
13    (if (cb-good-enough? nextguess guess lastguess x)
14        nextguess
15        (cbrt-iter nextguess guess x)))
16  (define (cbrt x)
17    (cbrt-iter 1.1 9999 x))
```

```
1   <<cbrt>>
2   <<try-these>>
3   (try-these cbrt 7 32 56 100 -2)
```

7	1.912931182772389
32	3.174802103936399
56	3.825862365544778
100	4.641588833612779
-2	-1.2599210498948732

2.17 1.2: Procedures and the Processes They Generate

Procedures define the **local evolution** of processes. We would like to be able to make statements about the **global** behavior of a process.

2.18 1.2.1: Linear Recursion and Iteration

Consider these two procedures for obtaining factorials:

```
(define (factorial-recursion n)
  (if (= n 1)
      1
      (* n
         (factorial-recursion (- n 1)))))

(define (factorial-iteration n)
  (define (fact-iter product counter max-count)
      (if (> counter max-count)
          product
          (fact-iter
                    (* counter product)
                    (+ counter 1)
                    max-count)))

  (fact-iter 1 1 n))
```

These two procedures reach the same answers, but form very different processes. The `factorial-recursion` version takes more computational **time** and **space** to evaluate, by building up a chain of deferred operations. This is a **recursive process**. As the number of steps needed to operate, and the amount of info needed to keep track of these operations, both grow linearly with n, this is a **linear recursive process**.

The second version forms an **iterative process**. Its state can be summarized with a fixed number of state variables. The number of steps required grow linearly with n, so this is a **linear iterative process**.

recursive procedure is a procedure whose definition refers to itself.

recursive process is a process that evolves recursively.

So `fact-iter` is a recursive *procedure* that generates an iterative *process*.

Many implementations of programming languages interpret all recursive procedures in a way that consume memory that grows with the number of procedure calls, even when the process is essentially iterative. These languages instead use looping constructs such as `do`, `repeat`, `for`, etc. Implementations that execute iterative processes in constant space, even if the procedure is recursive, are **tail-recursive**.

2.19 Exercise 1.9

2.19.1 Question

> Each of the following two procedures defines a method for adding two positive integers in terms of the procedures `inc`, which increments its argument by 1, and `dec`, which decrements its argument by 1.

```
(define (+ a b)
  (if (= a 0)
      b
      (inc (+ (dec a) b))))

(define (+ a b)
  (if (= a 0)
      b
      (+ (dec a) (inc b))))
```

> Using the substitution model, illustrate the process generated by each procedure in evaluating `(+ 4 5)`. Are these processes iterative or recursive?

2.19.2 Answer

The first procedure is recursive, while the second is iterative though tail-recursion.

1. recursive procedure

```
(+ 4 5)
(inc (+ 3 5))
(inc (inc (+ 2 5)))
(inc (inc (inc (+ 1 5))))
(inc (inc (inc (inc (+ 0 5)))))
(inc (inc (inc (inc 5))))
(inc (inc (inc 6)))
(inc (inc 7))
(inc 8)
9
```

2. iterative procedure

```
(+ 4 5)
(+ 3 6)
(+ 2 7)
(+ 1 8)
(+ 0 9)
9
```

2.20 Exercise 1.10

2.20.1 Question

The following procedure computes a mathematical function called Ackermann's function.

```
(define (A x y)
  (cond ((= y 0) 0)
        ((= x 0) (* 2 y))
        ((= y 1) 2)
        (else (A (- x 1)
                 (A x (- y 1))))))
```

What are the values of the following expressions?

```
(A 1 10)
(A 2 4)
(A 3 3)
```

(1 10)	1024
(2 4)	65536
(3 3)	65536

```
<<ackermann>>
(define (f n) (A 0 n))
(define (g n) (A 1 n))
(define (h n) (A 2 n))
(define (k n) (* 5 n n))
```

Give concise mathematical definitions for the functions computed by the procedures `f`, `g`, and `h` for positive integer values of n. For example, `(k n)` computes $5n^2$.

2.20.2 Answer

1. `f`

```
<<try-these>>
<<EX1-10-defs>>
(try-these f 1 2 3 10 15 20)
```

1	2
2	4
3	6
10	20
15	30
20	40

$$f(n) = 2n$$

2. g

```
<<try-these>>
<<EX1-10-defs>>
(try-these g 1 2 3 4 5 6 7 8)
```

1	2
2	4
3	8
4	16
5	32
6	64
7	128
8	256

$$g(n) = 2^n$$

3. h

```
<<try-these>>
<<EX1-10-defs>>
(try-these h 1 2 3 4)
```

1	2
2	4
3	16
4	65536

It took a while to figure this one out, just because I didn't know the term. This is repeated exponentiation. This operation is to exponentiation, what exponentiation is to multiplication. It's called either *tetration* or *hyper-4* and has no formal notation, but two common ways would be these:

$$h(n) = 2 \uparrow\uparrow n$$

$$h(n) = {}^n2$$

2.21 1.2.2: Tree Recursion

Consider a recursive procedure for computing Fibonacci numbers:

```
(define (fib n)
  (cond ((= n 0) 0)
        ((= n 1) 1)
        (else (+ (fib (- n 1))
                 (fib (- n 2))))))
```

The resulting process splits into two with every iteration, creating a tree of computations, many of which are duplicates of previous computations. This kind of pattern is called **tree-recursion**. However, this one is quite inefficient. The time and space required grows exponentially with the number of iterations requested.

Instead, it makes much more sense to start from `Fib(1) ~ 1` and `Fib(0) ~ 0` and iterate upwards to the desired value. This only requires a linear number of steps relative to the input.

```
(define (fib n)
  (fib-iter 1 0 n))
(define (fib-iter a b count)
  (if (= count 0) b (fib-iter (+ a b) a (- count 1))))
```

However, notice that the inefficient tree-recursive version is a fairly straightforward translation of the Fibonacci sequence's definition, while the iterative version required redefining the process as an iteration with three variables.

2.21.1 Example: Counting change

I should come back and try to make the "better algorithm" suggested.

2.22 Exercise 1.11

2.22.1 Question

A function f is defined by the rule that:

$$f(n) = n \text{ if } n < 3$$

and

$$f(n) = f(n-1) + 2f(n-2) + 3f(n-3) \text{ if } n \geq 3$$

Write a procedure that computes f by means of a recursive process.
Write a procedure that computes f by means of an iterative process.

2.22.2 Answer

1. Recursive

```
(define (fr n)
  (if (< n 3)
      n
      (+      (fr (- n 1))
         (* 2 (fr (- n 2)))
         (* 3 (fr (- n 3))))))
```

```
<<try-these>>
<<EX1-11-fr>>
(try-these fr 1 3 5 10)
```

1	1
3	4
5	25
10	1892

2. Iterative

   (a) Attempt 1

```
;; This seems like it could be better
(define (fi n)
  (define (formula l)
    (let ((a (car l))
          (b (cadr l))
          (c (caddr l)))
      (+ a
         (* 2 b)
         (* 3 c))))
  (define (iter l i)
    (if (= i n)
        (car l)
        (iter (cons (formula l) l)
              (+ 1 i))))
  (if (< n 3)
      n
      (iter '(2 1 0) 2)))
```

```
<<try-these>>
<<EX1-11-fi>>
(try-these fi 1 3 5 10)
```

1	1
3	4
5	25
10	1892

It works but it seems wasteful.

(b) Attempt 2

```
(define (fi2 n)
  (define (formula a b c)
      (+ a
         (* 2 b)
         (* 3 c)))
  (define (iter a b c i)
    (if (= i n)
        a
        (iter (formula a b c)
              a
              b
              (+ 1 i))))
  (if (< n 3)
      n
      (iter 2 1 0 2)))
```

```
<<try-these>>
<<EX1-11-fi2>>
(try-these fi2 1 3 5 10)
```

1	1
3	4
5	25
10	1892

I like that better.

2.23 Exercise 1.12

2.23.1 Question

The following pattern of numbers is called Pascal's triangle.

```
        1
      1   1
    1   2   1
  1   3   3   1
1   4   6   4   1
      . . .
```

The numbers at the edge of the triangle are all 1, and each number inside the triangle is the sum of the two numbers above it. Write a procedure that computes elements of Pascal's triangle by means of a recursive process.

2.23.2 Answer

I guess I'll rotate the triangle 45 degrees to make it the corner of an infinite spreadsheet.

```
(define (pascal x y)
  (if (or (= x 0)
          (= y 0))
      1
      (+ (pascal (- x 1) y)
         (pascal x (- y 1)))))
```

```
<<try-these>>
<<pascal-rec>>
(let ((l (iota 8)))
  (map (λ (row)
         (map (λ (xy)
                (apply pascal xy))
              row))
       (map (λ (x)
              (map (λ (y)
                     (list x y))
                   l))
            l)))
```

1	1	1	1	1	1	1	1
1	2	3	4	5	6	7	8
1	3	6	10	15	21	28	36
1	4	10	20	35	56	84	120
1	5	15	35	70	126	210	330
1	6	21	56	126	252	462	792
1	7	28	84	210	462	924	1716
1	8	36	120	330	792	1716	3432

The test code was much harder to write than the actual solution.

2.24 Exercise 1.13 optional

2.24.1 Question

Prove that Fib(n) is the closest integer to $\frac{\Phi}{n}\sqrt{5}$ where Φ is $\frac{1+\sqrt{5}}{2}$. Hint: let $\Upsilon = \frac{1-\sqrt{5}}{2}$. Use induction and the definition of the Fibonacci numbers to prove that

$$\text{Fib}(n) = \frac{\Phi^n - \Upsilon^n}{\sqrt{5}}$$

2.24.2 Answer

I don't know how to write a proof yet, but I can make functions to demonstrate it.

1. Fibonacci number generator

```
(define (fib-iter n)
  (define (iter i a b)
    (if (= i n)
        b
    (iter (+ i 1)
          b
          (+ a b))))
  (if (<= n 2)
      1
      (iter 2 1 1)))
```

2. Various algorithms relating to the question

```
<<sqrt>>
(define sqrt5
  (sqrt 5))
(define phi
  (/ (+ 1 sqrt5) 2))
(define upsilon
  (/ (- 1 sqrt5) 2))
(define (fib-phi n)
  (/ (- (expt phi n)
        (expt upsilon n))
     sqrt5))
```

```
(use-srfis '(1))
<<fib-iter>>
<<fib-phi>>
<<try-these>>

(let* ((vals (drop (iota 21) 10))
       (fibs (map fib-iter vals))
       (approx (map fib-phi vals)))
  (zip vals fibs approx))
```

10	55	54.99999999999999
11	89	89.0
12	144	143.99999999999997
13	233	232.99999999999994
14	377	377.00000000000006
15	610	610.0
16	987	986.9999999999998
17	1597	1596.9999999999998
18	2584	2584.0
19	4181	4181.0
20	6765	6764.999999999999

You can see they follow closely. Graphing the differences, it's just an exponential curve at very low values, presumably following the exponential increase of the Fibonacci sequence itself.



2.25 1.2.3: Orders of Growth

An **order of growth** gives you a gross measure of the resources required by a process as its inputs grow larger.

Let n be a parameter for the size of a problem, and $R(n)$ be the amount of resources required for size n. $R(n)$ has order of growth $\Theta(f(n))$

For example:

$\Theta(1)$ is constant, not growing regardless of input size.

$\Theta(n)$ is growth 1-to-1 proportional to the input size.

Some algorithms we've already seen:

Linear recursive is time and space $\Theta(n)$

Iterative is time $\Theta(n)$ space $\Theta(1)$

Tree-recursive means in general, time is proportional to the number of nodes, space is proportional to the depth of the tree. In the Fibonacci algorithm example, $\Theta(n)$ and time $\Theta(\Upsilon^n)$ where Υ is the golden ratio $\frac{1+\sqrt{5}}{2}$

Orders of growth are very crude descriptions of process behaviors, but they are useful in indicating how a process will change with the size of the problem.

2.26 Exercise 1.14

2.26.1 Text

Below is the default version of the `count-change` function. I'll be aggressively modifying it in order to get a graph out of it.

```
(define (count-change amount)
  (cc amount 5))

(define (cc amount kinds-of-coins)
  (cond ((= amount 0) 1)
        ((or (< amount 0)
             (= kinds-of-coins 0))
         0)
        (else
         (+ (cc amount (- kinds-of-coins 1))
            (cc (- amount (first-denomination
                           kinds-of-coins))
                kinds-of-coins)))))

(define (first-denomination kinds-of-coins)
  (cond ((= kinds-of-coins 1) 1)
        ((= kinds-of-coins 2) 5)
        ((= kinds-of-coins 3) 10)
        ((= kinds-of-coins 4) 25)
        ((= kinds-of-coins 5) 50)))
```

2.26.2 Question A

> Draw the tree illustrating the process generated by the `count-change` procedure of 1.2.2 in making change for 11 cents.

2.26.3 Answer

I want to generate this graph algorithmically.

```scheme
;; cursed global
(define bubblecounter 0)
;; Returns # of ways change can be made
;; "Helper" for (cc)
(define (count-change amount)
  (display "digraph {\n") ;; start graph
  (cc amount 5 0)
  (display "}\n") ;; end graph
  (set! bubblecounter 0))

;; GraphViz output
;; Derivative: https://stackoverflow.com/a/14806144
(define (cc amount kinds-of-coins oldbubble)
  (let ((recur (lambda (new-amount new-kinds)
                 (begin
                   (display "\"") ;; Source bubble
                   (display `(,oldbubble ,amount ,kinds-of-coins))
                   (display "\"")
                   (display " -> ") ;; arrow pointing from parent to
↪  child
                   (display "\"") ;; child bubble
                   (display `(,bubblecounter ,new-amount ,new-kinds))
                   (display "\"")
                   (display "\n")
                   (cc new-amount new-kinds bubblecounter)))))
    (set! bubblecounter (+ bubblecounter 1))
    (cond ((= amount 0) 1)
          ((or (< amount 0) (= kinds-of-coins 0)) 0)
          (else (+
                 (recur amount (- kinds-of-coins 1))
                 (recur (- amount
                           (first-denomination kinds-of-coins))
                        kinds-of-coins))))))

(define (first-denomination kinds-of-coins)
  (cond ((= kinds-of-coins 1) 1)
        ((= kinds-of-coins 2) 5)
        ((= kinds-of-coins 3) 10)
        ((= kinds-of-coins 4) 25)
        ((= kinds-of-coins 5) 50)))
```

I'm not going to include the full printout of the `(count-change 11)`, here's an example of what this looks like via `1`.

```scheme
<<count-change-graphviz>>
(count-change 1)
```

```
digraph {
"(0 1 5)" -> "(1 1 4)"
"(1 1 4)" -> "(2 1 3)"
"(2 1 3)" -> "(3 1 2)"
"(3 1 2)" -> "(4 1 1)"
"(4 1 1)" -> "(5 1 0)"
"(4 1 1)" -> "(6 0 1)"
"(3 1 2)" -> "(7 -4 2)"
"(2 1 3)" -> "(8 -9 3)"
"(1 1 4)" -> "(9 -24 4)"
"(0 1 5)" -> "(10 -49 5)"
}
```



So, the graph of `(count-change 11)` is:



2.26.4 Question B

> What are the orders of growth of the space and number of steps used by this process as the amount to be changed increases?

2.26.5 Answer B

Let's look at this via the number of function calls needed for value `n`. Instead of returning an integer, I'll return a pair where `car` is the number of ways to count change, and `cdr` is the number of function calls that have occurred down that branch of the tree.

```scheme
(define (count-calls amount)
  (cc-calls amount 5))

(define (cc-calls amount kinds-of-coins)
  (cond ((= amount 0) '(1 . 1))
        ((or (< amount 0)
             (= kinds-of-coins 0))
         '(0 . 1))
        (else
         (let ((a (cc-calls amount (- kinds-of-coins 1)))
               (b (cc-calls (- amount (first-denomination
                                       kinds-of-coins))
                            kinds-of-coins)))
           (cons (+ (car a)
                    (car b))
                 (+ 1
                    (cdr a)
                    (cdr b)))))))

(define (first-denomination kinds-of-coins)
  (cond ((= kinds-of-coins 1) 1)
        ((= kinds-of-coins 2) 5)
        ((= kinds-of-coins 3) 10)
        ((= kinds-of-coins 4) 25)
        ((= kinds-of-coins 5) 50)))
```

```scheme
(use-srfis '(1))
<<cc-calls>>
(let* ((vals (drop (iota 101) 1))
       (mine (map count-calls vals)))
  (zip vals (map car mine) (map cdr mine)))
```



I believe the space to be $\Theta(n + d)$ as the function calls count down the denominations before counting down the change. However I notice most answers describe $\Theta(n)$ instead, maybe I'm being overly pedantic and getting the wrong answer.

My issues came finding the time. The book describes the meaning and properties of Θ notation in Section 1.2.3. However, my lack of formal math education made realizing the significance of this passage difficult. For one, I didn't understand that $k_1 f(n) \leq R(n) \leq k_2 f(n)$ means "you can find the Θ by proving that a graph of the algorithm's resource usage is bounded by two identical functions multiplied by constants." So, the graph of resource usage for an algorithm with $\Theta(n^2)$ will by bounded by lines of $n^2 \times someconstant$, the top boundary's constant being larger than the small boundary. These are arbitrarily chosen constants, you're just proving that the function behaves the way you think it does.

Overall, finding the Θ and Ω and O notations (they are all different btw!) is about aggressively simplifying to make a very general statement about the behavior of the algorithm.

I could tell that a "correct" way to find the Θ would be to make a formula which describes the algorithm's function calls for given input and denominations. This is one of the biggest time sinks, although I had a lot of fun and learned a lot. In the end, with some help from Jach in a Lisp Discord, I had the following formula:

$$\sum_{i=1}^{ceil(n/val(d))} T(n - val(d) * i, d)$$

But I wasn't sure where to go from here. The graphs let me see some interesting trends, though I didn't get any closer to an answer in the process.

By reading on other websites, I knew that you could find Θ by obtaining a formula for $R(n)$ and removing constants to end up with a term of interest. For example, if your algorithm's resource usage is $\frac{n^2+7n}{5}$, this demonstrates $\Theta(n^2)$. So I know a formula **without** a $\sum$ would give me the answer I wanted. It didn't occur to me that it might be possible to use calculus to remove the $\sum$ from the equation. At this point I knew I was stuck and decided to look up a guide.

After seeing a few solutions that I found somewhat confusing, I landed on this awesome article from Codology.net. They show how you can remove the summation, and proposed this equation for count-change with 5 denominations:

$$T(n,5) = \frac{n}{50} + 1 + \sum_{i=0}^{n/50} T(n - 50i, 1)$$

Which, when expanded and simplified, demonstrates $\Theta(n^5)$ for 5 denominations.

Overall I'm relieved that I wasn't entirely off, given I haven't done math work like this since college. It's inspired me to restart my remedial math courses, I don't think I really grasped the nature of math as a tool of empowerment until now.

2.27 Exercise 1.15

2.27.1 Question A

> The sine of an angle (specified in radians) can be computed by making use of the approximation $\sin x \approx x$ if x is sufficiently small, and the trigonometric identity $\sin x = 3\sin\frac{x}{3} - 4\sin^3\frac{x}{3}$ to reduce the size of the argument of sin. (For purposes of this exercise an angle is considered "sufficiently small" if its magnitude is not greater than 0.1 radians.) These ideas are incorporated in the following procedures:

```
(define (cube x) (* x x x))
(define (p x) (- (* 3 x) (* 4 (cube x))))
(define (sine angle)
  (if (not (> (abs angle) 0.1))
      angle
      (p (sine (/ angle 3.0)))))
```

> How many times is the procedure `p` applied when `(sine 12.15)` is evaluated?

2.27.2 Answer A

Let's find out!

```
(define (cube x) (* x x x))
(define (p x) (- (* 3 x) (* 4 (cube x))))
(define (sine angle)
  (if (not (> (abs angle) 0.1))
      (cons angle 0)
      (let ((x (sine (/ angle 3.0))))
        (cons (p (car x)) (+ 1 (cdr x))))))
```

```
<<1-15-p-measure>>
(let ((xy (sine 12.15)))
  (list (car xy) (cdr xy)))
```

-0.39980345741334 5

`p` is evaluated 5 times.

2.27.3 Question B

> What is the order of growth in space and number of steps (as a function of `a`) used by the process generated by the sine procedure when `(sine a)` is evaluated?

2.27.4 Answer B

```
(use-srfis '(1))
<<1-15-p-measure>>
(let* ((vals (iota 300 0.1 0.1))
       (sines (map (λ (i)
                     (cdr (sine i)))
                   vals)))
  (zip vals sines))
```

```
(use-srfis '(1))
<<1-15-p-measure>>
(let* ((vals (iota 10 0.1 0.1))
       (sines (map (λ (i)
                     (cdr (sine i)))
                   vals)))
  (zip vals sines))
```

Example output:

0.1	0
0.2	1
0.30000000000000004	2
0.4	2
0.5	2
0.6	2
0.7000000000000001	2
0.8	2
0.9	2
1.0	3



This graph shows that the number of times `sine` will be called is logarithmic.

- 0.1 to 0.2 are divided once
- 0.3 to 0.8 are divided twice
- 0.9 to 2.6 are divided three times
- 2.7 to 8 are divided four times
- 8.5 to 23.8 are divided five times

Given that the calls to `p` get stacked recursively, like this:

```
(sine 12.15)
(p (sine 4.05))
(p (p (sine 1.35)))
```

```
(p (p (p (sine 0.45))))
(p (p (p (p (sine 0.15)))))
(p (p (p (p (p (sine 0.05))))))
(p (p (p (p (p 0.05)))))
(p (p (p (p 0.14950000000000002))))
(p (p (p 0.43513455050000005)))
(p (p 0.9758465331678772))
(p -0.7895631144708228)
-0.39980345741334
```

So I argue the space and time is $\Theta(\log(n))$

We can also prove this for the time by benchmarking the function:

```
;; This execution takes too long for org-mode, so I'm doing it
;; externally and importing the results
(use-srfis '(1))
(use-modules (ice-9 format))
(load "../../mattbench.scm")
<<1-15-deps>>
(let* ((vals (iota 300 0.1 0.1))
       (times (map (λ (i)
                     (mattbench (λ () (sine i)) 1000000))
                   vals)))
  (with-output-to-file "sine-bench.dat" (λ ()
     (map (λ (x y)
            (format #t "~s~/~s~%" x y))
          vals times))))
```

```
reset # helps with various issues in execution
set xtics 0.5
set xlabel 'values of x'
set logscale x
set key top left
set style fill solid 1.00 border
#set style function fillsteps below

f(x) = (log(x) * a) + b
fit f(x) 'Ex15/sine-bench.dat' using 1:2 via a,b

plot 'Ex15/sine-bench.dat' using 1:2 with fillsteps title
↪  'time to execute', \
     'Ex15/sine-bench.dat' using 1:(f($1)) with lines title
↪  sprintf('(log(x) * %.2f) + %.2f', a, b)
```



1. 1.2.4 Exponentiation

   Considering a few ways to compute the exponential of a given number.

```
(define (expt b n)
  (expt-iter b n 1))
(define (expt-iter b counter product)
  (if (= counter 0)
      product
      (expt-iter b (- counter 1) (* b product))))
```

   This iterative procedure is essentially equivalent to:

$$b^8 = b \cdot (b \cdot (b \cdot (b \cdot (b \cdot (b \cdot b)))))$$

   But note it could be approached faster with squaring:

$$b^2 = b \cdot b$$
$$b^4 = b^2 \cdot b^2$$
$$b^8 = b^4 \cdot b^4$$

2.28 Exercise 1.16

2.28.1 Text

```
(define (expt-rec b n)
  (if (= n 0)
      1
      (* b (expt-rec b (- n 1)))))

(define (expt-iter b n)
  (define (iter counter product)
    (if (= counter 0)
        product
        (iter (- counter 1)
              (* b product))))
  (iter n 1))

(define (fast-expt b n)
  (cond ((= n 0)
         1)
        ((even? n)
         (square (fast-expt b (/ n 2))))
        (else
         (* b (fast-expt b (- n 1))))))
```

2.28.2 Question

Design a procedure that evolves an iterative exponentiation process that uses successive squaring and uses a logarithmic number of steps, as does fast-expt. (Hint: Using the observation that $(b^{n/2})^2 = (b^2)^{n/2}$, keep, along with the exponent n and the base b, an additional state variable a , and define the state transformation in such a way that the product ab^n is unchanged from state to state. At the beginning of the process a is taken to be 1, and the answer is given by the value of a at the end of the process. In general, the technique of defining an *invariant quantity* that remains unchanged from state to state is a powerful way to think about the design of iterative algorithms.)

2.28.3 Diary

First I made this program which tries to use a false equivalence:

$$ab^2 = (a+1)b^{n-1}$$

```
<<square>>
(define (fast-expt-iter b n)
  (define (iter b n a)
    (format #t "~&|~s~/~/|~s~/~/|~s|~%" b n a)
```

```
    (cond ((= n 1) (begin (format #t "~&|~s~/~/|~s~/~/|~s|~%" (* b a) 1
↪  1)
                          (* b a)))
          ((even? n) (iter (square b)
                           (/ n 2)
                           a))
          (else (iter b (- n 1) (+ a 1)))))
  (format #t "|~a~/|~a~/|~a|~%" "base" "power" "variable")
  (format #t "~&|--|--|--|~%")
  (iter b n 1))
```

```
<<fast-expt-iter-fail1>>
<<try-these>>
(fast-expt-iter 2 6)
```

Here's what the internal state looks like during 2^6 (correct answer is 64):

base	power	variable
2	6	1
4	3	1
4	2	2
16	1	2
32	1	1

2.28.4 Answer

There are two key transforms to a faster algorithm. The first was already shown in the text:

$$ab^n \rightarrow a(b^2)^{n/2}$$

The second which I needed to deduce was this:

$$ab^n \rightarrow ((a \times b) \times b)^{n-1}$$

The solution essentially follows this logic:

- initialize a to 1
- If n is 1, return $b * a$
- else if n is even, halve n, square b, and iterate
- else n is odd, so subtract 1 from n and $a \rightarrow a \times b$

```
<<square>>
(define (fast-expt-iter b n)
  (define (iter b n a)
    (cond ((= n 1) (* b a))
          ((even? n) (iter (square b)
                           (/ n 2)
                           a))
          (else (iter b (- n 1) (* b a)))))
  (iter b n 1))
```

```
<<fast-expt-iter>>
<<try-these>>
(try-these (λ(x) (fast-expt-iter 3 x)) (cdr (iota 11)))
```

1	3
2	9
3	27
4	81
5	243
6	729
7	2187
8	6561
9	19683
10	59049

2.29 Exercise 1.17

2.29.1 Question

The exponentiation algorithms in this section are based on performing exponentiation by means of repeated multiplication. In a similar way, one can perform integer multiplication by means of repeated addition. The following multiplication procedure (in which it is assumed that our language can only add, not multiply) is analogous to the expt procedure:

```
(define (* a b)
  (if (= b 0)
      0
      (+ a (* a (- b 1)))))
```

This algorithm takes a number of steps that is linear in b. Now suppose we include, together with addition, operations double, which doubles an integer, and halve, which divides an (even) integer by 2. Using these, design a multiplication procedure analogous to fast-expt that uses a logarithmic number of steps.

2.29.2 Answer

```
(define (double x)
  (+ x x))
(define (halve x)
  (/ x 2))
(define (fast-mult-rec a b)
  (cond ((= b 0) 0)
        ((even? b)
         (double (fast-mult-rec a (halve b)))) ; This was kind of a stretch to think of.G
         ;(fast-mult (double a) (halve b))) <== My first instinct is iterative
        (else (+ a (fast-mult-rec a (- b 1))))))
```

Proof it works:

```
<<fast-mult-rec>>
<<try-these>>
(try-these (λ(x) (fast-mult-rec 3 x)) (cdr (iota 11)))
```

1	3
2	6
3	9
4	12
5	15
6	18
7	21
8	24
9	27
10	30

2.30 Exercise 1.18

2.30.1 Question

Using the results of Exercise 1.16 and Exercise 1.17, devise a procedure that generates an iterative process for multiplying two integers in terms of adding, doubling, and halving and uses a logarithmic number of steps.

2.30.2 Diary

1. Comparison benchmarks:

```
(load "../mattbench.scm")
<<fast-mult-iter>>
<<fast-mult-rec>>
```

```
<<print-table>>
(print-table (list (list "fast-mult-rec" "fast-mult-iter")
                   (list (mattbench (λ() (fast-mult-rec 32 32)) 10000000)
                         (mattbench (λ() (fast-mult 32 32)) 10000000)))
             #:colnames #t)
```

"fast-mult-rec"	"fast-mult-iter"
196.89	166.35

So the iterative version takes 0.84 times less to do 32×32.

2. Hall of shame

   Some of my *very* incorrect ideas:

$$ab = (a+1)(b-1)$$

$$ab = \left(a + \left(\frac{a}{2}\right)(b-1)\right)$$

$$ab + c = \big(a(b-1) + (b+c)\big)$$

2.30.3 Answer

```
(define (double x)
  (+ x x))
(define (halve x)
  (/ x 2))
(define (fast-mult a b)
  (define (iter a b c)
    (cond ((= b 0) 0)
          ((= b 1) (+ a c))
          ((even? b)
           (iter (double a) (halve b) c))
          (else (iter a (- b 1) (+ a c)))))
  (iter a b 0))
```

```
<<fast-mult-iter>>
<<try-these>>
(try-these (λ(x) (fast-mult 3 x)) (cdr (iota 11)))
```

1	3
2	6
3	9
4	12
5	15
6	18
7	21
8	24
9	27
10	30

2.31 Exercise 1.19

2.31.1 Question

There is a clever algorithm for computing the Fibonacci numbers in a logarithmic number of steps. Recall the transformation of the state variables a and b in the `fib-iter` process of section 1-2-2:

$$a < -a + b \text{ and } b < -a$$

Call this transformation T, and observe that applying T over and over again n times, starting with 1 and 0, produces the pair Fib(n+1) and Fib(n). In other words, the Fibonacci numbers are produced by applying T^n, the nth power of the transformation T, starting with the pair (1,0). Now consider T to be the special case of p = 0 and q = 1 in a family of transformations $T_{(pq)}$, where $T_{(pq)}$ transforms the pair (a,b) according to $a < -bq + aq + ap$ and $b < -bp + aq$. Show that if we apply such a transformation $T_{(pq)}$ twice, the effect is the same as using a single transformation $T_{(p'q')}$ of the same form, and compute p' and q' in terms of p and q. This gives us an explicit way to square these transformations, and thus we can compute T^n using successive squaring, as in the 'fast-expt' procedure. Put this all together to complete the following procedure, which runs in a logarithmic number of steps:

```
(define (fib n)
  (fib-iter 1 0 0 1 n))

(define (fib-iter a b p q count)
  (cond ((= count 0) b)
        ((even? count)
         (fib-iter a
                   b
                   <??>      ; compute p'
```

```
                    <??>      ; compute q'
                    (/ count 2)))
          (else (fib-iter (+ (* b q) (* a q) (* a p))
                          (+ (* b p) (* a q))
                          p
                          q
                          (- count 1)))))
```

2.31.2 Diary

More succinctly put:

$$\text{Fib}_n \begin{cases} a \leftarrow a + b \\ b \leftarrow a \end{cases}$$

$$\text{Fib-iter}_{abpq} \begin{cases} a \leftarrow bq + aq + ap \\ b \leftarrow bp + aq \end{cases}$$

`(T)` returns a transformation function based on the two numbers in the attached list. so `(T 0 1)` returns a fib function.

```
(define (T p q)
  (λ (a b)
    (cons (+ (* b q) (* a q) (* a p))
          (+ (* b p) (* a q)))))

(define T-fib
  (T 0 1))

;; Repeatedly apply T functions:
(define (Tr f n)
  (Tr-iter f n 0 1))
(define (Tr-iter f n a b)
  (if (= n 0)
      a
      (let ((l (f a b)))
        (Tr-iter f (- n 1) (car l) (cdr l)))))
```

$$\text{T}_{\text{pq}}\text{: a,b} \mapsto \begin{cases} a \leftarrow \text{bq} + \text{aq} + \text{ap} \\ b \leftarrow \text{bp} + \text{aq} \end{cases}$$

```
<<T-func>>
<<try-these>>
(try-these (λ (x) (Tr (T 0 1) x)) (cdr (iota 11)))
```

1	1
2	1
3	2
4	3
5	5
6	8
7	13
8	21
9	34
10	55

2.31.3 Answer

```
(define (fib-rec n)
  (cond ((= n 0) 0)
        ((= n 1) 1)
        (else (+ (fib-rec (- n 1))
                 (fib-rec (- n 2))))))
(define (fib n)
  (fib-iter 1 0 0 1 n))

(define (fib-iter a b p q count)
  (cond ((= count 0) b)
        ((even? count)
         (fib-iter a
                   b
                   (+ (* p p)
                      (* q q))      ; compute p'
                   (+ (* p q)
                      (* q q)
                      (* q p))      ; compute q'
                   (/ count 2)))
        (else (fib-iter (+ (* b q) (* a q) (* a p))
                        (+ (* b p) (* a q))
                        p
                        q
                        (- count 1)))))
```

"n"	"fib-rec"	"fib-iter"
1	1	1
2	1	1
3	2	2
4	3	3
5	5	5
6	8	8
7	13	13
8	21	21
9	34	34

2.32 1.2.5: Greatest Common Divisor

A greatest common divisor (or GCD) for two integers is the largest integer that divides both of them. A GCD can be quickly found by transforming the problem like so:

$$a\%b = r$$

$$\text{GCD}(a, b) = \text{GCD}(b, r)$$

This eventually produces a pair where the second number is 0. Then, the GCD is the other number in the pair. This is Euclid's Algorithm.

$$\begin{aligned}\text{GCD}(206, 40) &= \text{GCD}(40, 6)\\ &= \text{GCD}(6, 4)\\ &= \text{GCD}(4, 2)\\ &\text{GCD}(2, 0)\ 2\end{aligned}$$

Lamé's Theorem: If Euclid's Algorithm requires k steps to compute the GCD of some pair, then the smaller number in the pair must be greater than or equal to the k^{th}Fibonacci number.

2.33 Exercise 1.20

2.33.1 Text

```
(define (gcd a b)
  (if (= b 0)
      a
      (gcd b (remainder a b))))
```

2.33.2 Question

The process that a procedure generates is of course dependent on the rules used by the interpreter. As an example, consider the iterative `gcd` procedure given above. Suppose we were to interpret this procedure using normal-order evaluation, as discussed in 1.1.5: The Substitution Model for Procedure Application. (The normal-order-evaluation rule for **`if`** is described in Exercise 1.5.) Using the substitution method (for normal order), illustrate the process generated in evaluating `(gcd 206 40)` and indicate the `remainder` operations that are actually performed. How many `remainder` operations are actually performed in the normal-order evaluation of `(gcd 206 40)`? In the applicative-order evaluation?

2.33.3 Answer

I struggled to understand this, but the key here is that normal-order evaluation causes the unevaluated expressions to be duplicated, meaning they get evaluated multiple times.

1. Applicative order

```
call (gcd 206 40)
(if)
(gcd 40 (remainder 206 40))
eval remainder before call
call (gcd 40 6)
(if)
(gcd 6 (remainder 40 6))
eval remainder before call
call (gcd 6 4)
(if)
(gcd 2 (remainder 4 2))
eval remainder before call
call (gcd 2 0)
(if)
;; => 2
```

```
;; call gcd
(gcd 206 40)

;; eval conditional
(if (= 40 0)
    206
    (gcd 40 (remainder 206 40)))

;; recurse
(gcd 40 (remainder 206 40))

; encounter conditional
(if (= (remainder 206 40) 0)
    40
    (gcd (remainder 206 40)
         (remainder 40 (remainder 206 40))))

; evaluate 1 remainder
(if (= 6 0)
    40
    (gcd (remainder 206 40)
         (remainder 40 (remainder 206 40))))

; recurse
```

```
(gcd (remainder 206 40)
     (remainder 40 (remainder 206 40)))

; encounter conditional
(if (= (remainder 40 (remainder 206 40)) 0)
    (remainder 206 40)
    (gcd (remainder 40 (remainder 206 40))
         (remainder (remainder 206 40) (remainder 40 (remainder
↪  206 40)))))

; eval 2 remainder
(if (= 4 0)
    (remainder 206 40)
    (gcd (remainder 40 (remainder 206 40))
         (remainder (remainder 206 40) (remainder 40 (remainder
↪  206 40)))))

; recurse
(gcd (remainder 40 (remainder 206 40))
     (remainder (remainder 206 40) (remainder 40 (remainder 206
↪  40))))

; encounter conditional
(if (= (remainder (remainder 206 40) (remainder 40 (remainder 206
↪  40))) 0)
    (remainder 40 (remainder 206 40))
    (gcd (remainder (remainder 206 40) (remainder 40 (remainder
↪  206 40)))
         (remainder (remainder 40 (remainder 206 40)) (remainder
↪  (remainder 206 40) (remainder 40 (remainder 206 40))))))

; eval 4 remainders
(if (= 2 0)
    (remainder 40 (remainder 206 40))
    (gcd (remainder (remainder 206 40) (remainder 40 (remainder
↪  206 40)))
         (remainder (remainder 40 (remainder 206 40)) (remainder
↪  (remainder 206 40) (remainder 40 (remainder 206 40))))))

; recurse
(gcd (remainder (remainder 206 40) (remainder 40 (remainder 206
↪  40)))
     (remainder (remainder 40 (remainder 206 40)) (remainder
↪  (remainder 206 40) (remainder 40 (remainder 206 40)))))

; encounter conditional
(if (= (remainder (remainder 40 (remainder 206 40)) (remainder
↪  (remainder 206 40) (remainder 40 (remainder 206 40)))) 0)
```

```
    (remainder (remainder 206 40) (remainder 40 (remainder 206
↪   40)))
    (gcd (remainder (remainder 40 (remainder 206 40)) (remainder
↪   (remainder 206 40) (remainder 40 (remainder 206 40))))
↪   (remainder a  (remainder (remainder 40 (remainder 206 40))
↪   (remainder (remainder 206 40) (remainder 40 (remainder 206
↪   40)))))))

; eval 7 remainders
(if (= 0 0)
    (remainder (remainder 206 40) (remainder 40 (remainder 206
↪   40)))
    (gcd (remainder (remainder 40 (remainder 206 40)) (remainder
↪   (remainder 206 40) (remainder 40 (remainder 206 40))))
↪   (remainder a  (remainder (remainder 40 (remainder 206 40))
↪   (remainder (remainder 206 40) (remainder 40 (remainder 206
↪   40)))))))

; eval 4 remainders
(remainder (remainder 206 40) (remainder 40 (remainder 206 40)))
; => 2
```

So, in normal-order eval, remainder is called 18 times, while in applicative order it's called 5 times.

2.34 1.2.6: Example: Testing for Primality

Two algorithms for testing primality of numbers.

1. $\Theta(\sqrt{n})$: Start with $x = 2$, check for divisibility with n, if not then increment x by 1 and check again. If $x^2 > n$ and you haven't found a divisor, n is prime.

2. $\Theta(\log n)$: Given a number n, pick a random number $a < n$ and compute the remainder of a^n modulo n. If the result is not equal to a, then n is certainly not prime. If it is a, then chances are good that n is prime. Now pick another random number a and test it with the same method. If it also satisfies the equation, then we can be even more confident that n is prime. By trying more and more values of a, we can increase our confidence in the result. This algorithm is known as the Fermat test.

   Fermat's Little Theorem: If n is a prime number and a is any positive integer less than n, then a raised to the n^{th} power is congruent to a modulo n. [Two numbers are *congruent modulo* n if they both have the same remainder when divided by n.]

The Fermat test is a probabilistic algorithm, meaning its answer is likely to be correct rather than guaranteed to be correct. Repeating the test increases the likelihood of a correct answer.

2.35 Exercise 1.21

2.35.1 Text

```
<<square>>
(define (smallest-divisor n)
  (find-divisor n 2))

(define (find-divisor n test-divisor)
  (cond ((> (square test-divisor) n)
         n)
        ((divides? test-divisor n)
         test-divisor)
        (else (find-divisor
               n
               (+ test-divisor 1)))))

(define (divides? a b)
  (= (remainder b a) 0))
```

2.35.2 Question

Use the smallest-divisor procedure to find the smallest divisor of each of the following numbers: 199, 1999, 19999.

2.35.3 Answer

```
<<find-divisor-txt>>
(map smallest-divisor '(199 1999 19999))
```

199 1999 7

2.36 Exercise 1.22

2.36.1 Question

Most Lisp implementations include a primitive called `runtime` that returns an integer that specifies the amount of time the system has been running (measured, for example, in microseconds). The following `timed-prime-test` procedure, when called with an integer n, prints n and checks to see if n is prime. If n is prime, the procedure prints three asterisks followed by the amount of time used in performing the test.

```
<<find-divisor-txt>>
(define (prime? n)
  (= n (smallest-divisor n)))
```

```
<<prime-smallest-divisor>>
(define (timed-prime-test n)
  (newline)
  (display n) ;; Guile compatible \downarrow
  (start-prime-test n (get-internal-run-time)))
(define (start-prime-test n start-time)
  (if (prime? n)
      (begin
        (report-prime (- (get-internal-run-time)
                         start-time))
        n)
      #f))
(define (report-prime elapsed-time)
  (display " *** ")
  (display elapsed-time))
```

Using this procedure, write a procedure `search-for-primes` that checks the primality of consecutive odd integers in a specified range. Use your procedure to find the three smallest primes larger than 1000; larger than 10,000; larger than 100,000; larger than 1,000,000. Note the time needed to test each prime. Since the testing algorithm has order of growth of $\Theta(\sqrt{n})$, you should expect that testing for primes around 10,000 should take about $\sqrt{10}$ times as long as testing for primes around 1000. Do your timing data bear this out? How well do the data for 100,000 and 1,000,000 support the $\Theta(\sqrt{n})$ prediction? Is your result compatible with the notion that programs on your machine run in time proportional to the number of steps required for the computation?

2.36.2 Answer

1. Part 1

   So this question is a little funky, because modern machines are so fast that the single-run times can seriously vary.

```
<<timed-prime-test-txt>>
(define (search-for-primes minimum goal)
  (define m (if (even? minimum)
                (+ minimum 1)
                (minimum)))
  (search-for-primes-iter m '() goal))
(define (search-for-primes-iter n lst goal)
  (if (= goal 0)
      lst
      (let ((x (timed-prime-test n)))
        (if (not (equal? x #f))
```

```
            (search-for-primes-iter (+ n 2) (cons x lst) (- goal
↪  1))
            (search-for-primes-iter (+ n 2) lst goal)))))
```

```
<<search-primes-basic>>
(let ((lt1000-1 (search-for-primes 1000 3)))
  (list "Primes > 1000" lt1000-1))
```

```
1001
1003
1005
1007
1009 *** 1651
1011
1013 *** 1425
1015
1017
1019 *** 1375
```

There's proof it works. And here are the answers to the question:

```
<<search-primes-basic>>
(let ((lt1000-1 (search-for-primes 1000 3))
      (lt10000-1 (search-for-primes 10000 3))
      (lt100000-1 (search-for-primes 100000 3))
      (lt1000000000-1 (search-for-primes 1000000 3)))
  (list
   (list "Primes > 1000" (reverse lt1000-1))
   (list "Primes > 10000" (reverse lt10000-1))
   (list "Primes > 100000" (reverse lt100000-1))
   (list "Primes > 100000000" (reverse lt100000000-1))
   ))
```

Primes > 1000	(1009 1013 1019)
Primes > 10000	(10007 10009 10037)
Primes > 100000	(100003 100019 100043)
Primes > 100000000	(1000003 1000033 1000037)

2. Part 2

   Repeatedly re-running, it I see it occasionally jump to twice the time. I'm not happy with this, so I'm going to refactor to use the `mattbench2` utility from the root of the project folder.

```
(define (mattbench2 f n)
  ;; Executes "f" for n times, and returns how long it took.
  ;; f is a lambda that takes no arguments, a.k.a. a "thunk"

  ;; Returns a list with car(last execution results) and cadr(time taken divided by iterations n)

  (define (time-getter) (get-internal-run-time))
  (define start-time (time-getter))
  (define (how-long) (- (time-getter) start-time))

  (define (iter i)
    (f)
    (if (<= i 0)
        (f) ;; return the results of the last function call
        (iter (- i 1))))

  (list (iter n) ;; result of last call of f
        (/ (how-long) (* n 1.0))));; Divide by iterations so changed n has no effect
```

I'm going to get some more precise times. First, I need a prime searching variant that doesn't bother benchmarking. This will call `prime?`, which will be bound later since we'll be trying different methods.

```
(define (search-for-primes minimum goal)
  (define m (if (even? minimum)
                (+ minimum 1)
                (minimum)))
  (search-for-primes-iter m '() goal))
(define (search-for-primes-iter n lst goal)
  (if (= goal 0)
      lst
      (let ((x (prime? n)))
        (if (not (equal? x #f))
            (search-for-primes-iter (+ n 2) (cons n lst) (- goal 1))
            (search-for-primes-iter (+ n 2) lst goal)))))
```

I can benchmark these functions like so:

```
<<mattbench2>>
<<prime-smallest-divisor>>
<<search-for-primes-untimed>>
<<print-table>>

(define benchmark-iterations 1000000)
(define (testit f)
```

```
    (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
          (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
          (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
          (cadr (mattbench2 (λ() (f 1000000 3))
  ↪  benchmark-iterations))))

 (print-row
  (testit search-for-primes))
```

Here are the results (run externally from Org-Mode):

5425.223086	20772.332491	53577.240193	121986.712395



The plot for the square root function doesn't quite fit the real one and I'm not sure where the fault lies. I don't struggle to understand things like "this algorithm is slower than this other one," but when asked to find or prove the Θ notation I'm pretty clueless;

2.37 Exercise 1.23

2.37.1 Question

The `smallest-divisor` procedure shown at the start of this section does lots of needless testing: After it checks to see if the number is divisible by 2 there is no point in checking to see if it is divisible by any larger even numbers. This suggests that the values used for test-divisor should not be 2, 3, 4, 5, 6, ..., but rather 2, 3, 5, 7, 9, ...

To implement this change, define a procedure `next` that returns 3 if its input is equal to 2 and otherwise returns its input plus 2. Modify the smallest-divisor procedure to use (`next test-divisor`) instead of (`+ test-divisor 1`). With `timed-prime-test` incorporating this modified version of `smallest-divisor`, run the test for each of the 12 primes found in Exercise 1.22. Since this modification halves the number of test steps, you should expect it to run about twice as fast. Is this expectation confirmed? If not, what is the observed ratio of the speeds of the two algorithms, and how do you explain the fact that it is different from 2?

2.37.2 A Comedy of Error (just the one)

```
<<square>>
(define (smallest-divisor n)
  (find-divisor n 2))

(define (next n)
  (if (= n 2)
      3
      (+ n 1)))

(define (find-divisor n test-divisor)
  (cond ((> (square test-divisor) n)
         n)
        ((divides? test-divisor n)
         test-divisor)
        (else (find-divisor
               n
               (next test-divisor)))))

(define (divides? a b)
  (= (remainder b a) 0))
```

```
<<mattbench2>>
<<find-divisor-faster>>
(define (prime? n)
  (= n (smallest-divisor n)))
<<search-for-primes-untimed>>
<<print-table>>

(define benchmark-iterations 1000000)
(define (testit f)
  (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
```

```
        (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))

(print-row
 (testit search-for-primes))
```

6456.538118 25550.757304 66746.041644 148505.580638

min	(+1)	(next)
1000	5507.42497	6366.99462
10000	20913.71497	24845.9193
100000	53778.74737	64756.73693
1000000	122135.60511	143869.63561



So it's *slower* than before. Why?
Oh, that's why.

```
(define (next n)
  (if (= n 2)
      3
      (+ n 1))) ;; <-- D'oh.
```

2.37.3 Answer

Ok, let's try that again.

```
<<square>>
(define (smallest-divisor n)
  (find-divisor n 2))

(define (next n)
  (if (= n 2)
      3
      (+ n 2)))

(define (find-divisor n test-divisor)
  (cond ((> (square test-divisor) n)
         n)
        ((divides? test-divisor n)
         test-divisor)
        (else (find-divisor
               n
               (next test-divisor)))))

(define (divides? a b)
  (= (remainder b a) 0))
```

```
<<mattbench2>>
<<find-divisor-faster-real>>
(define (prime? n)
  (= n (smallest-divisor n)))
<<search-for-primes-untimed>>
<<print-table>>

(define benchmark-iterations 500000)
(define (testit f)
  (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))

(print-row
 (testit search-for-primes))
```

3863.7424	13519.209814	33520.676384	73005.539932

min	(+1)	(next-broken)	(next-fixed)
1000	5425.223086	6456.538118	3863.7424
10000	20772.332491	25550.757304	13519.209814
100000	53577.240193	66746.041644	33520.676384
1000000	121986.712395	148505.580638	73005.539932



I had a lot of trouble getting this one to compile, I have to restart Emacs in order to get it to render.

Anyways, there's the speedup that was expected. Let's compare the ratios.

Defining a new average that takes arbitrary numbers of arguments:

```
(define (average . args)
  (let ((len (length args)))
    (/ (apply + args) len)))
```

Using it for percentage comparisons:

```
<<average-varargs>>
(list (cons "% speedup for broken (next)"
            (cons (format #f "~2$%"
                          (apply average
                                 (map (λ (x y) (* 100 (/ x y)))
                                      (car smd) (car smdf))))
                  #nil))
      (cons "% speedup for real (next)"
            (cons (format #f "~2$%"
                          (apply average
                                 (map (λ (x y) (* 100 (/ x y)))
                                      (car smd) (car smdff))))
                  #nil)))
```

% speedup for broken (next)	81.93%
% speedup for real (next)	155.25%

Since this changed algorithm cuts out almost half of the steps, you might expect something more like a 200% speedup. Let's try optimizing it further. Two observations:

1. The condition (**divides? 2** n) only needs to be run once at the start of the program.

2. Because it only needs to be run once, it doesn't need to be a separate function at all.

```
<<square>>
(define (smallest-divisor n)
  (if (divides? 2 n)                  ;; check for division by 2
      2
      (find-divisor n 3)))            ;; start find-divisor at 3

(define (find-divisor n test-divisor)
  (cond ((> (square test-divisor) n)
         n)
        ((divides? test-divisor n)
         test-divisor)
        (else (find-divisor
               n
               (+ 2 test-divisor))))) ;; just increase by 2

(define (divides? a b)
  (= (remainder b a) 0))
```

```
<<mattbench2>>
<<find-divisor-faster-real2>>
(define (prime? n)
  (= n (smallest-divisor n)))
<<search-for-primes-untimed>>
<<print-table>>

(define benchmark-iterations 500000)
(define (testit f)
  (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))

(print-row
 (testit search-for-primes))
```

3151.259574 11245.20428 27803.067944 61997.275154

min	(+1)	(next-broken)	(next-fixed)	integrated
1000	5425.223086	6456.538118	3863.7424	3151.259574
10000	20772.332491	25550.757304	13519.209814	11245.20428
100000	53577.240193	66746.041644	33520.676384	27803.067944
1000000	121986.712395	148505.580638	73005.539932	61997.275154



% speedup for broken (next)	81.93%
% speedup for real (next)	155.25%
% speedup for optimized	186.59%

2.38 Exercise 1.24

2.38.1 Text

```
<<square>>
(define (expmod base exp m)
  (cond ((= exp 0) 1)
        ((even? exp)
         (remainder
          (square (expmod base (/ exp 2) m))
          m))
        (else
         (remainder
```

```
          (* base (expmod base (- exp 1) m))
          m))))
```

```
(define (fermat-test n)
  (define (try-it a)
    (= (expmod a n n) a))
  (try-it (+ 1 (random (- n 1)))))
```

```
(define (fast-prime? n times)
  (cond ((= times 0) #t)
        ((fermat-test n)
         (fast-prime? n (- times 1)))
        (else #f)))
```

2.38.2 Question

Modify the `timed-prime-test` procedure of 2.36 to use `fast-prime?` (the Fermat method), and test each of the 12 primes you found in that exercise. Since the Fermat test has $\Theta(\log n)$ growth, how would you expect the time to test primes near 1,000,000 to compare with the time needed to test primes near 1000? Do your data bear this out? Can you explain any discrepancy you find?

2.38.3 Answer

```
<<mattbench2>>
<<expmod>>
<<fermat-test>>
<<fast-prime>>
(define fermat-iterations 2)
(define (prime? n)
  (fast-prime? n fermat-iterations))
<<search-for-primes-untimed>>
<<print-table>>

(define benchmark-iterations 500000)
(define (testit f)
  (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))))

(print-row
 (testit search-for-primes))
```

11175.799722 23518.62116 32150.745642 32679.766448

min	(+1)	integrated	fermat (2 guesses)
1000	5425.223086	3151.259574	11175.799722
10000	20772.332491	11245.20428	23518.62116
100000	53577.240193	27803.067944	32150.745642
1000000	121986.712395	61997.275154	32679.766448



It definitely looks to be advancing much slower than the other methods. I'd like to see more of the function.

```
<<mattbench2>>
<<find-divisor-faster-real>>
(define (prime? n)
  (= n (smallest-divisor n)))
<<search-for-primes-untimed>>
<<print-table>>

(define benchmark-iterations 100000)
(define (testit f)
  (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000000 3)) benchmark-iterations))
```

```
        (cadr (mattbench2 (λ() (f 100000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000000000 3))
↪  benchmark-iterations))))

(print-row
 (testit search-for-primes))
```

```
<<mattbench2>>
<<expmod>>
<<fermat-test>>
<<fast-prime>>
(define fermat-iterations 100)
(define (prime? n)
  (fast-prime? n fermat-iterations))
<<search-for-primes-untimed>>
<<print-table>>

(define benchmark-iterations 100000)
(define (testit f)
  (list (cadr (mattbench2 (λ() (f 1000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 10000000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 100000000000 3)) benchmark-iterations))
        (cadr (mattbench2 (λ() (f 1000000000000 3))
↪  benchmark-iterations))))

(print-row
 (testit search-for-primes))
```



For the life of me I have no idea what that bump is. Maybe it needs more aggressive bignum processing there?

2.39 Exercise 1.25

2.39.1 Question

> Alyssa P. Hacker complains that we went to a lot of extra work in writing `expmod`. After all, she says, since we already know how to compute exponentials, we could have simply written

```
(define (expmod base exp m)
  (remainder (fast-expt base exp) m))
```

> Is she correct? Would this procedure serve as well for our fast prime tester? Explain.

2.39.2 Answer

In Alyssa's version of `expmod`, the result of the `fast-expt` operation is *extremely* large. For example, in the process of checking for divisors of 1,001, the number 455 will be tried. (`expt 455 1001`) produces an integer 2,661 digits long. This is just one of the thousands of exponentiations that `smallest-divisor` will perform. It's best to avoid this, so we use to our advantage the fact that we only need to know the remainder of the exponentiations. `expmod` breaks down

the exponentiation into smaller steps and performs `remainder` after every step, significantly reducing the memory requirements.

As an example, let's trace (some of) the execution of `(expmod 455 1001 1001)`:

```
(expmod 455 1001 1001)
> (even? 1001)
> #f
> (expmod 455 1000 1001)
> > (even? 1000)
> > #t
> > (expmod 455 500 1001)
> > > (even? 500)
> > > #t
;; ...
> > > x11 (expmod 455 2 1001)
> > > x11 > (even? 2)
> > > x11 > #t
> > > x11 > (expmod 455 1 1001)
> > > x11 > > (even? 1)
> > > x11 > > #f
> > > x11 > > (expmod 455 0 1001)
> > > x11 > > 1
> > > x11 > 455
> > > x11 > (square 455)
> > > x11 > 207025
> > > x11 819
;; ...
> > > (square 364)
> > > 132496
> > 364
> > (square 364)
> > 132496
> 364
455
```

You can see that the numbers remain quite manageable throughout this process. So taking these extra steps actually leads to an algorithm that performs better.

2.40 Exercise 1.26

2.40.1 Question

> Louis Reasoner is having great difficulty doing Exercise 1.24. His `fast-prime?` test seems to run more slowly than his `prime?` test. Louis calls his friend Eva Lu Ator over to help. When they examine Louis's code, they find that he has rewritten the `expmod` procedure to use an explicit multiplication, rather than calling `square`:

```
(define (expmod base exp m)
  (cond ((= exp 0) 1)
        ((even? exp)
         (remainder
          (* (expmod base (/ exp 2) m) ;; <== hmm.
             (expmod base (/ exp 2) m))
          m))
        (else
         (remainder
          (* base
             (expmod base (- exp 1) m))
          m))))
```

> "I don't see what difference that could make," says Louis. "I do." says Eva. "By writing the procedure like that, you have transformed the $\Theta(\log n)$ process into a $\Theta(n)$ process." Explain.

2.40.2 Answer

Making the same function call twice isn't the same as using a variable twice – Louis' version doubles the work, having two processes solving the exact same problem. Since the number of processes used increases exponentially, this turns $\log n$ into n.

2.41 Exercise 1.27

2.41.1 Question

> Demonstrate that the Carmichael numbers listed in Footnote 1.47 really do fool the Fermat test. That is, write a procedure that takes an integer n and tests whether a^n is congruent to a modulo n for every $a < n$, and try your procedure on the given Carmichael numbers.
>
> 561 1105 1729 2465 2821 6601

2.41.2 Answer

```
<<expmod>>
(define (car-test n)
  (define (check a)
    (= (remainder (expt a n) n)
       (remainder (modulo a n) n)))
  (every check
           (cddr (iota n))))
```

```
<<car-test>>
(list (car-test 12) ; <== false (not prime)
      (car-test 1009);<== true  (real prime)
      (car-test 561));<== true  (not prime,
                     ;      Carmichael number)
```

2.42 Exercise 1.28

2.42.1 Question

> One variant of the Fermat test that cannot be fooled is called the Miller-Rabin test (Miller 1976; Rabin 1980). This starts from an alternate form of Fermat's Little Theorem, which states that if n is a prime number and a is any positive integer less than n, then a raised to the $(n-1)$ -st power is congruent to 1 modulo n. To test the primality of a number n by the Miller-Rabin test, we pick a random number $a < n$ and raise a to the $(n-1)$ -st power modulo n using the `expmod` procedure. However, whenever we perform the squaring step in `expmod`, we check to see if we have discovered a "nontrivial square root of 1 modulo n," that is, a number not equal to 1 or $n-1$ whose square is equal to 1 modulo n. It is possible to prove that if such a nontrivial square root of 1 exists, then n is not prime. It is also possible to prove that if n is an odd number that is not prime, then, for at least half the numbers $a < n$, computing $an-1$ in this way will reveal a nontrivial square root of 1 modulo n. (This is why the Miller-Rabin test cannot be fooled.) Modify the `expmod` procedure to signal if it discovers a nontrivial square root of 1, and use this to implement the Miller-Rabin test with a procedure analogous to fermat-test. Check your procedure by testing various known primes and non-primes. Hint: One convenient way to make `expmod` signal is to have it return 0.

2.42.2 Analysis

For the sake of verifying this, I want to get a bigger list of primes and Carmichael numbers to verify against. I'll save them using Guile's built in read/write functions that save Lisp lists to text:

```
<<find-divisor-faster-real>>
(define (prime? n)
  (= n (smallest-divisor n)))
(call-with-output-file "Data/primes-1k_to_1mil.txt" (λ(port)
  (write (filter prime? (iota (- 1000000 1000) 1000))
         port)))
```

```
;; fermat prime test but checks *every* value from 2 to n-1
(define (fermat-prime? n)
  (define (iter a)
    (if (= a n)
        #f
        (if (= (expmod a n n) a)
            #t
            (iter (+ 1 a)))))
  (iter 2))
```

```
(use-srfis '(1))
<<expmod>>
<<fermat-prime?>>
<<find-divisor-faster-real>>
(define (prime? n)
  (= n (smallest-divisor n)))
(call-with-output-file "Data/carmichael-verification.txt" (λ(port)
     (write (filter
             (λ(x) (and (fermat-prime? x)
                        (not (prime? x))))
             (iota (- 1000000 1000) 1000))
            port)))
```

This will be useful in various future functions:

```
(define list-of-primes (call-with-input-file
↪  "Data/primes-1k_to_1mil.txt" read))
(define list-of-carmichaels (call-with-input-file "Data/carmichael.txt"
↪  read))
```

```
(use-srfis '(1))
<<expmod>>
<<fermat-prime?>>
<<find-divisor-faster-real>>
(define (prime? n)
  (= n (smallest-divisor n)))
<<get-lists-of-primes>>
(define prime-is-working
  (and (and-map prime? list-of-primes)
       (not (and-map prime? list-of-carmichaels))))
(format #t "(prime?) is working: ~a~%"
        (if prime-is-working
            "Yes"
            "No"))
(define fermat-is-vulnerable
  (and (and-map fermat-prime? list-of-primes)
       (and-map fermat-prime? list-of-carmichaels)))
```

```
(format #t "(fermat-prime?) is vulnerable: ~a~%"
        (if fermat-is-vulnerable
            "Yes"
            "No"))
```

```
(prime?) is working: Yes
(fermat-prime?) is vulnerable: Yes
```

2.42.3 Answer

```
<<square>>
(define (expmod-mr base exp m)
  (cond ((= exp 0) 1)
        ((even? exp)
         (let ((sqr
                (square (expmod-mr base (/ exp 2) m))))
           (if (= 1 (modulo sqr m))
               0
               (remainder sqr m))))
        (else
         (remainder
          (* base (expmod-mr base (- exp 1) m))
          m))))
```

```
(define (mr-test n)
  (define (try-it a)
    (let ((it (expmod-mr a n n)))
      (or (= it a)
          (= it 0))))
  (try-it (+ 1 (random (- n 1)))))
```

```
(define (mr-prime? n times)
  (cond ((= times 0) #t)
        ((mr-test n)
         (mr-prime? n (- times 1)))
        (else #f)))
```

```
<<expmod-mr>>
<<mr-test>>
<<mr-prime>>
(define mr-times 100)
<<get-lists-of-primes>>
(format #t
↪  "      mr detects primes: ~a~%mr false-positives Carmichaels: ~a~%"
```

```
        (and-map (λ(x)(mr-prime? x mr-times)) list-of-primes)
      (and-map (λ(x)(mr-prime? x mr-times)) list-of-carmichaels))
```

```
    mr detects primes: #t
mr false-positives Carmichaels: #t
```

Shoot. And I thought I did a very literal interpretation of what the book asked.

Ah, I see the problem. I need to keep track of what the pre-squaring number was and use that to determine whether the square is valid or not.

```
<<square>>
(define (expmod-mr base exp m)
  (cond ((= exp 0) 1)
        ((even? exp)
         ;; Keep result and remainder seperate
         (let* ((result (expmod-mr base (/ exp 2) m))
                (rem (remainder (square result) m)))
           (if (and (not (= result 1))
                    (not (= result (- m 1)))
                    (= 1 rem))
               0 ;; non-trivial sqrt mod 1 is found
               rem)))
        (else
         (remainder
          (* base (expmod-mr base (- exp 1) m))
          m))))
```

Unfortunately this one has the same problem. What's the issue?

Sadly, there's a massive issue in `mr-test`.

```
(define (mr-test n)
  (define (try-it a)
    (let ((it (expmod-mr a n n))) ;; Should be "a (- n 1) n"
      (or (= it a)    ;; Should be (= it 1)
          (= it 0)))) ;; Two strikes, you're out
  (try-it (+ 1 (random (- n 1)))))
```

One more time.

```
(define (mr-test n)
  (define (try-it a)
    (= 1 (expmod-mr a (- n 1) n)))
  (try-it (+ 1 (random (- n 1)))))
```

```
<<expmod-mr2>>
<<mr-test2>>
<<mr-prime>>
(define mr-times 100)
<<get-lists-of-primes>>
(format #t
↪  "      mr detects primes: ~a~%mr false-positives Carmichaels: ~a~%"
        (and-map (λ(x)(mr-prime? x mr-times)) list-of-primes)
      (and-map (λ(x)(mr-prime? x mr-times)) list-of-carmichaels))
```

```
     mr detects primes: #t
mr false-positives Carmichaels: #f
```

2.43 1.3: Formulating Abstractions with Higher-Order Procedures

Procedures that manipulate procedures are called *higher-order procedures*.

2.44 1.3.1: Procedures as Arguments

Let's say we have several different types of series that we want to sum. Functions for each of these tasks will look very similar, so we're better off defining a general function that expresses the *idea* of summation, that can then be passed specific functions to cause the specific behavior of the series. Mathematicians express this as $\sum$ ("sigma") notation.

For the program:

```
(define (sum term a next b)
  (if (> a b)
      0
      (+ (term a)
         (sum term (next a) next b))))
```

Which is equivalent to:

$$\sum_{n\ a}^{b} term(n)\ term(a) + term(next(a)) + term(next(next(a))) + \cdots + term(b)$$

We can pass integers to `a` and `b` and functions to `term` and `next`. Note that in order to simply sum integers, we'd need to define and pass an identity function to `term`.

2.45 Exercise 1.29

2.45.1 Text

```
(define (sum term a next b)
  (if (> a b)
      0
      (+ (term a)
         (sum term (next a) next b))))
```

```
(define (integral f a b dx)
  (define (add-dx x)
    (+ x dx))
  (* (sum f (+ a (/ dx 2.0)) add-dx b)
     dx))
```

2.45.2 Question

Simpson's Rule is a more accurate method of numerical integration than the method illustrated above. Using Simpson's Rule, the integral of a function f between a and b is approximated as

$$\frac{h}{3}(y_0 + 4y_1 + 2y_2 + 4y_3 + 2y_4 + \cdots + 2y_{n-2} + 4y_{n-1} + y_n)$$

where $h = (b - a)/n$, for some even integer n, and $y_k = f(a + kh)$. (Increasing n increases the accuracy of the approximation.) Define a procedure that takes as arguments f, a, b, and n and returns the value of the integral, computed using Simpson's Rule. Use your procedure to integrate `cube` between 0 and 1 (with $n = 100$ and $n = 1000$), and compare the results to those of the `integral` procedure shown above.

2.45.3 Answer

```
(define (int-simp f a b n)
  (define h
    (/ (- b a)
     n))
  (define (gety k)
    (f (+ a (* k h))))
  (define (series-y sum k) ;; start with sum = y_0
    (cond ((= k n) (+ sum (gety k)));; and k = 1
          ((even? k) (series-y
                      (+ sum (* 2 (gety k)))
                      (+ 1 k)))
          (else (series-y
                 (+ sum (* 4 (gety k)))
                 (+ 1 k)))))
  (define sum-of-series (series-y (gety a) 1)) ;; (f a) = y_0
```

```
  (* (/ h 3) sum-of-series))
```

Let's compare these at equal levels of computational difficulty.

```
<<mattbench2>>
<<print-table>>
(define (cube x)
  (* x x x))
<<sum>>
<<integral>>
<<int-simp>>

(define iterations 100000) ;; benchmark iterations
(define (run-test1)
  (integral cube 0.0 1.0 0.0008))
(define (run-test2)
  (int-simp cube 0.0 1.0 1000.0))
(print-table (list (list "integral dx:0.0008" "int-simp i:1000")
                   (list (run-test1) (run-test2))
                   (list (cadr (mattbench2 run-test1 iterations))
                         (cadr (mattbench2 run-test2 iterations))))
             #:colnames #t)
```

integral dx:0.0008	int-simp i:1000
0.24999992000001311	0.25000000000000006
321816.2755	330405.8918

So, more accurate for roughly the same effort or less.

2.46 Exercise 1.30

2.46.1 Question

The `sum` procedure above generates a linear recursion. The procedure can be rewritten so that the sum is performed iteratively. Show how to do this by filling in the missing expressions in the following definition:

2.46.2 Answer

```
(define (sum-iter term a next b)
  (define (iter a result)
    (if (> a b)
        result
        (iter (next a) (+ result (term a)))))
  (iter a 0))
```

Let's check the stats!

recursive	iterative
30051.080005	19568.685587

2.47 Exercise 1.31

2.47.1 Question A.1

> The `sum` procedure is only the simplest of a vast number of similar abstractions that can be captured as higher-order procedures. Write an analogous procedure called `product` that returns the product of the values of a function at points over a given range.

2.47.2 Answer A.1

```
(define (product-iter term a next b)
  (define (iter a result)
    (if (> a b)
        result
        (iter (next a) (* result (term a)))))
  (iter a 1)) ;; start at 1 so it's not always 0
```

2.47.3 Question A.2

> Show how to define `factorial` in terms of `product`.

2.47.4 Answer A.2

I was briefly stumped because `product` only counts upward. Then I realized that's just how it's presented and it can go either direction, since addition and multiplication are commutative. I look forward to building up a more intuitive sense of numbers.

```
<<product-iter>>
(define (identity x)
  x)
(define (inc x)
  (1+ x))

(define (factorial n)
  (product-iter identity 1 inc n))

(display (factorial 7))
```

2.47.5 Question A.3

> Also use `product` to compute approximations to π using the formula

$$\frac{\pi}{4} = \frac{2 \cdot 4 \cdot 4 \cdot 6 \cdot 6 \cdot 8 \cdots}{3 \cdot 3 \cdot 5 \cdot 5 \cdot 7 \cdot 7 \cdots}$$

2.47.6 Answer A.3

Once this equation is encoded, you just need to multiply it by two to get π.
Fun fact: the formula is slightly wrong, it should start the series with $\frac{1}{2}$.

```
<<product-iter>>
(define (pi-product n)
  (define (div x)
    (let ((x1 (- x 1))
          (x2 (+ x 1)))
      (* (/ x x1) (/ x x2))))
  (* 2.0 (product-iter div 2 (lambda (z) (+ z 2)) n)))

(display (pi-product 100000))
```

```
3.1415769458228726
```

2.47.7 Question B

> If your product procedure generates a recursive process, write one that generates an iterative process. If it generates an iterative process, write one that generates a recursive process.

2.47.8 Answer B

```
(define (product-rec term a next b)
  (if (> a b)
      1
      (* (term a)
         (product-rec term (next a) next b))))
```

```
<<mattbench2>>
<<print-table>>
<<product-iter>>
(define (pi-product n)
  (define (div x)
    (let ((x1 (- x 1))
          (x2 (+ x 1)))
      (* (/ x x1) (/ x x2))))
  (* 2.0 (product-iter div 2 (lambda (z) (+ z 2)) n)))
<<product-rec>>
(define (pi-product-rec n)
  (define (div x)
    (let ((x1 (- x 1))
          (x2 (+ x 1)))
      (* (/ x x1) (/ x x2))))
  (* 2.0 (product-rec div 2 (lambda (z) (+ z 2)) n)))

```

```
(define iterations 50000)
(print-table
 (list (list "iterative" "recursive")
       (list (cadr (mattbench2 (λ()(pi-product 1000)) iterations))
             (cadr (mattbench2 (λ()(pi-product-rec 1000)) iterations))))
 #:colnames #t)
```

iterative	recursive
1267118.0538	3067085.5323

2.48 Exercise 1.32

2.48.1 Question A

> Show that `sum` and `product` are both special cases of a still more general notion called `accumulate` that combines a collection of terms, using some general accumulation function:

```
(accumulate combiner null-value term a next b)
```

> `accumulate` takes as arguments the same term and range specifications as `sum` and `product`, together with a `combiner` procedure (of two arguments) that specifies how the current term is to be combined with the accumulation of the preceding terms and a `null-value` that specifies what base value to use when the terms run out. Write `accumulate` and show how `sum` and `product` can both be defined as simple calls to `accumulate`.

2.48.2 Answer A

When I first did this question, I struggled a lot before realizing `accumulate` was much closer to the exact definitions of sum/product than I thought.

```
(define (accumulate-iter combiner null-value term a next b)
  (define (iter a result)
    (if (> a b)
        result
        (iter (next a)
              (combiner result (term a)))))
  (iter a null-value))
```

```
<<accumulate-iter>>

;; here you can see definitions in terms of accumulate
(define (sum term a next b)
  (accumulate-iter + 0 term a next b))
(define (product term a next b)
```

```
  (accumulate-iter * 1 term a next b))

(define (identity x)
  x)
(define (inc x)
  (1+ x))

;; accumulate in action
(define (factorial n)
  (accumulate-iter * 1 identity 1 inc n))

(display (factorial 7))
```

```
5040
```

2.48.3 Question B

If your `accumulate` procedure generates a recursive process, write one that generates an iterative process. If it generates an iterative process, write one that generates a recursive process.

2.48.4 Answer B

```
(define (accumulate-rec combiner null-value term a next b)
  (if (> a b)
      null-value
      (combiner (term a)
         (accumulate-rec combiner null-value
                         term (next a) next b))))
```

2.49 Exercise 1.33

2.49.1 Question A

You can obtain an even more general version of `accumulate` by introducing the notion of a filter on the terms to be combined. That is, combine only those terms derived from values in the range that satisfy a specified condition. The resulting `filtered-accumulate` abstraction takes the same arguments as accumulate, together with an additional predicate of one argument that specifies the filter. Write `filtered-accumulate` as a procedure.

2.49.2 Answer A

```
(define (filtered-accumulate-iter
           predicate? combiner null-value
           term a next b)
  (define (iter a result)
```

```
    (cond ((> a b) result)
          ((predicate? a)
           (iter (next a)
                 (combiner result (term a))))
          (else (iter (next a)
                      result))))
  (iter a null-value))
```

2.49.3 Question B

Show how to express the following using `filtered-accumulate`:

1. A

   Find the sum of the squares of the prime numbers in the interval a to b (assuming that you have a `prime?` predicate already written)

   ```
   (load "mattcheck.scm")
   (define (square x)
     (* x x))
   <<filtered-accumulate-iter>>
   <<expmod-mr2>>
   <<mr-test2>>
   <<mr-prime>>
   (define mr-times 100)
   (define (prime? x)
     (mr-prime? x mr-times))
   (define (prime-sum a b)
     (filtered-accumulate-iter prime? + 0
                               square a 1+ b))

   (mattcheck-equal "1 prime correct"
                    (prime-sum 1008 1010)
                    (square 1009)) ;; 1009
   (mattcheck-equal "many primes correct"
                    (prime-sum 1000 2001)
                    (apply +
                           (map square
                                (filter prime? (iota (- 2001 1000)
                                                     1000)))))
   ```

   ```
   SUCCEED at 1 prime correct
   SUCCEED at many primes correct
   ```

2. B

   Find the product of all the positive integers less than n that are relatively prime to n (i.e., all positive integers $i < n$ such that $\text{GCD}(i, n) = 1$.

```
(load "mattcheck.scm")
(define (square x)
  (* x x))
(define (id x) x)
<<filtered-accumulate-iter>>
<<gcd>>
(define (relative-prime? x y)
  (= 1 (gcd x y)))

(define (Ex_1-33B n)
  (filtered-accumulate-iter
   (λ(i) (relative-prime? i n))
   * 1 id
   1 1+ (1- n)))

(define (alternate n)
  (apply *
         (filter (λ(i) (relative-prime? i n))
                 (iota (- n 1) 1))))

(mattcheck-equal "Ex_1-33B"
                 (Ex_1-33B 100)
                 (alternate 100))
```

```
SUCCEED at Ex_1-33B
```

2.50 1.3.2: Constructing Procedures Using lambda

A procedure that's only used once is more conveniently expressed as the special form **lambda**.

Variables that are only briefly used in a limited scope can be specified with the special form **let**. Variables in **let** blocks override external variables. The authors recommend using **define** for procedures and **let** for variables.

2.51 Exercise 1.34

2.51.1 Question

Suppose we define the procedure

```
(define (f g) (g 2))
```

Then, we have

```
(f square)
; 4
(f (lambda (z) (* z (+ z 1))))
; 6
```

What happens if we (perversely) ask the interpreter to evaluate the combination `(f f)`? Explain.

2.51.2 Answer

It ends up trying to execute `2` as a function.

```
;; Will be evaluated like this:
;;   (f f)
;;   (f 2)
;;   (2 2)
(define (f g) (g 2))
(f f)
```

```
ice-9/boot-9.scm:1685:16: In procedure raise-exception:
Wrong type to apply: 2
```

2.52 1.3.3 Procedures as General Methods

The **half-interval method**: if $f(a) < 0 < f(b)$, then f must have at least one 0 between a and b. To find 0, let x be the average of a and b, if $f(x) < 0$ then 0 must be between x and b, if $f(x) > 0$ than 0 must be between a and x.

The **fixed point** of a function satisfies the equation

$$f(x) = x$$

For some functions, we can locate a fixed point by beginning with an initial guess y and applying $f(y)$ repeatedly until the value doesn't change much.

Average damping can help converge fixed-point searches.

The symbol $\mapsto$ ("maps to") can be considered equivalent to a lambda. For example, $x \mapsto x + x$ is equivalent to `(lambda (x) (+ x x))`. In English, "the function whose value at y is x/y". *Though it seems like $\mapsto$ doesn't necessarily describe a function, but the value of a function at a certain point? Or maybe that would just be , ie $f(x)$ etc*

2.53 Exercise 1.35

2.53.1 Text

```
(define (close-enough? x y)
  (< (abs (- x y)) 0.001))
```

```
(define tolerance 0.00001)

(define (fixed-point f first-guess)
  (define (close-enough? v1 v2)
    (< (abs (- v1 v2))
       tolerance))
  (define (try guess)
    (let ((next (f guess)))
      (if (close-enough? guess next)
          next
          (try next))))
  (try first-guess))
```

2.53.2 Question

Show that the golden ratio φ is a fixed point of the transformation $x \mapsto 1 + 1/x$, and use this fact to compute φ by means of the `fixed-point` procedure.

2.53.3 Answer

```
<<close-enough>>
<<fixed-point-txt>>
(define golden-ratio
  (fixed-point (λ(x)(+ 1 (/ 1 x)))
               1.0))

(display golden-ratio)
```

```
1.6180327868852458
```

2.54 Exercise 1.36

2.54.1 Question

Modify `fixed-point` so that it prints the sequence of approximations it generates, using the `newline` and `display` primitives shown in Exercise 1.22. Then find a solution to $x^x = 1000$ by finding a fixed point of $x \mapsto \log(1000)/\log(x)$. (Use Scheme's primitive `log` procedure, which computes natural logarithms.) Compare the number of steps this takes with and without average damping. (Note that you cannot start `fixed-point` with a guess of 1, as this would cause division by $\log(1) = 0$.)

2.54.2 Answer

Using the `display` and `newline` functions at any great extent is pretty exhausting, so I'll use `format` instead.

```
(use-modules (ice-9 format))
(define tolerance 0.00001)

(define (fixed-point f first-guess)
  (define (close-enough? v1 v2)
    (< (abs (- v1 v2))
       tolerance))
  (define (try guess)
    (let ((next (f guess)))
      (format #t "~&~a~%" next)
      (if (close-enough? guess next)
          next
          (try next))))
  (try first-guess))
```

```
<<close-enough>>
<<fixed-point-debug>>
(fixed-point (λ(x) (/ (log 1000) (log x))) 1.1)
```

Undamped, fixed-point makes 37 guesses.

```
<<close-enough>>
<<fixed-point-debug>>
(define (average x y)
  (/ (+ x y) 2))
(fixed-point (λ(x) (average (log x) (/ (log 1000) (log x)))) 1.1)
```

Damped, it makes 21.

2.55 Exercise 1.37

2.55.1 Question A

An infinite continued fraction is an expression of the form

$$f = \cfrac{N_1}{D_1 + \cfrac{N_2}{D_2 + \cfrac{N_3}{D_3 + \ldots}}}$$

As an example, one can show that the infinite continued fraction expansion with the N_i and the D_i all equal to 1 produces $1/\varphi$, where φ is the golden ratio (described in 1.2.2). One way to approximate an infinite continued fraction is to truncate the expansion after a given number of terms. Such a truncation—a so-called k-term finite continued fraction}—has the form

$$\cfrac{N_1}{D_1 + \cfrac{N_2}{\ddots + \cfrac{N_k}{D_k}}}$$

Suppose that `n` and `d` are procedures of one argument (the term index i) that return the N_i and D_i of the terms of the continued fraction. Define a procedure `cont-frac` such that evaluating `(cont-frac n d k)` computes the value of the k-term finite continued fraction.

2.55.2 Answer A

A note: the "golden ratio" this code estimates is exactly `1.0` less than the golden ratio anyone else seems to be talking about.

```
(define (cont-frac n d k)
  (define (iter i result)
    (if (= i 0)
        result
        (iter (1- i) (/ (n i) (+ (d i) result)))))

  (iter (1- k) (/ (n k) (d k))))
```

2.55.3 Question B

Check your procedure by approximating $1/\varphi$ using

```
(cont-frac (lambda (i) 1.0)
           (lambda (i) 1.0)
           k)
```

for successive values of `k`. How large must you make `k` in order to get an approximation that is accurate to 4 decimal places?

2.55.4 Answer B

1	-0.3819660112501051
2	0.1180339887498949
3	-0.04863267791677173
4	0.018033988749894814
5	-0.0069660112501050975
6	0.0026493733652794837
7	-0.0010136302977241662
8	0.00038692992636546464
9	-0.00014782943192326314
10	5.6460660007306984e-05



k must be at least 10 to get precision of 4 decimal places.

2.55.5 Question C

If your `cont-frac` procedure generates a recursive process, write one that generates an iterative process. If it generates an iterative process, write one that generates a recursive process.

2.55.6 Answer C

```
(define (cont-frac-rec n d k)
  (define (rec i)
    (if (= i k)
        (/ (n i) (d i))
        (/ (n i) (+ (d i) (rec (1+ i))))))

  (rec 1))
```

```
<<cont-frac>>
<<cont-frac-rec>>
(define (golden-ratio k)
  (cont-frac (λ(i) 1.0)(λ(i)1.0) k))
(define (golden-ratio-rec k)
  (cont-frac-rec (λ(i) 1.0)(λ(i)1.0) k))

```

```
(load "mattcheck.scm")
(mattcheck-equal "cont-frac iter and recursive equivalence"
           (golden-ratio-rec 15)
           (golden-ratio 15))
```

```
SUCCEED at cont-frac iter and recursive equivalence
```

2.56 Exercise 1.38

2.56.1 Question

In 1737, the Swiss mathematician Leonhard Euler published a memoir *De Fractionibus Continuis*, which included a continued fraction expansion for $e - 2$, where e is the base of the natural logarithms. In this fraction, the N_i are all 1, and the D_i are successively 1, 2, 1, 1, 4, 1, 1, 6, 1, 1, 8, ... Write a program that uses your `cont-frac` procedure from Exercise 1.37 to approximate e, based on Euler's expansion.

2.56.2 Answer

```
<<cont-frac>>
(define (euler k)
  (+ 2
     (cont-frac (λ(i) 1.0)
             (λ(i) (let ((j (1+ i)))
                     (if (= 0 (modulo j 3))
                         (* 2 (/ j 3))
                         1)))
             k)))

(euler 100)
```

```
2.7182818284590455
```

2.57 Exercise 1.39

2.57.1 Question

A continued fraction representation of the tangent function was published in 1770 by the German mathematician J.H. Lambert:

$$\tan x = \cfrac{x}{1 - \cfrac{x^2}{3 - \cfrac{x^2}{5 - \ldots}}}$$

where x is in radians. Define a procedure (`tan-cf x k`) that computes an approximation to the tangent function based on Lambert's formula. `k` specifies the number of terms to compute, as in Exercise 1.37.

2.57.2 Answer

```
<<cont-frac>>
(define (tan-cf x k)
  (cont-frac (λ(i) (if (= i 1)
                       x
                       (* x x -1.0)))
             (λ(i) (if (= i 1)
                       1.0
                       (- (* i 2.0) 1.0)))
             k))

(tan-cf 55 101)
```

```
-45.1830879105221
```

2.58 1.3.4 Procedures as Returned Values

Procedures can return other procedures, which opens up new ways to express processes.

Newton's Method: $g(x) = 0$ is a fixed point of the function $x \mapsto f(x)$ where

$$f(x) = x - \frac{g(x)}{Dg(x)}$$

Where $x \mapsto g(x)$ is a differentiable function and $Dg(x)$ is the derivative of g evaluated at x.

2.59 Exercise 1.40

2.59.1 Text

```
(define (average-damp f)
  (lambda (x) (average x (f x))))
```

```
(define dx 0.00001)
```

```
(define (deriv g)
  (lambda (x) (/ (- (g (+ x dx)) (g x)) dx)))
```

```
(define (newton-transform g)
  (lambda (x) (- x (/ (g x) ((deriv g) x)))))
(define (newtons-method g guess)
  (fixed-point (newton-transform g) guess))
```

```
<<average>>
<<average-damp>>
<<dx>>
<<deriv>>
<<newtons-method>>
```

2.59.2 Question

Define a procedure `cubic` that can be used together with the `newtons-method` procedure in expressions of the form:

```
(newtons-method (cubic a b c) 1)
```

to approximate zeros of the cubic $x^3 + ax^2 + bx + c$.

2.59.3 Answer

```
(define (cubic a b c)
  (lambda (x)
    (+ (expt x 3)
       (* a (expt x 2))
       (* b x)
       c)))
```

```
(define (cubic-zero a b c)
  (newtons-method (cubic a b c) 1))
```

```
<<fixed-point-txt>>
<<newtons-method-txt>>
<<cubic>>
<<cubic-zero>>

(cubic-zero 2 3 4)
```

2.60 Exercise 1.41

2.60.1 Question

> Define a procedure `double` that takes a procedure of one argument as argument and returns a procedure that applies the original procedure twice. For example, if `inc` is a procedure that adds 1 to its argument, then `(double inc)` should be a procedure that adds 2. What value is returned by

```
(((double (double double)) inc) 5)
```

2.60.2 Answer

```
(define (double f)
  (λ (x)
    (f (f x))))
```

```
(define inc 1+)
<<double>>
<<Ex1-41>>
```

```
21
```

2.61 Exercise 1.42

2.61.1 Question

> Let f and g be two one-argument functions. The composition f after g is defined to be the function $x \mapsto f(g(x))$. Define a procedure `compose` that implements composition.

2.61.2 Answer

```
(define (compose f g)
  (λ(x)
    (f (g x))))
```

```
<<compose>>
<<square>>
(define inc 1+)
((compose square inc) 6)
```

```
49
```

2.62 Exercise 1.43

2.62.1 Question

If f is a numerical function and n is a positive integer, then we can form the n^{th} repeated application of f, which is defined to be the function whose value at x is $f(f(\ldots(f(x))\ldots))$. For example, if f is the function $x \mapsto x+1$, then the n^{th} repeated application of f is the function $x \mapsto x+n$. If f is the operation of squaring a number, then the n^{th} repeated application of f is the function that raises its argument to the 2^n-th power. Write a procedure that takes as inputs a procedure that computes f and a positive integer n and returns the procedure that computes the n^{th} repeated application of f.

2.62.2 Answer

```
<<compose>>
(define (repeated f n)
  (if (= n 1)
      f
      (repeated (compose f f)
                (- n 1))))
```

```
<<square>>
<<repeated>>
(if (= ((repeated square 2) 5) 625)
    "Success"
    "Fail")
```

```
Success
```

2.63 Exercise 1.44

2.63.1 Question

The idea of smoothing a function is an important concept in signal processing. If f is a function and dx is some small number, then the smoothed version of f is the function whose value at a point x is the average of $f(x-dx)$, $f(x)$, and $f(x+dx)$. Write a procedure `smooth` that takes as input a procedure that computes f and returns a procedure that computes the smoothed f. It is sometimes valuable to repeatedly smooth a function (that is, smooth the smoothed function, and so on) to obtain the n-fold smoothed function. Show how to generate the n-fold smoothed function of any given function using `smooth` and `repeated` from Exercise 1.43.

2.63.2 Answer

```
<<average-varargs>>
(define (smooth f)
  (λ(x)
    (average (f (- x dx))
             (f x)
             (f (+ x dx)))))
(define (smooth-n f n)
  ((repeated smooth n) f))
```

2.64 Exercise 1.45

2.64.1 Question

> We saw in 1.3.3 that attempting to compute square roots by naively finding a fixed point of $y \mapsto x/y$ does not converge, and that this can be fixed by average damping. The same method works for finding cube roots as fixed points of the average-damped $y \mapsto x/y^2$. Unfortunately, the process does not work for fourth roots—a single average damp is not enough to make a fixed-point search for $y \mapsto x/y^3$ converge. On the other hand, if we average damp twice (i.e., use the average damp of the average damp of $y \mapsto x/y^3$) the fixed-point search does converge. Do some experiments to determine how many average damps are required to compute n^{th} roots as a fixed-point search based upon repeated average damping of $y \mapsto x/y^{n-1}$. Use this to implement a simple procedure for computing n^{th} roots using `fixed-point~, ~average-damp`, and the `repeated` procedure of Exercise 1.43. Assume that any arithmetic operations you need are available as primitives.

2.64.2 Answer

So this is strange. Back in my original workthrough of this book, I'd decided that finding an nth root required $\lfloor\sqrt{n}\rfloor$ dampings. With a solution like this:

```
<<fixed-point-txt>>
<<repeated>>
<<average-damp>>
(define (sqrt n)
  (fixed-point
   (average-damp
    (lambda (y)
      (/ x y)))
   1.0))
(define (nth-root x n)
  (fixed-point
   ((repeated average-damp (ceiling (sqrt n)))
```

```
    (lambda (y)
      (/ x (expt y (- n 1)))))
   1.0))
```

While this solution appears to work fine, my experiments are suggesting that it takes *less* than $\lfloor \sqrt{n} \rfloor$. For example, I originally thought powers of 16 required four dampings, but this code isn't failing until it reaches powers of 32.

```
;; Version of "repeated" that can handle being asked to repeat zero
↪  times.
<<compose>>
<<identity>>
(define (repeated f n)
  (define (rec m)
  (if (= n 1)
      f
      (repeated (compose f f)
                (- n 1))))
  (if (= n 0)
      identity
      (rec n)))
```

```
;; version of "fixed-point" that will give up after a certain number of
↪  guesses.
(define (limited-fixed-point f first-guess)
  (define limit 5000)
  (define tolerance 0.00000001)
  (define (close-enough? v1 v2)
    (< (abs (- v1 v2))
       tolerance))
  (define (try guess tries)
    (if (= tries limit)
        "LIMIT REACHED"
        (let ((next (f guess)))
          (if (close-enough? guess next)
              next
              (try next (+ 1 tries))))))
    (try first-guess 1))
```

Let's automatically find how many dampings are necessary. We can make a program that finds higher and higher nth roots, and adds another layer of damping when it hits the error. It returns a list of nth roots along with how many dampings were needed to find them.

```
<<fixed-point-txt>>
<<limited-fixed-point>>
<<repeated>>
<<average-damp>>
```

```
<<average>>
<<print-table>>
(define (sqrt x)
  (fixed-point
   (average-damp
    (lambda (y) (/ x y)))
   1.0))
(define (nth-tester base n-max)
  (define (iter ll)
    (let ((n (+ 2 (length ll))))
      (define (try damps)
        (let ((x (limited-fixed-point
                  ((repeated average-damp damps)
                   (lambda (y)
                     (/ base (expt y (- n 1)))))
                  1.1)))
          (if (string? x)
              (try (1+ damps))
              (list base n x damps))))
      (if (> n n-max)
          ll
          (iter (cons (try 1) ll)))))

  (iter '()))
(let* ((t (reverse (nth-tester 3 65))))
  (cons '("root" "result" "damps needed" "floor(sqrt(root))"
↪  "floor(log2(root))")
        (map (λ(x)
               (append x
                       (list (floor (sqrt (car x)))
                             (floor (/ (log (car x))(log 2))))))
             (map cdr t))))
```



I've spent too much time on this problem already but I have to wonder about floating-point issues, given that they are the core of the `good-enough` procedure. I have to wonder whether a `fixed-point` version that replaces the `toleranc⌋ e` decision making, and instead retains the last three guesses and checks for a loop. (TODO)

2.65 Exercise 1.46

2.65.1 Question

> Several of the numerical methods described in this chapter are instances of an extremely general computational strategy known as *iterative improvement*. Iterative improvement says that, to compute something, we start with an initial guess for the answer, test if the guess is good enough, and otherwise improve the guess and continue the process using the improved guess as the new guess. Write a procedure `iterative-improve` that takes two procedures as arguments: a method for telling whether a guess is good enough and a method for improving a guess. `iterative-improve` should return as its value a procedure that takes a guess as argument and keeps improving the guess until it is good enough. Rewrite the `sqrt` procedure of 1.1.7 and the `fixed-point` procedure of 1.3.3 in terms of `iterative-imp⌋ rove`.

2.65.2 Answer

```
(define (iterative-improve good-enough? improve)
  (λ(first-guess)
    (define (iter guess)
      (let ((next (improve guess)))
        (if (good-enough? guess next)
            next
            (iter next))))
    (iter first-guess)))
```

```
<<iterative-improve>>
(define tolerance 0.00001)

(define (fixed-point-improve f first-guess)
  (define (close-enough? v1 v2)
    (< (abs (- v1 v2))
       tolerance))
  ((iterative-improve close-enough? f) first-guess))
```

```
<<average>>
<<iterative-improve>>
(define (improve guess x)
  (average guess (/ x guess)))
(define (good-enough? guess x)
   (= (improve guess x) guess))
(define (sqrt-improve x)
  ((iterative-improve
    (λ(guess) (improve guess x))
    (λ(guess) (good-enough? guess x)))
   1.0))
```

```
(load "mattcheck2.scm")
<<fixed-point-txt>>
<<fixed-point-improve>>
(mattcheck "fixed-point-improve still working"
                (fixed-point (λ(x)(+ 1 (/ 1 x))) 1.0)
                (fixed-point-improve (λ(x)(+ 1 (/ 1 x))) 1.0))
<<sqrt>>
<<sqrt-improve>>
(mattcheck "sqrt-improve still working"
                (sqrt 5)
                (sqrt 5))
```

```
SUCCEED at fixed-point-improve still working
SUCCEED at sqrt-improve still working
```

3 Chapter 2: Building Abstractions with Data

The basic representations of data we've used so far aren't enough to deal with complex, real-world phenomena. We need to combine these representations to form **compound data**.

The technique of isolating how data objects are *represented* from how they are *used* is called **data abstraction**.

3.1 2.1.1: Example: Arithmetic Operations for Rational Numbers

Lisp gives the procedures `cons`, `car`, and `cdr` to create **pairs**. This is an easy system for representing rational numbers.

Note that the system proposed for representing and working with rational numbers has **abstraction barriers** isolating different parts of the system. The parts that use rational numbers don't know how the constructors and selectors for rational numbers work, and the constructors and selectors use the underlying Lisp interpreter's pair functions without caring how they work.

Note that these abstraction layers allow the developer to change the underlying architecture without modifying the programs that depend on it.

3.2 Exercise 2.1

3.2.1 Text

```
(define (add-rat x y)
  (make-rat (+ (* (numer x) (denom y))
               (* (numer y) (denom x)))
            (* (denom x) (denom y))))

(define (sub-rat x y)
  (make-rat (- (* (numer x) (denom y))
               (* (numer y) (denom x)))
            (* (denom x) (denom y))))

(define (mul-rat x y)
  (make-rat (* (numer x) (numer y))
            (* (denom x) (denom y))))

(define (div-rat x y)
  (make-rat (* (numer x) (denom y))
            (* (denom x) (numer y))))

(define (equal-rat? x y)
  (= (* (numer x) (denom y))
     (* (numer y) (denom x))))
```

```
(define (make-rat n d) (cons n d))
(define (numer x) (car x))
(define (denom x) (cdr x))
```

```
(define (print-rat x)
  (newline)
  (display (numer x))
  (display "/")
  (display (denom x)))
```

```
(define one-half (make-rat 1 2))
(define one-third (make-rat 1 3))
(print-rat one-half)
(print-rat
 (mul-rat one-half one-third))
```

```
1/2
1/6
```

3.2.2 Question

Define a better version of `make-rat` that handles both positive and negative arguments. `make-rat` should normalize the sign so that if the rational number is positive, both the numerator and denominator are positive, and if the rational number is negative, only the numerator is negative.

3.2.3 Answer

```
<<abs>>
(define (make-rat n d)
  (cond ((not (or (< n 0)
                  (< d 0)))
         (cons n d))
        ((and (< n 0)
              (< d 0))
         (cons (- n) (- d)))
        (else
         (cons (- (abs n)) (abs d)))))
(define (numer x) (car x))
(define (denom x) (cdr x))

;; Bonus: an attempt to optimize
(define (make-rat-opt n d)
  (let ((nn (< n 0))
        (dn (< d 0)))
```

```
    (cond ((not (or nn dn))
           (cons n d))
          ((and nn dn)
           (cons (- n) (- d)))
          (else
           (cons (- (abs n)) (abs d))))))
```

```
<<make-rat>>
<<print-rat-txt>>
<<rat-ops-txt>>
(load "mattcheck2.scm")
(mattcheck "make-rat double negative"
           (cons 1 2)
           (make-rat -1 -2))
(mattcheck "make-rat numerator negative"
           (cons -1 2)
           (make-rat -1 2))
(mattcheck "make-rat denominator negative"
           (cons -1 2)
           (make-rat 1 -2))
(mattcheck "make-rat-opt double negative"
           (cons 1 2)
           (make-rat-opt -1 -2))
(mattcheck "make-rat-opt numerator negative"
           (cons -1 2)
           (make-rat-opt -1 2))
(mattcheck "make-rat-opt denominator negative"
           (cons -1 2)
           (make-rat-opt 1 -2))
```

```
SUCCEED at make-rat double negative
SUCCEED at make-rat numerator negative
SUCCEED at make-rat denominator negative
SUCCEED at make-rat-opt double negative
SUCCEED at make-rat-opt numerator negative
SUCCEED at make-rat-opt denominator negative
```

My "optimized" version shows no benefit at all:

```
unoptimized make-rat: ((1 . 2) 231.74267794)
optimized make-rat: ((1 . 2) 233.99087033)
```

3.3 Exercise 2.2

3.3.1 Question

Consider the problem of representing line segments in a plane. Each segment is represented as a pair of points: a starting point and an

ending point. Define a constructor `make-segment` and selectors `start-segment` and `end-segment` that define the representation of segments in terms of points. Furthermore, a point can be represented as a pair of numbers: the x coordinate and the y coordinate. Accordingly, specify a constructor `make-point` and selectors `x-point` and `y-point` that define this representation. Finally, using your selectors and constructors, define a procedure `midpoint-segment` that takes a line segment as argument and returns its midpoint (the point whose coordinates are the average of the coordinates of the endpoints). To try your procedures, you'll need a way to print points:

```
(define (print-point p)
  (newline)
  (display "(")
  (display (x-point p))
  (display ",")
  (display (y-point p))
  (display ")"))
```

3.3.2 Answer

```
<<average>>
(define (make-point x y)
  (cons x y))
(define (x-point p)
  (car p))
(define (y-point p)
  (cdr p))
(define (make-segment start end)
  (cons start end))
(define (start-segment seg)
  (car seg))
(define (end-segment seg)
  (cdr seg))
(define (midpoint-segment seg)
  (make-point (average (x-point (start-segment seg))
                       (x-point (end-segment seg)))
              (average (y-point (start-segment seg))
                       (y-point (end-segment seg)))))
(define (midpoint-segment-opt seg)
  (let ((ax (x-point (start-segment seg)))
        (bx (x-point (end-segment seg)))
        (ay (y-point (start-segment seg)))
        (by (y-point (end-segment seg))))
  (make-point (average ax
                       bx)
              (average ay
```

```
                        by))))
```

```
<<make-point>>
(load "mattcheck2.scm")
(mattcheck "make-point"
           (list 1 2)
           (let ((p (make-point 1 2)))
             (list (x-point p)
                   (y-point p))))
(let* ((p1 (make-point 1 2))
       (p2 (make-point -1 -2))
       (s (make-segment p1 p2)))
  (mattcheck "make-segment"
             (list p1 p2)
             (list (start-segment s)
                   (end-segment s)))
  (mattcheck "midpoint-segment"
              (make-point 0 0)
              (midpoint-segment s))
  (mattcheck "midpoint-segment-opt"
              (make-point 0 0)
              (midpoint-segment-opt s)))
```

```
SUCCEED at make-point
SUCCEED at make-segment
SUCCEED at midpoint-segment
SUCCEED at midpoint-segment-opt
```

And once again my bikeshedding is revealed:

```
unoptimized make-rat: ((0.0 . 0.0) 326.94653558)
optimized make-rat: ((0.0 . 0.0) 331.83410742)
```

3.4 Exercise 2.3

3.4.1 Question

> Implement a representation for rectangles in a plane. (Hint: You may want to make use of Exercise 2.2.) In terms of your constructors and selectors, create procedures that compute the perimeter and the area of a given rectangle. Now implement a different representation for rectangles. Can you design your system with suitable abstraction barriers, so that the same perimeter and area procedures will work using either representation?

3.4.2 Answer 1

I don't really like the "wishful thinking" process the book advocates but since this question specifically regards abstraction, I'll start by writing the two requested procedures first.

```
(define (rect-area R)
  (* (rect-height R)
     (rect-width R)))

(define (rect-peri R)
  (* 2
     (+ (rect-height R)
        (rect-width R))))
```

So my "wishlist" is just for `(rect-area R)` and `(rect-width R)`.

So, my first implementation of a rectangle will be of a list of 3 points ABC, with the fourth point D being constructed from the others. I haven't done geometry lessons in a while but logically I can deduce that D is as far from A as B is from C, and as far from C as A is from B. by experimentation I've figured out that $D = A + (C - B) = C + (A - B)$.

```
;;   AB = width
;;(0,1) (1,1)
;; A-----B
;; |     | BC = height
;; D-----C
;;(0,0) (1,0)
;; could be rotated any direction
<<square>>
<<make-point>>
(define (make-rect a b c)
  (cons (cons a b) c))
(define (rect-a R)
  (caar R))
(define (rect-b R)
  (cdar R))
(define (rect-c R)
  (cdr R))
;(define (rect-d R)
;  (make-point (x-point (rect-a R))
;              (y-point (rect-c R))))
;; Wait, this won't work if the rectangle is angled.

(define (sub-points a b)
  (make-point (- (x-point a)
                 (x-point b))
              (- (y-point a)
```

```
                  (y-point b))))

(define (add-points a b)
  (make-point (+ (x-point a)
                 (x-point b))
              (+ (y-point a)
                 (y-point b))))

(define (rect-d R)
  (let ((a (rect-a R))
        (b (rect-b R))
        (c (rect-c R)))
    (add-points a
                (sub-points c b))))
(define (rect-d-alt R) ; should be mathematically identical.
  (let ((a (rect-a R))
        (b (rect-b R))
        (c (rect-c R)))
    (add-points c
                (sub-points a b))))

;; this is incorrect
;(define (length-points a b)
;  (let ((diffP (sub-points a b)))
;    (+ (abs (x-point diffP))
;       (abs (y-point diffP)))))
(define (length-points a b)
  (let ((ax (x-point a))
        (ay (y-point a))
        (bx (x-point b))
        (by (y-point b)))
    (sqrt (+ (square (- ax bx))
             (square (- ay by))))))

(define (rect-height R)
  (abs (length-points (rect-b R)
                      (rect-c R))))
(define (rect-width R)
  (abs (length-points (rect-b R)
                      (rect-a R))))

(define (length-segment seg)
  (abs (length-points (start-segment seg)
                      (end-segment seg))))
```

```
(load "mattcheck2.scm")

<<rect-4pt>>
```

```
<<rect-area-peri>>

(let* ((a (make-point 13 14))
       (b (make-point 14 14))
       (c (make-point 14 13))
       (d (make-point 13 13))
       (ABC (make-rect a b c))
       (CDA (make-rect c d a))
       (w (make-point -2.0 -2.0))
       (x (make-point -0.5 -0.5))
       (y (make-point -1.5 0.5))
       (z (make-point -3.0 -1.0))
       (WXY (make-rect w x y)))
  (mattcheck "make-rect"
             ABC
             (cons (cons a b) c))
  (mattcheck "rect-d and rect-d-alt (ABCD)"
             (rect-d ABC)
             (rect-d-alt ABC)
             d)
  (mattcheck "rect-d and rect-d-alt (CDAB)"
             (rect-d CDA)
             (rect-d-alt CDA)
             b)
  (mattcheck "rect-d and rect-d-alt (WXYZ)"
             (rect-d WXY)
             (rect-d-alt WXY)
             z)
  (mattcheck "rect-d and rect-d-alt (XYZW)"
             (rect-d (make-rect x y z))
             w)
  (mattcheck "rect-height ABC"
             (rect-height ABC)
             1)
  (mattcheck "rect-width ABC"
             (rect-width ABC)
             1)
  (mattcheck "rect-height WXY"
             (rect-height WXY)
             1.4142135623730951)
  (mattcheck "rect-width WXY"
             (rect-width WXY)
             2.1213203435596424)
  (mattcheck "rect-area ABCD"
             (rect-area ABC)
             (rect-area CDA)
             1)
  (mattcheck "rect-area WXYZ"
```

```
              (rect-area WXY)
              3.0)
   (mattcheck "rect-peri ABCD"
              (rect-peri ABC)
              4)
   (mattcheck "rect-peri WXYZ"
              (rect-peri WXY)
              7.0710678118654755))
```

```
SUCCEED at make-rect
SUCCEED at rect-d and rect-d-alt (ABCD)
SUCCEED at rect-d and rect-d-alt (CDAB)
SUCCEED at rect-d and rect-d-alt (WXYZ)
SUCCEED at rect-d and rect-d-alt (XYZW)
SUCCEED at rect-height ABC
SUCCEED at rect-width ABC
SUCCEED at rect-height WXY
SUCCEED at rect-width WXY
SUCCEED at rect-area ABCD
SUCCEED at rect-area WXYZ
SUCCEED at rect-peri ABCD
SUCCEED at rect-peri WXYZ
```

3.4.3 Answer 2

My second implementation will be of a rectangle as an origin, height, width, and angle. Basically, height and width are two vectors originating from origin, with width going straight right and height offset 90 deg from width. Angle is added during conversion from Polar to Cartesian coordinates. In relation to my 1st implementation, point D is where the origin is.

```
<<make-point>>
;; origin is a (make-point), hwa are floats
(define (make-rect origin height width angle)
  (cons (cons origin height)
        (cons width angle)))

(define (rect-origin R)
  (caar R))
(define rect-d rect-origin)
(define (rect-height R)
  (cdar R))
(define (rect-width R)
  (cadr R))
(define (rect-angle R)
  (cddr R))

```

```
;; I underestimated how much math this would take.
(define (add-points a b)
  (make-point (+ (x-point a)
                 (x-point b))
              (+ (y-point a)
                 (y-point b))))

(define pi (* 4 (atan 1.0)))
(define (radian deg)
  (* deg (/ pi 180.0)))
(define (vector-to-xy distance angle)
      ;; rect-c: (cos(Theta),sin(Theta)) * width
      (make-point (* (cos (radian angle))
                     distance)
                  (* (sin (radian angle))
                     distance)))
      ;; could also be rotated by 90 degrees just by using
      ;;   (-sin(Theta),cos(Theta)) * height
(define (rect-c R)
  (add-points
   (rect-origin R)
   (vector-to-xy (rect-width R) (rect-angle R))))
(define (rect-a R)
  (add-points
   (rect-origin R)
   (vector-to-xy (rect-height R)
                 (+ 90 (rect-angle R)))))
(define (rect-b R)
  (add-points
   (rect-origin R)
   (add-points
    (vector-to-xy (rect-width R) (rect-angle R))
    (vector-to-xy (rect-height R)
                  (+ 90 (rect-angle R))))))
```

```
(load "mattcheck2.scm")

<<rect-ohwa>>
<<rect-area-peri>>

(let* ((a (make-point 13.0 14.0))
       (b (make-point 14.0 14.0))
       (c (make-point 14.0 13.0))
       (d (make-point 13.0 13.0))
       (ABC (make-rect d 1 1 0))
       (CDA (make-rect b 1 1 180))
       (w (make-point -2.0 -2.0))
       (x (make-point -2.5 1.5))
```

```
        (y (make-point -1.5 0.5))
        (z (make-point -3.0 -1.0))
        (wxy-height 1.4142135623730951)
        (wxy-width 2.1213203435596424)
        (WXY (make-rect z wxy-height wxy-width 45)))
    (mattcheck "make-rect"
               ABC
               (cons (cons d 1) (cons 1 0)))
    (mattcheck "rect-b (ABCD)"
               (rect-b ABC)
               b)
    (mattcheck "rect-b (CDAB)"
               (rect-b CDA)
               d)
    (mattcheck "rect-b (WXYZ)"
               (rect-b WXY)
               x)
    (mattcheck "rect-height"
               (rect-height WXY)
               wxy-height)
    (mattcheck "rect-width"
               (rect-width WXY)
               wxy-width)
    (mattcheck "rect-area ABCD"
               (rect-area ABC)
               (rect-area CDA)
               1)
    (mattcheck "rect-area WXYZ"
               (rect-area WXY)
               3.0)
    (mattcheck "rect-peri ABCD"
               (rect-peri ABC)
               4)
    (mattcheck "rect-peri WXYZ"
               (rect-peri WXY)
               7.0710678118654755))
```

```
SUCCEED at make-rect
SUCCEED at rect-b (ABCD)
SUCCEED at rect-b (CDAB)
SUCCEED at rect-b (WXYZ)
SUCCEED at rect-height
SUCCEED at rect-width
SUCCEED at rect-area ABCD
SUCCEED at rect-area WXYZ
SUCCEED at rect-peri ABCD
SUCCEED at rect-peri WXYZ
```

3.5 2.1.3: What Is Meant by Data?

We can consider data as being a collection of selectors and constructors, together with specific conditions that these procedures must fulfill in order to be a valid representation. For example, in the case of our rational number implementation, for rational number x made with numerator n and denominator d, dividing the result of `(numer x)` over the result of `(denom x)` should be equivalent to dividing n over d.

3.6 Exercise 2.4

3.6.1 Question

> Here is an alternative procedural representation of pairs. For this representation, verify that `(car (cons x y))` yields `x` for any objects `x` and `y`.

```
(define (cons x y)
  (lambda (m) (m x y)))
(define (car z)
  (z (lambda (p q) p)))
```

> What is the corresponding definition of `cdr`? (Hint: To verify that this works, make use of the substitution model of 1.1.5.)

3.6.2 Answer

First, let's explain with the substitution model.

```
(cons 0 1)
(lambda (m) (m 0 1))

(car (lambda (m) (m 0 1)))
((lambda (m) (m 0 1)) (lambda (p q) p))
(lambda (0 1) 0)
0
(cdr (lambda (m) (m 0 1)))
((lambda (m) (m 0 1)) (lambda (p q) q))
(lambda (0 1) 1)
1
```

Now for implementation.

```
(load "mattcheck2.scm")
<<alt-pairs-txt>>
(define (cdr z)
  (z (lambda (p q) q)))

```

```
(let ((pair (cons 0 1)))
  (mattcheck "car"
             (car pair)
             0)
  (mattcheck "cdr"
             (cdr pair)
             1))
```

```
| (0 . 0) | (0 . 1) | (0 . 2) | (0 . 3) | (0 . 4) | (0 . 5) | (0 . 6) |
| (1 . 0) | (1 . 1) | (1 . 2) | (1 . 3) | (1 . 4) | (1 . 5) | (1 . 6) |
| (2 . 0) | (2 . 1) | (2 . 2) | (2 . 3) | (2 . 4) | (2 . 5) | (2 . 6) |
| (3 . 0) | (3 . 1) | (3 . 2) | (3 . 3) | (3 . 4) | (3 . 5) | (3 . 6) |
| (4 . 0) | (4 . 1) | (4 . 2) | (4 . 3) | (4 . 4) | (4 . 5) | (4 . 6) |
| (5 . 0) | (5 . 1) | (5 . 2) | (5 . 3) | (5 . 4) | (5 . 5) | (5 . 6) |
| (6 . 0) | (6 . 1) | (6 . 2) | (6 . 3) | (6 . 4) | (6 . 5) | (6 . 6) |
```

3.7 Exercise 2.5 optional

3.7.1 Question

> Show that we can represent pairs of nonnegative integers using only numbers and arithmetic operations if we represent the pair a and b as the integer that is the product $2^a 3^b$. Give the corresponding definitions of the procedures `cons`, `car`, and `cdr`.

3.7.2 Answer

This one really blew my mind inside-out when I first did it. Basically, because the two numbers are coprime, you can factor out the unwanted number and be left with the desired one.

> Where x is the scrambled number, p is the base we want to remove, q is the base we want to retrieve from and y is the value exponentiating p, the original number is retrieved by dividing x by p for y number of times, and then applying $\log_q$ to the result.

First, let's make `cons`.

```
(define (cons-nnint a b)
  (* (expt 2 a) (expt 3 b)))
(define (cons-nnint-debug a b) ;; DEBUG
  (let* ((aa (expt 2 a))
         (bb (expt 3 b))
         (ab (* aa bb)))
    (display aa)
    (newline)
    (display bb)
    (newline)
```

```
    (display ab)
    (newline)
    ab))
```

Also, Guile doesn't have a function for custom logs so let's define that now.

```
(define (logn b p)
  (/ (log p) (log b)))
```

Let's do some analysis to see how these numbers are related.

```
<<cons-nnint>>
(let*
    ((tablesize 7)
     (inputs (map (λ(x)
                    (map (λ(y)
                           (cons x y))
                         (iota tablesize)))
                  (iota tablesize)))
     (outputs (map (λ(row)
                     (map (λ(col)
                            (cons-nnint (car col) (cdr col)))
                          row))
                   inputs)))
  outputs)
```

1	3	9	27	81	243	729
2	6	18	54	162	486	1458
4	12	36	108	324	972	2916
8	24	72	216	648	1944	5832
16	48	144	432	1296	3888	11664
32	96	288	864	2592	7776	23328
64	192	576	1728	5184	15552	46656

Here are our scrambled numbers.

```
;; To find a number of some base in some column,
;; First divide by unwantedbase for targetcol number of times
<<repeated>>
(let ((targetcol 2)
      (unwantedbase 3))
  (map (λ(row)
         (map (λ(item)
                ((repeated (λ(x)
                             (/ x unwantedbase)) targetcol)
                 item))
```

```
                row))
        data))
```

1/9	1/3	1	3	9	27	81
2/9	2/3	2	6	18	54	162
4/9	4/3	4	12	36	108	324
8/9	8/3	8	24	72	216	648
16/9	16/3	16	48	144	432	1296
32/9	32/3	32	96	288	864	2592
64/9	64/3	64	192	576	1728	5184

The numbers from our target column onwards are integers, with the target column being linearly exponentiated by 2 because the original numbers were linear.

```
<<logn>>
(let ((wantedbase 2))
  (map (λ(row)
         (map (λ(item)
                (format #f "~6,3f" (logn 2 item)))
              row))
       data))
```

-3.170	-1.585	0.000	1.585	3.170	4.755	6.340
-2.170	-0.585	1.000	2.585	4.170	5.755	7.340
-1.170	0.415	2.000	3.585	5.170	6.755	8.340
-0.170	1.415	3.000	4.585	6.170	7.755	9.340
0.830	2.415	4.000	5.585	7.170	8.755	10.340
1.830	3.415	5.000	6.585	8.170	9.755	11.340
2.830	4.415	6.000	7.585	9.170	10.755	12.340

Now the second column has recovered its original values. Although we didn't know what the original integer values were, we can now tell which column has the correct numbers by looking at which are integer values.

We can use this sign of a correct result in the proposed `car` and `cdr` procedures.

```
<<cons-nnint>>
<<logn>>
(use-srfis '(1))
(define (all-your-base ab unwanted wanted)
  (if (equal? (modulo ab unwanted) 0)
      (all-your-base (/ ab unwanted) unwanted wanted)
      (if (equal? (modulo ab wanted) 0)
          (round (logn wanted ab))
          "This number isn't a factor!")))
(define (car-nnint ab)
  (all-your-base ab 3 2))
```

```
(define (cdr-nnint ab)
  (all-your-base ab 2 3))

(let* ((initvalues '((2 3) (4 5) (7 2)))
       (conslist (map (λ(x)
                        (apply cons-nnint x))
                      initvalues))
       (carlist (map (λ(x)
                       (car-nnint x))
                     conslist))
       (cdrlist (map (λ(x)
                       (cdr-nnint x))
                     conslist)))
  (map (λ(x y) (cons x y))
       (list "pairs" "cons'd" "car" "cdr")
       (list initvalues conslist carlist cdrlist)))
```

pairs	(2 3)	(4 5)	(7 2)
cons'd	108	3888	1152
car	2.0	4.0	7.0
cdr	3.0	5.0	2.0

3.8 Exercise 2.6 optional

3.8.1 Question

> In case representing pairs as procedures wasn't mind-boggling enough, consider that, in a language that can manipulate procedures, we can get by without numbers (at least insofar as nonnegative integers are concerned) by implementing 0 and the operation of adding 1 as

```
(define zero (λ (f) (λ (x) x)))
(define (add-1 n)
  (λ (f) (λ (x) (f ((n f) x)))))
```

> This representation is known as *Church numerals*, after its inventor, Alonzo Church, the logician who invented the λ-calculus.
>
> Define `one` and `two` directly (not in terms of `zero` and `add-1`). (Hint: Use substitution to evaluate `(add-1 zero)`). Give a direct definition of the addition procedure `+` (not in terms of repeated application of `add-1`).

3.8.2 Answer

First, let's check out `(add-1 zero)`.

```
(define zero (λ (f) (λ (x) x)))
(define (add-1 n)
  (λ (f) (λ (x)
           (f ((n f) x)))))

(add-1 zero)
((λ (f) (λ (x)
          (f ((zero f) x)))))
((λ (f) (λ (x)
          (f ((λ (x) x) x)))))
((λ (f) (λ (x)
          (f x))))
```

So from this I believe the correct definition of one and two are:

```
(load "mattcheck2.scm")
(define one
  (λ (f) (λ (x)
           (f x))))
(define two
  (λ (f) (λ (x)
            (f (f x)))))

(mattcheck "1 = 1+0"
           1
           ((one 1+) 0))
(mattcheck "2 = 1+1+0"
           2
           ((two 1+) 0))

(define (add a b)
  (λ (f) (λ (x)
           ((a f) ((b f) x)))))

(mattcheck "3 = 1+2 = (1+0) + (1+1+0)"
           3
           (((add one two) 1+) 0))
```

```
SUCCEED at 1 = 1+0
SUCCEED at 2 = 1+1+0
SUCCEED at 3 = 1+2 = (1+0) + (1+1+0)
```

3.9 Exercise 2.7

3.9.1 Text

```
(define (add-interval x y)
  (make-interval (+ (lower-bound x) (lower-bound y))
                 (+ (upper-bound x) (upper-bound y))))
```

```
(define (mul-interval x y)
  (let ((p1 (* (lower-bound x) (lower-bound y)))
        (p2 (* (lower-bound x) (upper-bound y)))
        (p3 (* (upper-bound x) (lower-bound y)))
        (p4 (* (upper-bound x) (upper-bound y))))
    (make-interval (min p1 p2 p3 p4)
                   (max p1 p2 p3 p4))))
(define (div-interval x y)
  (mul-interval
   x
   (make-interval (/ 1.0 (upper-bound y))
                  (/ 1.0 (lower-bound y)))))
```

3.9.2 Question

Alyssa's program is incomplete because she has not specified the implementation of the interval abstraction. Here is a definition of the interval constructor:

```
(define (make-interval a b) (cons a b))
```

Define selectors `upper-bound` and `lower-bound` to complete the implementation.

3.9.3 Answer

```
<<interval-txt>>

;; Makes more sense to me to test
;; order in the constructor than selector
(define (make-interval a b)
  (if (> a b)
      (cons a b)
      (cons b a)))

(define (upper-bound i)
  (car i))
(define (lower-bound i)
  (cdr i))
```

3.10 Exercise 2.8

3.10.1 Question

Using reasoning analogous to Alyssa's, describe how the difference of two intervals may be computed. Define a corresponding subtraction procedure, called `sub-interval`.

3.10.2 Answer

I would argue that with one interval subtracted from the other, the lowest possible value is the lower of the first subtracted from the *upper* of the second, and the highest is the upper of the first subtracted from the lower of the second.

```
(define (sub-interval x y)
  (make-interval (- (lower-bound x) (upper-bound y))
                 (- (upper-bound x) (lower-bound y))))
```

3.11 Exercise 2.9

3.11.1 Question

> The *width* of an interval is half of the difference between its upper and lower bounds. The width is a measure of the uncertainty of the number specified by the interval. For some arithmetic operations the width of the result of combining two intervals is a function only of the widths of the argument intervals, whereas for others the width of the combination is not a function of the widths of the argument intervals. Show that the width of the sum (or difference) of two intervals is a function only of the widths of the intervals being added (or subtracted). Give examples to show that this is not true for multiplication or division.

3.11.2 Answer

My first interpretation of the question was that it asked whether width operations are *distributive*. For example, multiplication is distributive:

$$a(b+c) = (a \times b) + (a \times c)$$

For this I wrote the following tests:

```
(load "mattcheck2.scm")
<<make-interval>>
<<sub-interval>>

(define (halve x)
  (/ x 2))

(define (width-interval I)
  (halve (- (upper-bound I)
            (lower-bound I))))

(let* ((ia (make-interval 10.1 9.9))
       (ib (make-interval 5.2 4.8))
       (Aab (add-interval ia ib))
```

```
        (Sab (sub-interval ia ib))
        (Mab (mul-interval ia ib))
        (Dab (div-interval ia ib)))
    (mattcheck-float "ia width = roughly .1"
                     0.1
                     (width-interval ia))
    (mattcheck-float "ib width = roughly .2"
                     0.2
                     (width-interval ib))
    (mattcheck-float "width addition is distributive"
                     (width-interval Aab)
                     (+ (width-interval ia)
                        (width-interval ib)))
    (mattcheck-float "width subtraction is distributive"
                     (width-interval Sab)
                     (- (width-interval ia)
                        (width-interval ib)))
    (mattcheck-float "width multiplication is distributive"
                     (width-interval Mab)
                     (* (width-interval ia)
                        (width-interval ib)))
    (mattcheck-float "width division is distributive"
                     (width-interval Dab)
                     (/ (width-interval ia)
                        (width-interval ib))))
```

```
SUCCEED at ia width = roughly .1
SUCCEED at ib width = roughly .2
SUCCEED at width addition is distributive
FAIL at width subtraction is distributive
expected: -0.10000000000000053
returned: 0.2999999999999998
FAIL at width multiplication is distributive
expected: 0.01999999999999995
returned: 2.5
FAIL at width division is distributive
expected: 0.4999999999999978
returned: 0.10016025641025639
```

However upon rereading the question I see that it could be rephrased as "in what operations can you calculate the resulting interval's width with only the widths of the argument intervals?"

Basically, for argument interval x and y and result interval z:

IF $z = x + y$
THEN $z_{width} = x_{width} + y_{width}$
IF $z = x - y$
THEN $z_{width} = x_{width} + y_{width}$

Multiplied or divided widths cannot be determined from widths alone.

So, let's try that again.

```
(load "mattcheck2.scm")
<<make-interval>>
<<sub-interval>>

(define (halve x)
  (/ x 2))

(define (width-interval I)
  (halve (- (upper-bound I)
            (lower-bound I))))

(let* ((ia (make-interval 10.1 9.9))
       (ib (make-interval 5.2 4.8))
       (Aab (add-interval ia ib))
       (Sab (sub-interval ia ib)))
  (mattcheck-float "ia width = roughly .1"
                   0.1
                   (width-interval ia))
  (mattcheck-float "ib width = roughly .2"
                   0.2
                   (width-interval ib))
  (mattcheck-float "width(ia+ib) = width(ia) + width(ib)"
                   (width-interval Aab)
                   (+ (width-interval ia)
                      (width-interval ib)))
  (mattcheck-float "width(ia-ib) = width(ia) + width(ib)"
                   (width-interval Sab)
                   (+ (width-interval ia)
                      (width-interval ib))))
```

```
SUCCEED at ia width = roughly .1
SUCCEED at ib width = roughly .2
SUCCEED at width(ia+ib) = width(ia) + width(ib)
SUCCEED at width(ia-ib) = width(ia) + width(ib)
```

3.12 Exercise 2.10

3.12.1 Question

Ben Bitdiddle, an expert systems programmer, looks over Alyssa's shoulder and comments that it is not clear what it means to divide by an interval that spans zero. Modify Alyssa's code to check for this condition and to signal an error if it occurs.

3.12.2 Answer

```
(define (interval-spans-zero? I)
  (and (> upper-bound 0)
       (< lower-bound 0)))
(define (div-interval x y)
  (if (interval-spans-zero? y)
      "DIV-INTERVAL ERROR: denominator spans zero"
      (mul-interval
       x
       (make-interval (/ 1.0 (upper-bound y))
                      (/ 1.0 (lower-bound y))))))
```

3.13 Exercise 2.11

3.13.1 Question

> In passing, Ben also cryptically comments: "By testing the signs of the endpoints of the intervals, it is possible to break `mul-interval` into nine cases, only one of which requires more than two multiplications." Rewrite this procedure using Ben's suggestion.

3.13.2 Answer

This problem doesn't appear to have a beautiful, elegant answer.

Let's examine the nine cases.

```
(use-modules (ice-9 format))
(use-srfis '(1))
(load "mattcheck2.scm")
<<make-interval>>
<<sub-interval>>

(define (matt-examine-mult f)
  (let* ((pp (make-interval 3 2))
         (pn (make-interval 3 -5))
         (pn2 (make-interval 1 -0.5))
         (nn (make-interval -5 -7))
         (listofpairs (list
                       (list pp pp)
                       (list pp pn)
                       (list pp nn)
                       (list pn pp)
                       (list pn pn)
                       (list pn pn2) ;;<- edge case to catch incomplete
                       (list pn nn)  ;;   multiplication functions
                       (list nn pp)
                       (list nn pn)
                       (list nn nn)))
```

```
          (givesign (λ(x)
                      (if (negative? x)
                          "-"
                          "+")))
          (print-sign (λ(I)
                        (format #f "~a ~a"
                                (givesign (upper-bound I))
                                (givesign (lower-bound I)))))
          (print-int (λ(I)
                       (format #f "~a/~a"
                               (upper-bound I)
                               (lower-bound I))))
          (print-ints (λ(I J)
                        (format #f "~a times ~a"
                                (print-int I)
                                (print-int J))))
          (results (map (λ(p)
                          (apply f p))
                        listofpairs)))
     (list
      (map (λ(p)
             (apply print-ints p))
           listofpairs)
      (map print-int results)
      (map (λ(I)
             (print-sign I))
           results)
      (map (λ(p)
             (format #f "~a // ~a"
                     (print-sign (car p))
                     (print-sign (cadr p))))
           listofpairs))))
 (cons
  (list "problem" "result" "signs" "problem signs")
  (apply zip
         (matt-examine-mult mul-interval)))
```

problem	result	signs	problem signs
3/2 times 3/2	9/4	+ +	+ + // + +
3/2 times 3/-5	9/-15	+ -	+ + // + -
3/2 times -5/-7	-10/-21	- -	+ + // - -
3/-5 times 3/2	9/-15	+ -	+ - // + +
3/-5 times 3/-5	25/-15	+ -	+ - // + -
3/-5 times 1/-0.5	3.0/-5.0	+ -	+ - // + -
3/-5 times -5/-7	35/-21	+ -	+ - // - -
-5/-7 times 3/2	-10/-21	- -	- - // + +
-5/-7 times 3/-5	35/-21	+ -	- - // + -
-5/-7 times -5/-7	49/25	+ +	- - // - -

```
(define (mul-interval-opt x y)
  (let ((xu (upper-bound x))
        (xl (lower-bound x))
        (yu (upper-bound y))
        (yl (lower-bound y)))
    (define p? positive?)
    (define n? negative?)
    (define (check-signs? a b x y) ;; pass functions
      (and (a xu)
           (b xl)
           (x yu)
           (y yl)))
    (define (same-signs?)
      (or (check-signs? p? p? p? p?)
          (check-signs? n? n? n? n?)))
    (define (alt-signs?)
      (or (check-signs? p? p? n? n?)
          (check-signs? n? n? p? p?)))
    (cond ((same-signs?)
           (make-interval (* xu yu)
                          (* xl yl)))
          ((alt-signs?)
           (make-interval (* xl yu)
                          (* xu yl)))
          ((check-signs? p? p? p? n?)
           (make-interval (* xu yu)
                          (* xu yl)))
          ((check-signs? p? n? p? p?)
           (make-interval (* xu yu)
                          (* xl yu)))
          ((check-signs? p? n? p? n?)
           (let ((p1 (* xu yu))
                 (p2 (* xu yl))
                 (p3 (* xl yu))
                 (p4 (* xl yl)))
             (make-interval (max p1 p2 p3 p4)
                            (min p1 p2 p3 p4))))
          ((check-signs? p? n? n? n?)
           (make-interval (* xl yl)
                          (* xu yl)))
          ((check-signs? n? n? p? n?)
           (make-interval (* xl yl)
                          (* xl yu))))))
```

```
(use-modules (ice-9 format))
(use-srfis '(1))
(load "mattcheck2.scm")
```

```
<<make-interval>>
<<sub-interval>>
<<mul-interval-opt>>

(define (matt-mult-consistency f1 f2)
  (let* ((pp (make-interval 3 2))
         (pn (make-interval 3 -5))
         (pn2 (make-interval 1 -0.5))
         (nn (make-interval -5 -7))
         (listofpairs (list
                       (list "pp*nn" pp pp)
                       (list "pp*pn" pp pn)
                       (list "pp*nn" pp nn)
                       (list "pn*pp" pn pp)
                       (list "pn*pn" pn pn)
                       (list "pn*pn2" pn pn2) ;;<- edge case to catch incomplete
                       (list "pn*nn" pn nn)  ;;   multiplication functions
                       (list "nn*pp" nn pp)
                       (list "nn*pn" nn pn)
                       (list "nn*nn" nn nn))))
    (map (λ(l)
           (mattcheck (car l)
                      (apply f1 (cdr l))
                      (apply f2 (cdr l))))
         listofpairs)))

(matt-mult-consistency mul-interval mul-interval-opt)
```

```
SUCCEED at pp*nn
SUCCEED at pp*pn
SUCCEED at pp*nn
SUCCEED at pn*pp
SUCCEED at pn*pn
SUCCEED at pn*pn2
SUCCEED at pn*nn
SUCCEED at nn*pp
SUCCEED at nn*pn
SUCCEED at nn*nn


Unoptimized mul-interval: (5231.8421225)
Optimized mul-interval: (2526.5896437)
```

So as expected, about twice as fast!

3.14 Exercise 2.12

3.14.1 Question

After debugging her program, Alyssa shows it to a potential user, who complains that her program solves the wrong problem. He wants a program that can deal with numbers represented as a center value and an additive tolerance; for example, he wants to work with intervals such as 3.5 ± 0.15 rather than [3.35, 3.65]. Alyssa returns to her desk and fixes this problem by supplying an alternate constructor and alternate selectors:

```
(define (make-center-width c w)
  (make-interval (- c w) (+ c w)))
(define (center i)
  (/ (+ (lower-bound i) (upper-bound i)) 2))
(define (width i)
  (/ (- (upper-bound i) (lower-bound i)) 2))
```

Unfortunately, most of Alyssa's users are engineers. Real engineering situations usually involve measurements with only a small uncertainty, measured as the ratio of the width of the interval to the midpoint of the interval. Engineers usually specify percentage tolerances on the parameters of devices, as in the resistor specifications given earlier.

Define a constructor `make-center-percent` that takes a center and a percentage tolerance and produces the desired interval. You must also define a selector `percent` that produces the percentage tolerance for a given interval. The `center` selector is the same as the one shown above.

3.14.2 Answer

```
(define (reciprocal x)
  (/ 1 x))
```

```
<<reciprocal>>
<<make-interval>>
<<sub-interval>>
<<interval-center-width>>

(define (make-center-percent c pt)
  (let ((pp (* c
               (* pt 0.01))))
    (make-interval (- c pp) (+ c pp))))
(define (percent I)
  (* 100.0
     (/ (width I)
        (center I))))
```

```
<<interval-percent>>
(load "mattcheck2.scm")
  (define (roughly-eq? a b)
    ;; error size varies with magnitude of fp
    ;; so dx must vary too.
    (define dx (* a 0.000001))
    (and (> a (- b dx))
         (< a (+ b dx))))
(define (interval-roughly-eq? I J)
  (and (roughly-eq? (upper-bound I) (upper-bound J))
       (roughly-eq? (lower-bound I) (lower-bound J))))

(let* ((i1 (make-interval 105.0 95.0))
       (i2 (make-center-width 100.0 5))
       (i3 (make-center-percent 100.0 5))
       (i1a (upper-bound i1)))
  (mattcheck "make-center-width"
             i1
             i2)
  (mattcheck "make-center-percent"
             i1
             i3)
  (mattcheck "percent"
             (percent i1)
             (percent i3)
             5.0)
  (mattcheck+ "make-center-percent is consistent"
              (list i1 i3)
              #:eql? interval-roughly-eq?))
```

```
SUCCEED at make-center-width
SUCCEED at make-center-percent
SUCCEED at percent
SUCCEED at make-center-percent is consistent
```

3.15 Exercise 2.13 optional

3.15.1 Question

> Show that under the assumption of small percentage tolerances there is a simple formula for the approximate percentage tolerance of the product of two intervals in terms of the tolerances of the factors. You may simplify the problem by assuming that all numbers are positive.

3.15.2 Answer

I should've written this function a while ago.

```
(use-modules (ice-9 format))
(define (stringit . args)
  (string-append
   (apply string-append
    (map (lambda(x)
          (format #f "~:a " x))
         args))))
(define (echo . args)
  (format #t "~&~:a~%" (apply stringit args)))
```

Now, let's examine how interval percents relate to each other.

```
<<echo>>
<<interval-percent>>
<<mul-interval-opt>>

(let* ((i1 (make-center-width 100 5))
       (i2 (make-center-width 200 5))
       (M12 (mul-interval-opt i1 i2)))
  (echo "intervals 1 and 2:" i1 i2)
  (echo "width of 1 and 2:" (width i1) (width i2))
  (echo "percent of 1 and 2:" (percent i1) (percent i2))
  (echo "i1*i2 = " M12)
  (echo "width M12:" (width M12))
  (echo "percent M12:" (percent M12)))
```

```
intervals 1 and 2: (105 . 95) (205 . 195)
width of 1 and 2: 5 5
percent of 1 and 2: 5.0 2.5
i1*i2 =  (21525 . 18525)
width M12 1500
percent M12 7.490636704119851
```

Perhaps $\text{percent}(A \times B) = \text{percent}(A) + \text{percent}(B)$?

```
<<echo>>
<<interval-percent>>
<<mul-interval-opt>>

(let* ((i1 (make-center-percent 40 0.1))
       (i2 (make-center-percent 200 0.4))
       (M12 (mul-interval-opt i1 i2)))
  (echo "percent of 1 and 2:" (percent i1) (percent i2))
  (echo "percent M12:" (percent M12)))
```

```
percent of 1 and 2: 0.09999999999999788 0.40000000000000563
percent M12: 0.499998000008
```

3.16 Exercise 2.14

3.16.1 Question

After considerable work, Alyssa P. Hacker delivers her finished system. Several years later, after she has forgotten all about it, she gets a frenzied call from an irate user, Lem E. Tweakit. It seems that Lem has noticed that the formula for parallel resistors can be written in two algebraically equivalent ways:

$$\frac{R_1 R_2}{R_1 + R_2}$$

and

$$\frac{1}{\frac{1}{R_1} + \frac{1}{R_2}}$$

He has written the following two programs, each of which computes the parallel-resistors formula differently:

```
(define (par1 r1 r2)
  (div-interval (mul-interval r1 r2)
                (add-interval r1 r2)))

(define (par2 r1 r2)
  (let ((one (make-interval 1 1)))
    (div-interval
     one (add-interval (div-interval one r1)
                       (div-interval one r2)))))
```

Lem complains that Alyssa's program gives different answers for the two ways of computing. This is a serious complaint.

Demonstrate that Lem is right. Investigate the behavior of the system on a variety of arithmetic expressions. Make some intervals A and B, and use them in computing the expressions A/A and A/B. You will get the most insight by using intervals whose width is a small percentage of the center value. Examine the results of the computation in center-percent form (see Exercise 2.12).

3.16.2 Answer

```
<<echo>>
<<interval-percent>>
<<mul-interval-opt>>
<<par-resistors>>
```

```
5
6  (let* ((A (make-center-percent 10 1))
7         (B (make-center-percent 10 0.01))
8         (p1 (par1 A B))
9         (p2 (par2 A B)))
10   (echo "A,B:" A B)
11   (echo "par1(A,B):" p1)
12   (echo "par2(A,B):" p2)
13   (echo "percent(par1):" (percent p1))
14   (echo "percent(par2):" (percent p2))
15   (echo "center(par1):" (center p1))
16   (echo "center(par2):" (center p2)))
17
18 (echo "So these two have inconsistent effects on the width.")
19 (newline)
20 (echo "It should also be noted that floating-point errors accumulate.")
21 (echo "Take a look at the error on these (correct answer is 1)")
22
23 (let* ((A (make-center-percent 10 1))
24        (p1 (div-interval
25             (div-interval
26              (mul-interval A A)
27              A)
28             A))
29        (p2 (div-interval
30             (div-interval
31              (div-interval
32               (mul-interval
33                (mul-interval A A)
34                A)
35               A)
36              A)
37             A)))
38   (echo "p1:" (center p1))
39   (echo "p2:" (center p2)))
```

```
A,B: (10.1 . 9.9) (10.001 . 9.999)
par1(A,B): (5.076139504497713 . 4.924635590269141)
par2(A,B): (5.025128103079449 . 4.974626865671642)
percent(par1): 1.5149217214958663
percent(par2): 0.5050247487625606
center(par1): 5.000387547383427
center(par2): 4.999877484375546
So these two have inconsistent effects on the width.

It should also be noted that floating-point errors accumulate.
Take a look at the error on these (correct answer is 1)
p1: 1.0008001600240033
p2: 1.0018006601460259
```

3.17 Exercise 2.15

3.17.1 Question

> Eva Lu Ator, another user, has also noticed the different intervals computed by different but algebraically equivalent expressions. She says that a formula to compute with intervals using Alyssa's system will produce tighter error bounds if it can be written in such a form that no variable that represents an uncertain number is repeated. Thus, she says, `par2` is a "better" program for parallel resistances than `par1`. Is she right? Why?

3.17.2 Answer

If I am correct in understanding that "uncertain number" means "a number with an error tolerance", than `par2` *is* better – it only uses two instances of variables with error tolerance, while `par1` uses four.

It should be noted that this system does not directly translate to algebraic expressions. For example, take these expressions:

$$A + A = 2A$$

$$A - A = 0$$

$$A/A = 1$$

Note that these do not hold up in practice with uncertain numbers:

```
<<echo>>
<<interval-percent>>
<<mul-interval-opt>>

(define A (make-center-percent 10 1))
(echo "A+A = 2A !=" (add-interval A A))
(echo "A-A = 0 !=" (sub-interval A A))
(echo "A/A = 1 !=" (div-interval A A))
```

```
A+A = 2A != (20.2 . 19.8)
A-A = 0 != (0.1999999999999993 . -0.1999999999999993)
A/A = 1 != (1.02020202020202 . 0.9801980198019803)
```

3.18 Exercise 2.16 optional

3.18.1 Question

> Explain, in general, why equivalent algebraic expressions may lead to different answers. Can you devise an interval-arithmetic package that does not have this shortcoming, or is this task impossible? (Warning: This problem is very difficult.)

3.18.2 Answer

It is *indeed* very difficult, because from what I'm seeing online, no interval system without these issues exists. To avoid these issues, interval mathematics would need to satisfy the conditions for a **field** – and failing that, needs to only use each variable once, which becomes impossible as soon as you encounter an expression as simple as x^2.

GitHub user "`diiq`" has an incredible analysis of this, which can be found here: `https://gist.github.com/diiq/1f39df0e54b2137bb07e7e04b11cb075`

3.19 2.2: Hierarchical Data and the Closure Property

`cons` pairs can be used to construct more complex data-types.



Figure 2.3: Two ways to combine 1, 2, 3, and 4 using pairs.

The ability to combine things using an operation, then combine those results using the same operation, can be called the **closure property**. `cons` can create pairs whose elements are pairs, which satisfies the closure property. This property enables you to create hierarchical structures. We've already regularly used the closure property in creating procedures composed of other procedures.

> **Definitions of "closure"**
>
> The use of the word "closure" here comes from abstract algebra, where a set of elements is said to be closed under an operation if applying the operation to elements in the set produces an element that is again an element of the set. The Lisp community also (unfortunately) uses the word "closure" to describe a totally unrelated concept: A closure is an implementation technique for representing procedures with free variables. We do not use the word "closure" in this second sense in this book.

3.20 2.2.1: Representing Sequences

sequence An ordered collection of data objects.

list A sequence of `cons` pairs.

```
(cons 1
      (cons 2
            (cons 3
                  (cons 4 nil))))
(list 1 2 3 4)
;; both evaluate to '(1 2 3 4)
```

An aside: many parts of this book have covered ways to solve problems by splitting problems into simple recursive solutions. I may be getting ahead of myself, but I wanted to note how the `cons` pair system goes hand-in-hand with this. For example, when going over a list `l` with function `f`:

```
(define (map f l)
  (if (null? l)
      #nil
      (cons (f (car l))
            (map f (cdr l)))))

(map (λ(x)(* x 2))
     (list 1 2 3 4))
```

```
| 2 | 4 | 6 | 8 |
```

3.21 Exercise 2.17

3.21.1 Question

> Define a procedure `last-pair` that returns the list that contains only the last element of a given (nonempty) list:

```
(last-pair (list 23 72 149 34))
;; (34)
```

3.21.2 Answer

```
(define (last-pair l)
  (let ((a (car l))
        (d (cdr l)))
    (if (= 1 (length d))
        d
```

```
        (last-pair d))))
```

```
<<last-pair>>
(last-pair (list 23 72 149 34))
```

```
| 34 |
```

3.22 Exercise 2.18

3.22.1 Question

Define a procedure `reverse` that takes a list as argument and returns a list of the same elements in reverse order:

```
(reverse (list 1 4 9 16 25))
;; (25 16 9 4 1)
```

3.22.2 Answer

```
(define (reverse l)
  (define len (length l))
  (define (iter i result)
    (if (< (1- len) i)
        result
        (iter (+ i 1)
              (cons (list-ref l i)
                    result))))
  (iter 0 '()))
```

```
<<reverse>>
(reverse (list 23 72 149 34))
```

```
34  149  72  23
```

3.23 Exercise 2.19

3.23.1 Question

Consider the change-counting program of 1.2.2. It would be nice to be able to easily change the currency used by the program, so that we could compute the number of ways to change a British pound, for example. As the program is written, the knowledge of the currency is distributed partly into the procedure `first-denomination` and partly into the procedure `count-change` (which knows that there are five

kinds of U.S. coins). It would be nicer to be able to supply a list of coins to be used for making change.

We want to rewrite the procedure `cc` so that its second argument is a list of the values of the coins to use rather than an integer specifying which coins to use. We could then have lists that defined each kind of currency:

```
(define us-coins (list 50 25 10 5 1))
(define uk-coins (list 100 50 20 10 5 2 1 0.5))
```

We could then call `cc` as follows:

```
(cc 100 us-coins)
; 292
```

To do this will require changing the program `cc` somewhat. It will still have the same form, but it will access its second argument differently, as follows:

```
(define (cc amount coin-values)
  (cond ((= amount 0) 1)
        ((or (< amount 0) (no-more? coin-values)) 0)
        (else
         (+ (cc amount
                (except-first-denomination
                 coin-values))
            (cc (- amount
                   (first-denomination
                    coin-values))
                coin-values)))))
```

Define the procedures `first-denomination`, `except-first-denomination`, and `no-more?` in terms of primitive operations on list structures. Does the order of the list `coin-values` affect the answer produced by `cc`? Why or why not?

3.23.2 Answer

```
<<cc-lists>>

(define us-coins
  (list 50 25 10 5 1))
(define uk-coins
  (list 100 50 20 10 5 2 1 0.5))

(define first-denomination car)
```

```
(define except-first-denomination cdr)
(define no-more? null?)
```

```
<<Ex-2-19>>
(list
 (cc 100 us-coins)
 (cc 100 (reverse us-coins))
 (cc 100 (list 50 10 25 5 1)))
```

| 292 | 292 | 292 |

Apparently, the order of the list does *not* affect the value. However, it does effect the execution time, with small-to-large coin lists taking more time than large-to-small.

```
decreasing values: (357503.80704)
increasing values: (823460.64376)
```

3.24 Exercise 2.20

3.24.1 Question

> Use `[dotted-pair]` notation to write a procedure `same-parity` that takes one or more integers and returns a list of all the arguments that have the same even-odd parity as the first argument. For example,

```
(same-parity 1 2 3 4 5 6 7)
; (1 3 5 7)
(same-parity 2 3 4 5 6 7)
; (2 4 6)
```

3.24.2 Answer

```
(define (same-parity . rest)
  (define same?
    (if (even? (car rest))
        even?
        odd?))
  (define (iter l results)
    (if (null? l)
         results
        (let ((a (car l)))
          (iter (cdr l)
                (if (same? a)
                    (cons a results)
                    results)))))
  (iter (reverse rest) '()))
```

```
15
16  ;; Attempting to remove the reversing
17  (define (same-parity2 . args)
18    (define first (car args))
19    (define same?
20      (if (even? first)
21          even?
22          odd?))
23    (define (iter l results)
24      (if (null? l)
25          results
26          (let ((a (car l))
27                (d (cdr l)))
28            (if (same? a)
29                (iter d (append results
30                                (cons a #nil)))
31                (iter d results)))))
32    (iter (cdr args) (cons first #nil)))
```

```
1  <<same-parity>>
2  (list
3   (same-parity 1 2 3 4 5 6 7)
4   (same-parity2 2 3 4 5 6 7))
```

```
| 1 | 3 | 5 | 7 |
| 2 | 4 | 6 |   |

same-parity: (10003.483436)
same-parity2: (56007.042334)
```

Once again, my attempts to optimize are a complete failure. I'm guessing that the act of traversing the whole list in the call to `append` is the problem.

3.25 Exercise 2.21

3.25.1 Question

> The procedure `square-list` takes a list of numbers as argument and returns a list of the squares of those numbers.

```
1  (square-list (list 1 2 3 4))
2  ;; (1 4 9 16)
```

> Here are two different definitions of `square-list`. Complete both of them by filling in the missing expressions:

```
(define (square-list items)
  (if (null? items)
      nil
      (cons <??> <??>)))
(define (square-list items)
  (map <??> <??>))
```

3.25.2 Answer

```
<<square>>
(define (square-list-manual items)
  (if (null? items)
      #nil
      (cons (square (car items))
            (square-list-manual (cdr items)))))
(define (square-list items)
  (map square items))
```

```
<<square-list>>
(let ((l (list 2 3 4 5 6)))
  (list l
        (square-list-manual l)
        (square-list l)))
```

```
| 2 | 3 |  4 |  5 |  6 |
| 4 | 9 | 16 | 25 | 36 |
| 4 | 9 | 16 | 25 | 36 |
```

3.26 Exercise 2.22

3.26.1 Questions

Louis Reasoner tries to rewrite the first `square-list` procedure of Exercise 2.21 so that it evolves an iterative process:

```
(define (square-list items)
  (define (iter things answer)
    (if (null? things)
        answer
        (iter (cdr things)
              (cons (square (car things))
                    answer))))
  (iter items nil))
```

Unfortunately, defining `square-list` this way produces the answer list in the reverse order of the one desired. Why?

Louis then tries to fix his bug by interchanging the arguments to `cons`:

```
(define (square-list items)
  (define (iter things answer)
    (if (null? things)
        answer
        (iter (cdr things)
              (cons answer
                    (square (car things))))))
  (iter items nil))
```

This doesn't work either. Explain.

3.26.2 Answer

I'm positive I've made this exact mistake before, though this is likely not recorded.

The first form of `square-list` produces a correct list in reverse order:

```
(square-list (iota 6))
(25 16 9 4 1 0)
```

This is because he is prepending to the list every iteration.

While the second produces a broken list, which is literally backwards:

```
(square-list (iota 6))
((((((#nil . 0) . 1) . 4) . 9) . 16) . 25)
;; Equivalent to:
(cons (cons (cons (cons (cons (cons #nil
                                    0)
                              1)
                        4)
                  9)
            16)
      25)
```

Since Lisp was designed with the `cons pair` structure of list-building, it needed to define a "correct" direction for the pairs to go. Since the Western world thinks left-to-right, they made it so that the left (first) cell is for content, and the right is for the pointer to the next pair. However, this means that you can't append to a list without first traveling its length and changing the `nil` marking the end to a pointer to your new pair. Since that is a lot of list traveling, it makes more sense to `cons` your list together in reverse and then calling `reverse` only once at the end of the procedure.

3.27 Exercise 2.23

3.27.1 Question

> The procedure `for-each` is similar to `map`. It takes as arguments a procedure and a list of elements. However, rather than forming a list of the results, `for-each` just applies the procedure to each of the elements in turn, from left to right. The values returned by applying the procedure to the elements are not used at all—`for-each` is used with procedures that perform an action, such as printing. For example,

```
(for-each (lambda (x)
            (newline)
            (display x))
          (list 57 321 88))
;; 57
;; 321
;; 88
```

> The value returned by the call to `for-each` (not illustrated above) can be something arbitrary, such as true. Give an implementation of `for-each`.

3.27.2 Answer

```
(define (for-each-mine proc items)
  (define (iter l)
    (if (null? l)
        #t
        (begin (proc (car l))
               (iter (cdr l)))))
  (iter items))
```

```
<<for-each-mine>>
(for-each-mine (λ(x)(display x)(display " ")) (list "all" "your" "base"))
(for-each (λ(x)(display x)(display " ")) (list "are" "belong" "to" "us"))
```

```
all your base are belong to us
```

3.28 Exercise 2.24

3.28.1 Text Definitions

```
(define (count-leaves x)
  (cond ((null? x) 0)
        ((not (pair? x)) 1)
```

```
        (else (+ (count-leaves (car x))
                 (count-leaves (cdr x))))))
```

3.28.2 Question

Suppose we evaluate the expression `(list 1 (list 2 (list 3 4)))`. Give the result printed by the interpreter, the corresponding box-and-pointer structure, and the interpretation of this as a tree (as in Figure 2.6).

3.28.3 Answer

This is sort of a trick question – on first reading, I read it like a series of `cons` statements. Looking again, though, I can see that the correct formulation is as follows:

```
<<echo>>
(let ((l1 (list 1 (list 2 (list 3 4))))
      (l2 (cons 1
                (cons
                 (cons 2
                       (cons
                        (cons 3
                              (cons 4
                                    #nil))
                        #nil))
                 #nil))))
  (echo "textbook version:" l1)
  (echo "cons'd version:" l2))
```

```
textbook version: (1 (2 (3 4)))
cons'd version: (1 (2 (3 4)))
```

Dot and box version:



Tree version:



3.29 Exercise 2.25

3.29.1 Question

Give combinations of `car`s and `cdr`s that will pick 7 from each of the following lists:

```
(1 3 (5 7) 9)
((7))
(1 (2 (3 (4 (5 (6 7))))))
```

```
<<echo>>
(let ((l1 (list 1 3 (list 5 7) 9))
      (l2 (list (list 7)))
      (l3 (list 1 (list 2 (list 3 (list 4 (list 5 (list 6 7))))))))
  (echo (car (cdaddr l1))
        (caar l2)
        (cadadr (cadadr (cadadr l3)))))
```

```
7 7 7
```

3.30 Exercise 2.26

3.30.1 Question

Suppose we define x and y to be two lists:

```
(define x (list 1 2 3))
(define y (list 4 5 6))
```

What result is printed by the interpreter in response to evaluating each of the following expressions:

```
(append x y)
(cons x y)
(list x y)
```

3.30.2 Answer

```
<<echo>>
(let* ((x (list 1 2 3))
       (y (list 4 5 6))
       (e1 (append x y))
       (e2 (cons x y))
       (e3 (list x y)))
  (echo "(append x y):" e1)
  (echo "(cons x y):" e2)
  (echo "(list x y):" e3))
```

```
(append x y): (1 2 3 4 5 6)
(cons x y): ((1 2 3) 4 5 6)
(list x y): ((1 2 3) (4 5 6))
```

3.31 Exercise 2.27

3.31.1 Question

Modify your `reverse` procedure of Exercise 2.18 to produce a `deep-reverse` procedure that takes a list as argument and returns as its value the list with its elements reversed and with all sublists deep-reversed as well. For example,

```
(define x (list (list 1 2) (list 3 4)))
x
;; ((1 2) (3 4))
(reverse x)
;; ((3 4) (1 2))
(deep-reverse x)
;; ((4 3) (2 1))
```

3.31.2 Answer

```
(define (deep-reverse l)
  (define len (length l))
  (define (iter i result)
    (if (< (1- len) i)
        result
        (iter (+ i 1)
              (cons (let ((here (list-ref l i)))
                      (if (pair? here)
                          (deep-reverse here)
                          here))
                    result))))
  (iter 0 '()))
```

```
<<deep-reverse>>
(deep-reverse (list (list 1 2) (list 3 4)))
```

```
((4 3) (2 1))
```

3.32 Exercise 2.28

3.32.1 Question

Write a procedure `fringe` that takes as argument a tree (represented as a list) and returns a list whose elements are all the leaves of the tree arranged in left-to-right order. For example,

```
(define x (list (list 1 2) (list 3 4)))
(fringe x)
;; (1 2 3 4)
```

```
(fringe (list x x))
;; (1 2 3 4 1 2 3 4)
```

3.32.2 Answer

```
(define (fringe l)
  (if (null? l)
      #nil
      (let ((a (car l))
            (d (cdr l)))
        (append (if (pair? a)
                    (fringe a)
                    (list a))
                (fringe d)))))
```

```
<<fringe>>
(fringe (list (list 1 (list 2 3)) (list 4 5)))
```

```
(1 2 3 4 5)
```

3.33 Exercise 2.29: Binary Mobiles

3.33.1 Text Definitions

```
(define (make-mobile left right)
  (list left right))

(define (make-branch length structure)
  (list length structure))
```

3.33.2 Question A: Selectors

> Write the corresponding selectors `left-branch` and `right-branch`, which return the branches of a mobile, and `branch-length` and `br⌋ anch-structure`, which return the components of a branch.

3.33.3 Answer A

```
<<mobile-constructors-list>>
(define (left-branch mobile)
  (car mobile))
(define (right-branch mobile)
  (cadr mobile))
(define (branch-length branch)
  (car branch))
```

```
(define (branch-structure branch)
  (cadr branch))
```

3.33.4 Question B: total-weight

Using your selectors, define a procedure `total-weight` that returns the total weight of a mobile.

3.33.5 Answer B

```
(define (total-weight mobile)
  (let ((leftS (branch-structure (left-branch mobile)))
        (rightS (branch-structure (right-branch mobile))))
    (+ (if (number? leftS)
           leftS
           (total-weight leftS))
       (if (number? rightS)
           rightS
           (total-weight rightS)))))
```

```
<<mobile-selectors-list>>
<<mobile-total-weight>>

(let ((M1 (make-mobile
           (make-branch 5 5)
           (make-branch 1
                        (make-mobile (make-branch 2 2)
                                     (make-branch 2 3)))))
      (M2 (make-mobile
           (make-branch 2
                        (make-mobile
                         (make-branch 2 2)
                         (make-branch 2 2)))
           (make-branch 2
                        (make-mobile
                         (make-branch 2 2)
                         (make-branch 2 2))))))
  (list (total-weight M1)
        (total-weight M2)))
```

```
| 10 | 8 |
```

3.33.6 Question C: Balancing

A mobile is said to be **balanced** if the torque applied by its top-left branch is equal to that applied by its top-right branch (that is, if the length of the left rod multiplied by the weight hanging from that rod is equal to the corresponding

product for the right side) and if each of the submobiles hanging off its branches is balanced. Design a predicate that tests whether a binary mobile is balanced.

3.33.7 Answer C

I can imagine a ton of ways I could shoot myself in the foot by starting with optimization, so let's just try to nail down exactly what needs to happen.

```
(define (total-torque branch)
  (let ((len (branch-length branch))
        (struct (branch-structure branch)))
    (* len
       (if (number? struct)
           struct
           (let ((lbs (branch-structure (left-branch struct)))
                 (rbs (branch-structure (right-branch struct))))
             (+ (total-weight lbs)
                (total-weight rbs)))))))
(define (balanced? mobile)
  (let* ((l (left-branch mobile))
         (ls (branch-structure l))
         (r (right-branch mobile))
         (rs (branch-structure r))
         (l-balanced (if (number? ls)
                         #t
                         (balanced? ls)))
         (r-balanced (if (number? rs)
                         #t ;; Fixed: accidentally used ls again.
                         (balanced? rs))))
    (if (and l-balanced r-balanced)
        (= (total-torque l)
           (total-torque r))
        #f)))
```

I'll also need a modified `total-weight` that can notice when its argument is a non-mobile and just return the value.

```
(define (total-weight mobile)
  (if (number? mobile)
      mobile ;; this is a weight, just return it
      (let ((leftS (branch-structure (left-branch mobile)))
            (rightS (branch-structure (right-branch mobile))))
        (+ (if (number? leftS)
               leftS
               (total-weight leftS))
           (if (number? rightS)
               rightS
               (total-weight rightS))))))
```

```
<<echo>>
<<mobile-selectors-list>>
<<mobile-total-weight-2>>
<<mobile-balanced-dumb>>
(define M1 ;; all segments unbalanced
  (make-mobile
   (make-branch 5 5) ;; torque 25 = 5*5
   (make-branch 1 ;; torque 5 = (2 + 3) * 1
                (make-mobile (make-branch 2 2)
                             (make-branch 2 3)))))
(define M2 ;; all segments balanced as they are duplicates
  (make-mobile
   (make-branch 2 ;; torque 8 = 2*(2+2)
                (make-mobile
                 (make-branch 2 2)
                 (make-branch 2 2)))
   (make-branch 2
                (make-mobile
                 (make-branch 2 2)
                 (make-branch 2 2)))))
(define M3 ;; equal torque, but one segment is unbalanced.
  (make-mobile
   (make-branch 2 ;; torque 12 = 2*(4+2)
                (make-mobile
                 (make-branch 2 4) ;; torque 8 = 2*4
                 (make-branch 4 2))) ;torque 8 = 4*2
   (make-branch 3 ;; torque 12 = 3*(1+3)
                (make-mobile
                 (make-branch 1 1) ;; torque 1
                 (make-branch 1 3))))) ;; torque 3
(define M4 ;; equal torque
  (make-mobile
   (make-branch 2 ;; torque 12 = 2*(4+2)
                (make-mobile
                 (make-branch 2 4) ;; torque 8 = 2*4
                 (make-branch 4 2))) ;torque 8 = 4*2
   (make-branch 3 ;; torque 12 = 3*(1+3)
                (make-mobile
                 (make-branch 3 1) ;; torque 3
                 (make-branch 1 3))))) ;; torque 3
(define (isbalanced? Name Status)
  (define Success "is balanced!")
  (define Failure "is not balanced!")
  (if (eq? Status #t)
      (echo Name Success)
      (echo Name Failure)))
(isbalanced? "M1" (balanced? M1))
(isbalanced? "M2" (balanced? M2))
```

```
(isbalanced? "M3" (balanced? M3))
(isbalanced? "M4" (balanced? M4))
```

```
M1 is not balanced!
M2 is balanced!
M3 is not balanced!
M4 is balanced!
```

This one took quite some fiddling. First I struggled to figure out exactly how I should juggle of torque, weight, and balance. For example, a mobile is balanced if the torques of its branches are equal, and if every submobile is also balanced, with torque being defined as length×weight. Note that it's the *weight*, not its submobile's *torque*.

TODO: I'd like to come back and make an optimized version that doesn't have to crawl the tree multiple times. Maybe getting torque/weight/balanced status at the same time?

3.33.8 Question D: Implementation shakeup

> Suppose we change the representation of mobiles so that the constructors are

```
(define (make-mobile left right)
  (cons left right))

(define (make-branch length structure)
  (cons length structure))
```

> How much do you need to change your programs to convert to the new representation?

3.33.9 Answer D

Ideally I should only need to change the selectors, like this:

```
<<mobile-constructors-cons>>
(define (left-branch mobile)
  (car mobile))
(define (right-branch mobile)
  (cdr mobile))
(define (branch-length branch)
  (car branch))
(define (branch-structure branch)
  (cdr branch))
```

Now, if I run the same code, I should get the same result:

```
<<echo>>
<<mobile-selectors-cons>>
<<mobile-total-weight-2>>
<<mobile-balanced-dumb>>
(define M1 ;; all segments unbalanced
  (make-mobile
   (make-branch 5 5) ;; torque 25 = 5*5
   (make-branch 1 ;; torque 5 = (2 + 3) * 1
                (make-mobile (make-branch 2 2)
                             (make-branch 2 3)))))
(define M2 ;; all segments balanced as they are duplicates
  (make-mobile
   (make-branch 2 ;; torque 8 = 2*(2+2)
                (make-mobile
                 (make-branch 2 2)
                 (make-branch 2 2)))
   (make-branch 2
                (make-mobile
                 (make-branch 2 2)
                 (make-branch 2 2)))))
(define M3 ;; equal torque, but one segment is unbalanced.
  (make-mobile
   (make-branch 2 ;; torque 12 = 2*(4+2)
                (make-mobile
                 (make-branch 2 4) ;; torque 8 = 2*4
                 (make-branch 4 2))) ;torque 8 = 4*2
   (make-branch 3 ;; torque 12 = 3*(1+3)
                (make-mobile
                 (make-branch 1 1) ;; torque 1
                 (make-branch 1 3))))) ;; torque 3
(define M4 ;; equal torque
  (make-mobile
   (make-branch 2 ;; torque 12 = 2*(4+2)
                (make-mobile
                 (make-branch 2 4) ;; torque 8 = 2*4
                 (make-branch 4 2))) ;torque 8 = 4*2
   (make-branch 3 ;; torque 12 = 3*(1+3)
                (make-mobile
                 (make-branch 3 1) ;; torque 3
                 (make-branch 1 3))))) ;; torque 3
(define (isbalanced? Name Status)
  (define Success "is balanced!")
  (define Failure "is not balanced!")
  (if (eq? Status #t)
      (echo Name Success)
      (echo Name Failure)))
(isbalanced? "M1" (balanced? M1))
(isbalanced? "M2" (balanced? M2))
```

```
(isbalanced? "M3" (balanced? M3))
(isbalanced? "M4" (balanced? M4))
```

```
M1 is not balanced!
M2 is balanced!
M3 is not balanced!
M4 is balanced!
```

3.34 Exercise 2.30

3.34.1 Question

Define a procedure `square-tree` analogous to the `square-@/list` procedure of 3.25. That is, `square-tree` should behave as follows:

```
(square-tree
 (list 1
       (list 2 (list 3 4) 5)
       (list 6 7)))
;; (1 (4 (9 16) 25) (36 49))
```

Define `square-tree` both directly (i.e., without using any higher-order procedures) and also by using `map` and recursion.

3.34.2 Answer

```
<<square>>
(define (square-tree-discrete tree)
  (cond ((null? tree) '())
        ((not (pair? tree)) (square tree))
        (else (cons (square-tree-discrete (car tree))
                    (square-tree-discrete (cdr tree))))))

(define (square-tree-map tree)
  (map (lambda (sub-tree)
         (if (pair? sub-tree)
             (square-tree-map sub-tree)
             (square sub-tree)))
       tree))
```

```
(load "mattcheck2.scm")
<<square-tree>>
(let ((testlist
       (list 1
             (list 2 (list 3 4) 5)
             (list 6 7)))
      (answer
```

```
        (list 1
              (list 4 (list 9 16) 25)
              (list 36 49))))
   (mattcheck "square-tree-discrete"
              (square-tree-discrete testlist)
              answer)
   (mattcheck "square-tree-map"
              (square-tree-map testlist)
              answer))
```

```
SUCCEED at square-tree-discrete
SUCCEED at square-tree-map
```

While writing that, I ran headfirst into a lesson I've had to repeatedly learn: default Guile functions end their lists with `'()` which does not match equality with lists ended with `#nil`.

```
(let ((parens-list (cons 1 (cons 2 (cons 3 '()))))
      (nil-list (cons 1 (cons 2 (cons 3 #nil)))))
  (display parens-list)(display " <== ends with '()")
  (newline)
  (display nil-list)(display " <== ends with #nil")
  (newline)
  (display "Are these two lists equal? > ")
  (display (equal? parens-list nil-list))
  (newline)
  (display "Does Guile consider #nil and '() equal? > ")
  (display (equal? #nil '()))
  (newline)
  (display "What about #nil and #f? > ")
  (display (equal? #nil #f)))
```

```
(1 2 3) <== ends with '()
(1 2 3) <== ends with #nil
Are these two lists equal? > #f
Does Guile consider #nil and '() equal? > #f
What about #nil and #f? > #f
```

3.35 Exercise 2.31

3.35.1 Question

> Abstract your answer to 3.34 to produce a procedure `tree-map` with the property that `square-tree` could be defined as

```
(define (square-tree tree) (tree-map square tree))
```

3.35.2 Answer

```
(define (tree-map f tree)
  (cond ((null? tree) '())
        ((not (pair? tree)) (f tree))
        (else (cons (tree-map f (car tree))
                    (tree-map f (cdr tree))))))
```

```
(load "mattcheck2.scm")
<<tree-map>>
<<square>>
(define (square-tree-tm tree)
  (tree-map square tree))
(let ((testlist
       (list 1
             (list 2 (list 3 4) 5)
             (list 6 7)))
      (answer
       (list 1
             (list 4 (list 9 16) 25)
             (list 36 49))))
  (mattcheck "square-tree-tm"
             (square-tree-tm testlist)
             answer))
```

```
SUCCEED at square-tree-tm
```

3.36 Exercise 2.32

3.36.1 Question

> We can represent a set as a list of distinct elements, and we can represent the set of all subsets of the set as a list of lists. For example, if the set is `(1 2 3)`, then the set of all subsets is `(() (3) (2) (2 3) (1) (1 3) (1 2) (1 2 3))`. Complete the following definition of a procedure that generates the set of subsets of a set and give a clear explanation of why it works:

```
(define (subsets s)
  (if (null? s)
      (list nil)
      (let ((rest (subsets (cdr s))))
        (append rest (map <??> rest)))))
```

3.36.2 Answer

```
(define (subsets s)
  (if (null? s)
      (list '())
      (let ((rest (subsets (cdr s))))
        (append rest (map (λ(x)
                            (cons (car s) x))
                          rest)))))
```

```
<<subsets>>
(load "mattcheck2.scm")
(let ((answer
       (list '()
             (list 3)
             (list 2) (list 2 3)
             (list 1) (list 1 3)
             (list 1 2) (list 1 2 3))))
  (mattcheck "subsets"
             (subsets (list 1 2 3))
             answer))
```

```
SUCCEED at subsets
```

Essentially, `subsets` is rotating through members of the list in a similar way that a counter incrementing rotates through all numbers in its base. For a list with items 1 to n, `subsets` makes a list with the last item, $[n]$, then lists $[n-1]$ and $[n-1, n]$, then lists $[n-2][n-2, n-1][n-2, n]$, then $[n-3][n-3, n-2][n-3, n-1][n-3, n]$ and so on.

I'd like to try adding some debugging statements to subsets and see if it might help clarify the operation.

```
<<echo>>
(define (doit f x)
  (f x) ;; this probably has a formal lambda calculus name
  x)    ;; but I don't know what it is
(define (echo-return x)
  (doit echo x))
(define (echo-y-return-x x . y)
  (apply echo y)
  x)
```

```
<<subsets>>
<<echo-return>>
(define (subsets-debug s)
  (echo "Enter with" s "{")
  (let ((a1
         (if (null? s)
             (echo-return (list '()))
```

```
              (let ((rest (subsets-debug (cdr s))))
                (echo "[ iter" (car s) "over" rest "]")
                (append rest (map (λ(x)
                                    (let ((y (cons (car s) x)))
                                      (format #t "~a " y)
                                      y))
                                  rest))))))
    (echo "} end" s)
    a1))
  (let ((answer
         (list '()
               (list 3)
               (list 2) (list 2 3)
               (list 1) (list 1 3)
               (list 1 2) (list 1 2 3))))
    (subsets-debug (list 1 2 3))
    answer)
```

```
Enter with (1 2 3) {
Enter with (2 3) {
Enter with (3) {
Enter with () {
(())
} end ()
[ iter 3 over (()) ]
(3)
} end (3)
[ iter 2 over (() (3)) ]
(2) (2 3)
} end (2 3)
[ iter 1 over (() (3) (2) (2 3)) ]
(1) (1 3) (1 2) (1 2 3)
} end (1 2 3)
```

3.37 2.2.3: Sequences as Conventional Interfaces

Abstractions are an important part of making code clearer and more easy to understand. One beneficial manner of abstraction is making available conventional interfaces for working with compound data, such as `filter` and `map`.

This allows for easily making "signal-flow" conceptions of processes:



3.38 Exercise 2.33: The flexibility of `accumulate`

3.38.1 Text Definitions

```
(define (accumulate op initial sequence)
  (if (null? sequence)
      initial
      (op (car sequence)
          (accumulate op
                      initial
                      (cdr sequence)))))
```

3.38.2 Question

Fill in the missing expressions to complete the following definitions of some basic list-manipulation operations as accumulations.

```
(define (map p sequence)
  (accumulate (lambda (x y) <??>) nil sequence))
(define (append seq1 seq2)
  (accumulate cons <??> <??>))
(define (length sequence)
  (accumulate <??> 0 sequence))
```

3.38.3 Answer

```
(define (map-acc p sequence)
  (accumulate (lambda (x y)
                (cons (p x) y))
              '() sequence))
(define (append-acc seq1 seq2)
```

```
    (accumulate cons seq2 seq1))
  (define (length-acc sequence)
    (accumulate (λ(x y)
                  (1+ y))
                0 sequence))
```

```
  (load "mattcheck2.scm")
  <<accumulate>>
  <<accumulate-forms>>
  <<square>>
  (let ((l (list 1 2 3 4)))
    (mattcheck "map"
               (map square l)
               (map-acc square l))
    (mattcheck "append"
               (append l l)
               (append-acc l l))
    (mattcheck "length"
               (length l)
               (length-acc l)))
```

```
SUCCEED at map
SUCCEED at append
SUCCEED at length
```

3.39 Exercise 2.34

3.39.1 Question

Evaluating a polynomial in x at a given value of x can be formulated as an accumulation. We evaluate the polynomial

$$a_n x^n + a_{n-1} x^{n-1} + \cdots + a_1 x + a_0$$

using a well-known algorithm called *Horner's rule*, which structures the computation as

$$(\ldots (a_n x + a_{n-1}) x + \cdots + a_1) x + a_0.$$

In other words, we start with a_n, multiply by x, add a_{n-1}, multiply by x, and so on, until we reach a_0.

Fill in the following template to produce a procedure that evaluates a polynomial using Horner's rule. Assume that the coefficients of the polynomial are arranged in a sequence, from a_0 through a_n.

```
(define (horner-eval x coefficient-sequence)
  (accumulate (lambda (this-coeff higher-terms) <??>)
              0
              coefficient-sequence))
```

For example, to compute $1 + 3x + 5x^3 + x^5$ at $x = 2$ you would evaluate

```
(horner-eval 2 (list 1 3 0 5 0 1))
```

3.39.2 Answer

```
(define (horner-eval x coefficient-sequence)
  (accumulate (lambda (this-coeff higher-terms)
                (+ this-coeff
                   (* higher-terms x)))
              0
              coefficient-sequence))
```

```
(load "mattcheck2.scm")
<<accumulate>>
<<horner-eval>>
(mattcheck "horner-eval"
           (horner-eval 2 (list 1 3 0 5 0 1))
           79)
```

```
SUCCEED at horner-eval
```

This one was very satisfying. It essentially "delays" the exponentiation, carrying it out per stage, by rewriting this:

$1 + 3 \times 2 + 5 \times 2^3 + 2^5$

Into this operation, left to right:

$$0 + 1 * 2 + 0 * 2 + 0 * 2 + 3 * 2 + 1$$

3.40 Exercise 2.35

3.40.1 Question

Redefine `count-leaves` from 2.2.2 as an accumulation:

3.40.2 Answer

```
(define (count-leaves-acc t)
  (accumulate (λ(i total)
                (+ i total))
              0 (map (λ(x)
                       (if (pair? x)
                           (count-leaves-acc x)
                           1))
                     t)))
```

```
(load "mattcheck2.scm")
<<accumulate>>
<<count-leaves>>
<<count-leaves-acc>>
(let ((l (list (list (list 1 2 3) 4) (list 5 (list 6 7) 8))))
  (mattcheck "count-leaves-acc"
             (count-leaves l)
             (count-leaves-acc l)))
```

```
SUCCEED at count-leaves-acc
```

3.41 Exercise 2.36: Accumulate across multiple lists

3.41.1 Question

> The procedure `accumulate-n` is similar to `accumulate` except that it takes as its third argument a sequence of sequences, which are all assumed to have the same number of elements. It applies the designated accumulation procedure to combine all the first elements of the sequences, all the second elements of the sequences, and so on, and returns a sequence of the results. For instance, if `s` is a sequence containing four sequences, `((1 2 3) (4 5 6) (7 8 9) (10 11 12))`, then the value of `(accumulate-n + 0 s)` should be the sequence `(22 26 30)`. Fill in the missing expressions in the following definition of `accumulate-n`:

```
(define (accumulate-n op init seqs)
  (if (null? (car seqs))
      nil
      (cons (accumulate op init <??>)
            (accumulate-n op init <??>))))
```

3.41.2 Answers

```
(define (accumulate-n op init seqs)
  (if (null? (car seqs))
      '()
```

```
      (cons (accumulate op init
                        (map car seqs))
            (accumulate-n op init
                          (map cdr seqs)))))
```

```
(load "mattcheck2.scm")
<<accumulate>>
<<accumulate-n>>
(let ((s (list (list 1 2 3) (list 4 5 6) (list 7 8 9) (list 10 11 12))))
  (mattcheck "accumulate-n"
             (accumulate-n + 0 s)
             (list 22 26 30)))
```

```
SUCCEED at accumulate-n
```

3.42 Exercise 2.37: Enter the matrices

3.42.1 Question

See full quote in book.

Suppose we represent vectors as lists, and matrices as lists of vectors. For example:

$$\begin{pmatrix} 1 & 2 & 3 & 4 \\ 4 & 5 & 6 & 6 \\ 6 & 7 & 8 & 9 \end{pmatrix}$$

```
'((1 2 3 4) (4 5 6 6) (6 7 8 9))
```

Define these operations:

```
(define (dot-product v w)
  (accumulate + 0 (map * v w)))
(define (matrix-*-vector m v)
  (map <??> m))
(define (transpose mat)
  (accumulate-n <??> <??> mat))
(define (matrix-*-matrix m n)
  (let ((cols (transpose n)))
    (map <??> m)))
```

3.42.2 Answer

```
(define (dot-product v w)
  (accumulate + 0 (map * v w)))
(define (matrix-*-vector m v)
  (map (λ(row)
         (dot-product v row))
       m))
(define (transpose mat)
```

```
  (accumulate-n cons '() mat))
(define (matrix-*-matrix m n)
  (let ((cols (transpose n)))
    (map (λ(row)
           (map (λ(col)
                  (dot-product row col))
                cols))
         m)))
```

```
(load "mattcheck2.scm")
<<echo>>
<<accumulate>>
<<accumulate-n>>
<<matrix-ops>>
(let* ((v1 (list 1 2 3 4))
       (v2 (list 5 6 7 8))
       (m1 (list v1 v2))
       (m1t (list (list 1 5)
                  (list 2 6)
                  (list 3 7)
                  (list 4 8)))
       (m1-*-m1t (list (list 30 70)
                       (list 70 174))))
  (mattcheck "transpose"
             (transpose m1)
             m1t)
  (mattcheck "dot-product"
             (dot-product v1 v2)
             70)
  (mattcheck "matrix-*-vector"
             (matrix-*-vector m1 v1)
             (list 30 70))
  (mattcheck "matrix-*-matrix"
             (matrix-*-matrix m1 m1t)
             m1-*-m1t))
```

```
SUCCEED at transpose
SUCCEED at dot-product
SUCCEED at matrix-*-vector
SUCCEED at matrix-*-matrix
```

I struggled a lot with what order things should be processed and applied in. Some of that came from never having done matrix multiplication before now. I would probably still not understand it if I hadn't found Herb Gross' lecture regarding matrix operations.

The other issue is nested map operations. I find it easy to read Python-ish code like this:

```
# Pseudocode
for row in m1:
    for column in m2t:
        for a,b in row,column:
            answer[i:j] += a*b
```

But much harder to comprehend Lisp code like this:

```
(map (λ(row)
       (map (λ(col)
              (dot-product row col))
            m1t))
     m1)
```

I must have a mental block in the way I think about map operations.

3.43 Exercise 2.38: fold-right

3.43.1 Question A

The `accumulate` procedure is also known as `fold-right`, because it combines the first element of the sequence with the result of combining all the elements to the right. There is also a `fold-left`, which is similar to `fold-right`, except that it combines elements working in the opposite direction:

```
(define (fold-left op initial sequence)
  (define (iter result rest)
    (if (null? rest)
        result
        (iter (op result (car rest))
              (cdr rest))))
  (iter initial sequence))
```

What are the values of

```
(fold-right / 1 (list 1 2 3))
(fold-left / 1 (list 1 2 3))
(fold-right list nil (list 1 2 3))
(fold-left list nil (list 1 2 3))
```

3.43.2 Answer A

```
<<accumulate>>
(define fold-right accumulate)
<<fold-left>>
<<echo>>
```

```
(echo "(fold-right / 1 (list 1 2 3)):" (fold-right / 1 (list 1 2 3)))
(echo "(fold-left / 1 (list 1 2 3)):" (fold-left / 1 (list 1 2 3)))
(echo "(fold-right list nil (list 1 2 3))" (fold-right list '() (list 1
↪  2 3)))
(echo "(fold-left list nil (list 1 2 3))" (fold-left list '() (list 1 2
↪  3)))
```

```
(fold-right / 1 (list 1 2 3)): 3/2
(fold-left / 1 (list 1 2 3)): 1/6
(fold-right list nil (list 1 2 3)) (1 (2 (3 ())))
(fold-left list nil (list 1 2 3)) ((() 1) 2) 3)
```

3.43.3 Question B

> Give a property that `op` should satisfy to guarantee that `fold-right` and `fold-left` will produce the same values for any sequence.

3.43.4 Answer B

They would need to be commutative, like addition and multiplication.

```
(load "mattcheck2.scm")
<<accumulate>>
(define fold-right accumulate)
<<fold-left>>
<<echo>>

(let* ((l (list 1 2 3 4 5))
       (lr (reverse l)))
  (mattcheck "commutative addition"
             (fold-right + 0 l)
             (fold-right + 0 lr)
             (fold-left + 0 l)
             (fold-left + 0 lr))
  (mattcheck "commutative multiplication"
             (fold-right * 1 l)
             (fold-right * 1 lr)
             (fold-left * 1 l)
             (fold-left * 1 lr)))
```

```
SUCCEED at commutative addition
SUCCEED at commutative multiplication
```

3.44 Exercise 2.39: reverse via fold

3.44.1 Question

> Complete the following definitions of `reverse` (Exercise 2.18) in terms of `fold-right` and `fold-left` from Exercise 2.38:

```
(define (reverse sequence)
  (fold-right (lambda (x y) <??>) nil sequence))
(define (reverse sequence)
  (fold-left (lambda (x y) <??>) nil sequence))
```

3.44.2 Answer

First, I'd like to start using the SRFI folds instead. This is my little "compatibility module".

```
;; SICP compat
(use-srfis '(1))
(define accumulate fold-right)
(define fold-left fold)
```

Now to the problem.

```
<<fold-compat>>
(define (reverse-fr sequence)
  (fold-right (lambda (x y)
                (append  y (list x)))
              '() sequence))
(define (reverse-fl sequence)
  (fold-left (lambda (x y)
               (append (list x) y))
             '() sequence))
```

```
(load "mattcheck2.scm")
<<reverse-fold>>
(let ((l (iota 5))
      (lr (reverse (iota 5))))
  (mattcheck "reverse-fr"
             (reverse-fr l)
             lr)
  (mattcheck "reverse-fl"
             (reverse-fl l)
             lr))
```

```
SUCCEED at reverse-fr
SUCCEED at reverse-fl
```

3.45 Exercise 2.40: `unique-pairs`

3.45.1 Text Definitions

```
(define (enumerate-interval low high)
  (if (> low high)
      '()
      (cons low
            (enumerate-interval
             (+ low 1)
             high))))
```

```
<<fold-compat>>
(define (flatmap proc seq)
  (accumulate append '() (map proc seq)))
```

```
<<flatmap>>
<<enumerate-interval>>
<<prime-smallest-divisor>>
(define (prime-sum? pair)
  (prime? (+ (car pair) (cadr pair))))
(define (make-pair-sum pair)
  (list (car pair) (cadr pair) (+ (car pair) (cadr pair))))
(define (prime-sum-pairs n)
  (map make-pair-sum
       (filter prime-sum? (flatmap
                           (lambda (i)
                             (map (lambda (j) (list i j))
                                  (enumerate-interval 1 (- i 1))))
                           (enumerate-interval 1 n)))))
```

3.45.2 Question

Define a procedure `unique-pairs` that, given an integer n, generates the sequence of pairs (i, j) with $1 \leq j < i \leq n$. Use `unique-pairs` to simplify the definition of `prime-sum-pairs` given above.

3.45.3 Answer

```
<<fold-compat>>
<<enumerate-interval>>
<<flatmap>>
(define (unique-pairs n)
  (flatmap
   (lambda (i)
     (map (lambda (j) (list i j))
          (enumerate-interval 1 (- i 1))))
   (enumerate-interval 1 n)))
```

```
(load "mattcheck2.scm")
<<unique-pairs>>
(let ((target 5)
      (answer (list (list 2 1)(list 3 1)(list 3 2)(list 4 1)
                    (list 4 2)(list 4 3)(list 5 1)(list 5 2)
                    (list 5 3)(list 5 4))))
  (mattcheck "unique-pairs"
             (unique-pairs target)
             answer))
<<echo>>
<<prime-sum-txt>>
(define (prime-sum-pairs-mine n)
  (map make-pair-sum
       (filter prime-sum?
               (unique-pairs n))))
(let ((answer (list (list 2 1 3) (list 3 2 5)
                    (list 4 1 5) (list 4 3 7)
                    (list 5 2 7)) ))
  (mattcheck "prime-sum-pairs with unique-pairs"
             (prime-sum-pairs-mine 5)
             answer))
```

```
SUCCEED at unique-pairs
SUCCEED at prime-sum-pairs with unique-pairs
```

3.46 Exercise 2.41: Ordered triples of positive integers

3.46.1 Question

> Write a procedure to find all ordered triples of distinct positive integers i, j, and k less than or equal to a given integer n that sum to a given integer s.

3.46.2 Answer

```
<<fold-compat>>
<<enumerate-interval>>
<<flatmap>>
(define (unique-triplets n)
  (flatmap (λ(i)
             (flatmap (λ(j)
                        (map (λ(k) (list i j k))
                             (enumerate-interval 1 (- j 1))))
                      (enumerate-interval 1 (- i 1))))
           (enumerate-interval 1 n)))
```

```
(define (triplets-sum n s)
  (filter (λ(triplet)
            (= s (fold + 0 triplet)))
          (unique-triplets n)))
```

```
(load "mattcheck2.scm")
<<unique-triplets>>
<<triplets-sum>>
(let ((answer-a
       (list (list 3 2 1) (list 4 2 1)
             (list 4 3 1) (list 4 3 2)
             (list 5 2 1) (list 5 3 1)
             (list 5 3 2) (list 5 4 1)
             (list 5 4 2) (list 5 4 3)))
      (answer-b
       (list (list 5 3 2) (list 5 4 1)
             (list 6 3 1) (list 7 2 1))))
  (mattcheck "unique-triplets"
             (unique-triplets 5)
             answer-a)
  (mattcheck "triplets-sum"
             (triplets-sum 7 10)
             answer-b))
```

```
SUCCEED at unique-triplets
SUCCEED at triplets-sum
```

3.47 Exercise 2.42: Eight Queens

3.47.1 Question

> The "eight-queens puzzle" asks how to place eight queens on a chessboard so that no queen is in check from any other.

```
<<flatmap>>
<<enumerate-interval>>
(define (queens board-size)
  (define (queen-cols k)
    (if (= k 0)
        (list empty-board)
        (filter
         (lambda (positions) (safe? k positions))
         (flatmap
          (lambda (rest-of-queens)
            (map (lambda (new-row)
                   (adjoin-position
                    new-row k rest-of-queens))
```

```
                (enumerate-interval 1 board-size)))
          (queen-cols (- k 1))))))
  (queen-cols board-size))
```

Complete the program by writing the following:

- representation for sets of board positions, including:
  - `adjoin-position`, which adjoins a new row-column position to a set of positions
  - `empty-board`, which represents an empty set of positions.
- `safe?`, which determines for a set of positions, whether the queen in the k^{th} column is safe with respect to the others. (Note that we need only check whether the new queen is safe—the other queens are already guaranteed safe with respect to each other.)

3.47.2 Answer

```
(define empty-board '())
(define (adjoin-position new-row column rest-of-queens)
  (cons (list new-row column) rest-of-queens))
(define (newer-position board)
  (car board))
(define (older-positions board)
  (cdr board))
(define (get-row position)
  (car position))
(define (get-column position)
  (cadr position))
(define (safe? k board)
  (define (same-row? a b)
    (= (get-row a) (get-row b)))
  (define (diagonal? a b)
    (let ((row-diff (abs (- (get-row a) (get-row b))))
          (col-diff (abs (- (get-column a) (get-column b)))))
      (= row-diff col-diff)))
  (let* ((new (newer-position board))
         (compare (older-positions board)))
    (and-map (λ(pos)
               (not (or (same-row? new pos)
                        (diagonal? new pos))))
             compare)))
```

```
<<queens-txt>>
<<queens-mine>>
(load "mattcheck2.scm")
(let ((q4 '((3 4) (1 3) (4 2) (2 1))
```

```
            ((2 4) (4 3) (1 2) (3 1))))
        (q11l 2680))
  (mattcheck "queens"
             (list (queens 4)
                   (length (queens 11)))
             (list q4 q11l)))
```

```
SUCCEED at queens
```

3.48 Exercise 2.43: Louis' **queens**

3.48.1 Question

> Louis Reasoner is having a terrible time doing Exercise 2.42. His `queens` procedure seems to work, but it runs extremely slowly. (Louis never does manage to wait long enough for it to solve even the 6×6 case.) When Louis asks Eva Lu Ator for help, she points out that he has interchanged the order of the nested mappings in the `flatmap`, writing it as

```
(flatmap
 (lambda (new-row)
   (map (lambda (rest-of-queens)
          (adjoin-position new-row k rest-of-queens))
        (queen-cols (- k 1))))
 (enumerate-interval 1 board-size))
```

> Explain why this interchange makes the program run slowly. Estimate how long it will take Louis's program to solve the eight-queens puzzle, assuming that the program in Exercise 2.42 solves the puzzle in time T.

3.48.2 Answer

The biggest contributor to the slowdown is likely the location of the `queen-cols` recursive call. This call being inside of the loop means it is being called k more times, all returning the same answer. But my math reasoning skills limit me from going further .Let's check with benchmarks.

```
<<queens-txt>>
<<queens-mine>>
(load "../mattbench.scm")

(define (queens-louis board-size)
  (define (queen-cols k)
    (if (= k 0)
        (list empty-board)
        (filter
```

```
          (lambda (positions) (safe? k positions))
          (flatmap
           (lambda (new-row)
             (map (lambda (rest-of-queens)
                    (adjoin-position new-row k rest-of-queens))
                  (queen-cols (- k 1))))
           (enumerate-interval 1 board-size)))))
    (queen-cols board-size))

  (define (test size)
    (format #t "~&normal queens x~a: ~a" size
            (mattbench (λ()(queens size)) 10000))
    (format #t "~&swapped queens x~a: ~a" size
            (mattbench (λ()(queens-louis size)) 1000)))

  (map (λ(n)
         (test n))
       (enumerate-interval 5 8))
```

```
normal queens x5: 135424.6151
swapped queens x5: 5487381.643
normal queens x6: 538798.604
swapped queens x6: 85704466.218
normal queens x7: 2210394.5659
swapped queens x7: 1255288880.717
normal queens x8: 8067290.5992
swapped queens x8: 25384464494.259
```



So that's 40 times worse at 5x5, 159 times worse at 6x6, 568 times worse at 7x7, and 3146 times at 8x8.

3.49 2.2.4: Example: A Picture Language

Authors describe a possible implementation of a "picture language" that tiles, patterns, and warps images according to a specification. This language consists of:

- a **painter** which makes an image within a specified parallelogram shaped frame. This is the most primitive element.

- **Operations** which make new painters from other painters. For example:

  - *beside* takes two painters, producing a new painter that puts one in the left half and one in the right half.

  - *flip-horiz* takes one painter and produces another to draw its image right-to-left reversed. These are defined as Scheme procedures and therefore have all the properties of Scheme procedures.

I'm going to have to get a little messy in order to make the picture language a reality. First I need a support library called guile-picture-language. I installed Guix on my system and ran `guix shell guile-picture-language guile`. Now the `pict` module is availale to me.

```
(use-modules (pict))
(pict->file (triangle 50 70)
            "2/pict/test.svg")
```



So that works.
How about this?

```
(use-modules (pict))
(let* ((p (pict-from-file "2/pict/Potato.png"))
       (pr (rotate p 180)))
(pict->file
 (scale (ht-append (vl-append pr p) pr p (vc-append pr p))
```

```
        0.5)
  "2/pict/Rotato.svg")) ;; must be SVG
```

And I just realized there's no flip/mirror function in this library. Racket's picture language is more complete, but it won't integrate with org-mode how I need. Maybe I can mock up a text-based version, similar to the binary mobile, and simultaneously check my work with Racket.

```
;; tilting at windmills trying to make a text-only picture language
;; TODO, probably won't finish
(define pl-defaultsize 32)
(define (make-painter orientation height width)
  (cons orientation (cons a b)))
(define (p-orientation p)
  (car p))
(define (p-height p)
  (cadr p))
(define (p-width p)
  (cddr p))
(define (opposite orientation)
  (cond ((eq? orientation 'down) 'up)
        ((eq? orientation 'up ) 'down)
        ((eq? orientation 'left) 'right)
        ((eq? orientation 'right) 'left)))
(define (flip-p p)
  (make-painter (opposite (p-orientation p))))
(define (below a b)
  (make-painter 'down a b))
(define (beside a b)
  (make-painter 'right a b))
(define (flip-vert p)
```

```
    (if (pair? p)
        (make-painter ))
  (define (paint pict)
    (define (rect p x y)
      (cond ((string? p)
             )))
    (rec p pl-defaultsize pl-defaultsize))
  (define (beside a b)
    (hc-append a b))
  (define (below a b)
    (vc-append a b))
```

3.50 Exercise 2.44: `up-split`

3.50.1 Text Definitions

```
(define (right-split painter n)
  (if (= n 0)
      painter
      (let ((smaller (right-split painter
                                  (- n 1))))
        (beside painter
                (below smaller smaller)))))
```

3.50.2 Question

Define the procedure `up-split` used by `corner-split`. It is similar to `right-split`, except that it switches the roles of `below` and `beside`.

3.50.3 Answer

```
(define (up-split painter n)
  (if (= n 0)
      painter
      (let ((smaller (up-split painter
                               (- n 1))))
        (below painter
               (beside smaller smaller)))))
```

3.51 Exercise 2.45: Generalized splitting

3.51.1 Question

`right-split` and `up-split` can be expressed as instances of a general splitting operation. Define a procedure `split` with the property that evaluating

```
(define right-split (split beside below))
(define up-split (split below beside))
```

produces procedures `right-split` and `up-split` with the same behaviors as the ones already defined.

3.51.2 Answer

```
(define (split f1 f2)
  (define (rec painter n)
    (if (= n 0)
        painter
        (let ((smaller (rec painter
                            (- n 1))))
          (f1 painter
              (f2 smaller smaller)))))
  rec)
(define right-split (split beside below))
;; This one goes down for some reason?
;(define up-split (split below beside))
(define (up-split painter n)
  (if (= n 0)
      painter
      (let ((smaller (up-split painter
                               (- n 1))))
        (below painter
               (beside smaller smaller)))))
```

3.52 Exercise 2.46: Defining vectors

3.52.1 Question

A two-dimensional vector **v** running from the origin to a point can be represented as a pair consisting of an x-coordinate and a y-coordinate. Implement a data abstraction for vectors by giving a constructor `make-vect` and corresponding selectors `xcor-vect` and `ycor-vect`. In terms of your selectors and constructor, implement procedures `add-vect`, `sub-vect`, and `scale-vect` that perform the operations vector addition, vector subtraction, and multiplying a vector by a scalar:

$$
\begin{aligned}
(x_1, y_1) + (x_2, y_2) &= (x_1 + x_2, y_1 + y_2), \\
(x_1, y_1) - (x_2, y_2) &= (x_1 - x_2, y_1 - y_2), \\
s \cdot (x, y) &= (sx, sy).
\end{aligned}
$$

3.52.2 Answer

This is pretty close to the `make-point` work done in Exercise 2.2, as well as my 2nd implementation of a rectangle in Exercise 2.3. Let's start there. Also, *aren't*

these points and not vectors since they don't have direction??

```
;; Guarentee x is a float
(define (float x)
  (if (inexact? x)
      x
      (exact->inexact x)))
```

```
<<force-float>>
(define (make-vect x y)
  (cons (float x)
        (float y))) ;; make-point
(define (xcor-vect v)
  (car v)) ;; x-point
(define (ycor-vect v)
  (cdr v)) ;; y-point
(define (add-vect v w)
  (make-vect (+ (xcor-vect v)
                (xcor-vect w))
             (+ (ycor-vect v)
                (ycor-vect w))))
(define (sub-vect v w)
  (make-vect (- (xcor-vect v)
                (xcor-vect w))
             (- (ycor-vect v)
                (ycor-vect w))))
(define (scale-vect s v)
  (make-vect (* s (xcor-vect v))
             (* s (ycor-vect v))))
```

I originally got the definition for `scale-vect` wrong by making it `(scale v s)`, then modified `frame-coord-map` when that broke. I realized my mistake once I went to write `beside`.

3.53 Exercise 2.47: Defining frames

3.53.1 Question

Here are two possible constructors for frames:

```
(define (make-frame origin edge1 edge2)
  (list origin edge1 edge2))
(define (make-frame origin edge1 edge2)
  (cons origin (cons edge1 edge2)))
```

For each constructor supply the appropriate selectors to produce an implementation for frames.

3.53.2 Answer

```
(define (make-frame origin edge1 edge2)
  (list origin edge1 edge2))
(define (origin-frame F)
  (car F))
(define (edge1-frame F)
  (cadr F))
(define (edge2-frame F)
  (caddr F))
```

```
(define (make-frame origin edge1 edge2)
  (cons origin (cons edge1 edge2)))
(define (origin-frame F)
  (car F))
(define (edge1-frame F)
  (cadr F))
(define (edge2-frame F)
  (cddr F))
```

3.54 Exercise 2.48: Line segments

3.54.1 Question

> A directed line segment in the plane can be represented as a pair of vectors—the vector running from the origin to the start-point of the segment, and the vector running from the origin to the end-point of the segment. Use your vector representation from Exercise 2.46 to define a representation for segments with a constructor `make-segment` and selectors `start-segment` and `end-segment`.

3.54.2 Answer

Again reminding me of Exercise 2.2.

```
(define (make-segment start end)
  (cons start end))
(define (start-segment seg)
  (car seg))
(define (end-segment seg)
  (cdr seg))
```

3.55 Exercise 2.49: Primitive painters

3.55.1 Text Definitions

```
(define (frame-coord-map frame)
  (lambda (v)
    (add-vect
     (origin-frame frame)
     (add-vect
      (scale-vect (xcor-vect v)
                  (edge1-frame frame))
      (scale-vect (ycor-vect v)
                  (edge2-frame frame))))))
(define (segments->painter segment-list)
  (lambda (frame)
    (for-each
     (lambda (segment)
       (draw-line
        ((frame-coord-map frame)
         (start-segment segment))
        ((frame-coord-map frame)
         (end-segment segment))))
     segment-list)))
```

3.55.2 Question

Use `segments->painter` to define the following primitive painters:

1. The painter that draws the outline of the designated frame.
2. The painter that draws an `X` by connecting opposite corners of the frame.
3. The painter that draws a diamond shape by connecting the midpoints of the sides of the frame.
4. The `wave` painter.

3.55.3 Answer

In the past I would need to constantly execute my code to "see" what I'm doing. But I'm starting to think in larger chunks and need the feedback less.

```
<<continuous-lines>>
(define outline
  (segments->painter (list
   (make-segment (make-vect 0 0)
                 (make-vect 0 1))
   (make-segment (make-vect 0 1)
                 (make-vect 1 1))
   (make-segment (make-vect 1 1)
                 (make-vect 1 0))
   (make-segment (make-vect 1 0)
                 (make-vect 0 0)))))
```

```
(define frame-X
  (segments->painter (list
   (make-segment (make-vect 0 0)
                 (make-vect 1 1))
   (make-segment (make-vect 0 1)
                 (make-vect 1 0)))))
(define diamond
  (segments->painter (list
   (make-segment (make-vect 0.5 0)
                 (make-vect 1 0.5))
   (make-segment (make-vect 1 0.5)
                 (make-vect 0.5 1))
   (make-segment (make-vect 0.5 1)
                 (make-vect 0 0.5))
   (make-segment (make-vect 0 0.5)
                 (make-vect 0.5 0)))))
```

I'll make the wave painter once I have some graph paper in front of me.

Ok, how can I verify this? I guess I could rewrite `segments->painter` for `pict`. Since this isn't a drawing system with imperative procedures, I should make it return a list of lines to be superimposed.

```
(use-modules (pict))
<<make-frame>>
<<make-vect>>
<<make-segment>>
(define (frame-coord-map frame)
  ;; Returns a function for adjusting a frame by a vector
  (lambda (v)
    (add-vect
     (origin-frame frame)
     (add-vect
      (scale-vect (xcor-vect v)
                  (edge1-frame frame))
      (scale-vect (ycor-vect v)
                  (edge2-frame frame))))))
(define (draw-line start end)
  ;; take two vectors, returns a line SVG object for pict
  (line (xcor-vect start)
        (ycor-vect start)
        (xcor-vect end)
        (ycor-vect end)))
(define (segments->painter segment-list)
  ;; takes a list of segments, returns a "painter" lambda, which applies
  ↪  a frame
  ;; to those segments and then maps over the result with draw-line to
  ↪  make a
  ;; list of SVG line objects which pict can combine.
```

```
  (lambda (frame)
    (map
     (lambda (segment)
       (draw-line
        ((frame-coord-map frame)
         (start-segment segment))
        ((frame-coord-map frame)
         (end-segment segment))))
     segment-list)))
;; NOTE: in the text, draw-line is a function which triggers an action in
;; some graphics driver, and returns nothing. Because of this, (map) was
;; originally (for-each). Thus the final result would have been thrown
↪  away.

<<painters>>

(define (paint-lines painter)
  ;; use pict to compile an SVG with the elements described by painter
  (let ((Frame (make-frame (make-vect 0 0)
                           (make-vect 500 0)
                           (make-vect 0 500))))
    (apply lt-superimpose
           (painter Frame))))
```

```
<<linepainter-pict>>
(pict->file (paint-lines diamond)
            "2/pict/testline.svg")
; #:maxw 100 #:maxh 100) <- FIXME: This procedure should take these
↪  arguments
; but I can't get it to work.
```



Holy moly it actually works.

3.56 Exercise 2.50: Transforming painters

3.56.1 Text Definitions

```
(define (transform-painter
         painter origin corner1 corner2)
  (lambda (frame)
    (let ((m (frame-coord-map frame)))
      (let ((new-origin (m origin)))
        (painter (make-frame new-origin
                             (sub-vect (m corner1)
                                       new-origin)
                             (sub-vect (m corner2)
                                       new-origin)))))))

(define (flip-vert painter)
  (transform-painter
```

```
    painter
    (make-vect 0.0 1.0)   ; new origin
    (make-vect 1.0 1.0)   ; new end of edge1
    (make-vect 0.0 0.0))) ; new end of edge2

(define (rotate90 painter)
  (transform-painter painter
                     (make-vect 1.0 0.0)
                     (make-vect 1.0 1.0)
                     (make-vect 0.0 0.0)))

(define (squash-inwards painter)
  (transform-painter painter
                     (make-vect 0.0 0.0)
                     (make-vect 0.65 0.35)
                     (make-vect 0.35 0.65)))
(define (beside painter1 painter2)
  (let ((split-point (make-vect 0.5 0.0)))
    (let ((paint-left  (transform-painter
                        painter1
                        (make-vect 0.0 0.0)
                        split-point
                        (make-vect 0.0 1.0)))
          (paint-right (transform-painter
                        painter2
                        split-point
                        (make-vect 1.0 0.0)
                        (make-vect 0.5 1.0))))
      (lambda (frame)
        (append
         (paint-left frame)
         (paint-right frame))))))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
(define topleft-tri
  (segments->painter (list
                      (make-segment (make-vect 0 0)
                                    (make-vect 0 0.4))
                      (make-segment (make-vect 0 0.4)
                                    (make-vect 0.4 0))
                      (make-segment (make-vect 0.4 0)
                                    (make-vect 0 0)))))
(let ((picture (beside (beside topleft-tri
                               topleft-tri)
                       (rotate90 topleft-tri))))
```

```
(pict->file (paint-lines picture)
            "2/pict/rotatetest.svg"))
```

3.56.2 Question

Define the transformation `flip-horiz`, which flips painters horizontally, and transformations that rotate painters counterclockwise by 180 degrees and 270 degrees.

3.56.3 Answer

```
(define (flip-horiz painter)
  (transform-painter
   painter
   (make-vect 1.0 0.0)   ; new origin
   (make-vect 0.0 0.0)   ; new end of edge1
   (make-vect 1.0 1.0))) ; new end of edge2

(define (rotate180 painter)
  (transform-painter painter
                     (make-vect 1.0 1.0)
                     (make-vect 0.0 1.0)
                     (make-vect 1.0 0.0)))
(define (rotate270 painter)
  (transform-painter painter
                     (make-vect 0.0 1.0)
                     (make-vect 0.0 0.0)
                     (make-vect 1.0 1.0)))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
(define topleft-tri
  (segments->painter (list
                      (make-segment (make-vect 0 0)
```

```
                                    (make-vect 0 0.4))
                        (make-segment (make-vect 0 0.4)
                                      (make-vect 0.8 0))
                        (make-segment (make-vect 0.8 0)
                                      (make-vect 0 0)))))
  (let ((picture (beside (beside
                          (flip-vert topleft-tri)
                          (rotate90 topleft-tri))
                         (beside
                          (rotate180 topleft-tri)
                          (rotate270 topleft-tri)))))
    (pict->file (paint-lines picture)
                "2/pict/rotate2.svg"))
```



3.57 Exercise 2.51

3.57.1 Question

Define the `below` operation for painters. `below` takes two painters as arguments. The resulting painter, given a frame, draws with the first painter in the bottom of the frame and with the second painter in the top. Define `below` in two different ways—first by writing a procedure that is analogous to the `beside` procedure given above, and again in terms of `beside` and suitable rotation operations (from @ref{Exercise 2.50}).

3.57.2 Answer

```
(define (below painter1 painter2)
  (let ((split-point (make-vect 0.0 0.5)))
    (let ((paint-left  (transform-painter
                        painter1
                        (make-vect 0.0 0.0)
                        (make-vect 1.0 0.0)
                        split-point))
          (paint-right (transform-painter
                        painter2
                        split-point
                        (make-vect 1.0 0.5)
                        (make-vect 0.0 1.0))))
      (lambda (frame)
        (append
         (paint-left frame)
         (paint-right frame))))))
(define (below-rotate painter1 painter2)
  (rotate270 (beside
              (rotate90 painter2)
              (rotate90 painter1))))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
<<below>>
(define topleft-tri
  (segments->painter (list
                      (make-segment (make-vect 0 0)
                                    (make-vect 0 0.4))
                      (make-segment (make-vect 0 0.4)
                                    (make-vect 0.8 0))
                      (make-segment (make-vect 0.8 0)
                                    (make-vect 0 0)))))
(let ((p1 (below (beside
```

```
                        topleft-tri
                        (rotate90 topleft-tri))
                       (beside
                        (rotate180 topleft-tri)
                        (rotate270 topleft-tri))))
      (p2 (below-rotate (beside
                        topleft-tri
                        (rotate90 topleft-tri))
                       (beside
                        (rotate180 topleft-tri)
                        (rotate270 topleft-tri)))))
  (pict->file (paint-lines (beside p1 p2))
              "2/pict/rotate3.svg"))
```





An aside: I'm beginning to see what makes Lisp-style programming different from C style. In C, the pictures would be described with separate data structures specified up front, but in Lisp you can use the code to create the data structure. Off the top of my head, the biggest players here would be first-level functions, and how statements evaluate to specific values rather than being imperative commands that cause something to happen elsewhere.

3.58 2.2.4 continued

stratified design is the notion that complex systems should be structured as a sequence of levels with a sequence of languages. See how electronic components are described in EE terms, the binary gates they form are described in digital logic terms, the programs they run described in programming language terms, the networks of programs described in network architecture terms, etc.

This stratified design can be seen in our picture language. We use lines and points to specify painters, use painters to make arrangements with beside/below, arrange these arrangements into higher-level arrangements like `up-split`.

3.59 Exercise 2.52

3.59.1 Question A

Make changes to the square limit of `wave` shown in Figure 2.9 by working at each of the levels described above. In particular:

- Add some segments to the primitive `wave` painter of Exercise 2.49 (to add a smile, for example).

3.59.2 Answer A

I need to do what I've been slacking off on: actually making the `wave` painter. First, I want a helper function to make inputting shapes easier.

```
(define (unwrap-if-needed args)
  ;; commonly needed in vararg functions
  (if (and (= 1 (length args))
           (list? (car args)))
      (car args) ; assume we were passed a list, unwrap it
      args))
```

```
<<unwrap-if-needed>>
(define (continuous-lines . vectors)
  ;; Given a list of vectors, return a series of line segments
  ;; that continuously follow the vectors until there are no more,
  ;; at which point connect the last vector to the first.
  (define (iter vecs lines)
    (if (> 2 (length vecs))
        lines
        (let* ((first (car vecs))
               (rest (cdr vecs))
               (second (car rest))
               (new-line (make-segment first second)))
          (iter rest (cons new-line lines)))))
```

```
  (let ((input (unwrap-if-needed vectors)))
    (iter (append input (list (car input))) ;; Connect to start
          '())))
(define (numbers-to-vectors . numbers)
  (define (iter input output)
    (if (> 2 (length input))
        output
        (let* ((first (car input))
               (d (cdr input))
               (second (car d))
               (dd (cdr d)))
          (iter dd
                (cons (make-vect first second)
                      output)))))
  (reverse (iter numbers '())))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
<<below>>
<<continuous-lines>>
(define topleft-tri
  (segments->painter (list
                      (make-segment (make-vect 0 0)
                                    (make-vect 0 0.4))
                      (make-segment (make-vect 0 0.4)
                                    (make-vect 0.8 0))
                      (make-segment (make-vect 0.8 0)
                                    (make-vect 0 0)))))
(define topleft-tri2
  (segments->painter (continuous-lines
                      (make-vect 0 0)
                      (make-vect 0 0.4)
                      (make-vect 0.8 0))))
(define topleft-tri3
  (segments->painter (continuous-lines
                      (numbers-to-vectors
                       0 0
                       0 0.4
                       0.8 0))))
(let ((p1 (below (beside topleft-tri topleft-tri2)
                 (beside topleft-tri topleft-tri3))))
  (pict->file (paint-lines p1)
              "2/pict/cl-test.svg"))
```



Now, let's write the wave painter.

```
<<continuous-lines>>
;; Should be raising their left hand and
;; lowering their right. (I screwed up the axes)
(define wave
  (segments->painter
   (continuous-lines
    (numbers-to-vectors
    0.55 0
    0.5 0.15 ;; center-left side of head
    0.55 0.3
    0.45 0.275
    0.25 0.35
    0 0.25
    0 0.3
    0.25 0.45
    0.45 0.35
    0.5 0.65
    0.4 1
    0.45 1
    0.6 0.7
    0.75 1
    0.8 1
    0.7 0.65
    0.75 0.35
```

```
      1 0.675
      1 0.6
      0.75 0.275
      0.65 0.3
      0.7 0.15 ;; center-right side of head
      0.65 0))))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
<<below>>
<<continuous-lines>>
<<wave-painter>>
(let ((p1 (beside wave (flip-vert wave))))
  (pict->file (paint-lines p1)
              "2/pict/wave-test.svg"))
```

The question needs me to overlay something new. So I need some way to add more segments to a painter after it's already been written.



```
<<unwrap-if-needed>>
(define (append-painters-rec . args)
  (lambda (frame)
    (define (rec painters)
      (if (null? painters)
          '()
          (append ((car painters) frame)
                  (rec (cdr painters)))))
    (rec (unwrap-if-needed args))))
(define (append-painters . args)
  (lambda (frame)
    (define (iter segments to-paint)
      (if (null? to-paint)
          segments
          (iter (append ((car to-paint) frame)
                        segments)
                        (cdr to-paint))))
    (iter '()
          (unwrap-if-needed args))))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
<<below>>
<<continuous-lines>>
<<painters>>
<<wave-painter>>
<<append-painters>>
(let ((p1 (append-painters
           (beside wave (flip-vert wave))
           frame-X
           outline)))
  (pict->file (paint-lines p1)
              "2/pict/append-test.svg"))
```



Now let's (try to) add a smile.

```
<<wave-painter>>
<<append-painters>>
(define wave-smile
  (append-painters
   wave
   (segments->painter
    (continuous-lines
     (numbers-to-vectors
      0.55 0.2
      0.6 0.225
      0.65 0.2)))))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
<<below>>
<<continuous-lines>>
```

```
<<wave-smile>>
(pict->file (paint-lines wave-smile)
            "2/pict/wave-smile-test.svg")
```



3.59.3 Question B

Change the pattern constructed by `corner-split` (for example, by using only one copy of the `up-split` and `right-split` images instead of two).

```
(define (corner-split painter n)
  (if (= n 0)
      painter
      (let ((up (up-split painter (- n 1)))
            (right (right-split painter
                                (- n 1))))
        (let ((top-left (beside up up))
              (bottom-right (below right
                                   right))
              (corner (corner-split painter
```

```
                                    (- n 1))))
          (beside (below painter top-left)
                  (below bottom-right
                         corner))))))

(define (corner-split-mine painter n)
  (if (= n 0)
      painter
      (let ((up (up-split painter (- n 1)))
            (right (right-split painter
                                (- n 1))))
        (let ((top-left up)
              (bottom-right right)
              (corner (corner-split-mine painter
                                         (- n 1))))
          (beside (below painter top-left)
                  (below bottom-right
                         corner))))))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
<<below>>
<<continuous-lines>>
<<painters>>
<<wave-smile>>
<<append-painters>>
<<splits-generalized>>
<<corner-split>>
(let ((p1 (beside (flip-horiz (corner-split wave-smile 5))
                  (corner-split-mine wave-smile 5))))
;(let ((p1 (up-split frame-X 5)))
  (pict->file (paint-lines p1)
              "2/pict/corner-splits.svg"))
```



So my modified version doesn't split at the `corner-split` level, as predicted. However `up-split` and `right-split` do, so the effect is only delayed by one level. But more importantly:

Why does it go down instead of up? I don't get it.

`corner-split` and `up-split` are even the same code as every answer online. Could my `paint-lines` procedure be what's causing issues?

3.59.4 Question C

> Modify the version of `square-limit` that uses `square-of-four` so as to assemble the corners in a different pattern. (For example, you might make the big Mr. Rogers look outward from each corner of the square.)

3.59.5 Textbook Definitions

```
(define (square-of-four tl tr bl br)
  (lambda (painter)
    (let ((top (beside (tl painter)
```

```
                        (tr painter)))
          (bottom (beside (bl painter)
                          (br painter))))
      (below bottom top))))
(define (square-limit painter n)
  (let ((combine4
         (square-of-four flip-horiz
                         identity
                         rotate180
                         flip-vert)))
    (combine4 (corner-split painter n))))
```

3.59.6 Answer C

```
(define (square-limit-mine painter n)
  (let ((combine4
         (square-of-four identity
                         flip-horiz
                         flip-vert
                         rotate180)))
    (combine4 (corner-split painter n))))
```

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
<<below>>
<<continuous-lines>>
<<wave-smile>>
<<splits-generalized>>
<<corner-split>>
<<square-of-four-txt>>
<<square-limit-mine>>
(pict->file (paint-lines (beside (square-limit wave-smile 5)
                                 (square-limit-mine wave-smile 5)))
            "2/pict/square-limits.svg")
```



Ok, I'm confused. Shouldn't the shape of `square-limit` be the same regardless of the operators passed to it? It looks like the operators effect each chunk separately.

```
<<linepainter-pict>>
<<frame-transforms-txt>>
<<flip-rotate2>>
<<below>>
<<continuous-lines>>
<<wave-smile>>
<<splits-generalized>>
<<corner-split>>
<<square-of-four-txt>>
(define (square-limit-mine painter n)
  (let ((combine4
         (square-of-four identity
                         identity
                         identity
                         identity)))
```

```
    (combine4 (corner-split painter n))))
(pict->file (paint-lines (square-limit-mine wave-smile 5))
            "2/pict/square-limit-identity.svg")
```



I expected the operators would change the *image orientation*, not the *structure*. Maybe this is because my `draw-lines` makes recursively nested stacks of painter objects, rather than imperative calls to drawing procedures? Looking on the internet I see someone else having the same results.

3.60 2.3.1: Quotation

Syntactic vs semantic use in Lisp:

```
(define dog 123)
(+ dog dog) ;; semantic usage
(quote dog) ;; syntactic usage
```

This was introduced by famous LISP scientists, Abbott and Costello.

3.61 Exercise 2.53

3.61.1 Question

What would the interpreter print in response to evaluating each of the following expressions?

```
(list 'a 'b 'c)
(list (list 'george))
(cdr '((x1 x2) (y1 y2)))
(cadr '((x1 x2) (y1 y2)))
(pair? (car '(a short list)))
(memq 'red '((red shoes) (blue socks)))
(memq 'red '(red shoes blue socks))
```

3.61.2 Answer

```
(list 'a 'b 'c) ;; (a b c)
(list (list 'george)) ;; ((george))
(cdr '((x1 x2) (y1 y2))) ;; ((y1 y2))
(cadr '((x1 x2) (y1 y2))) ;; (y1 y2)
(pair? (car '(a short list))) ;; false
(memq 'red '((red shoes) (blue socks))) ;; false
(memq 'red '(red shoes blue socks)) ;; (red shoes blue socks)
```

3.62 Exercise 2.54

Two lists are said to be `equal?` if they contain equal elements arranged in the same order. For example,

```
(equal? '(this is a list) '(this is a list))
```

is true, but

```
(equal? '(this is a list) '(this (is a) list))
```

is false. To be more precise, we can define `equal?` recursively in terms of the basic `eq?` equality of symbols by saying that `a` and `b` are `equal?` if they are both symbols and the symbols are `eq?`, or if they are both lists such that `(car a)` is `equal?` to `(car b)` and `(cdr a)` is `equal?` to `(cdr b)`. Using this idea, implement `equal?` as a procedure.

3.62.1 Answer

```
(define (equal? a b)
  (cond ((and (symbol? a)
              (symbol? b))
         (eq? a b))
        ((and (list? a)
              (list? b))
         (and (equal? (car a) (car b))
              (equal? (cdr a) (cdr b))))
        (else #f)))
```

```
(load "mattcheck2.scm")
(let ((a '(this is a list))
      (aa '(this is a list))
      (b '(this (quote is a) list)))
  (mattcheck "equal? true"
             (equal? a a))
  (mattcheck "equal? trick question"
             (equal? a a))
  (mattcheck "equal? false"
             (equal? a b)
             #f))
```

```
SUCCEED at equal? true
SUCCEED at equal? trick question
SUCCEED at equal? false
```

3.63 Exercise 2.55

3.63.1 Question

Eva Lu Ator types to the interpreter the expression

```
(car ''abracadabra)
```

To her surprise, the interpreter prints back **quote**. Explain.

3.63.2 Answer

You're quoting quote, silly! Who's on first?

3.64 2.3.2: Example: Symbolic differentiator

I needed to get some help, but I think I have some understanding now.

The **derivative** of an expression E, relative to the variable x, describes the rate of change (or δ **delta**) of that expression with relation to x. Mathematicians symbolize it like this:

$$\frac{dE}{dx}$$

Two stumbling blocks about this notation:

1. This is not division. I understand why this notation would make sense to an experienced mathematician, since you factor out things by dividing an expression. For example:

   $$\frac{3x}{x} = 3$$

   $$\frac{3x}{3} = x$$

   But it's still confusing.

2. dx is not some variable d times some variable x. $\frac{d(x+3)}{dx}$ means "the derivative of $x + 3$ with respect to x."

The process of finding the derivative is called **differentiation**.
The rules the book puts forward, rephrased in plain English, are these:

- If the expression is a constant, the derivative must be 0.
- If the expression is x, the derivative must be 1.
- If the expression is $u + v$, the derivative is the sum of two derivatives:
  1. The derivative of u with respect to x.
  2. The derivative of v with respect to x.
- If the expression is $u \times v$, the derivative is the sum of:
  1. u times the derivative of v with respect to x
  2. v times the derivative of u with respect to x

Later:

- If the expression is u^0, the result is 1, which is a constant, so the rate of change is 0.
- If the expression is u^1, the result is u, so the rate of change is 1.
- Else, if the expression is u^n, the derivative is the product of these:
  1. n
  2. u to the power of $n - 1$
  3. The derivative of u with respect to x.

3.65 Exercise 2.56: Differentiating exponentiation

3.65.1 Text definitions

```
(define (variable? x) (symbol? x))
(define (same-variable? v1 v2)
  (and (variable? v1) (variable? v2) (eq? v1 v2)))
(define (sum? x) (and (pair? x) (eq? (car x) '+)))
(define (addend s) (cadr s))
(define (augend s) (caddr s))
(define (product? x) (and (pair? x) (eq? (car x) '*)))
(define (multiplier p) (cadr p))
(define (multiplicand p) (caddr p))
(define (make-sum a1 a2)
  (cond ((=number? a1 0) a2)
        ((=number? a2 0) a1)
        ((and (number? a1) (number? a2))
         (+ a1 a2))
        (else (list '+ a1 a2))))
(define (=number? exp num)
  (and (number? exp) (= exp num)))
(define (make-product m1 m2)
  (cond ((or (=number? m1 0) (=number? m2 0)) 0)
        ((=number? m1 1) m2)
        ((=number? m2 1) m1)
        ((and (number? m1) (number? m2)) (* m1 m2))
        (else (list '* m1 m2))))
(define (deriv exp var)
  (cond ((number? exp) 0)
        ((variable? exp) (if (same-variable? exp var) 1 0))
        ((sum? exp) (make-sum (deriv (addend exp) var)
                              (deriv (augend exp) var)))
        ((product? exp)
         (make-sum
           (make-product (multiplier exp)
                         (deriv (multiplicand exp) var))
           (make-product (deriv (multiplier exp) var)
                         (multiplicand exp))))
        (else
         (error "unknown expression type: DERIV" exp))))
```

3.65.2 Question

Show how to extend the basic differentiator to handle more kinds of expressions. For instance, implement the differentiation rule

$$\frac{d\left(u^{n}\right)}{dx} = nu^{n-1}\frac{du}{dx}$$

by adding a new clause to the `deriv` program and defining appropriate procedures `exponentiation?`, `base`, `exponent`, and `make-exponentiation`. (You may use the symbol `**` to denote exponentiation.) Build in the rules that anything raised to the power 0 is 1 and anything raised to the power 1 is the thing itself.

3.65.3 Answer

```
(define (exponentiation? x) (and (pair? x) (eq? (car x) '**)))
(define (base e) (cadr e))
(define (exponent e) (caddr e))
(define (make-exponentiation b e)
  (cond ((=number? base 1) 1)
        ((=number? e 0) 1)
        ((=number? e 1) b)
        ((and (number? b) (number? e)) (expt b e))
        (else (list '** b e))))
(define (deriv exp var)
  (cond ((number? exp) 0)
        ((variable? exp) (if (same-variable? exp var) 1 0))
        ((sum? exp) (make-sum (deriv (addend exp) var)
                              (deriv (augend exp) var)))
        ((product? exp)
         (make-sum
          (make-product (multiplier exp)
                        (deriv (multiplicand exp) var))
          (make-product (deriv (multiplier exp) var)
                        (multiplicand exp))))
        ((exponentiation? exp)
         (make-product
          (make-product
           (exponent exp)
           (make-exponentiation (base exp)
                                (make-sum (exponent exp) -1)))
          (deriv (base exp) var)))
        (else
         (error "unknown expression type: DERIV" exp))))
```

First I'll make sure the textbook examples work as expected.

```
<<deriv-basic-txt>>
<<deriv-exp>>
(load "mattcheck2.scm")
(mattcheck "deriv: basic addition"
           (deriv '(+ x 3) 'x)
           1)
(mattcheck "deriv: basic multiplication"
           (deriv '(* x y) 'x)
```

```
            'y)
  (mattcheck "deriv: bad simplification"
             (deriv '(* (* x y) (+ x 3)) 'x)
             '(+ (* x y) (* y (+ x 3))))
  ;; Now, did I do my job right?
  ;; Checking against this guy's results:
  ;;   http://jots-jottings.blogspot.com/2011/11/sicp-exercise-256-differe
  ↪  ntiating.html
  (mattcheck "make-exponentiation 1"
             (make-exponentiation 1 12)
             1)
  (mattcheck "make-exponentiation 2"
             (make-exponentiation 2 12)
             4096)
  (mattcheck "make-exponentiation 3"
             (make-exponentiation 'x 12)
             '(** x 12))
  (mattcheck "deriv: exponentiation 1"
             (deriv (make-exponentiation 'a 5) 'a)
             '(* 5 (** a 4)))
  (mattcheck "deriv: exponentiation 2"
             (deriv (make-exponentiation 'a 'b) 'a)
             '(* b (** a (+ b -1))))
```

```
SUCCEED at deriv: basic addition
SUCCEED at deriv: basic multiplication
SUCCEED at deriv: bad simplification
SUCCEED at make-exponentiation 1
SUCCEED at make-exponentiation 2
SUCCEED at make-exponentiation 3
SUCCEED at deriv: exponentiation 1
SUCCEED at deriv: exponentiation 2
```

3.66 Exercise 2.57

3.66.1 Question

Extend the differentiation program to handle sums and products of arbitrary numbers of (two or more) terms. Then the last example above could be expressed as

```
(deriv '(* x y (+ x 3)) 'x)
```

Try to do this by changing only the representation for sums and products, without changing the `deriv` procedure at all. For example, the `addend` of a sum would be the first term, and the `augend` would be the sum of the rest of the terms.

3.66.2 Answer

```
(define (sum? x) (and (pair? x) (eq? (car x) '+)))
(define (addend s) (cadr s))
(define (augend s)
  (let ((rest (cddr s)))
    (if (null? (cdr rest))
        (car rest)
;        (make-sum (addend rest)
;                  (augend (cons '+ rest))))))
        (cons '+ rest))))
(define (product? x) (and (pair? x) (eq? (car x) '*)))
(define (multiplier p) (cadr p))
(define (multiplicand p)
  (let ((rest (cddr p)))
    (if (null? (cdr rest))
        (car rest)
;        (make-product (multiplier rest)
;                      (multiplicand (cons '* rest))))))
        (cons '* rest))))
(define (make-sum a1 a2)
  (cond ((=number? a1 0) a2)
        ((=number? a2 0) a1)
        ((and (number? a1) (number? a2))
         (+ a1 a2))
        (else (list '+ a1 a2))))
(define (make-product m1 m2)
  (cond ((or (=number? m1 0) (=number? m2 0)) 0)
        ((=number? m1 1) m2)
        ((=number? m2 1) m1)
        ((and (number? m1) (number? m2))
         (* m1 m2))
        (else (list '* m1 m2))))
```

```
<<deriv-basic-txt>>
<<deriv-exp>>
<<deriv-longer>>
(load "mattcheck2.scm")
(mattcheck "deriv: longer addition"
           (deriv '(+ x 3 6 z) 'x)
           1)
(mattcheck "deriv: longer multiplication"
           (deriv '(* x y (+ x 3)) 'x)
           '(+ (* x y) (* y (+ x 3))))
```

```
SUCCEED at deriv: longer addition
SUCCEED at deriv: longer multiplication
```

I had to look up the solution for this one. The commented sections in `augend` and `multiplicand` were the answers I was trying to make work:

```
(define (augend s)
  (let ((rest (cddr s)))
    (if (null? (cdr rest))
        (car rest)
;        (make-sum (addend rest)
;                  (augend (cons '+ rest))))))
        (cons '+ rest))))
```

It's taking a bunch of steps that weren't ultimately helping, but it didn't occur to me that the solution was to go simpler rather than more complicated. I'll have to keep watch for problem-solving dead-ends like this.

3.67 Exercise 2.58

3.67.1 Question

Suppose we want to modify the differentiation program so that it works with ordinary mathematical notation, in which `+` and `*` are infix rather than prefix operators. Since the differentiation program is defined in terms of abstract data, we can modify it to work with different representations of expressions solely by changing the predicates, selectors, and constructors that define the representation of the algebraic expressions on which the differentiator is to operate.

3.67.2 Part 1

Show how to do this in order to differentiate algebraic expressions presented in infix form, such as `(x + (3 * (x + (y + 2))))`. To simplify the task, assume that `+` and `*` always take two arguments and that expressions are fully parenthesized.

3.67.3 Answer 1

```
(define (variable? x) (symbol? x))
(define (same-variable? v1 v2)
  (and (variable? v1) (variable? v2) (eq? v1 v2)))
(define (sum? x) (and (pair? x) (eq? (cadr x) '+)))
(define (addend s) (car s))
(define (augend s) (caddr s))
(define (product? x) (and (pair? x) (eq? (cadr x) '*)))
(define (multiplier p) (car p))
(define (multiplicand p) (caddr p))
(define (make-sum a1 a2)
  (cond ((=number? a1 0) a2)
        ((=number? a2 0) a1)
        ((and (number? a1) (number? a2))
         (+ a1 a2))
        (else (list a1 '+ a2))))
```

```
(define (=number? exp num)
  (and (number? exp) (= exp num)))
(define (make-product m1 m2)
  (cond ((or (=number? m1 0) (=number? m2 0)) 0)
        ((=number? m1 1) m2)
        ((=number? m2 1) m1)
        ((and (number? m1) (number? m2)) (* m1 m2))
        (else (list m1 '* m2))))
(define (deriv exp var)
  (cond ((number? exp) 0)
        ((variable? exp) (if (same-variable? exp var) 1 0))
        ((sum? exp) (make-sum (deriv (addend exp) var)
                              (deriv (augend exp) var)))
        ((product? exp)
         (make-sum
           (make-product (multiplier exp)
                         (deriv (multiplicand exp) var))
           (make-product (deriv (multiplier exp) var)
                         (multiplicand exp))))
        (else
         (error "unknown expression type: DERIV" exp))))
```

```
<<deriv-infix>>
(load "mattcheck2.scm")
(mattcheck "deriv: basic addition"
           (deriv '(x + 3) 'x)
           1)
(mattcheck "deriv: basic multiplication"
           (deriv '(x * y) 'x)
           'y)
(mattcheck "deriv: bad simplification"
           (deriv '((x * y) * (x + 3)) 'x)
           '((x * y) + (y * (x + 3))))
```

```
SUCCEED at deriv: basic addition
SUCCEED at deriv: basic multiplication
SUCCEED at deriv: bad simplification
```

3.67.4 Part 2

The problem becomes substantially harder if we allow standard algebraic notation, such as `(x + 3 * (x + y + 2))`, which drops unnecessary parentheses and assumes that multiplication is done before addition. Can you design appropriate predicates, selectors, and constructors for this notation such that our derivative program still works?

3.67.5 Answer 2

Ok, I think I can do the long-form list objective, since that would be a combination of two earlier exercises. But as I write I feel clueless how to make multiplication happen before addition – my first impulse is that this would mean I would need to either:

1. Change how Lisp's evaluation works (obviously overkill)
2. Make statements be evaluated twice, once for multiplication and then once for addition. And I think that would require modifications to `deriv`.

I'll just start working on the long-form feature and see if I can think of anything.

```
(define (variable? x) (symbol? x))
(define (same-variable? v1 v2)
  (and (variable? v1) (variable? v2) (eq? v1 v2)))
(define (sum? x) (and (pair? x) (eq? (cadr x) '+)))
(define (addend s) (car s))
(define (augend s)
  (let ((rest (cddr s)))
    (if (null? (cdr rest))
        (car rest)
        rest)))
(define (product? x) (and (pair? x) (eq? (cadr x) '*)))
(define (multiplier p) (car p))
(define (multiplicand p)
  (let ((rest (cddr p)))
    (if (null? (cdr rest))
        (car rest)
        rest)))
(define (make-sum a1 a2)
  (cond ((=number? a1 0) a2)
        ((=number? a2 0) a1)
        ((and (number? a1) (number? a2))
         (+ a1 a2))
        ((product? a2)
         (list a1 '+ ;; cross your fingers!
               (make-product (multiplier a2)
                             (multiplicand a2))))
        (else (list a1 '+ a2))))
(define (=number? exp num)
  (and (number? exp) (= exp num)))
(define (make-product m1 m2)
  (cond ((or (=number? m1 0) (=number? m2 0)) 0)
        ((=number? m1 1) m2)
        ((=number? m2 1) m1)
        ((and (number? m1) (number? m2)) (* m1 m2))
        (else (list m1 '* m2))))
```

```
(define (deriv exp var)
  (cond ((number? exp) 0)
        ((variable? exp) (if (same-variable? exp var) 1 0))
        ((sum? exp) (make-sum (deriv (addend exp) var)
                              (deriv (augend exp) var)))
        ((product? exp)
         (make-sum
          (make-product (multiplier exp)
                        (deriv (multiplicand exp) var))
          (make-product (deriv (multiplier exp) var)
                        (multiplicand exp))))
        (else
         (error "unknown expression type: DERIV" exp))))
```

```
<<deriv-algebraic>>
(load "mattcheck2.scm")
(mattcheck "deriv: basic addition"
           (deriv '(x + 3) 'x)
           1)
(mattcheck "deriv: basic multiplication"
           (deriv '(x * y) 'x)
           'y)
(mattcheck "deriv: bad simplification"
           (deriv '((x * y) * (x + 3)) 'x)
           '((x * y) + (y * (x + 3))))
(mattcheck "deriv: longer addition"
           (deriv '(x + 3 + 6 + z) 'x)
           1)
(mattcheck "deriv: longer multiplication"
           (deriv '(x * y * (x + 3)) 'x)
           '((x * y) + (y * (x + 3))))

;; Test cases from:
;; http://community.schemewiki.org/?sicp-ex-2.58
(mattcheck "deriv: mult prioritization 1"
           (deriv '(x + 3 * (x + y + 2)) 'x)
           4)
(mattcheck "deriv: mult prioritization 2"
           (deriv '(x * (y * (x + 3))) 'x)
           '((x * y) + (y * (x + 3))))
(mattcheck "deriv: mult prioritization 3"
           (deriv '((x * y) * (x + 3)) 'x)
           '((x * y) + (y * (x + 3))))
```

```
SUCCEED at deriv: basic addition
SUCCEED at deriv: basic multiplication
SUCCEED at deriv: bad simplification
SUCCEED at deriv: longer addition
```

```
SUCCEED at deriv: longer multiplication
SUCCEED at deriv: mult prioritization 1
SUCCEED at deriv: mult prioritization 2
SUCCEED at deriv: mult prioritization 3
```

My solution (which turned out to work) was by modifying `make-sum` to check the operator *after* the current one, see if it was a multiplication, and if so to evaluate it with `make-product` before doing the current addition. The "immutable variable" ways of problem solving makes this pretty easy to solve:

```
;; I want to do f to the first item of x, unless some-condition? in the
↪  second
;; item, in which case do g to the 2nd item before x to the first
(define (foo x)
  (if (some-condition? (cdr x))
      (f (cons (car x)
               (g (cdr x))))
      (f x)))
```

If I were doing this with mutable variables, where I needed to perform these modifications in one data structure, I'd be concerned.

Still, I was stuck on this problem for a while, only to find my *test cases* were wrong.

3.68 Exercise 2.59: Representing sets

3.68.1 Text definitions

```
(define (element-of-set?-manual x set)
  (cond ((null? set) #f)
        ((equal? x (car set)) #t)
        (else (element-of-set? x (cdr set)))))
(define element-of-set? member) ; builtins are faster :)

(define (adjoin-set x set)
  (if (element-of-set? x set)
      set
      (cons x set)))

(define (intersection-set set1 set2)
  (cond ((or (null? set1) (null? set2))
         '())
        ((element-of-set? (car set1) set2)
         (cons (car set1)
               (intersection-set (cdr set1)
                                 set2)))
        (else (intersection-set (cdr set1)
                                set2))))
```

3.68.2 Question

Implement the `union-set` operation for the unordered-list representation of sets.

3.68.3 Answer

`union-set` at a first glance looks like the opposite of `intersection-set`, since the logic looks like "merge two sets, and if an element exists in both than don't include it". However it's actually just avoiding putting a symbol in twice.

```
(define (union-set-rec set1 set2)
  (cond ((or (null? set1) (null? set2))
         set2)
        ((element-of-set? (car set1) set2)
         (union-set-rec (cdr set1)
                        set2))
        (else (cons (car set1)
                    (union-set-rec (cdr set1)
                                   set2)))))
(define (union-set set1 set2)
    (define (iter s result)
      (cond ((null? s)
             (reverse result))
            ((element-of-set? (car s) set2)
             (iter (cdr s) result))
            (else
             (iter (cdr s)
                   (cons (car s)
                         result)))))
    (append (iter set1 '())
            set2))
```

```
<<sets-txt>>
<<union-set>>
(load "mattcheck2.scm")
(let ((set1 (list 1 2 3 4 5 6))
      (set2 (list 4 5 6 7 8 9))
      (set-union (list 1 2 3 4 5 6 7 8 9)))
  (mattcheck "union-set-rec"
             (union-set-rec set1 set2)
             set-union)
  (mattcheck "union-set"
             (union-set set1 set2)
             set-union))
```

```
SUCCEED at union-set-rec
SUCCEED at union-set
```

```
<<enumerate-interval>>
<<sets-txt>>
<<union-set>>
(load "../mattbench.scm")

(let ((set1 (enumerate-interval 1 1000))
      (set2 (enumerate-interval 500 1500)))
  (define (test)
    (format #t "~&recursive union-sets: ~a"
            (cadr (mattbench2 (λ()(union-set-rec set1 set2)) 10000)))
    (format #t "~&iterative union-sets: ~a"
            (cadr (mattbench2 (λ()(union-set set1 set2)) 10000))))

  (test))
```

```
recursive union-sets: 4704343.6974
iterative union-sets: 4755592.2304
```

Man, my "optimized" versions never work. I would blame it on the Scheme compiler but I'm a new programmer so it's probably a skill issue.

3.69 Exercise 2.59: Sets with duplicates

3.69.1 Question

We specified that a set would be represented as a list with no duplicates. Now suppose we allow duplicates. For instance, the set $\{1, 2, 3\}$ could be represented as the list `(2 3 2 1 3 2 2)`. Design procedures `element-of-set?`, `adjoin-set`, `union-set`, and `intersection-set` that operate on this representation. How does the efficiency of each compare with the corresponding procedure for the non-duplicate representation? Are there applications for which you would use this representation in preference to the non-duplicate one?

3.69.2 Answer

`element-of-set` can be left unchanged. The others are a matter of getting sloppy: `adjoin-set` can just be a `cons`, and `union-set` can be an append. Now the remaining question is, how to make `intersection-set` keep the duplicates?

```
(define (adjoin-set-dupes x set)
  (cons x set))

(define (union-set-dupes set1 set2)
  (append set1 set2))

(define (intersection-set-dupes set1 set2)
  (let ((inter (intersection-set set1 set2)) ;; yes, we're calling the
  ↪  non-duplicate version
```

```
        (union (union-set-dupes set1 set2)))
    (filter (lambda(x) (element-of-set? x inter))
            union)))
(define (intersection-set-dupes2 set1 set2)
  (let ((inter (intersection-set set1 set2)) ;; yes, we're calling the
↪  non-duplicate version
        (inter2 (intersection-set set2 set1)))
    (append inter inter2)))
```

```
<<sets-txt>>
<<union-set>>
<<set-dupes>>
(load "mattcheck2.scm")
(let ((set1 (list 1 2 3 4 5 6))
      (set2 (list 4 5 6 7 8 9))
      (set-union (list 1 2 3 4 5 6 7 8 9))
      (set-union-dupes
       (list 1 2 3 4 5 6 4 5 6 7 8 9))
      (set-intersection (list 4 5 6))
      (set-intersection-dupes
       (list 4 5 6 4 5 6)))
  (mattcheck "union-set-dupes"
             (union-set-dupes set1 set2)
             set-union-dupes)
  (mattcheck "union-set"
             (union-set set1 set2)
             set-union)
  (mattcheck "intersection-set"
             (intersection-set set1 set2)
             set-intersection)
  (mattcheck "intersection-set-dupes"
             (intersection-set-dupes set1 set2)
             set-intersection-dupes)
  (mattcheck "intersection-set-dupes2"
             (intersection-set-dupes2 set1 set2)
             set-intersection-dupes))
```

```
SUCCEED at union-set-dupes
SUCCEED at union-set
SUCCEED at intersection-set
SUCCEED at intersection-set-dupes
SUCCEED at intersection-set-dupes2
```

```
<<enumerate-interval>>
<<sets-txt>>
<<union-set>>
<<set-dupes>>
```

```
(use-modules (ice-9 format))
(load "../mattbench.scm")

(let ((set1 (enumerate-interval 1 1000))
      (set2 (enumerate-interval 500 1500)))
  (define (test)
    (format #t "~&union-sets: ~a"
            (cadr (mattbench2 (λ()(union-set set1 set2)) 10000)))
    (format #t "~&union-sets-dupes: ~a"
            (cadr (mattbench2 (λ()(union-set-dupes set1 set2)) 10000)))
    (format #t "~&intersection-set: ~a"
            (cadr (mattbench2 (λ()(intersection-set set1 set2)) 10000)))
    (format #t "~&intersection-set-dupes: ~a"
            (cadr (mattbench2 (λ()(intersection-set-dupes set1 set2))
↪   10000)))
    (format #t "~&intersection-set-dupes2: ~a"
            (cadr (mattbench2 (λ()(intersection-set-dupes2 set1 set2))
↪   10000)))
    )

  (test))
```

```
union-sets: 4734892.3798
union-sets-dupes: 40132.52
intersection-set: 4673425.9196
intersection-set-dupes: 10325053.8432
intersection-set-dupes2: 10872996.2555
```

So for `union-sets` a significant speedup, while in algorithms that need to check for duplicates like `intersection-set` it's much more time. Also, in a no-duplicate implementation `element-of-set`, it wouldn't be wasting time checking duplicates.

3.70 Exercise 2.61: Ordered sets

3.70.1 Question

Give an implementation of `adjoin-set` using the ordered representation. By analogy with `element-of-set?` show how to take advantage of the ordering to produce a procedure that requires on the average about half as many steps as with the unordered representation.

3.70.2 Answer

```
(define (element-of-set?-ordered x set)
  (cond ((null? set) #f)
        ((= x (car set)) #t)
        ((< x (car set)) #f)
```

```
        (else (element-of-set?-ordered x (cdr set)))))

(define (adjoin-set-ordered x set)
  (define (iter checked rest)
    (cond ((null? rest)
           (append checked (list x)))
          ((= x (car rest)) set)
          ((> x (car rest))
           (iter (cons (car rest)
                       checked)
                 (cdr rest)))
          (else (append (reverse checked)
                        (cons x rest)))))
  (iter '() set))
```

```
<<sets-txt>>
<<union-set>>
<<set-ordered>>
(load "mattcheck2.scm")
(let ((set (list 1 2 4 5 6 7 8 9))
      (answer (list 1 2 3 4 5 6 7 8 9)))
  (mattcheck "adjoin-set-ordered"
             (adjoin-set-ordered 3 set)
             answer))
```

```
SUCCEED at adjoin-set-ordered
```

3.71 Exercise 2.62: `union-set` ordered

3.71.1 Question

Give a $\Theta(n)$ implementation of `union-set` for sets represented as ordered lists.

3.71.2 Answer

```
(define (union-set-ordered set1 set2)
  (define (iter s1 s2 result)
    (cond ((null? s1)
           (append (reverse result) s2))
          ((null? s2)
           (append (reverse result) s1))
          (else
           (let ((s1a (car s1))
                 (s2a (car s2)))
             (cond ((= s1a s2a)
                    (iter (cdr s1) (cdr s2)
                          (cons s1a result)))
                   ((< s1a s2a)
```

```
                    (iter (cdr s1) s2
                          (cons s1a result)))
                   ((> s1a s2a)
                    (iter s1 (cdr s2)
                          (cons s2a result))))))))
  (iter set1 set2 '()))
```

```
<<sets-txt>>
<<union-set>>
<<set-ordered>>
<<union-set-ordered>>
(load "mattcheck2.scm")
(let ((set1 (list 1 2 3 4 5))
      (set2 (list 4 5 6 7 8 9))
      (answer (list 1 2 3 4 5 6 7 8 9)))
  (mattcheck "union-set-ordered"
             (union-set-ordered set1 set2)
             answer))
```

```
SUCCEED at union-set-ordered
```

```
<<enumerate-interval>>
<<sets-txt>>
<<union-set>>
<<union-set-ordered>>
(use-modules (ice-9 format))
(load "../mattbench.scm")

;; http://community.schemewiki.org/?sicp-ex-2.62
(define (union-set-alt set1 set2)
  (cond  ((null? set1) set2)
         ((null? set2) set1)
         (else
          (let ((x1 (car set1))
                (x2 (car set2)))
            (cond ((= x1 x2) (cons x1 (union-set-alt (cdr set1) (cdr set2))))
                  ((< x1 x2) (cons x1 (union-set-alt (cdr set1) set2)))
                  (else (cons x2 (union-set-alt set1 (cdr set2)))))))))
(define (union-set-alt2 set1 set2)
  (cond ((null? set1) set2)
        ((null? set2) set1)
        (else
         (let ((x1 (car set1))
               (x2 (car set2)))
           (cons (min x1 x2)
                 (union-set-alt2 (if (> x1 x2)
```

```
                                    set1
                                    (cdr set1))
                                (if (> x2 x1)
                                    set2
                                    (cdr set2))))))))

(let ((set1 (enumerate-interval 1 1000))
      (set2 (enumerate-interval 500 1500)))
  (define (test)
    (format #t "~&union-set: ~a"
            (cadr (mattbench2 (λ()(union-set set1 set2)) 10000)))
    (format #t "~&union-set-ordered: ~a"
            (cadr (mattbench2 (λ()(union-set-ordered set1 set2)) 50000)))
    (format #t "~&union-set-alt: ~a"
            (cadr (mattbench2 (λ()(union-set-alt set1 set2)) 500000)))
    (format #t "~&union-set-alt2: ~a"
            (cadr (mattbench2 (λ()(union-set-alt2 set1 set2)) 500000)))
    )

  (test))
```

```
union-set: 4758999.4984
union-set-ordered: 107784.3673
union-set-alt: 45441.7717
union-set-alt2: 66262.4295
```

3.72 Exercise 2.63: binary trees

3.72.1 Text definitions

```
(define (entry tree) (car tree))
(define (left-branch tree) (cadr tree))
(define (right-branch tree) (caddr tree))
(define (make-tree entry left right)
  (list entry left right))
```

3.72.2 Question A

Each of the following two procedures converts a binary tree to a list.

```
<<make-tree>>
(define (tree->list-1 tree)
  (if (null? tree)
      '()
      (append (tree->list-1 (left-branch tree))
              (cons (entry tree)
                    (tree->list-1
                      (right-branch tree))))))
```

```
(define (tree->list-2 tree)
  (define (copy-to-list tree result-list)
    (if (null? tree)
        result-list
        (copy-to-list (left-branch tree)
                      (cons (entry tree)
                            (copy-to-list
                              (right-branch tree)
                              result-list)))))
  (copy-to-list tree '()))
```

1. Do the two procedures produce the same result for every tree? If not, how do the results differ? What lists do the two procedures produce for the trees in Figure 2.16?

3.72.3 Answer A

First let's check whether the arrangement of the input tree impacts the output list. I'd like a couple functions for generating lists.

```
(define (rightward-tree list)
  (if (null? list)
      '()
      (make-tree (car list) '()
                 (rightward-tree (cdr list)))))
(define (leftward-tree list)
  (define (rec list)
    (if (null? list)
        '()
        (make-tree (car list) (rec (cdr list))
                   '())))
  (rec list))
(define (nested-tree list)
  (define (rec list)
    (cond ((= 1 (length list))
           (make-tree (car list) '() '()))
          ((= 2 (length list))
           (make-tree (cadr list)
                      (make-tree (car list) '() '())
                      '()))
          (else
           (let ((halfway (truncate/ (length list)
                                     2)))
                 (make-tree (list-ref list halfway)
                            (rec (list-head list halfway))
                            (rec (list-tail list (1+ halfway))))))))
  (rec list))
```

```
<<echo>>
<<tree-to-list>>
<<tree-makers>>

(let* ((nil '())
       (1-to-4 (list 1 2 3 4))
       (tree-right
        (rightward-tree 1-to-4))
       (tree-left
        (leftward-tree 1-to-4))
       (tree-middle
        (nested-tree (iota 10))))
  (echo "tree-right:" tree-right)
  (echo "tree-right list 1:" (tree->list-1 tree-right))
  (echo "tree-right list 2:" (tree->list-2 tree-right))
  (newline)
  (echo "tree-left:" tree-left)
  (echo "tree-left list 1:" (tree->list-1 tree-left))
  (echo "tree-left list 2:" (tree->list-2 tree-left))
  (newline)
  (echo "tree-middle:" tree-middle)
  (echo "tree-middle list 1:" (tree->list-1 tree-middle))
  (echo "tree-middle list 2:" (tree->list-2 tree-middle)))
```

```
tree-right: (1 () (2 () (3 () (4 () ()))))
tree-right list 1: (1 2 3 4)
tree-right list 2: (1 2 3 4)

tree-left: (1 (2 (3 (4 () ()) ()) ()) ())
tree-left list 1: (4 3 2 1)
tree-left list 2: (4 3 2 1)

tree-middle: (5 (2 (1 (0 () ()) ()) (4 (3 () ()) ())) (8 (7 (6 () ()) ()) (9 () ())))
tree-middle list 1: (0 1 2 3 4 5 6 7 8 9)
tree-middle list 2: (0 1 2 3 4 5 6 7 8 9)
```

So, the lists are the same for both.

Fun fact: Emacs org-babel has suddenly stopped accepting the λ character in source code.

3.72.4 Question B

Do the two procedures have the same order of growth in the number of steps required to convert a balanced tree with n elements to a list? If not, which one grows more slowly?

3.72.5 Answer B

```
;; tree->list-1 evaluating the following tree:
;;     7
;;    / \
;;   3   9
;;  / \   \
;; 1   5   11
(append (append (append nil
                        (cons 1
                              nil))
                (cons 3
                      (append nil
                              (cons 5
                                    nil))))
        (cons 7
              (append nil
                      (cons 9
                            (append nil
                                    (cons 11
                                          nil))))))
;; holy cow that is wasteful. So many appends of nothing.

;; tree->list-2 evaluation (iterative)
(copy-to-list [full tree] nil)
(copy-to-list [tree 3 1 5]
              (cons 7
                    (copy-to-list
                     [tree 9 11]
                     nil)))
(copy-to-list [tree 3 1 5]
              (cons 7
                    (copy-to-list
                     nil
                     (cons 9
                           (copy-to-list
                            [tree 11]
                            nil)))))
(copy-to-list [tree 3 1 5]
              (cons 7
                    (copy-to-list
                     nil
                     (cons 9
                           (copy-to-list
                            nil
                            (cons 11
                                  (copy-to-list nil nil)))))))
(copy-to-list [tree 3 1 5]
              '(7 9 11))
(copy-to-list [tree 1]
```

```
               (cons 3
                     (copy-to-list
                      nil
                      (cons 5
                            (copy-to-list
                             nil
                             '(7 9 11))))))
(copy-to-list
 nil
 (cons 1
       (copy-to-list
        nil
        '(3 5 7 9 11))))
'(1 3 5 7 9 11)
```

So, assuming the evaluation isn't as slow as operations like `append`, the second is definitely faster – it's basically just evaluating to a series of `cons` statements. The internet suggests that the first is $\Theta(n \log n)$ while the second is $\Theta(n)$.

3.73 Exercise 2.64: Making a balanced binary tree

The following procedure `list->tree` converts an ordered list to a balanced binary tree. The helper procedure `partial-tree` takes as arguments an integer n and list of at least n elements and constructs a balanced tree containing the first n elements of the list. The result returned by `partial-tree` is a pair (formed with `cons`) whose `car` is the constructed tree and whose `cdr` is the list of elements not included in the tree.

```
<<make-tree>>
(define (list->tree elements)
  (car (partial-tree elements (length elements))))
(define (partial-tree elts n)
  (if (= n 0)
      (cons '() elts)
      (let ((left-size (quotient (- n 1) 2)))
        (let ((left-result
               (partial-tree elts left-size)))
          (let ((left-tree (car left-result))
                (non-left-elts (cdr left-result))
                (right-size (- n (+ left-size 1))))
            (let ((this-entry (car non-left-elts))
                  (right-result
                   (partial-tree
                    (cdr non-left-elts)
                    right-size)))
              (let ((right-tree (car right-result))
                    (remaining-elts
```

```
                    (cdr right-result)))
                (cons (make-tree this-entry
                                 left-tree
                                 right-tree)
                      remaining-elts))))))))
```

3.73.1 Question A

Write a short paragraph explaining as clearly as you can how `partial-tree` works. Draw the tree produced by `list->tree` for the list `(1 3 5 7 9 11)`.

3.73.2 Answer A

For my own sake, I'll reorganize this with `let*`.

```
(define (partial-tree elts n)
  (if (= n 0)
      (cons '() elts)
      (let* ((left-size (quotient (- n 1) 2))
             (left-result (partial-tree
                           elts
                           left-size))
             (left-tree (car left-result))
             (non-left-elts (cdr left-result))
             (right-size (- n (+ left-size 1)))
             (this-entry (car non-left-elts))
             (right-result (partial-tree
                            (cdr non-left-elts)
                            right-size))
             (right-tree (car right-result))
             (remaining-elts (cdr right-result)))
        (cons (make-tree this-entry
                         left-tree
                         right-tree)
              remaining-elts))))

```

Say we're evaluating `'(1 3 5 7 9 11)`. We define `left-size` as $\lfloor n/2 \rfloor$. `partial-tree` recurses until hitting $n = 0$, at which point it returns the starting list with `'()` prepended to it. This is used as the empty left branch for the second-to-the-bottom row of the tree, in this case for the tree node of `1`. That empty tree branch is saved in `left-tree`, `this-entry` (`1`) is defined, then we recurse to the right branch, finding a bottom to place `3`. Finally, the tree with only 3 is returned, it becomes the right branch off `1`, making `(make-tree 1 '() (make-tree 3 '() '()))`. This gets `cons`'d to the remaining list, `'(5 7 9 11)`. Upon returning the new list, this will become the `left-tree` of its parent process. All in all, this is a recursive algorithm that kind of solves itself with minimal logic. Note its evaluation always comes back to `cons` operations.

This is very elegant. However, I am left wondering how clear this is in comparison with some C routine that does the same thing.



3.73.3 Question B

What is the order of growth in the number of steps required by `list->tree` to convert a list of n elements?

3.73.4 Answer B

I'm going to estimate that it's $\Theta(n)$ because of the `cons` operations being the cornerstone of the procedure. The internet appears to agree in this case.

3.74 Exercise 2.65: Sets as binary trees

3.74.1 Textbook Definitions

```
(define (element-of-set?-tree x set)
  (cond ((null? set) #f)
        ((= x (entry set)) #t)
        ((< x (entry set))
         (element-of-set?-tree
          x
          (left-branch set)))
        ((> x (entry set))
         (element-of-set?-tree
          x
          (right-branch set)))))
(define (adjoin-set-tree x set)
  (cond ((null? set) (make-tree x '() '()))
        ((= x (entry set)) set)
        ((< x (entry set))
         (make-tree
          (entry set)
          (adjoin-set-tree x (left-branch set))
          (right-branch set)))
        ((> x (entry set))
         (make-tree
          (entry set)
          (left-branch set)
          (adjoin-set-tree x (right-branch set))))))
```

3.74.2 Question

Use the results of Exercise 2.63 and Exercise 2.64 to give $\Theta(n)$ implementations of `union-set` and `intersection-set` for sets implemented as (balanced) binary trees.

3.74.3 Answer

Ok, now that's stretching my brain for sure. Hmm...

I think I need to look closer at how `tree->list` works.

```
(define (tree->list-debug tree)
  (define (copy-to-list tree result-list)
    (if (null? tree)
        result-list
        (let ((result
               (begin (echo ">> entering tree at" (entry tree) "results"
↪  result-list)
                      (copy-to-list (left-branch tree)
                                    (begin (echo "we'll cons" (entry
↪  tree))
                                           (let ((pair
                                                  (cons (entry tree)
```

```
                                                    (copy-to-list
                                                     (right-branch
  ↪  tree)
                                                     result-list))))
                                            (echo "  cons'd" pair)
                                            pair))))))
          (echo "<< leaving tree at" (entry tree) "results" result-list)
          result)))
    (copy-to-list tree '()))
```

```
>> entering tree at 5 results ()
we'll cons 5
>> entering tree at 9 results ()
we'll cons 9
>> entering tree at 11 results ()
we'll cons 11
  cons'd (11)
<< leaving tree at 11 results ()
  cons'd (9 11)
>> entering tree at 7 results (9 11)
we'll cons 7
  cons'd (7 9 11)
<< leaving tree at 7 results (9 11)
<< leaving tree at 9 results ()
  cons'd (5 7 9 11)
>> entering tree at 1 results (5 7 9 11)
we'll cons 1
>> entering tree at 3 results (5 7 9 11)
we'll cons 3
  cons'd (3 5 7 9 11)
<< leaving tree at 3 results (5 7 9 11)
  cons'd (1 3 5 7 9 11)
<< leaving tree at 1 results (5 7 9 11)
<< leaving tree at 5 results ()
(1 3 5 7 9 11)
```

What's really holding me up is the $\Theta(n)$ complexity. Since I'm probably not going to figure this out easily, let's try doing it the "wrong" way first, by using previously made algorithms.

```
<<make-tree>>
<<list-to-tree>>
<<tree-to-list>>
<<tree-sets-txt>>

(define (tree->list-debug tree)
  (define (copy-to-list tree result-list)
```

```
    (if (null? tree)
        result-list
        (let ((result
               (begin (echo ">> entering tree at" (entry tree) "results"
↪  result-list)
                      (copy-to-list (left-branch tree)
                                    (begin (echo "we'll cons" (entry
↪  tree))
                                           (let ((pair
                                                  (cons (entry tree)
                                                        (copy-to-list
                                                         (right-branch
↪  tree)
                                                         result-list))))
                                             (echo "  cons'd" pair)
                                             pair))))))
          (echo "<< leaving tree at" (entry tree) "results" result-list)
          result)))
  (copy-to-list tree '()))

(define (union-set-tree-wrong1 set1 set2)
  (define (copy-to-list tree held-tree result-list)
    (define (swap-trees)
      (copy-to-list held-tree tree result-list))
    (define (advance-tree-no-cons)
      (copy-to-list (left-branch tree)
                    held-tree
                    (copy-to-list
                     (right-branch tree)
                     held-tree
                     result-list)))
    (define (advance-tree-cons)
      (copy-to-list (left-branch tree)
                    held-tree
                    (cons (entry tree)
                          (copy-to-list
                           (right-branch tree)
                           held-tree
                           result-list))))
    (define (advance-both-cons)
      (copy-to-list (left-branch tree)
                    held-tree
                    (cons (entry tree)
                          (copy-to-list
                           (right-branch tree)
                           held-tree
                           result-list))))
    (cond ((and (null? tree)
```

```
                (null? held-tree))
           result-list)
          ((null? tree)
           (swap-trees))
          ((= (entry tree)
              (car result-list))
           (advance-tree-no-cons))
          ((= (entry tree) (entry held-tree))
           (advance-both-cons))
          ((< (entry tree) (entry held-tree))
           (advance-tree-cons))
          ((> (entry tree) (entry held-tree))
           (swap-trees))))
  (copy-to-list set1 set2 '()))

(define (union-set-tree-wrong2 set1 set2)
  (define (copy-to-tree tree result-tree)
    (if (null? tree)
        result-tree
        (copy-to-tree (left-branch tree)
                      (adjoin-set-tree (entry tree)
                                       (copy-to-tree
                                        (right-branch tree)
                                        result-tree)))))
  (copy-to-tree set1 set2))

(define (intersection-set-tree-wrong1 set1 set2)
  (define (copy-to-tree tree result-tree)
    (if (null? tree)
        result-tree
        (let ((adjoin-or-not
               (lambda (rest)
                 (if (element-of-set?-tree (entry tree)
                                           set2)
                     (adjoin-set-tree (entry tree)
                                      rest)
                     rest))))
          (copy-to-tree (left-branch tree)
                        (adjoin-or-not
                         (copy-to-tree
                          (right-branch tree)
                          result-tree))))))
    (copy-to-tree set1 '()))
```

```
<<sets-as-trees>>
<<echo>>
(load "mattcheck2.scm")
(let ((set1 (list->tree '(1 3 5 7 9 11)))
```

```
;; (5 (1 () (3 () ())) (9 (7 () ()) (11 () ())))
      (set2 (list->tree '(2 4 6 7 8 11)))
;; (6 (2 () (4 () ())) (8 (7 () ()) (11 () ())))
      (union (list->tree '(1 2 3 4 5 6 7 8 9 11)))
;; (5 (2 (1 () ()) (3 () (4 () ()))) (8 (6 () (7 () ())) (9 () (11 ()
↪  ()))))
      (union-unbal '(6 (2 (1 () ()) (4 (3 () ()) (5 () ()))) (8 (7 ()
↪  ()) (11 (9 () ()) ()))))
      (int-unbal '(11 (7 () ()) ()))
      (int (list->tree '(7 11))))
;; (7 () (11 () ()))
  (mattcheck "element-of-set?-tree true"
             (element-of-set?-tree 1 set1)
             #t)
  (mattcheck "element-of-set?-tree true"
             (element-of-set?-tree 7 set1)
             #t)
  (mattcheck "element-of-set?-tree false"
             (element-of-set?-tree 2 set1)
             #f)
  (mattcheck "union-set-tree-wrong2"
             (union-set-tree-wrong2 set1 set2)
             union-unbal)
  (mattcheck "intersection-set-tree-wrong1"
             (intersection-set-tree-wrong1 set1 set2)
             int-unbal)
  (echo (tree->list-debug set1)))
```

```
<unknown-location>: warning: possibly unbound variable `mattcheck'
SUCCEED at element-of-set?-tree true
SUCCEED at element-of-set?-tree true
SUCCEED at element-of-set?-tree false
SUCCEED at union-set-tree-wrong2
SUCCEED at intersection-set-tree-wrong1
```

I can't really think of a better way to do it. Time to look up the answer!

And... To my surprise, the answer most internet people have given is what I called the "wrong" one. In their case, using `tree->list` and `list->tree` to apply the list-based `union-set` and `intersection-set`, saying it's $\Theta(n)$. I also notice that they didn't use `element-of-set?` and `adjoin-to-set` like I did.

Let's compare.

```
(use-modules (ice-9 format))
(load "../mattbench.scm")
<<sets-txt>>
<<sets-as-trees>>
<<enumerate-interval>>
(define tree->list tree->list-2)
```

```
(define (union-set set1 set2)
  (cond ((or (null? set1) (null? set2))
         set2)
        ((element-of-set? (car set1) set2)
         (union-set (cdr set1)
                    set2))
        (else (cons (car set1)
                    (union-set (cdr set1)
                               set2)))))

;; from http://community.schemewiki.org/?sicp-ex-2.65
(define (union-set-l2t set1 set2)
  (cond ((null? set1) set2)
        ((null? set2) set1)
        (else (list->tree (union-set (tree->list set1) (tree->list
↪  set2))))))

(define (intersection-set-l2t set1 set2)
  (cond ((null? set1) '())
        ((null? set2) '())
        (else (list->tree (intersection-set (tree->list set1)
↪  (tree->list set2))))))

(define (make-2sets max)
  (let* ((a-start 0)
         (a-end 0.6)
         (b-start 0.4)
         (b-end 1))
    (cons (list->tree
           (enumerate-interval
            0
            (inexact->exact (* a-end max))))
          (list->tree
           (enumerate-interval
            (inexact->exact (* b-start max))
            (inexact->exact (* b-end max)))))))

(define (test-with 2sets repeats)
  (format #t "~&union-set-tree-wrong2: ~a"
          (cadr (mattbench2
                 (λ()(union-set-tree-wrong2
                      (car 2sets)
                      (cdr 2sets)))
                 repeats)))
  (format #t "~&union-set-l2t: ~a"
          (cadr (mattbench2
                 (λ()(union-set-l2t (car 2sets)
                                    (cdr 2sets)))
```

```
                repeats)))
  (format #t "~&intersection-set-tree-wrong1: ~a"
          (cadr (mattbench2
                 (λ()(intersection-set-tree-wrong1
                      (car 2sets)
                      (cdr 2sets)))
                 repeats)))
  (format #t "~&intersection-set-l2t: ~a"
          (cadr (mattbench2
                 (λ()(intersection-set-l2t (car 2sets)
                                           (cdr 2sets)))
                 repeats))))
(format #t "~&~a" (make-2sets 10))
(define base-repeats 10000)
(test-with (make-2sets 100)
           base-repeats)
(test-with (make-2sets 1000)
           (/ base-repeats 2))
(test-with (make-2sets 10000)
           (/ base-repeats 4))
```

```
union-set-tree-wrong2: 159831.3273
union-set-l2t: 49633.5962
intersection-set-tree-wrong1: 42577.4905
intersection-set-l2t: 33261.5268
union-set-tree-wrong2: 12648028.8732
union-set-l2t: 2196015.1536
intersection-set-tree-wrong1: 3165935.4476
intersection-set-l2t: 2015603.052
union-set-tree-wrong2: 1393227860.116
union-set-l2t: 195546567.7616
intersection-set-tree-wrong1: 361923468.7844
intersection-set-l2t: 195366120.9524
```



Not really enough datapoints, but you can at least see that the `l2t` variants are faster.

3.75 Exercise 2.66: binary tree `lookup`

3.75.1 Question

Implement the `lookup` procedure for the case where the set of records is structured as a binary tree, ordered by the numerical values of the keys.

3.75.2 Answer

First let's define what these records look like.

```
(define (make-record key value)
  (cons key value))
(define (key record)
  (car record))
(define (value record)
  (cdr record))
```

```
<<make-tree>>
<<make-record>>
(define (lookup given-key tree-of-records)
  (cond ((null? tree-of-records) #f)
        ((= given-key
```

```
            (key (entry tree-of-records)))
          (entry tree-of-records))
         ((< given-key
             (key (entry tree-of-records)))
          (lookup
           given-key
           (left-branch tree-of-records)))
         ((> given-key
             (key (entry tree-of-records)))
          (lookup
           given-key
           (right-branch tree-of-records)))))
```

```
<<lookup-tree>>
<<list-to-tree>>
<<echo>>
(load "mattcheck2.scm")
(let ((set1 (list->tree
             (list (make-record 1 'a)
                   (make-record 3 'b)
                   (make-record 5 'c)
                   (make-record 7 'd)
                   (make-record 9 'e)
                   (make-record 11 'f)))))
  (mattcheck "lookup retrieves correct record"
             (lookup 9 set1)
             (make-record 9 'e)))
```

```
SUCCEED at lookup retrieves correct record
```

3.76 Exercise 2.67: decoding Huffman tree messages

3.76.1 Text definitions

```
;; huffman-trees-txt
(define (make-leaf symbol weight)
  (list 'leaf
        symbol
        weight))
(define (leaf? object)
  (eq? (car object)
       'leaf))
(define (symbol-leaf x)
  (cadr x))
(define (weight-leaf x)
  (caddr x))

(define (make-code-tree left right)
```

```
  (list left
        right
        (append (symbols left) (symbols right))
        (+ (weight left) (weight right))))

(define (left-branch  tree)
  (car tree))
(define (right-branch tree)
  (cadr tree))
(define (symbols tree)
  (if (leaf? tree)
      (list (symbol-leaf tree))
      (caddr tree)))
(define (weight tree)
  (if (leaf? tree)
      (weight-leaf tree)
      (cadddr tree)))
(define (decode bits tree)
  (define (decode-1 bits current-branch)
    (if (null? bits)
        '()
        (let ((next-branch
               (choose-branch (car bits) current-branch)))
          (if (leaf? next-branch)
              (cons (symbol-leaf next-branch)
                    (decode-1 (cdr bits) tree))
              (decode-1 (cdr bits) next-branch)))))
  (decode-1 bits tree))
(define (choose-branch bit branch)
  (cond ((= bit 0) (left-branch branch))
        ((= bit 1) (right-branch branch))
        (else (error "bad bit: CHOOSE-BRANCH" bit))))
(define (adjoin-set x set)
  (cond ((null? set) (list x))
        ((< (weight x) (weight (car set))) (cons x set))
        (else (cons (car set)
                    (adjoin-set x (cdr set))))))
(define (make-leaf-set pairs)
  (if (null? pairs)
      '()
      (let ((pair (car pairs)))
        (adjoin-set (make-leaf (car pair)
                               (cadr pair))
                    (make-leaf-set (cdr pairs))))))
```

3.76.2 Question

Define an encoding tree and a sample message. Use the `decode` procedure to decode the message, and give the result.

3.76.3 Answer

```
<<huffman-trees-txt>>
<<echo>>
(define sample-tree
  (make-code-tree (make-leaf 'A 4)
                  (make-code-tree
                   (make-leaf 'B 2)
                   (make-code-tree
                    (make-leaf 'D 1)
                    (make-leaf 'C 1)))))
(define whos-on-first-tree
  (make-code-tree
   (make-leaf 'who 8)
                  (make-code-tree
                   (make-leaf 'what 4)
                   (make-code-tree
                    (make-leaf 'Idontknow 2)
                    (make-leaf 'why 2)))))
(define sample-message '(0 1 1 0 0 1 0 1 0 1 1 1 0))
(define whos-on-first-message '(0 1 0 1 1 0 1 1 1))
(echo (decode sample-message sample-tree))
(echo (decode whos-on-first-message whos-on-first-tree))
```

```
(A D A B B C A)
(who what Idontknow why)
```

3.77 Exercise 2.68: encoding Huffman tree messages

3.77.1 Question

The `encode` procedure takes as arguments a message and a tree and produces the list of bits that gives the encoded message.

```
;; encode-txt
(define (encode message tree)
  (if (null? message)
      '()
      (append (encode-symbol (car message) tree)
              (encode (cdr message) tree))))
```

`encode-symbol` is a procedure, which you must write, that returns the list of bits that encodes a given symbol according to a given tree. You should design `encode-symbol` so that it signals an error if the symbol is not in the tree at all. Test your procedure by encoding the result you obtained in Exercise 2.67 with the sample tree and seeing whether it is the same as the original sample message.

3.77.2 Answer

```
;; encode-symbol
(define element-of-set? member)
(define (encode-symbol symbol tree)
  (define (rec t)
    (cond ((and (leaf? t)
                (eq? (symbol-leaf t)
                     symbol))
           '())
          ((element-of-set? symbol
                            (symbols (left-branch t)))
           (cons 0
                 (rec (left-branch t))))
          ((element-of-set? symbol
                            (symbols (right-branch t)))
           (cons 1
                 (rec (right-branch t))))
          (else (error "encode-symbol: logic error"))))
  (if (element-of-set? symbol (symbols tree))
      (rec tree)
      (error "encode-symbol: symbol not in tree")))
```

```
<<huffman-trees-txt>>
<<encode-txt>>
<<encode-symbol>>
(load "mattcheck2.scm")
(define sample-tree
  (make-code-tree (make-leaf 'A 4)
                  (make-code-tree
                   (make-leaf 'B 2)
                   (make-code-tree
                    (make-leaf 'D 1)
                    (make-leaf 'C 1)))))
(define whos-on-first-tree
  (make-code-tree
   (make-leaf 'who 8)
                  (make-code-tree
                   (make-leaf 'what 4)
                   (make-code-tree
                    (make-leaf 'Idontknow 2)
                    (make-leaf 'why 2)))))
(define sample-message '(0 1 1 0 0 1 0 1 0 1 1 1 0))
(define whos-on-first-message '(0 1 0 1 1 0 1 1 1))
(mattcheck "sample encoded message"
           sample-message
           (encode '(A D A B B C A)
                   sample-tree))
```

```
(mattcheck "who's on first encoded message"
           whos-on-first-message
           (encode '(who what Idontknow why)
                   whos-on-first-tree))
```

```
SUCCEED at sample encoded message
SUCCEED at who's on first encoded message
```

3.78 Exercise 2.69: Generating Huffman trees

3.78.1 Question

The following procedure takes as its argument a list of symbol-frequency pairs (where no symbol appears in more than one pair) and generates a Huffman encoding tree according to the Huffman algorithm.

```
(define (generate-huffman-tree pairs)
  (successive-merge (make-leaf-set pairs)))
```

`make-leaf-set` is the procedure given above that transforms the list of pairs into an ordered set of leaves. `successive-merge` is the procedure you must write, using `make-code-tree` to successively merge the smallest-weight elements of the set until there is only one element left, which is the desired Huffman tree. (This procedure is slightly tricky, but not really complicated. If you find yourself designing a complex procedure, then you are almost certainly doing something wrong. You can take significant advantage of the fact that we are using an ordered set representation.)

3.78.2 Answer

This one took a while. The whole in-place-reorganization of the list seems tricky, since adding branches together makes a new branch which is higher value than the later entries. In a C language I'd be solving the problem by "moving" a play-head forward and back, but here I'll have to coerce the evaluator into doing it.

```
;; generate-huffman-tree
(use-srfis '(1))
(define element-of-set? member)
(define (generate-huffman-tree pairs)
  (successive-merge (make-leaf-set pairs)))

(define (successivemerge-iter delayed leaves)
  (let ((dl (length delayed))
        (ll (length leaves)))
    (cond ((= ll 0)
           (successivemerge-iter '()
                                 delayed))
```

```
          ((= dl 0)
           (if (= ll 1)
               (car leaves)
               (successivemerge-iter (cons (car leaves) '())
                                     (cdr leaves))))
          ((and (= ll 1)
                (= dl 1))
           (make-code-tree (car delayed)
                           (car leaves)))
          ((<= (weight (car delayed))
               (weight (car leaves)))
           (successivemerge-iter (cdr delayed)
                                 (cons (make-code-tree (car delayed)
                                                       (car leaves))
                                       (cdr leaves))))
          ((> (weight (car delayed))
              (weight (car leaves)))
           (successivemerge-iter (cons (car leaves)
                                       delayed)
                                 (cdr leaves))))))

  (define (successive-merge leaves)
    (successivemerge-iter (cons (car leaves) '())
                          (cdr leaves)))
```

```
<<huffman-trees-txt>>
<<encode-txt>>
<<encode-symbol>>
<<generate-huffman-tree>>
<<echo>>
(load "mattcheck2.scm")

(let ((pairs
       (make-leaf-set '((A 8)(B 3)(C 1)(D 1)
                        (E 1)(F 1)(G 1)(H 1))))
      (answer
       (make-code-tree (make-leaf 'A 8)
                       (make-code-tree
                        (make-code-tree
                         (make-leaf 'B 3)
                         (make-code-tree
                          (make-leaf 'C 1)
                          (make-leaf 'D 1)))
                        (make-code-tree
                         (make-code-tree
                          (make-leaf 'E 1)
                          (make-leaf 'F 1))
                         (make-code-tree
```

```
                          (make-leaf 'G 1)
                          (make-leaf 'H 1)))))))
  (mattcheck "successive-merge"
             (successive-merge pairs)
             answer)
  (echo (successive-merge pairs)))
```

```
<unknown-location>: warning: possibly unbound variable `mattcheck'
FAIL at successive-merge
expected: ((leaf A 8) (((leaf B 3) ((leaf C 1) (leaf D 1) (C D) 2) (B C D) 5) (# …) …) …)
returned: (((((leaf H 1) (leaf G 1) (H G) 2) ((leaf F 1) (leaf E 1) (F E) 2) (…) …) …) …)
(((((leaf H 1) (leaf G 1) (H G) 2) ((leaf F 1) (leaf E 1) (F E) 2) (H G F E) 4)
  (((leaf D 1) (leaf C 1) (D C) 2) (leaf B 3) (D C B) 5) (H G F E D C B) 9)
    (leaf A 8) (H G F E D C B A) 17)
```

And the tree is *backwards*. But it is correct. Sigh.

Time to be embarrassed by the internet's solution.

```
;; see 2.67
(define (adjoin-set x set)
  (cond ((null? set) (list x))
        ((< (weight x) (weight (car set)))
         (cons x set))
        (else (cons (car set)
                    (adjoin-set x (cdr set))))))
;; for my own edification
(define (adjoin-set-iter x set)
  (define (iter less more)
    (cond ((null? more)
           (reverse (cons x less)))
          ((< (weight x) (weight (car more)))
           (append (reverse (cons x less))
                   more))
          (else (adjoin-set-iter (cons (car more) less)
                                 (cdr more)))))

  (iter '() set))

;; https://codereview.stackexchange.com/a/117980
(define (successive-merge-small leaves)
  (if (null? (cdr leaves))
      (car leaves)
      (successive-merge-small
       (adjoin-set
        (make-code-tree (car leaves)
                        (cadr leaves))
        (cddr leaves)))))
```

Comparing my above code with the code for `adjoin-set`, mine is definitely unnecessarily complicated. How does the performance stack up?

```
Mine: 4453.0773347
Theirs: 4554.0141235
```

So for all that unnecessary complexity it's still about the same.

The lesson I keep failing to learn is remembering past work (i.e. `adjoin-set`), since this book loves to integrate past exercises.

3.79 Exercise 2.70

3.79.1 Question

The following eight-symbol alphabet with associated relative frequencies was designed to efficiently encode the lyrics of 1950s rock songs. (Note that the symbols of an alphabet need not be individual letters.)

A	2	NA	16
BOOM	1	SHA	3
GET	2	YIP	9
JOB	2	WAH	1

Use `generate-huffman-tree` to generate a corresponding Huffman tree, and use `encode` to encode the following message:

```
Get a job
Sha na na na na na na na na

Get a job
Sha na na na na na na na na

Wah yip yip yip yip
yip yip yip yip yip
Sha boom
```

How many bits are required for the encoding? What is the smallest number of bits that would be needed to encode this song if we used a fixed-length code for the eight-symbol alphabet?

3.79.2 Answer

```
(define hippie-set
  '((NA 16)(YIP 9)(SHA 3)(A 2)
    (GET 2)(JOB 2)(BOOM 1)(WAH 1)))
(define hippie-tree
  (generate-huffman-tree hippie-set))
(define hippie-plaintext
  '(GET A JOB
    SHA NA NA NA NA NA NA NA NA
    GET A JOB
    SHA NA NA NA NA NA NA NA NA
```

```
    WAH YIP YIP YIP YIP
    YIP YIP YIP YIP YIP
    SHA BOOM))
(define hippie-encoded
  (encode hippie-plaintext
          hippie-tree))
```

```
<<huffman-trees-txt>>
<<encode-txt>>
<<encode-symbol>>
<<generate-huffman-tree>>
(define (successive-merge leaves)
  (if (null? (cdr leaves))
      (car leaves)
      (successive-merge
       (adjoin-set
        (make-code-tree (car leaves)
                        (cadr leaves))
        (cddr leaves)))))
<<hippie-trees>>
<<echo>>
(load "mattcheck2.scm")

(echo "Length of unencoded message is" (length hippie-plaintext)
↪  "words.")
(echo "Length of encoded message is" (length hippie-encoded) "bits")
(echo "Unencoded message (newlines added):" (decode hippie-encoded
↪  hippie-tree))
```

```
Length of unencoded message is 36 words.
Length of encoded message is 84 bits
Unencoded message (newlines added):
(GET A JOB
SHA NA NA NA NA NA NA NA NA
GET A JOB SHA NA NA NA NA NA NA NA NA
WAH YIP YIP YIP YIP
YIP YIP YIP YIP YIP
SHA BOOM)
```

A fixed-length alphabet would take 3 bits per word, or 108 bits. So the VLE is 77% the size of anything else.

```
<<huffman-trees-txt>>
<<encode-txt>>
<<encode-symbol>>
<<generate-huffman-tree>>
(define (successive-merge leaves)
```

```
  (if (null? (cdr leaves))
      (car leaves)
      (successive-merge
       (adjoin-set
        (make-code-tree (car leaves)
                        (cadr leaves))
        (cddr leaves)))))
<<hippie-trees>>
(define (split-list-every-x list x)
  (define (rec ll)
    (if (< (length ll) x)
        (cons ll '())
        (cons (list-head ll x)
              (rec (list-tail ll x)))))
  (rec list))
(split-list-every-x hippie-encoded 7)
```

1	1	1	1	1	1	1
0	0	1	1	1	1	0
1	1	1	0	0	0	0
0	0	0	0	0	1	1
1	1	1	1	1	0	0
1	1	1	1	0	1	1
1	0	0	0	0	0	0
0	0	0	1	1	0	1
0	1	0	1	0	1	0
1	0	1	0	1	0	1
0	1	0	1	0	1	1
1	0	1	1	0	1	1



3.80 Exercise 2.71

3.80.1 Questions

Suppose we have a Huffman tree for an alphabet of n symbols, and that the relative frequencies of the symbols are $1, 2, 4, \ldots, 2^{n-1}$. Sketch the tree for $n = 5$; for $n = 10$. In such a tree (for general n) how many bits are required to encode the most frequent symbol? The least frequent symbol?

3.80.2 Answers

In this type of tree, the most frequent symbol takes 1 bit. The nth symbol takes n bits. If it's the least frequent symbol and n is even, it takes $n - 1$ bits.