

# SICP Chapter 1 Answers

ProducerMatt

## HOW THIS DOCUMENT IS MADE

### TODO

```
1 (define (foo a b)
2   (+ a (* 2 b)))
3
4 (foo 5 3)
```

11

^ Dynamically evaluated when you press "enter" on the BEGIN\_SRC block!

### Also consider:

- `:results` output for what the code prints
- `:exports` code or `:exports results` to just get one or the other

$a + (\pi \times b)$  <~ inline Latex btw :)

### Current command for conversion

```
pandoc --from org --to latex 1.org -o 1.tex -s; xelatex 1.tex
```

## Helpers for org-mode tables

### try-these

Takes function `f` and list `testvals` and applies `f` to each item `i`. For each `i` returns a list with `i` and the result. Useful for making tables with a column for input and a column for output.

```
1 ;; Surely this could be less nightmarish
2 (define (try-these f . testvals)
3   (let ((l (if (and (= 1 (length testvals))
4                     (list? (car testvals)))
5                 (car testvals)
6                 testvals)))
7     (map (lambda (i) (cons i
```

```

8             (cons (if (list? i)
9                     (apply f i)
10                    (f i))
11                  #nil)))
12         l)))

```

### transpose-list

"Rotate" a list, for example from '(1 2 3) to '('(1) '(2) '(3))

```

1 (define (transpose-list l)
2   (map list l))

```

### print-as-rows

For manually printing items in rows to stdout. Not currently used.

```

1 (define (p-nl a)
2   (display a)
3   (newline))
4 (define (print-spaced args)
5   (let ((a (car args))
6         (d (cdr args)))
7     (if (null? d)
8         (p-nl a)
9         (begin (display a)
10                (display " ")
11                (print-spaced d))))))
12 (define (print-as-rows . args)
13   (let ((a (car args))
14         (d (cdr args)))
15     (if (list? a)
16         (if (= 1 (length args))
17             (apply print-as-rows a)
18             (print-spaced a))
19         (p-nl a))
20     (if (null? d)
21         '()
22         (apply print-as-rows d))))

```

## Exercise 1.1

### Question

Below is a sequence of expressions. What is the result printed by the interpreter in response to each expression? Assume that the sequence is to be evaluated in the order in which it is presented.

## Answer

```
1 10 ;; 10
2 (+ 5 3 4) ;; 12
3 (- 9 1) ;; 8
4 (/ 6 2) ;; 3
5 (+ (* 2 4) (- 4 6)) ;; 6
6 (define a 3) ;; a=3
7 (define b (+ a 1)) ;; b=4
8 (+ a b (* a b)) ;; 19
9 (= a b) ;; false
10 (if (and (> b a) (< b (* a b)))
11     b
12     a) ;; 4
13 (cond ((= a 4) 6)
14       ((= b 4) (+ 6 7 a))
15       (else 25)) ;; 16
16 (+ 2 (if (> b a) b a)) ;; 6
17 (* (cond ((> a b) a)
18      ((< a b) b)
19      (else -1))
20    (+ a 1)) ;; 16
```

## Exercise 1.2

### Question

Translate the following expression into prefix form:

$$\frac{5 + 2 + (2 - 3 - (6 + \frac{4}{5}))}{3(6 - 2)(2 - 7)}$$

### Answer

```
1 (/ (+ 5 2 (- 2 3 (+ 6 (/ 4 5)))))
2    (* 3 (- 6 2) (- 2 7)))
```

1/75

## Exercise 1.3

### Text

```
1 (define (square x)
2   (* x x))
```

## Question

Define a procedure that takes three numbers as arguments and returns the sum of the squares of the two larger numbers.

## Answer

```
1 <<square>>
2 (define (sum-square x y)
3   (+ (square x) (square y)))
4 (define (square-2of3 a b c)
5   (cond ((and (>= a b) (>= b c)) (sum-square a b))
6         ((and (>= a b) (> c b)) (sum-square a c))
7         (else (sum-square b c)))

1 <<EX1-3>>
2 <<try-these>>
3 (try-these square-2of3 '(7 5 3)
4                          '(7 3 5)
5                          '(3 5 7))
```

---

(7 5 3)	74
(7 3 5)	74
(3 5 7)	74

---

## Exercise 1.4

### Question

Observe that our model of evaluation allows for combinations whose operators are compound expressions. Use this observation to describe the behavior of the following procedure:

```
1 (define (a-plus-abs-b a b)
2   ((if (> b 0) + -) a b))
```

### Answer

This code accepts the variables `a` and `b`, and if `b` is positive, it adds `a` and `b`. However, if `b` is zero or negative, it subtracts them. This decision is made by using the `+` and `-` procedures as the results of an `if` expression, and then evaluating according to the results of that expression. This is in contrast to a language like Python, which would do something like this:

```
if b > 0: a + b
else: a - b
```

## Exercise 1.5

### Question

Ben Bitdiddle has invented a test to determine whether the interpreter he is faced with is using applicative-order evaluation or normal-order evaluation. He defines the following two procedures:

```
1 (define (p) (p))
2
3 (define (test x y)
4   (if (= x 0)
5       0
6       y))
```

Then he evaluates the expression

```
1 (test 0 (p))
```

What behavior will Ben observe with an interpreter that uses applicative-order evaluation? What behavior will he observe with an interpreter that uses normal-order evaluation? Explain your answer. (Assume that the evaluation rule for the special form `if` is the same whether the interpreter is using normal or applicative order: The predicate expression is evaluated first, and the result determines whether to evaluate the consequent or the alternative expression.)

### Answer

In either type of language, `(define (p) (p))` is an infinite loop. However, a normal-order language will encounter the special form, return `0`, and never evaluate `(p)`. An applicative-order language evaluates the arguments to `(test 0 (p))`, thus triggering the infinite loop.

## Exercise 1.6

### Text code

```
1 (define (abs x)
2   (if (< x 0)
3       (- x)
4       x))

1 (define (average x y)
2   (/ (+ x y) 2))

1 <<average>>
2 (define (improve guess x)
3   (average guess (/ x guess)))
4
```

```

5 <<square>>
6 <<abs>>
7 (define (good-enough? guess x)
8   (< (abs (- (square guess) x)) 0.001))
9
10 (define (sqrt-iter guess x)
11   (if (good-enough? guess x)
12       guess
13       (sqrt-iter (improve guess x) x)))
14
15 (define (sqrt x)
16   (sqrt-iter 1.0 x))

```

## Question

Exercise 1.6: Alyssa P. Hacker doesn't see why `if` needs to be provided as a special form. "Why can't I just define it as an ordinary procedure in terms of `cond`?" she asks. Alyssa's friend Eva Lu Ator claims this can indeed be done, and she defines a new version of `if`:

```

1 (define (new-if predicate
2               then-clause
3               else-clause)
4   (cond (predicate then-clause)
5         (else else-clause)))

```

Eva demonstrates the program for Alyssa:

```

1 (new-if (= 2 3) 0 5)
2 ;; => 5
3
4 (new-if (= 1 1) 0 5)
5 ;; => 0

```

Delighted, Alyssa uses `new-if` to rewrite the square-root program:

```

1 (define (sqrt-iter guess x)
2   (new-if (good-enough? guess x)
3           guess
4           (sqrt-iter (improve guess x) x)))

```

What happens when Alyssa attempts to use this to compute square roots? Explain.

## Answer

Using Alyssa's `new-if` leads to an infinite loop because the recursive call to `sqrt-iter` is evaluated before the actual call to `new-if`. This is because `if` and

cond are special forms that change the way evaluation is handled; whichever branch is chosen leaves the other branches unevaluated.

## Exercise 1.7

### Text

```
1 (define (mean-square x y)
2   (average (square x) (square y)))
```

### Question

The good-enough? test used in computing square roots will not be very effective for finding the square roots of very small numbers. Also, in real computers, arithmetic operations are almost always performed with limited precision. This makes our test inadequate for very large numbers. Explain these statements, with examples showing how the test fails for small and large numbers. An alternative strategy for implementing good-enough? is to watch how guess changes from one iteration to the next and to stop when the change is a very small fraction of the guess. Design a square-root procedure that uses this kind of end test. Does this work better for small and large numbers?

### Diary

#### Solving

My original answer was this, which compares the previous iteration until the new and old are within an arbitrary  $dx$ .

```
1 <<txt-sqrt>>
2 (define (inferior-good-enough? guess lastguess)
3   (<=
4     (abs (-
5           (/ lastguess guess)
6             1))
7     0.00000000000001)) ; dx
8 (define (new-sqrt-iter guess x lastguess) ;; Memory of previous value
9   (if (inferior-good-enough? guess lastguess)
10       guess
11       (new-sqrt-iter (improve guess x) x guess)))
12 (define (new-sqrt x)
13   (new-sqrt-iter 1.0 x 0))
```

This solution can correctly find small and large numbers:

```
1 <<inferior-good-enough>>
2 (new-sqrt 1000000000000000)
```

3162277.6601683795

```
1 <<try-these>>
2 <<inferior-good-enough>>
3 (try-these new-sqrt '(0.01 0.0001 0.000001 0.00000001 0.0000000001))
```

0.01	0.1
0.0001	0.01
1e-06	0.001
1e-08	9.999999999999999e-05
1e-10	9.999999999999999e-06

However, I found this solution online that isn't just simpler but automatically reaches the precision limit of the system:

```
1 <<txt-sqrt>>
2 (define (best-good-enough? guess x)
3   (= (improve guess x) guess))
```

### Improving (sqrt) by avoiding extra (improve) call

#### 1. Non-optimized

```
1 (use-modules (ice-9 format))
2 (load "../mattbench.scm")
3 (define (average x y)
4   (/ (+ x y) 2))
5 (define (improve guess x)
6   (average guess (/ x guess)))
7 (define (good-enough? guess x)
8   (= (improve guess x) guess)) ;; improve call 1
9 (define (sqrt-iter guess x)
10  (if (good-enough? guess x)
11      guess
12      (sqrt-iter (improve guess x) x))) ;; call 2
13 (define (sqrt x)
14  (sqrt-iter 1.0 x))
15 (newline)
16 (display (mattbench (lambda () (sqrt 69420)) 4000000000))
17 (newline)
18 ;; 4731.30 <- Benchmark results
```

#### 2. Optimized

```
1 (use-modules (ice-9 format))
2 (load "../mattbench.scm")
3 (define (average x y)
```



```

4      (/ (+ x y) 2))
5      (define (improve guess x)
6        (average guess (/ x guess)))
7      (define (good-enough? guess nextguess x)
8        (= nextguess guess))
9      (define (sqrt-iter guess x)
10       (let ((nextguess (improve guess x)))
11         (if (good-enough? guess nextguess x)
12             guess
13             (sqrt-iter nextguess x))))
14      (define (sqrt x)
15        (sqrt-iter 1.0 x))
16      (newline)
17      (display (mattbench (lambda () (sqrt 69420)) 400000000))
18      (newline)

```

### 3. Benchmark results

Unoptimized	4731.30
Optimized	2518.44

## Answer

The current method has decreasing accuracy with smaller numbers. Notice the steady divergence from correct answers here (should be decreasing powers of 0.1):

```

1 <<txt-sqrt>>
2 <<try-these>>
3 (try-these sqrt 0.01 0.0001 0.000001 0.00000001 0.0000000001)

```

0.01	0.10032578510960605
0.0001	0.03230844833048122
1e-06	0.031260655525445276
1e-08	0.03125010656242753
1e-10	0.03125000106562499

And for larger numbers, an infinite loop will eventually be reached.  $10^{12}$  can resolve, but  $10^{13}$  cannot.

```

1 <<txt-sqrt>>
2 (sqrt 1000000000000)

```

1000000.0

So, my definition of sqrt:

```

1 <<average>>
2 (define (improve guess x)
3   (average guess (/ x guess)))
4 (define (good-enough? guess x)
5   (= (improve guess x) guess))
6 (define (sqrt-iter guess x)
7   (if (good-enough? guess x)
8       guess
9       (sqrt-iter (improve guess x) x)))
10 (define (sqrt x)
11   (sqrt-iter 1.0 x))

1 <<try-these>>
2 <<sqrt>>
3 (try-these sqrt '(0.01 0.0001 0.000001 0.00000001 0.0000000001))

```

0.01	0.1
0.0001	0.01
1e-06	0.001
1e-08	9.999999999999999e-05
1e-10	9.999999999999999e-06

## Exercise 1.8

### Question

Newton's method for cube roots is based on the fact that if  $y$  is an approximation to the cube root of  $x$ , then a better approximation is given by the value:

$$\frac{\frac{x}{y^2} + 2y}{3} \quad (1)$$

Use this formula to implement a cube-root procedure analogous to the square-root procedure. (In 1.3.4 we will see how to implement Newton's method in general as an abstraction of these square-root and cube-root procedures.)

### Diary

My first attempt works, but needs an arbitrary limit to stop infinite loops:

```

1 <<square>>
2 <<try-these>>
3 (define (cb-good-enough? guess x)
4   (= (cb-improve guess x) guess))
5 (define (cb-improve guess x)
6   (/

```

```

7      (+
8        (/ x (square guess))
9        (* guess 2))
10     3))
11 (define (cbrt-iter guess x counter)
12   (if (or (cb-good-enough? guess x) (> counter 100))
13       guess
14       (begin
15         (cbrt-iter (cb-improve guess x) x (+ 1 counter))))))
16 (define (cbrt x)
17   (cbrt-iter 1.0 x 0))
18
19 (try-these cbrt 7 32 56 100)

```

---

7	1.912931182772389
32	3.174802103936399
56	3.825862365544778
100	4.641588833612779

---

However, this will hang on an infinite loop when trying to run `(cbrt 100)`. I speculate it's a floating point precision issue with the "improve" algorithm. So to avoid it I'll just keep track of the last guess and stop improving when there's no more change occurring. Also while researching I discovered that (again due to floating point) `(cbrt -2)` loops forever unless you initialize your guess with a slightly different value, so let's do 1.1 instead.

## Answer

```

1 <<square>>
2 (define (cb-good-enough? nextguess guess lastguess x)
3   (or (= nextguess guess)
4       (= nextguess lastguess)))
5 (define (cb-improve guess x)
6   (/
7     (+
8       (/ x (square guess))
9       (* guess 2))
10    3))
11 (define (cbrt-iter guess lastguess x)
12   (define nextguess (cb-improve guess x))
13   (if (cb-good-enough? nextguess guess lastguess x)
14       nextguess
15       (cbrt-iter nextguess guess x)))
16 (define (cbrt x)
17   (cbrt-iter 1.1 9999 x))

```

```

1 <<cbrr>>
2 <<try-these>>
3 (try-these cbrr 7 32 56 100 -2)

```

---

7	1.912931182772389
32	3.174802103936399
56	3.825862365544778
100	4.641588833612779
-2	-1.2599210498948732

---

## Exercise 1.9

### Question

Each of the following two procedures defines a method for adding two positive integers in terms of the procedures `inc`, which increments its argument by 1, and `dec`, which decrements its argument by 1.

```

1 (define (+ a b)
2   (if (= a 0)
3       b
4       (inc (+ (dec a) b))))
5
6 (define (+ a b)
7   (if (= a 0)
8       b
9       (+ (dec a) (inc b))))

```

Using the substitution model, illustrate the process generated by each procedure in evaluating `(+ 4 5)`. Are these processes iterative or recursive?

### Answer

The first procedure is recursive, while the second is iterative though tail-recursion.

#### recursive procedure

```

1 (+ 4 5)
2 (inc (+ 3 5))
3 (inc (inc (+ 2 5)))
4 (inc (inc (inc (+ 1 5))))
5 (inc (inc (inc (inc (+ 0 5)))))
6 (inc (inc (inc (inc 5))))
7 (inc (inc (inc 6)))
8 (inc (inc 7))

```

```

9  (inc 8)
10 9

```

**iterative procedure**

```

1  (+ 4 5)
2  (+ 3 6)
3  (+ 2 7)
4  (+ 1 8)
5  (+ 0 9)
6  9

```

## Exercise 1.10

### Question

The following procedure computes a mathematical function called Ackermann's function.

```

1  (define (A x y)
2    (cond ((= y 0) 0)
3          ((= x 0) (* 2 y))
4          ((= y 1) 2)
5          (else (A (- x 1)
6                    (A x (- y 1))))))

```

What are the values of the following expressions?

```

1  (A 1 10)
2  (A 2 4)
3  (A 3 3)

```

(1 10)	1024
(2 4)	65536
(3 3)	65536

```

1  <<ackermann>>
2  (define (f n) (A 0 n))
3  (define (g n) (A 1 n))
4  (define (h n) (A 2 n))
5  (define (k n) (* 5 n n))

```

Give concise mathematical definitions for the functions computed by the procedures  $f$ ,  $g$ , and  $h$  for positive integer values of  $n$ . For example,  $(k\ n)$  computes  $5n^2$ .

## Answer

**f**

```
1 <<try-these>>
2 <<EX1-10-defs>>
3 (try-these f 1 2 3 10 15 20)
```

---

1	2
2	4
3	6
10	20
15	30
20	40

---

$$f(n) = 2n$$

**g**

```
1 <<try-these>>
2 <<EX1-10-defs>>
3 (try-these g 1 2 3 4 5 6 7 8)
```

---

1	2
2	4
3	8
4	16
5	32
6	64
7	128
8	256

---

$$g(n) = 2^n$$

**h**

```
1 <<try-these>>
2 <<EX1-10-defs>>
3 (try-these h 1 2 3 4)
```

---

1	2
2	4
3	16

---

---


$$4 \quad 65536$$


---

It took a while to figure this one out, just because I didn't know the term. This is repeated exponentiation. This operation is to exponentiation, what exponentiation is to multiplication. It's called either *tetration* or *hyper-4* and has no formal notation, but two common ways would be these:

$$h(n) = 2 \uparrow\uparrow n$$

$$h(n) = {}^n 2$$

## Exercise 1.11

### Question

A function  $f$  is defined by the rule that:

$$f(n) = n \text{ if } n < 3$$

and

$$f(n) = f(n-1) + 2f(n-2) + 3f(n-3) \text{ if } n \geq 3$$

Write a procedure that computes  $f$  by means of a recursive process. Write a procedure that computes  $f$  by means of an iterative process.

### Answer

#### Recursive

```

1 (define (fr n)
2   (if (< n 3)
3       n
4       (+ (fr (- n 1))
5          (* 2 (fr (- n 2)))
6          (* 3 (fr (- n 3))))))
1 <<try-these>>
2 <<EX1-11-fr>>
3 (try-these fr 1 3 5 10)
```

---

1	1
3	4
5	25
10	1892

---

## Iterative

### 1. Attempt 1

```
1 ;; This seems like it could be better
2 (define (fi n)
3   (define (formula l)
4     (let ((a (car l))
5           (b (cadr l))
6           (c (caddr l)))
7       (+ a
8          (* 2 b)
9          (* 3 c))))
10  (define (iter l i)
11    (if (= i n)
12        (car l)
13        (iter (cons (formula l) l)
14              (+ 1 i))))
15  (if (< n 3)
16      n
17      (iter '(2 1 0) 2)))

1 <<try-these>>
2 <<EX1-11-fi>>
3 (try-these fi 1 3 5 10)
```

---

1	1
3	4
5	25
10	1892

---

It works but it seems wasteful.

### 2. Attempt 2

```
1 (define (fi2 n)
2   (define (formula a b c)
3     (+ a
4        (* 2 b)
5        (* 3 c)))
6   (define (iter a b c i)
7     (if (= i n)
8         a
9         (iter (formula a b c)
10              a
11              b
12              (+ 1 i))))
```



```

13      (if (< n 3)
14          n
15          (iter 2 1 0 2)))

1  <<try-these>>
2  <<EX1-11-fi2>>
3  (try-these fi2 1 3 5 10)

```

1	1
3	4
5	25
10	1892

I like that better.

## Exercise 1.12

### Question

The following pattern of numbers is called Pascal's triangle.

*Pretend there's a Pascal's triangle here.*

The numbers at the edge of the triangle are all 1, and each number inside the triangle is the sum of the two numbers above it. Write a procedure that computes elements of Pascal's triangle by means of a recursive process.

### Answer

I guess I'll rotate the triangle 45 degrees to make it the top-left corner of an infinite spreadsheet.

```

1  (define (pascal x y)
2    (if (or (= x 0)
3            (= y 0))
4        1
5        (+ (pascal (- x 1) y)
6            (pascal x (- y 1)))))

1  <<try-these>>
2  <<pascal-rec>>
3  (let ((l (iota 8)))
4    (map (λ (row)
5          (map (λ (xy)
6                (apply pascal xy))
7                row))
8          (map (λ (x)

```

```

9      (map (λ (y)
10            (list x y))
11            l))
12      l)))

```

---

1	1	1	1	1	1	1	1
1	2	3	4	5	6	7	8
1	3	6	10	15	21	28	36
1	4	10	20	35	56	84	120
1	5	15	35	70	126	210	330
1	6	21	56	126	252	462	792
1	7	28	84	210	462	924	1716
1	8	36	120	330	792	1716	3432

---

The test code was much harder to write than the actual solution.

## Exercise 1.13

### Question

Prove that  $\text{Fib}(n)$  is the closest integer to  $\frac{\phi^n}{\sqrt{5}}$  where  $\phi$  is  $\frac{1+\sqrt{5}}{2}$ . Hint: let  $\Upsilon = \frac{1-\sqrt{5}}{2}$ . Use induction and the definition of the Fibonacci numbers to prove that

$$\text{Fib}(n) = \frac{\phi^n - \Upsilon^n}{\sqrt{5}}$$

### Answer

I don't know how to write a proof yet, but I can make functions to demonstrate it.

#### Fibonacci number generator

```

1  (define (fib-iter n)
2    (define (iter i a b)
3      (if (= i n)
4          b
5          (iter (+ i 1)
6                b
7                (+ a b))))
8    (if (<= n 2)
9        1
10       (iter 2 1 1)))

```

### Various algorithms relating to the question

```


1 <<sqrt>>
2 (define sqrt5
3   (sqrt 5))
4 (define phi
5   (/ (+ 1 sqrt5) 2))
6 (define epsilon
7   (/ (- 1 sqrt5) 2))
8 (define (fib-phi n)
9   (/ (- (expt phi n)
10         (expt epsilon n))
11       sqrt5))

1 (use-srfis '(1))
2 <<fib-iter>>
3 <<fib-phi>>
4 <<try-these>>
5
6 (let* ((vals (drop (iota 21) 10))
7        (fibs (map fib-iter vals))
8        (approx (map fib-phi vals)))
9   (zip vals fibs approx))

```

10	55	54.99999999999999
11	89	89.0
12	144	143.99999999999997
13	233	232.99999999999994
14	377	377.00000000000006
15	610	610.0
16	987	986.9999999999998
17	1597	1596.9999999999998
18	2584	2584.0
19	4181	4181.0
20	6765	6764.999999999999

You can see they follow closely. Graphing the differences, it's just an exponential curve at very low values, presumably following the exponential increase of the Fibonacci sequence itself.



## Exercise 1.14

Below is the default version of the count-change function. I'll be aggressively modifying it in order to get a graph out of it.

```

1 (define (count-change amount)
2   (cc amount 5))
3
4 (define (cc amount kinds-of-coins)
5   (cond ((= amount 0) 1)
6         ((or (< amount 0)
7              (= kinds-of-coins 0))
8          0)
9         (else
10          (+ (cc amount (- kinds-of-coins 1))
11              (cc (- amount (first-denomination
12                           kinds-of-coins))
13                  kinds-of-coins))))))
14
15 (define (first-denomination kinds-of-coins)
16   (cond ((= kinds-of-coins 1) 1)
17         ((= kinds-of-coins 2) 5)
18         ((= kinds-of-coins 3) 10)

```

```

19      ((= kinds-of-coins 4) 25)
20      ((= kinds-of-coins 5) 50)))

```

## Question

Draw the tree illustrating the process generated by the count-change procedure of 1.2.2 in making change for 11 cents.

## Answer

I want to generate this graph algorithmically.

```

1  ;; cursed global
2  (define bubblecounter 0)
3  ;; Returns # of ways change can be made
4  ;; "Helper" for (cc)
5  (define (count-change amount)
6    (display "digraph {\n") ;; start graph
7    (cc amount 5 0)
8    (display "}\n") ;; end graph
9    (set! bubblecounter 0))
10
11  ;; GraphViz output
12  ;; Derivative: https://stackoverflow.com/a/14806144
13  (define (cc amount kinds-of-coins oldbubble)
14    (let ((recur (lambda (new-amount new-kinds)
15                  (begin
16                    (display "\"") ;; Source bubble
17                    (display `(\,oldbubble ,amount ,kinds-of-coins))
18                    (display "\"")
19                    (display " -> ") ;; arrow pointing from parent to child
20                    (display "\"") ;; child bubble
21                    (display `(\,bubblecounter ,new-amount ,new-kinds))
22                    (display "\"")
23                    (display "\n")
24                    (cc new-amount new-kinds bubblecounter))))))
25    (set! bubblecounter (+ bubblecounter 1))
26    (cond ((= amount 0) 1)
27          ((or (< amount 0) (= kinds-of-coins 0)) 0)
28          (else (+
29                (recur amount (- kinds-of-coins 1))
30                (recur (- amount
31                        (first-denomination kinds-of-coins))
32                        kinds-of-coins))))))
33
34  (define (first-denomination kinds-of-coins)

```

```

35     (cond ((= kinds-of-coins 1) 1)
36           ((= kinds-of-coins 2) 5)
37           ((= kinds-of-coins 3) 10)
38           ((= kinds-of-coins 4) 25)
39           ((= kinds-of-coins 5) 50)))

```


I'm not going to include the full printout of the (count-change 11), here's an example of what this looks like via 1.

```


1 <<count-change-graphviz>>
2 (count-change 1)

digraph {
  "(0 1 5)" -> "(1 1 4)"
  "(1 1 4)" -> "(2 1 3)"
  "(2 1 3)" -> "(3 1 2)"
  "(3 1 2)" -> "(4 1 1)"
  "(4 1 1)" -> "(5 1 0)"
  "(4 1 1)" -> "(6 0 1)"
  "(3 1 2)" -> "(7 -4 2)"
  "(2 1 3)" -> "(8 -9 3)"
  "(1 1 4)" -> "(9 -24 4)"
  "(0 1 5)" -> "(10 -49 5)"
}

```



So, the graph of (count-change 11) is:



## Question 2

What are the orders of growth of the space and number of steps used by this process as the amount to be changed increases?

## Answer 2

Let's look at this via the number of function calls needed for value  $n$ . Instead of returning an integer, I'll return a pair where `car` is the number of ways to



count change, and cdr is the number of function calls that have occurred down that branch of the tree.

```

1 (define (count-calls amount)
2   (cc-calls amount 5))
3
4 (define (cc-calls amount kinds-of-coins)
5   (cond ((= amount 0) '(1 . 1))
6         ((or (< amount 0)
7              (= kinds-of-coins 0))
8          '(0 . 1))
9         (else
10          (let ((a (cc-calls amount (- kinds-of-coins 1)))
11                (b (cc-calls (- amount (first-denomination
12                               kinds-of-coins))
13                               kinds-of-coins)))
14            (cons (+ (car a)
15                     (car b))
16                  (+ 1
17                     (cdr a)
18                     (cdr b)))))))
19
20 (define (first-denomination kinds-of-coins)
21   (cond ((= kinds-of-coins 1) 1)
22         ((= kinds-of-coins 2) 5)
23         ((= kinds-of-coins 3) 10)
24         ((= kinds-of-coins 4) 25)
25         ((= kinds-of-coins 5) 50)))

```

---


1	1	(1 . 11)
2	1	(1 . 13)
3	1	(1 . 15)
4	1	(1 . 17)
5	2	(2 . 19)
6	2	(2 . 25)
7	2	(2 . 29)
8	2	(2 . 33)
9	2	(2 . 37)
10	4	(4 . 41)

---

```

1 (use-srfis '(1))
2 <<cc-calls>>
3 (let* ((vals (drop (iota 101) 1))
4         (mine (map count-calls vals)))
5   (zip vals (map car mine) (map cdr mine)))

```



I believe the space to be  $\Theta(n+d)$  as the function calls count down the denominations before counting down the change. However I notice most answers describe  $\Theta(n)$  instead, maybe I'm being overly pedantic and getting the wrong answer.

My issues came finding the time. The book describes the meaning and properties of  $\Theta$  notation in Section 1.2.3. However, my lack of formal math education made realizing the significance of this passage difficult. For one, I didn't understand that  $k_1 f(n) \leq R(n) \leq k_2 f(n)$  means "you can find the  $\Theta$  by proving that a graph of the algorithm's resource usage is bounded by two identical functions multiplied by constants." So, the graph of resource usage for an algorithm with  $\Theta(n^2)$  will be bounded by lines of  $n^2 \times \text{someconstant}$ , the top boundary's constant being larger than the small boundary. These are arbitrarily chosen constants, you're just proving that the function behaves the way you think it does.

Overall, finding the  $\Theta$  and  $\Omega$  and  $O$  notations (they are all different btw!) is about aggressively simplifying to make a very general statement about the behavior of the algorithm.

I could tell that a "correct" way to find the  $\Theta$  would be to make a formula which describes the algorithm's function calls for given input and denominations. This is one of the biggest time sinks, although I had a lot of fun and learned a lot. In the end, with some help from Jach in a Lisp Discord, I had the following formula:

$$\sum_{i=1}^{\text{ceil}(n/\text{val}(d))} T(n - \text{val}(d) * i, d)$$

But I wasn't sure where to go from here. The graphs let me see some interesting trends, though I didn't get any closer to an answer in the process.

By reading on other websites, I knew that you could find  $\Theta$  by obtaining a formula for  $R(n)$  and removing constants to end up with a term of interest. For example, if your algorithm's resource usage is  $\frac{n^2+7n}{5}$ , this demonstrates  $\Theta(n^2)$ . So I know a formula **without** a  $\sum$  would give me the answer I wanted. It didn't occur to me that it might be possible to use calculus to remove the  $\sum$  from the equation. At this point I knew I was stuck and decided to look up a guide.

After seeing a few solutions that I found somewhat confusing, I landed on this awesome article from Codology.net. They show how you can remove the summation, and proposed this equation for count-change with 5 denominations:

$$T(n, 5) = \frac{n}{50} + 1 + \sum_{i=0}^{n/50} T(n - 50i, 1)$$

Which, when expanded and simplified, demonstrates  $\Theta(n^5)$  for 5 denominations.

Overall I'm relieved that I wasn't entirely off, given I haven't done math work like this since college. It's inspired me to restart my remedial math courses, I don't think I really grasped the nature of math as a tool of empowerment until now.

## Exercise 1.15

### Question 1

The sine of an angle (specified in radians) can be computed by making use of the approximation  $\sin x \approx x$  if  $x$  is sufficiently small, and the trigonometric identity  $\sin x = 3 \sin \frac{x}{3} - 4 \sin^3 \frac{x}{3}$  to reduce the size of the argument of  $\sin$ . (For purposes of this exercise an angle is considered "sufficiently small" if its magnitude is not greater than 0.1 radians.) These ideas are incorporated in the following procedures:

```

1 (define (cube x) (* x x x))
2 (define (p x) (- (* 3 x) (* 4 (cube x))))
3 (define (sine angle)
4   (if (not (> (abs angle) 0.1))
5       angle
6       (p (sine (/ angle 3.0)))))

```

How many times is the procedure `p` applied when `(sine 12.15)` is evaluated?

## Answer 1

Let's find out!

```
1 (define (cube x) (* x x x))
2 (define (p x) (- (* 3 x) (* 4 (cube x))))
3 (define (sine angle)
4   (if (not (> (abs angle) 0.1))
5       (cons angle 0)
6       (let ((x (sine (/ angle 3.0))))
7         (cons (p (car x)) (+ 1 (cdr x))))))
1 <<1-15-p-measure>>
2 (let ((xy (sine 12.15)))
3   (list (car xy) (cdr xy)))
```

---

-0.39980345741334	5
-------------------	---

---

p is evaluated 5 times.

## Question 2

What is the order of growth in space and number of steps (as a function of a) used by the process generated by the sine procedure when (sine a) is evaluated?

## Answer 2

```
1 (use-srfis '(1))
2 <<1-15-p-measure>>
3 (let* ((vals (iota 300 0.1 0.1))
4        (sines (map (lambda (i)
5                      (cdr (sine i)))
6                      vals)))
7   (zip vals sines))
```

Example output:

---

0.1	0
0.2	1
0.30000000000000004	2
0.4	2
0.5	2
0.6	2
0.7000000000000001	2
0.8	2
0.9	2

---

1.0	3
-----	---

---

```

reset # helps with various issues in execution
set xlabel 'values of x'
set logscale x
set key top left
set style fill solid 1.00 border
set style function fillsteps below


```

$f(x) = \log(x) + 2.3$

```

plot data using 1:2 with fillsteps title 'function calls', \
    data using 1:(f($1)) with lines title 'log(x) + 2.3'

```



This graph shows that the number of times sine will be called is logarithmic.

- 0.1 to 0.2 are divided once
- 0.3 to 0.8 are divided twice
- 0.9 to 2.6 are divided three times
- 2.7 to 8 are divided four times
- 8.5 to 23.8 are divided five times

Given that the calls to p get stacked recursively, like this:

```

(sine 12.15)
(p (sine 4.05))
(p (p (sine 1.35)))
(p (p (p (sine 0.45))))
(p (p (p (p (sine 0.15)))))
(p (p (p (p (p (sine 0.05))))))
(p (p (p (p (p 0.05)))))
(p (p (p (p 0.14950000000000002))))
(p (p (p 0.43513455050000005)))
(p (p 0.9758465331678772))
(p -0.7895631144708228)
-0.39980345741334

```

So I argue the space and time is  $\Theta(\log(n))$

We can also prove this for the time by benchmarking the function:

```

1 ;; This execution takes too long for org-mode, so I'm doing it
2 ;; externally and importing the results
3 (use-srfis '(1))
4 (use-modules (ice-9 format))
5 (load "../mattbench.scm")
6 <<1-15-deps>>
7 (let* ((vals (iota 300 0.1 0.1))
8         (times (map (lambda (i)
9                       (mattbench (lambda () (sine i)) 1000000))
10                      vals)))
11       (with-output-to-file "sine-bench.dat" (lambda ()
12         (map (lambda (x y)
13               (format #t "~s~/~s~%" x y))
14               vals times))))

```

```

reset # helps with various issues in execution
set xtics 0.5
set xlabel 'values of x'
set logscale x
set key top left
set style fill solid 1.00 border
#set style function fillsteps below

```

```


f(x) = (log(x) * a) + b
fit f(x) 'Ex15/sine-bench.dat' using 1:2 via a,b

```

```

plot 'Ex15/sine-bench.dat' using 1:2 with fillsteps title 'time to execute', \
      'Ex15/sine-bench.dat' using 1:(f($1)) with lines title sprintf('(log(x) * %.2f) + %.2f', a, b)

```



## Exercise 1.16

### Text

```

1 (define (expt-rec b n)
2   (if (= n 0)
3       1
4       (* b (expt-rec b (- n 1)))))
5
6 (define (expt-iter b n)
7   (define (iter counter product)
8     (if (= counter 0)
9         product
10        (iter (- counter 1)
11               (* b product))))
12   (iter n 1))
13
14 (define (fast-expt b n)
15   (cond ((= n 0)
16         1)
17         ((even? n)
18          (square (fast-expt b (/ n 2))))
19         (else
20          (* b (fast-expt b (/ n 2))))))

```

20

```
(* b (fast-expt b (- n 1))))))
```

## Question

Design a procedure that evolves an iterative exponentiation process that uses successive squaring and uses a logarithmic number of steps, as does fast-expt. (Hint: Using the observation that  $(b^{n/2})^2 = (b^2)^{n/2}$ , keep, along with the exponent  $n$  and the base  $b$ , an additional state variable  $a$ , and define the state transformation in such a way that the product  $ab^n$  is unchanged from state to state. At the beginning of the process  $a$  is taken to be 1, and the answer is given by the value of  $a$  at the end of the process. In general, the technique of defining an *invariant quantity* that remains unchanged from state to state is a powerful way to think about the design of iterative algorithms.)

## Diary

```
1 <<square>>
2 (define (fast-expt-iter b n)
3   (define (iter b n a)
4     (format #t "~&|~s~/~/|~s~/~/|~s|~%" b n a)
5     (cond ((= n 1) (begin (format #t "~&|~s~/~/|~s~/~/|~s|~%" (* b a) 1 1)
6                           (* b a)))
7           ((even? n) (iter (square b)
8                             (/ n 2)
9                             a))
10          (else (iter b (- n 1) (+ a 1)))))
11   (format #t "|~a~/|~a~/|~a|~%" "base" "power" "variable")
12   (format #t "~&|--|--|--|~%" )
13   (iter b n 1))

1 <<fast-expt-iter-fail1>>
2 <<try-these>>
3 (try-these (lambda (x) (fast-expt-iter 2 x)) (cdr (iota 11)))
```

1	2
2	4
3	8
4	16
5	32
6	32
7	48
8	256
9	512
10	512

---



base	power	variable
2	6	1
4	3	1
4	2	2
16	1	2
32	1	1